
DS2言語について
DS2は、高度なデータ操作に適した SAS独自のプログラミング言語です。DS2は
SAS Viyaプラットフォームに含まれており、SAS DATAステップと交差します。ま
た、ANSI SQLデータ型を含む追加のデータ型、追加のプログラミング構造要素、お
よびユーザー定義のメソッドとパッケージも含まれています。
いくつかの DS2言語要素は、埋め込まれた FedSQL構文を受け入れ、ランタイムで
生成されたクエリは、DS2とサポートされているデータベース間でインタラクティ
ブにデータを交換できます。これにより、入力テーブルの SQL前処理が可能にな
り、2つの言語の能力が効果的に組み合わされます。
SAS Viyaプラットフォームでは、DS2プログラムは次の環境で実行できます。
n SAS Cloud Analytic Services (CAS) Server

n SAS Compute Server

n SAS Event Stream Processing (ESP) Server

n SAS Federation Server

n SAS Micro Analytic Service

SAS Federation Serverは、マルチユーザーの標準ベースのフェデレーションデータ
アクセスを提供する、個別にライセンスされたサーバーです。SAS Micro Analytic 
Serviceは、インストリームデータ分析とディシジョンの自動化をサポートする、メ
モリに常駐する高性能のプログラム実行プラットフォームです。マイクロサービス
として配置することも、ESP、CAS、Computeサーバーなどのサーバープロセス内
で実行することもできます。このサービスは、SAS Event Stream Processingソリ
ューションに加えて、SAS Intelligent Decisionおよび SAS Model Managerで使用
できます。
DS2プロシジャを使用すると、SAS Studioから DS2言語ステートメントをサブミッ
トできます。PROC DS2の詳細については、Base SASプロシジャガイドを参照して
ください。

注: DS2言語は多くのデータソースで使用できるため、データ要素を説明するため
に行、列、およびテーブルという用語が使用されます。これを SAS DATAステップ
の用語と比較すると、行はオブザベーションに対応し、列は変数に対応し、テーブ
ルはデータセットに対応します。

DS2プログラムの実行
DS2プログラムは、次のいずれかの方法でサブミットできます。
n DS2プロシジャを使用して SAS Studioを介する方法。DS2プロシジャは、SAS 

Compute Serverまたは CASサーバーで DS2コードを実行するために使用でき
ます。1つの PROC DS2ステップに、複数の DS2プログラムを含めることがで
きます。
詳細については、“DS2 プロシジャ” (Base SASプロシジャガイド)を参照してくだ
さい。

n SAS Studioで、CASサーバーで runDS2アクションを使用します。runDS2アク
ションは、CASプロシジャと組み合わせて使用されます。

注: Pythonまたは Luaを使用している場合を除き、PROC DS2を使用して DS2
コードを CASサーバーにサブミットすることをお勧めします。

詳細については、“DS2 Action Set” (SAS Viya Platform: System Programming 
Guide)および 19章, “CASの DS2: 概念” (297ページ)を参照してください。

n SAS Federation Serverで SAS Federation Server用の SAS LIBNAMEエンジン
を使用します。詳細については、SAS LIBNAME Engine for SAS Federation Server: 
ユーザーガイドを参照してください。

n SAS Event Stream Processing、SAS Intelligent Decisions、および SAS Model 
Managerで、SAS Micro Analytic Serviceを使用します。詳細については、SAS 
Micro Analytic Service: プログラミングと管理ガイドの About Using SAS Micro 
Analytic Serviceおよび DS2 Programming for SAS Micro Analytic Serviceを参
照してください。

サポートされているデータソース
DS2は、次のデータソースにアクセスできます。
n Amazon Redshift

n CASテーブル
n UNIX動作環境用の DB2

n Google BigQuery

n Greenplum

n Hadoop (Hive)

n Impala

n JDBCに準拠しているデータベース(PostgreSQLなど)

n メモリデータストア(MDS)

n Microsoft SQL Server

n MongoDB *

n MySQL (MySQL 8サーバーを含む)

n Netezza

n ODBCに準拠しているデータベース(Microsoft SQL Serverなど)

n Oracle

n PostgreSQL

n Salesforce

n SAP (読み取り専用)

n SAP HANA

n SAP IQ

n SASデータセット
n SAS Scalable Performance Data Engine (SPD Engine)データセット
n UNIX動作環境の SAS Scalable Performance Data Server (SPD Server)テーブ
ル

n SingleStore

n Snowflake

n Spark

n UNIX動作環境用の Teradata

n Vertica

n Yellowbrick

サポートされているデータソース 5



注: 次のデータソースはサポートされていません。
n Informix

n OLEDB

n SAP ASE

DS2は、SASプログラミングランタイム環境または CASサーバー上のMDSおよび
SPD Serverを除くすべてのデータソースで動作できます。
*MongoDBデータソースには、テーブルを作成するための特別な要件があります。
データに応じて、ルートテーブル、親テーブル、または子テーブルを作成すること
をお勧めします。DS2ではルートテーブルのみを作成できます。親テーブルと子テ
ーブルを作成するには、FedSQLを使用する必要があります。SAS/ACCESS for 
Relational Databases: リファレンスのMongoDBに関する情報を参照してくださ。
DS2プロシジャと SAS Federation Serverは、異なるデータソースをサポートしま
す。それぞれがサポートするデータソースについては、Base SASプロシジャガイド
および SAS Federation Server: 管理者ガイドを参照してください。
データソースへのアクセスについては、必要に応じて SAS/ACCESS for Relational 
Databases: リファレンスおよび SAS/ACCESS for Nonrelational Databases: リファレ
ンスを参照してください。

対象とするユーザー
このドキュメントの情報は、これらの役割を実行する次のユーザーを対象としてい
ます。
n クライアントアプリケーションを作成するアプリケーション開発者。テーブル
の作成、テーブルのバルクロード、テーブルの操作、データのクエリを行うアプ
リケーションを作成します。

n クライアント/サーバー環境を設計および実装するデータベース管理者。データ
ベースを設計し、データソースのメタデータを設定することでデータを管理しま
す。つまり、データベース管理者はデータモデルを構築します。

n 数値精度の向上、CPUバインドプロセスの並列計算、DATAステッププロセスへ
の直接入力としての明示的なパススルー SQLクエリの使用、ビッグデータ環境
での in-databas処理など、DS2言語の高度な機能を利用したいか、または利用
する必要がある SAS プログラマ。

DS2を使用する場合
通常、DS2プログラムは、次のアクションを実行するアプリケーション用に作成さ
れています。
n サポートされている新しいデータ型を使用した結果の精度を必要とする
n 変数スコープ、ユーザー定義のメソッドとパッケージ、スレッド処理のサポート
など、DS2言語の新しいプログラミング要素を活用する

n DS2プログラム内から SAS FedSQLを実行する必要がある
n SASセッションの外部で実行する必要がある(たとえば、SAS Federation Server
上)

DATAステッププログラムから DS2プロ
グラムへの変換

DSTODS2プロシジャを使用して、DATAステップコードを DS2に変換することがで
きます。すべての DATAステップコードの変換がサポートされているわけではな
く、手動での変換が必要になる場合があります。変換できないコード行はコメント
に入れられます。詳細については、“DSTODS2 プロシジャ” (Base SASプロシジャガ
イド)を参照してください。
DATAステップコードを DS2に変換し、プログラムが構文的に完全であることを確
認したら、DS2プロシジャを使用して、SAS Viyaプラットフォームでプログラムを
実行できます。

DS2言語の構文規則
文字体裁の規則

DS2言語構文のドキュメントで使用される書体には、特別な意味があります。
UPPERCASE BOLD
ステートメントや関数の名前などの DS2キーワードを識別します(例: PUT)。

UPPERCASE ROMAN (大文字のローマン体)
リテラルの引数と値を識別します(例: FROM)。

italic
指定する引数または値を識別します。斜体の項目は、次のいずれかのユーザー指
定の値を表すことができます。
n 引数に割り当てられた非リテラル値(例: ALTER=alter-password)。
n 非リテラル引数(例: KEEP=(column-list)。
斜体の項目を複数使用できる場合、その項目は item [, …item]として表されます。

monospace
SASコードの例を識別します。

構文規則
このドキュメントでは、バッカスナウア記法(BNF)を使用しています。具体的には、
Jim Meltonが SQL:1999 Understanding Relational Language Componentsで使用し
ているのと同じ構文表記法です。
従来の SAS構文と DS2言語リファレンスドキュメントで使用されている構文の主
な違いは、オプションの構文引数の表示方法にあります。従来の SAS構文では、オ
プションの構文を示すために山かっこ(< >)が使用されます。DS2言語の構文では、
角かっこ([ ])がオプションの構文を示すために使用され、山かっこは非終端の構成要
素を示すために使用されます。
DS2言語の構文では、次の記号が使用されます。
::=
この記号は、"で構成される"または"として定義されている"と解釈できます。

<>
山かっこは、非終端の構成要素(つまり、下位レベルの構文文法にさらに分解でき
る構文の構成要素を示します。

[ ]
角かっこはオプション引数を示します。角かっこで囲まれていない引数はすべ
て必須引数です。角かっこは、その前にリテラルであることを示すバックスラッ
シュ(\)が付いていない限り、入力しないでください。

{ }
中かっこは、必要な複数単語の引数を区別する方法を提供します。中かっこは、
その前にリテラルであることを示すバックスラッシュ(\)が付いていない限り、入
力しないでください。

|
縦棒は、グループの中から 1つの値を選択できることを示しています。棒で区切
られた値は相互に排他的です。

…
省略記号は、省略記号に続く引数または引数のグループが何度でも繰り返される
ことを示しています。省略記号と次の引数が角かっこで囲まれている場合はオ
プションです。

\
バックスラッシュは、次の文字がリテラルであることを示します。

次の例は、このセクションで説明されている構文規則を示しています。これらの例
には、完全な構文ではなく、選択された構文要素が含まれています。

1 SET 2 <table-reference> [… [<table-reference>] [INDSNAME=variable];
   [ 3 BY [DESCENDING] 4 column [ 5 ...[DESCENDING]column];
   6 <table-reference>::=
   {table (table-options)} 7  | 8 \{sql-text 8 \}

1 SETはステートメントの名前であるため、大文字の太字で示されています。
2 <table-reference>は非終端引数であり、さらに下位レベルの構文文法に分解さ
れるため、山かっこで囲まれています。少なくとも 1つの<table-reference>を
指定する必要があります。

3 BYと DESCENDINGはリテラル引数であるため、大文字のローマン体で表記さ
れます。DESCENDINGはオプションの引数であるため、角かっこで囲まれてい
ます。

4 columnは、指定できる引数であるため、斜体で示しています。
5 columnの 2番目のインスタンスを囲む角かっこと省略記号は、引数がカンマで
区切られている限り、この引数を何度でも繰り返すことができることを示してい
ます。

6 <table-reference>::=非終端引数の構文は次のように読み取られます: <table-
reference>はテーブル名とテーブルオプション、または埋め込み SQLテキスト
で構成されます。

7 縦棒(|)は、table [table-options]または sql-textのいずれかを指定できますが、両
方は指定できないことを示します。

8 sql-textを囲む中かっこの前のバックスラッシュ(\)は、それらの中かっこがリテ
ラルであり、入力する必要があることを示します。

DS2プログラミングセマンティ

DS2識別子のスコープ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 17
プログラミングブロック . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 17
変数ルックアップ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 19

基本的な DS2プログラムの構文
DS2プログラムは、宣言のリストとそれに続くメソッドステートメントのリストで
構成されます。簡単な宣言リストの例を次に示します。
declare int x;
dcl double d; 

簡単なメソッドステートメントリストの例を次に示します。
method init();
   put 'in init';
end;

method run();
   put 'in run';
end;

method term();
   put 'in term';
end; 

2つのリストを組み合わせると、単純な DS2プログラムが作成されます。
declare int x;
declare double d;

method init();
end;

method run();
end;

method term();
end;

通常、DS2プログラムはより複雑ですが、この単純なプログラムにはいくつかの構
文要素が含まれています。
キーワード:

n DECLARE/DCL

n DOUBLE

n METHOD

n END

識別子:

n x

n d

n INIT

n RUN

n TERM

字句区切り記号:

n (

n )

n ;

このプログラムは、DECLAREステートメントまたはMETHODステートメントで識
別子を宣言する方法を示しています。また、DS2プログラムの高レベル構造が、一
連の変数宣言とそれに続く一連のMETHODステートメントで構成されていること
も示しています。次のセクションでは、これらの用語が DS2で何を意味するかを説
明します。

基本的な DS2プログラムのセマンティク
ス
変数宣言ステートメント

変数宣言はメモリ空間を割り当て、変数名と呼ばれる識別子でそのメモリを識別し
ます。宣言は、明示的または暗黙的に、変数にメモリを割り当て、そのメモリ位置
に保存できるデータの型を指定します。DS2では、DECLAREステートメントを使用
して変数を宣言します。DECLAREステートメントは、次のアクションを実行しま
す。
n 識別子をメモリ位置に割り当てます。その識別子が変数名になります。
n 変数が保持できるデータの型を指定します。
n 指定された量のメモリを変数に割り当てます。
1つの DECLAREステートメントで複数の変数を宣言できます。詳細については、
“DECLARE ステートメント” (SAS DS2言語リファレンス)を参照してください。

メソッド
メソッドは基本的なプログラム実行ユニットです。DS2では、メソッド構造を使用
して、関連するプログラムステートメントを 1つの構文的に識別可能な場所にグル
ープ化します。メソッド内のステートメントのグループは、簡単に複数回呼び出し
たり、実行したりできます。
DS2メソッドは、Cなどの言語の関数や Javaのメソッドに似ています。従来の
DATAステッププログラミングでは、LINKおよび RETURNステートメントは、DS2
データプログラムのメソッド定義と同様に使用されていました。ただし、LINKと
RETURNはローカルスコープの変数を作成できないため、再利用可能なコードモジ
ュールの配布にはあまり役に立ちません。FCMPプロシジャを使用すると、ローカ
ルスコープで再利用可能な関数を作成できます。これは、DS2メソッドと同等です。
メソッドはスコープブロックを定義します。したがって、メソッド本体のパラメー
ターと変数宣言は、メソッドに対してローカルです。

ヒント DS2メソッドは、最大 1000個の引数をサポートできます。1000を
超える引数を持つ DS2メソッドは、コンパイルエラーを生成する可能性があ
ります。

型シグネチャ(または単にシグネチャ)は、メソッドのパラメーター型の順序付きリス
トとして定義されます。2つのメソッド定義の名前が同じで、型シグネチャが異な
る場合、そのメソッドはオーバーロードされます。2つのメソッド定義が同じ名前
と同じ型シグネチャを持つ場合、エラーが発生します。

注: セッションエンコーディングに複数バイトが必要な場合、CHARおよび NCHAR
データ型のみに基づくメソッドをオーバーロードすることはできません。セッショ
ンエンコーディングで文字ごとに複数のバイトが必要な場合(UTF-8など)、CHARと
NCHARは同じ型であり、どちらも NCHARを使用します。したがって、2つのメソ
ッド定義は同じものと見なされます。

DS2には、システムメソッドとユーザー定義メソッドの 2種類のメソッドがありま
す。
DS2プログラムには、次の 3つのシステムメソッドの少なくとも 1つが含まれてい
る必要があります。
method init();
   end;
method run();
   end; 
method term();
   end; 

これらのメソッドは、SAS DATAステップの暗黙的なループの概念よりも構造化され
たフレームワークを提供することを目的としています。Base SASでは、DATAステ
ッププログラム全体が暗黙的なループに含まれていました。DS2では、暗黙的なル
ープは RUNメソッドによって表され、INITメソッドと TERMメソッドがそれぞれ
初期化コードとファイナライズコードを提供します。
DS2プログラムを実行した場合の結果を次に示します。
1 INITメソッドが実行されます。初期化が行われます。
2 保持されていないプログラムデータベクトル内の変数は、適切な欠損値に設定さ
れます。詳細については、“RETAIN ステートメント” (SAS DS2言語リファレンス)
を参照してください。

3 RUNメソッドが実行されます。
4 実行制御は、RUNメソッドの入力ステートメントのステータスに依存します。
現在、DS2の唯一の入力ステートメントは SETステートメントです。RUNメソ
ッドがこれらの条件のいずれかを満たす場合、処理はステップ 5に進みます。そ
れ以外の場合、RUNメソッドを再度実行できるように、処理はステップ 2に進
みます。
n 入力ステートメントなし
n 実行が完了した入力ステートメント

5 TERMメソッドが実行され、すべての最終ステートメントが実行されます。
INIT、RUN、および TERMメソッドは、パラメーターおよび戻り値なしで定義する
必要があります。INIT、RUN、または TERMメソッドにパラメーターを指定すると、
エラーが発生します。

DS2プログラムで OUTPUTステートメントを指定しない場合、DS2コンパイラは、
RUNメソッドの最後に実行されるパラメーターなしのステートメントを提供しま
す。RUNメソッドを記述していない場合は、DS2コンパイラによって暗黙的に作成
されます。したがって、最小の DS2プログラム形式は、次のようにデータプログラ
ム内の単一の宣言です。
data;
dcl double x;
enddata;

このコードは、次の 2つのコードセットと機能的に同等です。
data;
dcl double x;
method run();
end;
enddata;

data;
dcl double x;
method run();
   output;
end;
enddata;

各コードセットは、出力ウィンドウに単一の初期化されていない xの値を生成しま
す。
DS2プログラムから INIT、RUN、または TERMメソッドを直接呼び出そうとする
と、エラーが発生します。
ユーザー定義メソッドは、1回以上実行させるステートメントをMETHODステート
メントと ENDステートメントで囲むことで作成できます。ユーザー定義メソッド
の詳細については、“METHOD ステートメント” (SAS DS2言語リファレンス)を参照
してください。

注: PROC DS2を使用する場合、DS2プログラムは RUNステートメントで区切られ
ます。RUNステートメントの後に追加の DS2コードが見つかった場合、このコード
は、前の RUNステートメントの前の DS2プログラムとは別の新しい DS2プログラ
ムを構成します。

DS2識別子のスコープ
プログラミングブロック

プログラミングブロックは、順序付けられたキーワードのペアで始まり、終了する
コードのセクションです。次のキーワードは、プログラミングブロックを作成しま
す。

n DATA...ENDDATA

n PACKAGE...ENDPACKAGE

n THREAD...ENDTHREAD

n DO...END

n METHOD...END

このドキュメントでは、これらの用語はプログラミングブロックに使用されます。
n データプログラミングブロックまたはデータプログラムは、DATA…ENDDATAス
テートメントで囲まれたコードを指します。

n パッケージプログラミングブロックまたはパッケージは、PACKAGE…
ENDPACKAGEステートメントによって囲まれた変数およびメソッドの保存され
たライブラリを指します。パッケージの変数とメソッドは、DS2プログラム、ス
レッド、またはその他のパッケージで使用できます。

n スレッドプログラミングブロックまたはスレッドプログラムは、THREAD…
ENDTHREADステートメントで囲まれたストアドプログラムを指します。スレ
ッドプログラムは、DS2プログラムまたはパッケージの SET FROMステートメ
ントによって呼び出すことができます。

n DOプログラミングブロックまたは DO ループは、DOステートメントと ENDス
テートメントで囲まれたプログラミングステートメントのサブブロックを指し
ます。

n メソッドプログラミングブロックまたはメソッドブロックは、METHODステー
トメントと ENDステートメントで囲まれたプログラミングステートメントのサ
ブブロックを指します。

一部のブロックはネストできます。この例では、DATAおよび ENDDATAステートメ
ントによって定義された 1つのデータプログラムと、3つのメソッドステートメン
トによって定義された 3つのネストされたメソッドブロックがあります。
data _null_; 
   declare int x;
   
   method init();
      declare double d;
   end;

   method run();
   end;

   method term();
   end;
enddata;

最も外側のプログラミングブロックで宣言された変数は、グローバル変数、または
グローバルスコープを持つ変数と呼ばれます。ネストされたブロックで宣言された
変数は、ローカル変数、またはそのブロックに対してローカルなスコープを持つ変
数と呼ばれます。DS2は、宣言されていない変数にもグローバルスコープを割り当
てます。前の例では、Xはグローバル変数であり、Dはネストされた INITメソッド
に対してローカルな変数です。

注: DS2プログラムには、複数のサブプログラムとそれに続くオプションのデータ
プログラムを含めることができます。次の制限が適用されます。

18 2章 / DS2プログラミングセマンティクス



n データプログラムは 1つしか存在できず、データプログラムは最後のサブプログ
ラムである必要があります。

n ENDPACKAGE、ENDTHREAD、または ENDDATAステートメントは、DS2プロ
グラムの最後のサブプログラムではオプションですが、適切なプログラミング形
式にすることをお勧めします。これらのステートメントは、他のすべてのサブプ
ログラムでは必須です。

変数ルックアップ
変数が参照されると、DS2は参照のブロックで始まる変数の宣言を常に検索します。
次に、そこで見つからない場合は、外側のブロックまたはプログラムを連続して検
索します。この例では、INITメソッドでの Xへの参照は、Xのグローバル宣言を参
照しています。
declare int x;

method init();
end; 

メソッドはブロックであるため、宣言自体を含めることができます。この例では、
INITメソッドでの Xへの参照は Xのローカル宣言を参照していますが、RUNメソッ
ドでの Xへの参照は Xのグローバル宣言を参照しています。
declare int x;

method init();
   declare int x;
end;

method run();
end;  

スコープの定義
スコープは、識別子の属性と見なすことができます。識別子は、メソッド名、関数、
データ名、ラベル、プログラム変数など、多くのプログラムエンティティを参照で
きます。このセクションではプログラム変数を例として使用しますが、すべての識
別子はスコーピングルールに従います。
スコープは、変数にアクセスできるプログラム内の場所を表します。グローバル変
数にはグローバルスコープがあり、プログラムのどこからでもアクセスできます。
ローカル変数にはローカルスコープがあり、変数が宣言されたプログラムまたはブ
ロック内からのみアクセスできます。
DS2では、プログラムの実行が変数のスコープ内で行われている場合にのみ、変数
にアクセスできます。つまり、変数のスコープ内のステートメントがアクティブに
実行されている場合にのみ、変数の値にアクセスできます。

data;
   declare int x;    /* global x in global scope */

   method init();
      x = 5;          /* global x assigned 5 */
   end;

   method run();
   end;

   method term();
   end;

enddata;

特定のスコープ内の各変数には一意の名前を付ける必要がありますが、異なるスコ
ープ内の変数に同じ名前を付けることができます。スコープがネストされている場
合、外側のスコープの変数が内側のスコープの変数と同じ名前を持つ場合、外側の
スコープ内の変数は内側のスコープ内の変数によって隠されます。たとえば、次の
プログラムでは、2つの異なる変数が同じ名前 Xを共有しています。グローバル変
数 Xにはグローバルスコープがあり、ローカル変数 Xにはローカルスコープがあり
ます。メソッド INITのローカルスコープ内では、ローカル変数 Xはグローバル変数
Xを隠します。したがって、割り当てステートメントはローカル変数 Xに 5を割り
当てます。
data;
    declare int x;    /* global x in global scope */

    method init();
       declare int x;  /* local x in local scope */

       x = 5;          /* local x assigned 5 */
    end;

    method run();
    end;
    
    method term();
    end;

enddata;

変数有効期間
変数の有効期間は、変数が存在する期間です。グローバル変数は、プログラムの期
間中存在します。ローカル変数は、変数が宣言されたブロックの期間中存在します。
グローバル変数の値は、そのグローバル変数が現在のプログラムブロックの RETAIN
ステートメント内に出現しない限り、RUNメソッドに入る前に欠損値または null値
に設定されます。
次の例では、変数 Xは INITメソッドの実行中にのみ存在し、変数 Yはデータプログ
ラムの実行中に存在します。

data;
   declare double y;

   method init();
      declare double x;
   end;

enddata;

変数の有効期間中、次の例のように、同名のローカルで宣言された変数によって重
要性が低下する可能性があります。
declare int x;

method init();
   declare int x;
end;
 
method run();
end; 

グローバル変数 Xにはプログラム全体の有効期間がありますが、この例では、INIT
メソッドで Xがローカルに宣言されているため、INITメソッドから直接アクセスす
ることはできません。

DS2変数の概要
DS2プログラム変数の特性は、名前、スコープ、およびデータ型を持つことです。
名前または識別子は、変数に与えられる 1つ以上のトークンまたはシンボルです。
名前については、6章, “DS2識別子” (57ページ)で説明されています。
変数は、変数が宣言されている場所に応じて、グローバルスコープまたはローカル
スコープを持つことができます。詳細については、“変数スコープ” (34ページ)を参
照してください。

変数のデータ型は、宣言方法に応じて明示的または暗黙的に割り当てられます。詳
細については、“変数宣言” (24ページ)を参照してください。

注: "データ型"という用語には、精度、文字セットエンコーディング、長さなどのデ
ータ型属性が含まれます。データ型の詳細については、5章, “DS2データ型” (47ペ
ージ)を参照してください。

注: このセクションのコード例の DS2ステートメントは、PROC DS2を使用して
SASに送信されます。

変数宣言
変数を宣言する方法

変数とそのデータ型を宣言するには、次の 3つの方法があります。
n DECLAREステートメントを使用した明示的な宣言

DECLAREステートメントは、データ型を変数リストまたは配列内の各変数に関
連付けます。メソッド外で DECLAREステートメントを使用すると、グローバル
変数が作成されます。メソッド内で DECLAREステートメントを使用すると、ロ
ーカル変数が作成されます。メソッド内では、DECLAREステートメントはメソ
ッドステートメントの前に置く必要があります。それ以外の場合は、エラーが発
生します。
明示的に宣言された変数には、Variable属性が割り当てられます。
3つのグローバル値を明示的に宣言し、それらに値を割り当てるコード例を次に
示します。
proc ds2;
data input1 (overwrite=yes);
   dcl double x y z;
   method init();
      x=1; y=2; z=3;
      output;
   end;
enddata;
run;
quit;

詳細については、“変数スコープ” (34ページ)および“DECLARE ステートメント” 
(SAS DS2言語リファレンス)を参照してください。

n SETまたはMERGEステートメントを使用した暗黙的な宣言

SETまたはMERGEステートメントは、指定された各テーブルの列情報を読み取
ります。各テーブルの各列に対して、SETまたはMERGEステートメントは、列
と同じデータ型を持つグローバル変数を DS2プログラムに作成します。
暗黙的に宣言された変数には、Label属性が割り当てられます。
SETステートメントを使用して変数を暗黙的に宣言するコード例を次に示しま
す。
proc ds2;
data results (overwrite=yes);
   method run();
      set input1;
   end;
 enddata;
 run;
 quit;

PROC COMPAREからの次の出力は、テーブル Input1の列に variable属性があ
り、テーブル Resultsの列に label属性があることを示しています。

proc compare data=input1 compare=results error note criterion=1e-4;
run;

図 3.1 PROC COMPARE出力

詳細については、“SETステートメントを使用したデータの読み取り” (250ペー
ジ)、“MERGE ステートメント” (SAS DS2言語リファレンス)、および“SET ステー
トメント” (SAS DS2言語リファレンス)を参照してください。

n プログラミングブロックで宣言されていない変数を使用した暗黙的な宣言

注: すべての変数を明示的に宣言することを強くお勧めします。そうすること
で、データソース間のデータ型の不一致を回避できます。

変数を宣言せずに使用すると、DS2は変数の属性を次のように割り当てます。
o 割り当てステートメントの左側にある宣言されていない変数のデータ型は、
割り当てステートメントの右側にある値のデータ型によって決定されます。
詳細は次のとおりです。
n 割り当てステートメントの右側の値のデータ型が数値(DOUBLE、BIGINT、

DECIMALなど)、ANSI NULL、または SAS数値欠損値である場合、DOUBLE
型が左側の変数に割り当てられます。

n 割り当てステートメントの右側の値のデータ型が文字(CHAR、VARCHAR、
NCHARなど)または SAS文字欠損値である場合、固定長文字が左側の変数
に割り当てられます。

n 割り当てステートメントの右側の値のデータ型がバイナリ(BINARY、
VARBINARY)の場合、左側の変数には固定長バイナリが割り当てられます。

n 割り当てステートメントの右側の値のデータ型が DATE、TIME、または
TIMESTAMPの場合、左側の変数には同じ型が割り当てられます。

o 文字変数の長さは、右側の値が SAS文字欠損値でない限り、右側の値の長さ
です。SAS文字欠損値の長さは 200文字です。
割り当てステートメントの右側が式の場合、左側の変数の長さは右側の式の
結果の長さになります。この例では、変数 bの長さは 9です。これは、xと
yの文字列の長さを加算した結果です。

x='abc';
y='rstuvw';
b=x || y;

o 割り当てステートメントの右側にある宣言されていない変数のデータ型は
DOUBLEです。

o 暗黙的に宣言された文字型には、変数の文字セットのセッションエンコーデ
ィングが割り当てられます。

デフォルトでは、宣言されていない変数が作成されると、SASログに警告が送信
されます。警告は、宣言されていない変数に割り当てられているデータ型、長
さ、および必要に応じて精度を示します。

ヒント DS2SCONDシステムオプションまたは PROC DS2 SCONDオプ
ションを使用して、警告のかわりにエラーまたは注をログに発行できま
す。オプションを使用すると、宣言されていない変数についてレポートし
ないように指定することもできます。

割り当てステートメントの詳細については、“DS2による宣言されていない変数
の処理方法の制御” (27ページ)を参照してください。

次の例は、一部の変数を明示的に定義し、一部の変数を暗黙的に定義する方法を示
しています。FIRSTNAMEと LASTNAMEは、データ型が CHAR(20)の変数として明
示的に宣言されています。PNUMは割り当てによって暗黙的に宣言されます。ログ
に書き込まれた警告からわかるように、DS2は割り当ての右側の値(CHAR(11))に基
づいて PNUMにデータ型を割り当てます。
proc ds2;
data phonenums;
   dcl char(20) firstName lastName;
   method init();
      firstName='Sam';
      lastName='Alesski';
      pnum = '19192223454';
   end;
enddata;
run;
quit;

WARNING: Line 57: No DECLARE for assigned-to variable pnum; creating it as a
   global variable of type char(11).

割り当てと式を使用する場合は、型変換と欠損値に注意する必要があります。次の
例では、等値オペランドが DOUBLEオペランドと VARCHARオペランドで評価され
ます。DOUBLEオペランドは VARCHARオペランドよりも優先されるため、
VARCHARオペランドは DOUBLEに変換されます。文字列'on1'は有効な数値では
ないため、変換の結果、右側のオペランドの SAS欠損値が数値になります。変数 in2
はプログラムによって初期化されていないため、in2も SAS欠損値に設定されます。
SASモードでは、DOUBLEデータ型のmissing = missingは真と評価されます。
proc ds2;
data _null_;
   dcl varchar(100) "results";
   dcl varchar(100) "in1";

   method run();
      "in1"='onlystring2';

      /* Mistype in2 instead of in1 */
      if ("in2"='onl') then do;
         "results" = 'Rule is fired AGAIN';
      end;
      put 'here are the results';
      put "results";
      end;
enddata;
run;
quit;

次の行が SASログに書き込まれます。

here are the results
 Rule is fired AGAIN
 WARNING: Line 106: No DECLARE for referenced variable in2; creating it as a global variable of type double.

DS2による宣言されていない変数の処理方法の制
御

DS2が宣言されていない変数を処理する方法を制御するには、DS2SCONDシステム
オプションまたは DS2プロシジャの SCONDオプションを使用します。
表 3.1 変数宣言を制御するための設定

DS2SCOND/
SCOND設定 変数宣言への影響
WARNING 割り当てによる宣言が発生します。警告メッセージが SASログに

書き込まれます。これがデフォルトの動作です。

DS2SCOND/
SCOND設定 変数宣言への影響
NONE 割り当てによる宣言が発生します。SASログにメッセージは書き込

まれません。
NOTE 割り当てによる宣言が発生します。SASログに注が書き込まれま

す。
ERROR 割り当てによる宣言は発生しません。SASログにエラーメッセージ

が書き込まれます。これは、変数宣言の strictモードとも呼ばれま
す。

詳細については、“DS2SCOND= システムオプション” (SAS DS2言語リファレンス)お
よび“DS2 プロシジャ” (Base SASプロシジャガイド)を参照してください。

DS2変数リスト
変数リストの概要

DS2には、変数のリストを操作する方法が 2つあります。
n DS2は、省略構文を使用した、類似した名前を持つか、または類似した型の複数
のグローバル変数の選択をサポートします。この省略構文は、その構文で変数名
のリストを指定する必要があるステートメント(DROPや KEEPなど)で役立ちま
す。詳細については、“変数名を指定するための省略構文”を参照してください。

n DS2は、変数をグループ化するためのデータ構造をサポートしています。DS2
は、同種データ型のグローバル変数をグループ化するための配列をサポートして
います。DS2は、異種データ型のグローバル変数をグループ化するための varlist
をサポートしています。これらのデータ構造により、入力として変数のグループ
を必要とする DS2メソッド、パッケージ、および関数に変数グループ化をサブ
ミットできます。データ構造を使用すると、グループ化の要素のデータ値を設
定、取得、および渡すことができます。詳細については、11章, “DS2配列” (121
ページ)および 12章, “DS2 varlist” (145ページ)を参照してください。

変数名を指定するための省略構文
注: 省略された変数選択構文は、グローバル変数から選択します。省略形は、DOT
演算子でアクセスされるローカル変数、メソッドパラメーター変数、またはパッケ
ージ変数を参照しません。

多くの DS2ステートメントとテーブルオプションは、複数の変数名を指定する省略
構文をサポートしています。たとえば、KEEP、DROP、VARARRAY、VARLISTステ
ートメントでは、省略形を使用して、保持、削除、または参照する変数の名前を指
定できます。省略構文の例を次に示します。
keep name address city state zip phone;
 drop rep1-rep5;
 vararray int grades[*] assignment: quiz: exam:;

DS2は、次の省略リストをサポートしています。
n 名前変数リスト
n 番号付き範囲変数リスト
n 名前範囲変数リスト
n 名前接頭辞変数リスト
n 型変数リスト
n 特殊名変数リスト
1つの変数リストの中に、異なる形式の省略変数リストを混在させることができま
す。例えば、次のような変数リストが有効です。
u x1-x3 u:

この変数リストでは:

n uは、uという名前の単一の変数を参照する名前リストです。
n x1-x3は、x1、x2、および x3という名前の変数を参照する番号付き範囲リストで
す。

n u:は、名前が文字列"u"で始まるすべての変数を参照する名前接頭辞リストです。
したがって、 u:は、u、unit、user、uniformを指します。

上述の変数リストは u x1 x2 x3 u unit user uniformに展開されます。変数リストの展
開では、1つの変数を複数回参照できることに注意してください。
詳細については、“省略変数リストの種類” (30ページ)を参照してください。
varlistは、KEEP、DROP、VARARRAY、VARLISTステートメントでは指定できませ
ん。

省略変数リストの種類
表 3.2 省略変数リストの種類のまとめ

種類 定義 例
名前リスト グローバル変数名のリスト。 drop location pressure temperature;

番号付き範囲リス
ト

同じ名前の接頭辞が付いている
が、最後の文字の数値が異なる
グローバル変数で構成されるリ
スト。値は連続している必要が
あります。

keep x1-x5; 

これは、範囲として指定され
ていない次の番号付きリス
トと同等です。
keep x1 x2 x3 x4 x5;

名前範囲リスト 変数定義の順序に依存する名前
範囲リスト。

keep sales_jan--sales_mar;

すべての列をプログラムデ
ータベクトルの順序どおり
に指定します(keep sales_jan
から sales_marまで)。

名前接頭辞リスト 指定した文字列で始まるすべて
のグローバル変数を参照するリ
スト。

keep sales:;

sales_jan、sales_feb、sales_mar
など、名前が"sales"で始まる
すべての変数に展開されま
す。

型リスト 指定された DS2データ型のすべ
てのグローバル変数を含むリス
ト。複数の型を指定できます。
詳細については、“型変数リスト” 
(31ページ)を参照してくださ
い。

keep double;

DOUBLE型のすべてのグロ
ーバル変数を DATAブロッ
クに含めるように展開しま
す。
varlist a [smallint int];

SMALLINTおよび INT型の
すべてのグローバル変数を
DATAブロックに含めるよう
に展開します。

特殊名リスト 読み取り専用のグローバル変数
の特定のグループを参照するた
めに展開されるリスト。_ALL_
変数がサポートされています。
_ALL_は、DS2プログラム内のす
べてのグローバル変数を参照し
ます。

varlist x [ _all_ ];

SETステートメントで指定さ
れたテーブルの変数を含む、
DS2プログラムのすべての
グローバル変数を varlistに
含め、グループ化に名前 xを
割り当てます。

種類 定義 例
詳細については、“事前定義され
た DS2変数” (36ページ)を参
照してください。

注: 名前範囲リストでは、2つのハイフン( - - )を使用して変数間の範囲を指定し、番
号付き範囲リストでは単一のハイフン( - )を使用して範囲を指定することに注意し
てください。

型変数リスト
型変数リストは展開され、指定された型のすべてのグローバル変数を参照します。
型変数リストの構文は次のとおりです。
data-type

型変数リストでサポートされている型は次のとおりです。
n BIGINT

n BINARY (BINARYおよび VARBINARYに一致)

n CHAR (CHAR、VARCHAR、および CHARACTERに一致)

n CHARACTER (CHAR、VARCHAR、および CHARACTERに一致)

n DATE

n DECIMAL

n DOUBLE (DOUBLEおよび FLOATに一致)

n FLOAT (DOUBLEおよび FLOATに一致)

n INTEGER

n NCHAR (NCHARおよび NVARCHARに一致)

n NUMERIC

n REAL

n SMALLINT

n TIME

n TIMESTAMP

n TINYINT

型変数リストを宣言するコードの例を次に示します。
proc ds2;
  data _null_;
  dcl double x y z;
  vararray double a[*] double;

  ...
  enddata;
  run;
quit;

DOUBLE型のすべてのグローバル変数(x、y、および z)は、変数配列 aに含まれま
す。
VARARRAYステートメントと VARLISTステートメントで型リストを指定する方法
の例を次に示します。

vararray decimal(8,4) va[*] decimal;
varlist vl[numeric];

例: 省略変数リストの混在
このプログラム例では、KEEPステートメントで型変数リスト(double)と名前変数リ
ストを指定します。

proc ds2;
data OnlyDoubleAndC / overwrite=yes;
  dcl double x y;
  dcl char(10) a b c;
  keep double c;
  method init();
    x = 10; y = 20;
    a = 'apple'; b = 'banana'; c = 'cherry';
    output;
  end;
enddata;
run;
quit;

proc print data=OnlyDoubleAndC;
run;

SASログに書き込まれる出力は次のとおりです。

Obs     x     y      c
1     10    20    cherry

出力には、DOUBLEとして定義された変数(xおよび y)が含まれます。さらに、名前
変数 cが含まれます。

関数の引数として省略変数リストを指定する
OF演算子を使用すると、変数リストを一部の関数の引数として指定できます。OF
演算子は、省略変数リストの指定を変数名のカンマ区切りリストに変換し、これが
関数の引数になります。

注: OF演算子の使用にはルールと制限があります。詳細については、“OF演算子” 
(91ページ)を参照してください。

名前接頭辞変数リストを引数として SUM関数にサブミットする例を次に示します。
proc ds2;
data X (overwrite=yes);
   dcl double x A1 A2 A4 y;
   method run();
      A1 = 1.0;
      A4 = 3.0;
      y = 1;
      lab:
      if (y = 2) then
         X = SUM(OF A: );
         A2 = 2.0;
      if (y = 1) then 
      do;
         y = 2;
      goto lab;
      end;
      put X=;
   end;    
enddata;
run;
quit;

SASログに書き込まれる出力は次のとおりです。
x=6

この例では、OF演算子は、名前が文字 Aで始まるすべてのグローバル変数を計算に
含めるよう SUM関数に指示します。

省略変数リストの展開
省略変数リストの指定は、省略形に一致するグローバルスカラー変数を参照するよ
うに展開されます。ローカル変数が変数リスト展開に含まれることはありません。
次の例では、変数 x1はグローバル変数であり、変数 x2はメソッド INITに対してロ
ーカルです。したがって、ローカル変数 x2はテーブルに書き込まれません。
proc ds2;
data example (overwrite=yes);
   declare double x1;
   keep x:;
 
   method init();
      declare double x2;
   end;
enddata;
run;
quit;

DS2変数リスト 33



変数リストの展開では、宣言されていない変数が考慮されます。次の例では、KEEP
ステートメントの_ALL_が展開され、グローバル変数 x1、x2、および x3を参照しま
す。DS2は、DECLAREステートメントで指定されていないため、変数 x3をグロー
バル変数として作成します。したがって、変数 x1、x2、および x3はテーブル example
に書き込まれます。
proc ds2;
data example (overwrite=yes);
   declare double x1;
   keep _all_;
   declare double x2;
 
   method init();
      x3=17;
   end;
enddata;
run;
quit;

変数スコープ
グローバル変数

グローバルスコープを持つ変数であるグローバル変数は、次の 3つの方法のいずれ
かで宣言されます。DS2プログラムの最も外側のプログラミングブロックで
DECLAREステートメントを使用する方法、プログラミングブロック内で SETステー
トメントを使用して暗黙的に宣言する方法、またはプログラミングブロック内で宣
言されていない変数を使用して暗黙的に宣言する方法です。グローバルスコープを
持つ変数は、プログラムのどこからでもアクセスでき、プログラムの期間中存在し
ます。グローバル変数は、任意のプログラムブロックの THIS式で使用できます。
DS2変数の宣言については、“変数宣言” (24ページ)を参照してください。THIS式の
使用については、“THIS式” (97ページ)を参照してください。

ローカル変数
ローカルスコープを持つ変数であるローカル変数は、DECLAREステートメントを使
用して、メソッドやパッケージなどの内側のプログラミングブロック内で宣言され
ます。ローカルスコープを持つ変数は、変数が宣言されている内側のプログラミン
グブロック内でのみ認識されます。詳細については、“プログラミングブロック” (17
ページ)を参照してください。

DS2出力のグローバル変数とローカル変数
デフォルトでは、グローバル変数のみが出力に含まれます。プログラムループとイ
ンデックスに使用されるローカル変数は、出力から明示的に削除する必要はありま
せん。ローカル変数は、RUNメソッドの暗黙的なループの反復など、メソッド呼び
出しの開始時に常に作成され、各呼び出しの最後に破棄されます。したがって、RUN
メソッドでローカル変数をアキュムレータ変数として使用することはお勧めしませ
ん。
すべてのグローバル変数は、プログラムデータベクトル(PDV)で名前が付けられま
す。PDVは、DS2が行を書き込むときに出力テーブルに書き込まれる値のセットで
す。PDVの詳細については、“Processing a DATA Step: A Walk-Through” (SAS 
Programmer’s Guide: Essentials)を参照してください。

グローバル変数とローカル変数の例
次のプログラムは、グローバル変数(A、B、および TOTAL)とローカル変数(C)の両方
を示しています。

proc ds2;
data;
   dcl int a;                 /* 1 */
   method init();
      dcl int c;              /* 2  */
      a = 1;                  /* 3 */
      b = 2;                  /* 4 */
      c = a + b; 
      this.total = a + b + c; /* 5 */
   end;
enddata;
run;
quit;

1 Aは、最も外側の DS2プログラムで宣言されているため、グローバル変数です。
2 Cは、メソッドブロックMETHOD INIT()内で宣言されているため、ローカル変数
です。

3 Aはグローバル変数であるため、メソッドブロックMETHOD INIT()内で参照でき
ます。

4 BはMETHOD INIT()で宣言されていないため、デフォルトでグローバル変数にな
ります。DS2は Bに DOUBLEのデータ型を割り当てます。Bは PDVと出力テ
ーブルに表示されます。

5 THIS.TOTALは同時に、変数 TOTALをデータ型 DOUBLEのグローバル変数とし
て宣言し、A、B、および Cの値に基づいて値を割り当てます。

事前定義された DS2変数
事前定義されたメソッド変数

事前定義された変数は、メソッドブロック内で自動的に宣言され、メソッドが完了
すると破棄される変数です。事前定義された変数の値は、RUNメソッドのある反復
から次の反復まで保持されます。これらの変数はプログラムデータベクトルに追加
されますが、作成中のテーブルには書き込まれません。事前定義された変数は一時
的なものであり、データとともに保存されることはありません。事前定義された変
数はメソッドブロックで使用でき、自分で宣言した変数と同じように使用できます。
2つの事前定義された変数が作成されます。
_N
初期値は INITメソッドによって 0に設定されます。RUNメソッドが実行され
るたびに、値は 1ずつ増加します。_Nのデータ型は BIGINTです。これは読み
取り専用の変数です。_Nに値を割り当てることはできません。

_N_
初期値は INITメソッドによって 1に設定されます。RUNメソッドが実行され
るたびに、値は 1ずつ増加します。つまり、RUNメソッドが初めて実行される
とき、_N_の値は 1です。この値は、プログラムがメソッドをループした回数を
示します。この変数は行をカウントするために使用できますが、反復ごとに 1つ
のレコードが読み取られる場合にのみ、適切な行カウンターになります。_N_の
データ型は BIGINTです。_N_に数値を割り当てることはできますが、RUNメソ
ッドの次の反復では、値はプログラムによって割り当てられた値に戻ります。

事前定義された変数は出力変数ではありませんが、PUTステートメントに_ALL_引数
を指定すると、事前定義された変数の値を SASログに出力できます。
次の単純な DS2プログラムは、PUT _ALL_ステートメントを使用して、_Nおよび
_N_変数の値を SASログに出力する方法を示しています。
proc ds2;
data inp /overwrite=yes;
   dcl double a;
   method init();
      a = 1; output;
   end;
enddata; run;
    
data;
   method init();
     put 'init' _n=; put _ALL_;
   end;
   method run();
     set inp;
     put 'run' _n=; put _ALL_;

   end;
   method term();
      put 'term' _n=; put _ALL_;
   end;
enddata;
run;
quit;

例のコード 3.1 _Nおよび_N_変数値を含む SASログ

   .
   .
   .
NOTE: Execution succeeded. One row affected.
init _n=0
a=. _n_=1
run _n=1
a=1 _n_=1
term _n=2
a=1 _n_=2

DS2スレッド変数
次の自動変数は、DS2スレッド間で問題をサブセット化するために使用されます。
これらの自動変数は、PUTステートメントを使用してデバッグするときにコンテキ
ストを提供するのにも役立ちます。
n _HOSTNAME_

n _LOCALNTHREADS_

n _LOCALTHREADID_

n _NTHREADS_

n _THREADID_

詳細については、“DS2スレッドで役立つ自動変数” (245ページ)を参照してくださ
い。

定数の定義
定数は、固定値を示す数値、文字列、2進数、日付、時間、タイムスタンプのいず
れかです。DS2定数の例を次に示します。
107
'Trends in Business'
date '2008–01–01'
b'10011001'
x'FFE3546F'

数値定数
数値定数は、整数定数または小数定数のいずれかである負または正の数値です。

整数定数
小数部分のない数値、つまり整数です。整数定数値は正確な数値として保存され
ます。Nまたは n接尾辞のない整数定数は、BIGINT、INTEGER、SMALLINT、
TINYINTのいずれかのデータ型の値です。Nまたは n接尾辞が付いた整数定数
は、DECIMALまたは NUMERICデータ型の値です。整数定数の形式は次のとお
りです。ここで、integerは 0から 9までの 1つ以上の数字のシーケンスです。
integer
integerN|n

次の値は、有効な整数定数です。

0
124
124N
+124n
-124n

小数定数
小数部分を持つ数値です。小数定数値は、正確な数値または近似数値として保存
されます。Nまたは n接尾辞のない小数定数は、近似数値として保存される
REALまたは DOUBLEデータ型の値です。Nまたは n接尾辞が付いた小数定数
は、データ型の最大精度までの正確な数値として保存される DECIMALまたは
NUMERICデータ型の値です。近似分数定数は、標準表記または指数(E)表記をサ
ポートします。正確な小数定数は、標準表記のみをサポートします。小数定数の
形式は次のとおりです。ここで、integer、fraction、および exponentは、0から
9までの 1つ以上の数字のシーケンスです。
integer.

integer.N|n
integer.fraction
.fraction
integer.fractionE|e[+|-]exponent
integerE|e[+|-]exponent
.fractionE|e[+|-]exponent
integer.fractionN|n
.fractionN|n

次の値は、有効な小数定数です。

0.
124.0
+124.0
-57.33
.5
1E100
99.99e4
-.33e-2
-57.33n
123456789012345678901234567890.12N

文字定数
文字定数の概要

文字定数は、単一引用符で囲まれた一連の文字であり、次の形式を使用して記述で
きます。
'character-string'

n'character-string'

各国文字を含む文字列の場合は、NCHAR定数 n'character-string'を使用してくださ
い。各国文字は、複数バイトのストレージを使用できます。
文字定数と NCHAR定数には、複数行にわたる文字定数の改行を含めることができ
ます。

注: 二重引用符で囲まれた一連の文字は文字定数ではなく、識別子です。詳細につい
ては、“区切り識別子” (58ページ)を参照してください。

有効な文字定数は次のとおりです。
'PHONE'
n'STÄDTE'
'Phone 
Number'  /* constant that spans more than one line */

注: 文字列に単一引用符(')またはアポストロフィ(')を含めるには、追加の引用符を使
用します。次に例を示します。

'Isn''t life beautiful'

長い文字定数
長い文字列定数(NCHAR、NVARCHAR、および VARCHARデータ型)は、割り当て時
に二重角かっこ([[ ]])で囲むことができます。これは、LUAプログラミング言語が文
字列を区切る方法に似ています。
長い文字定数は、[[、[=[、[==[など(つまり、0個以上の等号が間にある 2つの左角
かっこ)で始まります。長い文字定数は、一致する]]、]=]、]==]など(つまり、同じ数の
等号が間にある 2つの右角かっこ)で終わります。終了トークンは、開始トークンの

パターンと一致する必要があります。たとえば、[=[は]]または]==]では閉じません。
一致する]=]トークンでのみ閉じます。次に例を示します。
proc ds2;
data;
   dcl nvarchar(1000) cc;
   method init();
      cc = [====[
         proc ds2; 
         data;
         dcl nchar(42) cc;
         method init();
         cc = [=[
         this is a test
         ]=];
         put '|' cc= '|';
         end;
         enddata;
      ]====];
      put '|' cc= '|';
   end;
enddata;
run; 
quit;

アウトプット 4.1 定数角かっこの例からの出力

注: 長い文字定数には改行が含まれる場合がありますが、改行は必須ではありませ
ん。これは、プログラムの動作に影響を与えずに n'xyzzy'を[==[xyzzy]==]に置き換
えることができることを意味します。

注意
二重角かっこ内のコメントは破棄されます。

注意
角かっこで囲まれた長い文字定数の中に"run;"または"quit;"を入れないでくださ
い。 角かっこで囲まれた長い文字定数の中に"run;"または"quit;"を入れると、コンパイ
ルエラーが発生し、プログラムが失敗します。

DS2定数の文字エンコーディング
すべての DS2文字データには、文字エンコーディングが関連付けられています。文
字定数についても同様です。DS2文字定数値は、DS2セッションエンコーディング

または DS2国別エンコーディングのいずれかでエンコードされます。各国文字変
数には、DS2セッションエンコーディングに関係なく、文字エンコーディングとし
て常に UTF-8が割り当てられます。
DS2セッションエンコーディングを使用してエンコードされる文字定数を指定する
場合、DS2プログラムテキストで定数の接頭辞を指定しないでください。
'Hello World'
'你好世界'

DS2各国文字エンコーディングを使用してエンコードされる文字定数を指定するに
は、DS2プログラムテキストで定数の前に nを付けます。
n'Hello World'
n'你好世界'

DS2プログラムテキストのエンコーディングは、DS2プログラムの処理中に文字定
数に割り当てられたエンコーディングとは異なる場合があります。たとえば、DS2
プログラムテキストの文字エンコーディングが UTF-8で、DS2セッションエンコー
ディングが latin-1の場合、次の PUTステートメントの文字定数は、プログラムの
処理中にトランスコードに失敗します。
put '你好世界';

文字定数は各国文字定数として指定されていないため、文字定数には DS2セッショ
ンエンコーディングの latin-1が割り当てられます。latin-1文字セットは、漢字'你
好世界'をサポートしていません。したがって、UTF-8 (DS2プログラムテキストエン
コーディング)から latin-1 (DS2セッションエンコーディング)へのトランスコード
は失敗します。トランスコードの失敗は、DS2セッションエンコーディングを漢字
をサポートするエンコーディング(EUC-CNや UTF-8など)に変更するか、DS2プログ
ラムテキストで nを前に付けて DS2各国文字エンコーディングを使用してエンコ
ードする文字定数を指定することで解消できます。

文字定数でのマクロ変数の参照
文字定数でマクロ変数を参照するには、SASマクロ関数%TSLITを使用します。これ
により、定数値を単一引用符で囲む必要がなくなります。たとえば、次のステート
メントでは%TSLIT関数を使用してマクロ変数&profitを参照します。
put %tslit(PROFIT: &profit);

%TSLITマクロ関数は、デフォルトの自動呼び出しマクロライブラリに格納されてい
ます。構文は次のとおりです。
%TSLIT(literal-text);

%TSLITマクロ関数は、リテラルテキスト引数でマクロ変数を参照し、結果のテキス
ト値を単一引用符で囲みます。これにより、DS2文字定数値が作成されます。

2進定数
2進定数は、次の形式を使用して、ビット文字列または 16進文字列として記述でき
ます。
b'bit-string'

x'hexadecimal-string'

bit-stringは、0と 1の 2進数のシーケンスです。
hexadecimal-stringは、0 - 9および A - Fの 16進文字の文字列です。
bit-stringと hexadecimal-stringの両方を単一引用符で囲む必要があります。
DS2は、DS2プログラムで使用できるように、ビット定数と 16進定数の両方を同
等の 2進数に変換します。2進定数と 16進定数は、バイト境界までゼロで埋められ
ます。
2進定数の例を次に示します。
b'11100011'
x'FF143E99'

16進定数を文字変数に割り当てる場合、割り当てを有効にするには、16進定数がセ
ッションエンコーディングと同じエンコーディングである必要があります。テキス
ト値は、文字エンコーディングを使用して列のデータ値にエンコードされます。特
定のテキスト値を格納するために使用されるバイトは、エンコーディングによって
異なります。たとえば、UTF-8および Latin-1文字エンコーディングでの文字列 ágê
の 16進表現を考えてみましょう。
表 4.1 文字列 ágêの UTF-8および Latin-1エンコーディング

文字エンコーディング 16進数のテキストデータ
UTF-8 C3 A1 67 C3 AA

Latin-1 E1 67 EA

UTF-8セッションで Latin-1エンコーディングの 16進値を指定すると、エラーが発
生するか、単に正しくない結果になる可能性があります。UTF-8セッションに適切
な 16進値をサブミットする DS2コード例のログ出力を次に示します。

1          proc ds2;
2             data _null_;
3                method init();
4                dcl varchar(100) s1;
5                s1 = x'C3A167C3AA';
6                put s1=;
7                end;
8             enddata;
9          run;
s1=ágê
NOTE: Execution succeeded. No rows affected.
quit;

注: 入力値から空白スペースを削除する必要があります。

DS2には、数値定数の 16進表記がありません。

日付と時間の定数
日付と時間の定数は、次の形式を使用して、日付、時間、またはタイムスタンプの
文字列として記述できます。
date 'yyyy-mm-dd'

time 'hh:mm:ss'

time 'hh:mm:ss.fraction'

timestamp 'yyyy-mm-dd hh:mm:ss'

timestamp 'yyyy-mm-dd hh:mm:ss.fraction'

yyyyは 4桁の年です。
mmは 2桁の月です。
ddは 2桁の日です。
hhは 2桁の時間です。
mmは 2桁の分です。
ssは 2桁の秒です。
fractionは、秒の端数を表す一連の数値です。
日付と時間の定数はすべて、単一引用符で囲む必要があります。年、月、日はハイ
フンで区切る必要があります。時、分、秒はコロンで区切る必要があります。秒と
秒の端数はピリオドで区切る必要があります。タイムスタンプ定数では、日付と時
間をスペースで区切ります。
日付定数の長さは、年、月、日を区切るハイフンを含めて 10文字にする必要があり
ます。
端数を除いた時間定数の長さは、時、分、および秒を区切るコロンを含めて 8文字
にする必要があります。端数を含む時間定数の長さは、秒の端数によって変化する
可能性があります。

端数を除いたタイムスタンプ定数の長さは、日付と時間の構成要素を区切るハイフ
ンとコロンを含めて 19文字にする必要があります。端数を含むタイムスタンプの
長さは、秒の端数によって変化する可能性があります。
日付と時間の定数の例を次に示します。
date '2008-01-01'
time '11:59:59'
timestamp '2007-09-30 02:33:31.59'

定数リスト
定数リストは、単純な配列割り当ておよび IN式で使用できる、カンマまたは空白で
区切られた、かっこで囲まれた一連の定数値です。定数リストには、ネストされた
定数リストを含めることもできます。定数リストの例は(2, '33', –5, (1, '3'))です。この
例では、(1, '3')はネストされた定数リストです。
各定数値またはネストされたリストには、反復子を接頭辞として付けることができ
ます。反復子とは、数字の後にアスタリスク( * )を付けたものです。反復子は、アス
タリスクに続く定数または定数リストを繰り返す回数を示します。たとえば、定数
リスト(2, '33', 3*–5, 2*(1, '3'))では、リスト項目 3* –5'の結果は–5, –5, —5、およびリス
ト項目 2*(1,'3')の結果は 1, '3', 1, '3'になります。反復子"5*"を使用することと、リスト
要素を 5回繰り返すことに違いはありません。

注: IN式で値が重複すると処理が遅くなる可能性があるため、反復子が IN式で役立
つ可能性はほとんどありません。

46 4章 / DS2定数



5
DS2データ型

データ型とは . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 47

データ型の特性 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 51
数値データ型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 51
文字データ型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 52
日付と時間のデータ型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 54
CASのバイナリデータ型 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 55

列のデータ型を定義する . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 55

DOUBLEおよび REALデータ型を使用するエラーメッセージ . . . . . . . . . . . . . . . . . . . . . . . . 56

データ型とは
データ型は、列が格納するデータの型を指定するすべての列の属性です。データ型
は、データの一部を文字列、整数、浮動小数点数、日付または時間として識別する
特性です。データ型は、列の値に割り当てるメモリの量も決定します。
次の表に、DS2でサポートされている一連のデータ型を示します。各データソース
のテーブルストレージにすべてのデータ型を使用できるわけではないことに注意し
てください。
表 5.1 DS2データ型

データ型 説明
BIGINT 19桁の精度で大きな符号付きの正確な整数を格納しま

す。整数の範囲は-9,223,372,036,854,775,807から
9,223,372,036,854,775,807までです。整数データ型は小
数値を保存しません。小数部分は破棄されます。

BINARY(n) 固定長のバイナリデータを格納します。ここで、nは格納
する最大バイト数です。値の実際のサイズに関係なく、各
値を格納するには最大バイト数が必要です。

データ型 説明
CHAR(n) 固定長の文字列を格納します。ここで、nは格納する最大

文字数です。値の実際のサイズに関係なく、各値を格納す
るには最大文字数が必要です。char(10)が指定され、文字
列の長さが 5文字のみの場合、値の右側にスペースが埋め
込まれます。

DATE カレンダー日付を格納します。日付リテラルは、yyyy-mm-
ddの形式で指定されます。これは、4桁の年(0001 - 
9999)、2桁の月(01 - 12)、および 2桁の日(01 - 31)です。
たとえば、1975年 9月 24日の日付は 1975-09-24と指定
されます。
DS2は、日付に関する ANSI SQL:1999標準に準拠してい
ます。ただし、すべてのデータソースが日付の全範囲をサ
ポートしているわけではありません。たとえば、
0001-01-01から 1582-12-31までの日付は、SASデータセ
ットまたは SPDデータセットの有効な日付ではありませ
ん。

DECIMAL|NUMERIC(p,s) ユーザー指定の精度とスケールで、符号付きの正確な固定
小数点 10進数を格納します。精度とスケールによって小
数点の位置が決まります。精度は、小数点の左右に格納で
きる最大桁数で、範囲は 1 - 52です。スケールは、小数点
以下に格納できる最大桁数です。スケールは精度以下に
する必要があります。たとえば、decimal(9,2)は、
1234567.89のように、2桁の固定小数点小数部分を含む
9桁までの 10進数を格納します。

DOUBLE 符号付き近似倍精度浮動小数点数を格納します。大きな
数値を許容し、小数点以下の桁数の精度を必要とする計算
を可能にします。

FLOAT 符号付き近似倍精度浮動小数点数を格納します。FLOAT
として定義されたデータは、DOUBLEと同じように扱われ
ます。

INTEGER 10桁の精度で通常サイズの符号付きの正確な整数を格納
します。整数の範囲は-2,147,483,647から
2,147,483,647までです。整数データ型は小数値を保存し
ません。小数部分は破棄されます。
注: ゼロによる整数除算は、すべてのオペレーティングシ
ステムで同じ結果になるわけではありません。ゼロによ
る整数除算は避けることをお勧めします。

NCHAR(n) CHARのような固定長の文字列を格納しますが、Unicode
各国語文字セットを使用します。nは、格納するマルチバ
イト文字の最大数です。プラットフォームに応じて、

データ型 説明
Unicode文字は 1文字あたり 2バイトまたは 4バイトを
使用し、すべての国際文字をサポートします。

NVARCHAR(n) VARCHARのような可変長の文字列を格納しますが、
Unicode各国語文字セットを使用します。nは、格納する
マルチバイト文字の最大数です。プラットフォームに応
じて、Unicode文字は 1文字あたり 2バイトまたは 4バ
イトを使用し、すべての国際文字をサポートできます。

REAL 符号付き近似単精度浮動小数点数を格納します。
SMALLINT 5桁の精度で小さな符号付きの正確な整数を格納します。

整数の範囲は-32,767から 32,767までです。整数データ
型は小数値を保存しません。小数部分は破棄されます。

TIME(p) 時間値を格納します。時間リテラルは、
hh:mm:ss[.nnnnnnnnn]の形式で指定します。これは、2桁
の時間 00から 23、2桁の分 00から 59、2桁の秒、およ
びオプションの小数値です。たとえば、午前 6:30は
06:30:00と指定されます。データソースでサポートされて
いる場合、pパラメーターは秒の精度を指定します。pは、
最大 9桁の長さのオプションの小数値です。

TIMESTAMP(p) 日付と時間の両方の値を格納します。タイムスタンプリ
テラルは、yyyy-mm-dd hh:mm:ss[.nnnnnnnnn]の形式で指
定します。これは、4桁の年 0001から 9999、2桁の月 01
から 12、2桁の日 01から 31、2桁の時間 00から 23、2
桁の分 00から 59、2桁の秒、およびオプションの小数値
です。たとえば、1975年 9月 24日午前 6:30という日付
と時間は 1975-09-24 06:30:00と指定されます。データソ
ースでサポートされている場合、pパラメーターは秒の精
度を指定します。pは、最大 9桁の長さのオプションの小
数値です。

TINYINT 3の精度で非常に小さな符号付きの正確な整数を格納し
ます。整数の範囲は-127から 127までです。整数データ
型は小数値を保存しません。小数部分は破棄されます。

VARBINARY(n) 可変長のバイナリデータを格納します。ここで、nは格納
する最大バイト数です。各値を格納するために最大バイ
ト数は必要ありません。varbinary(10)が指定され、バイナ
リ文字列が 5バイトのみを使用する場合、列には 5バイト
のみが格納されます。

VARCHAR(n) 可変長の文字列を格納します。ここで、nは格納する最大
文字数です。各値を格納するために最大文字数は必要あ
りません。varchar(10)が指定され、文字列の長さが 5文字
のみの場合、列には 5文字のみが格納されます。

データ型とは 49



次のプログラムは、DS2で使用可能な多くのデータ型の宣言の例です。プログラム
を正常に実行するには、これらのデータ型をサポートするデータソースに接続する
必要があることに注意してください。
data ds2data (overwrite=yes);
    dcl int           v01;
    dcl bigint        v02;
    dcl smallint      v03;
    dcl tinyint       v04;
    dcl double        v05;
    dcl char(8)       v06;
    dcl varchar(8)    v07;
    dcl nchar(8)      v08;
    dcl nvarchar(8)   v09;
    dcl date          v10;
    dcl time          v11;
    dcl timestamp     v12;
    dcl binary(8)     v13;
    dcl varbinary(32) v14;
    dcl decimal(5,2)  v15;

    method init();
      v01 = 1;
      v02 = 1;
      v03 = 1;
      v04 = 1;
      v05 = 1.0;
      v06 = 'aa';
      v07 = 'aa';
      v08 = 'aa';
      v09 = 'aa';
      v10 = date'2001-01-01';
      v11 = time'01:01:01';
      v12 = timestamp'2001-01-01 01:01:01';
      v13 = x'0123456789ABCDEF';
      v14 = x'0123456789ABCDEF';
      v15 = 32.23n;
      output;
    end;
enddata;
run;

PROC FEDSQL DESCRIBE TABLEステートメントを使用して、ds2dataテーブルの
内容の要約を作成できます。

CREATE TABLE DS2DATA (
     "v01"     INTEGER,
     "v02"     BIGINT,
     "v03"     INTEGER,
     "v04"     INTEGER,
     "v05"     DOUBLE PRECISION,
     "v06"     WCHAR(8),
     "v07"     WVARCHAR(8),
     "v08"     WCHAR(8),
     "v09"     WVARCHAR(8),
     "v10"     DATE,
     "v11"     TIME(0),
     "v12"     TIMESTAMP(4),
     "v13"     BINARY(8),
     "v14"     VARBINARY(32),
     "v15"     NUMERIC
);

データ型の特性
数値データ型

数値データ型には、数値が格納されます(数量や通貨の値など)。数値データ型の選択
は、格納される数値の種類と数値の使用方法によって異なります。
数値データ型を選択する際に考慮すべき特性は次のとおりです。
n 正確な数値データ型を使用するか、近似数値データ型を使用するか。BIGINT、

DECIMAL、INTEGER、NUMERIC、TINYINTなどの正確な数値データ型は、値を
正確に表します。REAL、DOUBLE、FLOATなどの近似数値データ型は、多数の
数値に対して指定された正確な値を格納しません。多くのアプリケーションで
は、正確な数値の動作が必要でない限り、指定された値と格納された近似値のわ
ずかな違いは目立ちません。

n 整数または小数のどちらを格納するか。整数データ型の場合、小数値が挿入され
ると、小数部分が破棄されることに注意してください。たとえば、2.7を挿入す
ると、格納される値は 2になります。

n 小数点付きの数値を格納する方法。浮動小数点形式は、小数点が固定されていな
い数値です。浮動小数点データ型は、より長い 10進値や、広範囲の数値をすば
やく計算するために使用されます。浮動小数点データ型の場合、大きすぎる整数
を入力するとエラーが発生します。浮動小数点表記を使用して値を入力します。
固定小数点とは、小数点以下が 1列に並ぶ数値の格納方法です。固定小数点デー
タ型 DECIMALは、通貨などの小数部分の正確な精度に使用されます。

n データ型の精度。データ型の精度は、数値に含めることができる合計桁数を指定
します。桁数が多いほど、値の精度が高くなります。値が範囲外の場合、エラー
が発生します。たとえば、BIGINTデータ型の場合、9,223,372,036,854,775,807
より大きい値に対してエラーが発生します。

データ型の特性 51



エラーや不正確な結果を避けるために、特に整数データ型の数値に対して演算を
実行するときは、結果を考慮する必要があります。INTEGERデータ型の精度は
10桁で、範囲は–2,147,483,647から 2,147,483,647までです。2つの大きな整
数を乗算し(たとえば、33432*79879)、結果が 2,147,483,647より大きい場合、
オーバーフローエラーが発生します。データ型の精度または範囲よりも大きい
結果が予想される場合は、結果を DOUBLEや BIGINTなどのより大きなデータ
型として割り当てるか、式を倍精度浮動小数点数として評価するように強制する
ために、整数ではなく、倍精度定数(たとえば、33432.0*79879.0)として入力で
きます。
結果を考慮する必要がある場合の具体例は、DS2で除算を実行する場合です。次
の例では、80/100は整数除算を実行し、結果は 0です。80.0/100の浮動小数点
除算は 0.8になります。
data; 
   dcl double x y; 
   method run(); 
      /* this gives an integer value result of 0 */
      x=80/100; 
      put x=;
      
      /* this gives the result of .8 */
      y=80.0/100;  
      put y=; 
   end; 
enddata; 
run;

文字データ型
文字データ型の概要
DS2には、文字列(テキスト)データを格納する複数の文字データ型が用意されていま
す。文字データ型には、英字、0 - 9の数字、およびその他の特殊文字を含めること
ができます。

注: 文字列に数値のみが含まれている場合、DS2は自動的にそれを数値型に変換し、
その数値を計算に使用します。

各文字データ型には、最大文字数を指定するためのパラメーターがあります。
DS2およびほとんどの SQLデータベースでは、すべての文字データ型の文字長は文
字列の文字数です。固定長データ型(CHARおよび NCHAR)の場合、文字長は文字列
内の正確な文字数です。可変長データ型(NVARCHARおよび VARCHAR)の場合、文
字長は文字列で許可される最大文字数です。
SASデータセットまたは CASテーブルを作成するときの DATAステップおよび
PROC SQLの場合、CHAR型の文字長はバイト単位で測定されます。

違いを理解するために、長さが 10で UTF-8エンコーディングの固定長文字列を想
定してください。DS2は、10の長さを 10文字として解釈します。DS2列には、10
個の 4バイト文字を含む任意の 10個の UTF-8文字を格納できます。これには 40
バイトのストレージが必要です。DATAステップと PROC SQLは、10の長さを 10
バイトとして解釈します。DATAステップと PROC SQLによって 10個の 4バイト
文字が書き込まれると、切り捨てエラーが発生します。これは、その 10文字が、デ
ータ用に SASデータセットまたは CASテーブルに割り当てられた 10バイトのスト
レージを超えるためです。

文字データ型とエンコーディング
すべての DS2文字データには、文字エンコーディングが関連付けられています。
DS2は、同じプログラム内の複数の文字エンコーディングでのデータ処理をサポー
トしていますが、各文字変数には、DS2プログラムの実行中に固定される 1つの関
連付けられた文字エンコーディングがあります。SETステートメントまたは割り当
てステートメントのいずれかによって文字変数に割り当てられたデータは、そのデ
ータが別の文字エンコーディングである場合、変数の文字エンコーディングにトラ
ンスコードされます。
文字変数の文字エンコーディングは、DECLAREステートメントの CHARACTER SET
オプションで指定できます。次のステートメントを考えてみましょう。
declare varchar(1024) character set utf8 s1;
declare varchar(1024) character set  latin1 s2;
declare varchar(1024) s3;
declare nvarchar(1024) s4;

文字変数 s1は、文字エンコーディング UTF-8で宣言されています。文字変数 s2
は、文字エンコーディング latin-1で宣言されています。文字変数 s3は、明示的な
文字エンコーディングなしで宣言されています。したがって、DS2は DS2セッショ
ンエンコーディングを変数 s3の文字エンコーディングとして割り当てます。PROC 
DS2を使用して実行する場合、DS2セッションエンコーディングは SASセッション
エンコーディングです。DS2セッションエンコーディングが EUC-CNの場合、文字
変数 s3には文字エンコーディング EUC-CNが割り当てられます。各国文字変数は、
DECLAREステートメントで CHARACTER SETオプションをサポートしない特殊な
ケースです。各国文字変数には、DS2セッションエンコーディングに関係なく、文
字エンコーディングとして常に UTF-8が割り当てられます。したがって、各国文字
変数 s4 には、文字エンコーディングとして UTF-8が割り当てられます。
NCHAR[n]宣言は CHARACTER[n] CHARACTER SET UTF-8と同等であり、
NVARCHAR[n]宣言は CHARACTER VARYING[n] CHARACTER SET UTF-8と同等で
す。
SETまたはMERGEステートメントによって暗黙的に宣言された文字変数には、デー
タソース内の対応する列のエンコーディングが割り当てられます。SETまたは
MERGEステートメント以外のステートメントによって暗黙的に宣言された文字変
数には、変数の文字エンコーディングに DS2セッションエンコーディングが割り当
てられます。

注: 出力データソースが SASデータセットなどの複数のエンコーディングをサポー
トしていない場合、さまざまなエンコーディングがテーブルの共通エンコーディン
グに抽出されます。

デフォルトのエンコーディングは、オペレーティングシステムとロケールによって
異なります。文字セットエンコーディング値の完全なリストについては、SAS各国
語サポート(NLS): リファレンスガイドの"Character Sets for Encoding in NLS"を参
照してください。

文字データ型の制限と考慮事項
文字変数に関する次の制限と考慮事項に注意してください。
n CHAR、VARCHAR、NCHAR、NVARCHARのデータ型の列には、読み取りまたは
生成される列に 10485760文字という長さの制限があります。文字はマルチバ
イト文字(UTF-8など)にすることができます。

n CHARデータ型と VARCHARデータ型の両方の比較演算では、末尾の空白は無視
されます。末尾の空白は、CHARと VARCHARの両方について、メソッドに渡さ
れる引数などとして、割り当て、連結、出力中に無視されません。コピー時に末
尾の空白は無視されません。VARCHAR(10)値'abc'が VARCHAR(20)にコピーされ
る場合、値は末尾の空白を含めてコピーされます。

n SASデータセットの読み取りまたは更新を行う場合、ENCODING=システムオプ
ションはサポートされていません。CHARACTER SET構文で複数のエンコーデ
ィング値が指定されている場合、DS2は特定の ENCODING=値を最小公分母エン
コーディングに変換します。使用される ENCODING=値は、データセットで実際
に作成される値とは異なる場合があります。この競合により、トランスコーディ
ングエラーが発生する可能性があります。

n DS2は、データキャストに関して ANSI SQL規格に準拠しています。たとえば、
CHARデータ型の変数を VARCHARデータ型の新しい変数に割り当てる場合、
CHAR変数の内容が収まる場合は VARCHAR変数に転送されます。次に例を示
します。
dcl char(8) u;
dcl varchar(6) v;
method run();
u='eighteen';
v=u;

割り当て後、vの結果の値は'eighte'で、長さは 6です。CHAR変数の内容が収ま
らない場合は、空白文字であるかどうかに関係なく切り捨てられます。
VARCHAR変数に値を割り当てる際に、変数に格納されている CHAR値から末尾
の空白文字をコピーしない場合は、TRIM関数を使用して、割り当て前に値から
末尾の空白文字を削除します。

日付と時間のデータ型
DS2は、日時を格納するという特定の目的のために、いくつかのデータ型をサポー
トしています。日付と時間の範囲は、データソース固有です。

CASのバイナリデータ型
CASで実行される DS2プログラムは、VARBINARYデータ型の列の作成、CASテー
ブルからの VARBINARY列の読み取り、VARBINARY列の CASテーブルへの書き込
みを行うことができます。
CASで実行される DS2プログラムは、BINARYデータ型の列を作成できます。ただ
し、CASテーブルから BINARYデータ列を読み取ったり、BINARY列を CASテーブ
ルに書き込んだりすることはできません。DS2プログラムが、CASテーブルから
BINARY列を読み取ったり、BINARY列を CASテーブルに書き込んだりすると、エ
ラーが発生します。このエラーを回避するには、CASテーブルに対して読み取りま
たは書き込みを行う前に、BINARY列を削除します。

列のデータ型を定義する
データ型を定義するときは、DS2でサポートされているデータ型のデータ型キーワ
ードを使用します。
データを保存するには、そのデータソースのデータストレージでデータ型を使用で
きる必要があることに注意してください。DS2はいくつかのデータ型をサポートし
ていますが、特定のテーブルに定義できるデータ型はデータソースによって異なり
ます。これは、各データソースが必ずしもすべての DS2データ型をサポートしてい
るわけではないためです。さらに、データソースは標準 SQLデータ型のバリエーシ
ョンをサポートします。つまり、指定した特定のデータ型が別のデータ型にマッピ
ングされる可能性があり、基になるデータソースで別の属性を持つ可能性もありま
す。これは、データソースが特定のデータ型をネイティブにサポートしていないが、
同様のデータ型のデータ値をデータ損失なしで変換できる場合に発生します。たと
えば、INTEGERデータ型をサポートするために、SASデータセットはデータ型定義
を DOUBLEである SAS数値にマッピングします。
各データ型のデータソースの実装の詳細については、“データ型リファレンス” (SAS 
DS2言語リファレンス)を参照してください。
さらに、データ型のマッピング方法を制御する CT_PRESERVE=接続引数は、データ
型を定義できるかどうかに影響する可能性があります。値 FORCE (デフォルト)およ
び FORCE_COL_SIZEは、データ型を定義できるかどうかには影響しません。要求さ
れたデータ型がデータソースにネイティブでない場合、または指定された精度また
はスケールがデータソースの範囲内にない場合、値 STRICTおよび SAFEはエラーに
なる可能性があります。CT_PRESERVE=接続引数の詳細については、SAS Federation 
Server: Administrator's Guideを参照してください。

DOUBLEおよび REALデータ型を使用す
るエラーメッセージ

エラーメッセージに DOUBLEおよび REALデータ型の数値が含まれている場合、こ
れらの値は BESTw.出力形式を使用してログに書き込まれます。BESTw.出力形式
は、数値を書き込むためのデフォルトの出力形式です。BESTw.を使用すると、SAS
は、使用可能なフィールド幅に従って、値に関する最も多くの情報を提供する出力
形式を選択します。
SASは BESTw.出力形式を使用するため、DOUBLEおよび REALデータ型の値は、プ
ログラムで使用する値とまったく同じではない可能性があります。BESTw.は値を
丸め、SASが指定された幅内で小数点以下の部分に少なくとも 1つの有効数字を表
示できる場合、BESTw.は結果を 10進数として生成します。それ以外の場合は、指
数表記で結果が生成されます。SASでは、表現に使用する出力形式にかかわらず、
常に完全な値を保存します。

DS2識別子

識別子の概要
識別子は、テーブル名や列名などのデータソースオブジェクトだけでなく、変数、
メソッド名、パッケージ名、配列などのプログラミング言語エンティティに名前を
付ける 1つ以上のトークンまたはシンボルです。
DS2言語は、通常の識別子と区切り識別子の両方について ANSI SQL:1999標準をサ
ポートしています。
通常の識別子は、ほとんどのプログラミング言語で見られるタイプの識別子です。
大文字と小文字が区別されないため、識別子 Nameは NAMEおよび nameと同じで
す。通常の識別子では、特定の文字のみが許可されます。
区切り識別子は、二重引用符を必要とする識別子、つまりテーブル名とスキーマ名
に対してのみ大文字と小文字が区別されます。その他の区切り識別子は、大文字と
小文字が区別されません。区切り識別子には任意の文字を使用でき、二重引用符で
囲む必要があります。変数名"Name"、"NAME" Name、name、NaMeはすべて同じ変
数を表します。
ANSI SQL:1999識別子をサポートすることにより、DS2言語は、ANSI SQL:1999識
別子もサポートするデータソースと互換性があります。

注: SASおよび SPDデータセットの識別子は、32文字に制限されています。

通常の識別子
通常の識別子に名前を付けるときは、次の規則を使用します。
n 通常の識別子の長さは、1文字から 255文字までです。
n 通常の識別子の最初の文字は、文字またはアンダースコアにする必要がありま
す。後続の文字には、文字、数字、またはアンダースコアを使用できます。たと
えば、識別子が引用符で囲まれていない限り、識別子にスラッシュやかっこを使
用することはできません。通常の識別子を 2つのアンダースコアで始めること
はできないことに注意してください。

n 通常の識別子は大文字と小文字を区別しません。
次の通常の識別子が有効です。
firstName
lastName
_phonenum
phone_num1

通常の識別子の文字は、大文字として内部的に格納されるため、大文字と小文字を
区別せずに文字を書くことができます。たとえば、引用符で囲まれていない入力列
名である phone_num1、Phone_Num1、また PHONE_NUM1は、出力では大文字の
PHONE_NUM1として表示されます。特殊文字やスペースを含まず、二重引用符で囲
まれた識別子は、引用符が存在しないかのように通常の識別子として扱われます。

注: 各データソースには独自の命名規則があり、そのすべてが DS2言語でサポート
されています。プログラムに特定のデータソースに固有の識別子が含まれている場
合は、そのデータソースの命名規則に従う必要があります。詳細については、
SAS/ACCESS for Relational Databases: リファレンスのデータソースの命名規則に関
するトピックを参照してください。

区切り識別子
区切り識別子に名前を付けるときは、次の規則に従います。
n 区切り識別子を二重引用符で開始および終了します。
n 区切り識別子の長さは、1文字から 253文字までです。(最大 255文字のうち 2
文字は引用符用に予約されています。)

n 区切り識別子は、先頭と末尾の二重引用符の間の任意の文字列(スペースや特殊文
字を含む)で構成されます。

n 大文字と小文字が区別される区切り識別子は、テーブル名とスキーマ名のみで
す。その他の区切り識別子は、大文字と小文字が区別されません。

n 区切り識別子の特殊文字は、識別子に必要な開始および終了の二重引用符に加え
て、対応する二重引用符で囲む必要があります。

n 区切り識別子にスラッシュ(/)を使用しないでください。これにより、一部の動作
環境でエラーが発生する可能性があります。たとえば、"/"は UNIXシステムでは
サブディレクトリを意味します。

二重引用符で囲まれた文字列は、文字定数ではなく識別子として解釈されます。文
字定数は、単一引用符でのみ囲むことができます。
次の区切り識別子が有効です。
"x&y&z"
"Ü1"
"(area)phone_num"
"a**B"

区切り識別子は、予約語になる可能性のある用語に使用できます。たとえば、日付
宣言以外で"date"という用語を使用するには、区切り識別子"date"として使用しま
す。次に例を示します。
/* In DATA step, there are no reserved words for variables. */
/* So, this doesn't cause an error.  However, how do you    */
/*  use such a variable in DS2?                             */
data a;
   date = mdy(8,14,2012);
run;

/* This program gives an error because the reserved word DATE */
/*  is used when a variable is expected.   */
proc ds2;
data b(overwrite=yes);
   method run();
      set a;
      if month(date) = 8 then
         put 'August';
      else
         put 'Not August';
   end;
enddata;
run; quit;

/* One solution is to quote the reserved word */
proc ds2;
data b(overwrite=yes);
   method run();
      set a;
      if month("date") = 8 then
         put 'August';
      else
         put 'Not August';
   end;
enddata;
run; quit;

/* Another solution is to rename the variable on input and output */
/*  to avoid the reserved word in DS2 code.                       */
proc ds2;
data b(overwrite=yes rename=(sas_date="date"));
   method run();
      set a(rename=("date"=sas_date));
      if month(sas_date) = 8 then
         put 'August';
      else
         put 'Not August';
   end;
enddata;
run; quit;

大文字と小文字が区別されないデータソースにサブミットされる DS2プログラム
で参照される、区切られた列名で作成されたテーブルに対して警告が発行されます。
大文字と小文字を区別しないデータソースでは、引用符が削除され、列名が区切ら
れていないものとして扱われます。
区切り識別子は、SAS DATAステップ言語の名前リテラルに似ていますが、同じでは
ありません。

注: 各データソースには独自の命名規則があり、そのすべてが DS2言語でサポート
されています。プログラムに特定のデータソースに固有の識別子が含まれている場
合は、そのデータソースの命名規則に従う必要があります。詳細については、
SAS/ACCESS for Relational Databases: リファレンスのデータソースの命名規則に関
するトピックを参照してください。

注意
LIBNAMEステートメントで PRESERVE_TAB_NAMES=NOオプションを使用する
と、予期しない結果が生じる可能性があります。

非ラテン文字のサポート
DS2言語では、通常の識別子としてラテン文字のみがサポートされています。非ラ
テン文字を使用するには、識別子を二重引用符で区切る必要があります。

存在しないデータの DS2モード
DS2は、存在しないデータを処理する 2つのモードをサポートしています。
n Oracleや DB2などの ANSI SQL互換データベースなど、存在しないデータを処
理する ANSIモード。ANSIモードは、すべてのデータ型の存在しないデータを
ANSI SQL null値として扱います。ANSIシステムでは、NULLは、データ値がな
いことを示すためだけに使用される特別なマークによって表されます。使用可
能な情報をエンコードしません。

n Base SASのように存在しないデータを処理する SASモード。SASモードは、固
定文字列および DOUBLE列の存在しないデータを欠損値として扱います。空白
スペースは、欠損文字データを表すために使用されます。デフォルトでは、欠損
数値データはピリオド(.)で表されます。ただし、特殊文字._ (ピリオドの後にアン
ダースコアが続く)および Aから Zを使用して、数値データを表すこともできま
す。SASモードでは、存在しないデータには既知の値があります。

null値と SAS欠損値の処理には大きな違いがあります。目的の結果が得られるよう
にモードが設定されていない場合、データは失われます。(詳細については、“ANSI
から SASへの変換と SASから ANSIへの変換について”を参照してください。)した
がって、DS2が SAS欠損値を処理する方法を理解することが重要です。

要点は次のとおりです。
n 存在しないデータの処理モードは、データソースへの接続方法によって異なりま
す。
o DS2プロシジャ、HPDS2プロシジャ、および DS2 CASアクションでサブミ
ットされる DS2コードは、デフォルトで SASモードを使用してデータを処理
します。

o SAS Federation Serverにサブミットされる DS2コードは、ANSIモードでデ
ータを処理します。

n 欠損値を持つことができるのは固定文字列と DOUBLE列だけです。DOUBLEま
たは CHAR(n)以外の DS2データ型の変数に欠損値が割り当てられている場合、
それらの値は自動的に null値に変換されます。その他の DS2データ型は、常に
ANSI null値として扱われます。

n MODE=オプションは、PROC DS2の ANSIMODEプロシジャオプション、DS2 
CASアクションの nullBehavior=パラメーター、および DS2_OPTIONSステート
メントの SASオプションに取って代わります。MODE=オプションは、混乱を減
らし、存在しない値の処理をインターフェイス間で一貫させるために提供されま
した。

n MODE=プロシジャオプションとアクションパラメーターは、DS2プログラム全
体の処理プリファレンスを指定します。DS2_OPTIONSステートメントを使用
すると、DS2プログラム内の特定のプログラムブロック(および指定されたプロ
グラムブロックから動作を継承するすべてのプログラムブロック )の処理プリフ
ァレンスを指定できます。DS2プログラムの残りの部分は、アクティブなプロシ
ジャまたはアクション設定を使用して処理されます。

n ほとんどの場合、存在しないデータを処理するためにモードを変更する必要はあ
りません。

n モードを変更する場合の例を次に示します。
o クライアントアプリケーションが SASデータセットを処理し、存在しないデ
ータのモードが ANSIモードの場合

o SASデータセットの処理が完了し、クライアントアプリケーションが ANSIモ
ードに戻る準備ができたとき

o クライアントアプリケーションが ANSIテーブルを処理し、存在しないデータ
のモードが SASモードの場合

o ANSIテーブルの処理が完了し、クライアントアプリケーションが SASモード
に戻る準備ができたとき

n 場合によっては、モードによって異なる結果が返されます。

null値の処理と SAS欠損値の処理の違い
SAS欠損値の処理は、ANSI null値の処理とは異なり、次の状況で重要な意味を持ち
ます。

n データをフィルタリングする場合(たとえば、WHERE句、HAVING句、サブセッ
ト化 IFステートメント、または外部結合 ON句)。SASモードは、各 null値を既
知の値である SAS欠損値(.)として解釈します。ANSIモードは、null値を不明な
値として解釈します。

n ANSIモードで外部結合をサブミットすると、内部処理によって中間結果テーブ
ルに nullが生成される場合があります。DS2は、中間結果テーブルの SASモー
ドで SAS欠損値を生成する場合があります。したがって、中間結果テーブルの場
合、ANSIモードでは nullは不明な値として解釈され、SASモードでは欠損値は
既知の値として解釈されます。

n 空白スペースだけで構成される文字値を比較すると、SASモードはその値を
MISSINGと解釈します。ANSIモードでは、完全に空白スペースで構成される文
字値は、空白文字で構成される文字列として扱われます(特別な意味はありませ
ん)。

n 次の場合、DS2は nullを SAS欠損値として解釈します。
o SASモードで、浮動小数点数または固定長文字値を含む計算または割り当て
で nullが使用されている場合。可変長文字列の欠損値は、ANSI null値のよ
うに扱われます。

o DOUBLEまたは CHARデータ型を想定している SAS出力形式または関数に
nullが渡された場合。

注: SAS Federation Server LIBNAMEエンジンを使用してデータにアクセスしてい
る場合、DS2_SASMISSING環境変数が TRUEに設定されていると、FedSQL CALL呼
び出しで ANSI null値が SAS欠損値に変換されます。

null値と SAS欠損値の属性と動作の違いを次に示します。
表 7.1 ANSI SQL null値と SAS欠損値の属性と動作の違い

属性または動作 ANSI SQL null値 SAS欠損値
内部表現 メタデータ 浮動小数点または文字
論理演算子による評価 解決値が True、False、お

よび nullである 3値ロジ
ックを使用して比較される
不明な値です。たとえば、
WHERE col1 = nullは nullを
返します。
注: ANSIモードでは、
null=nullは nullです。

比較するとブール値に解決
される既知の値です。たと
えば、WHERE col1 =.Aは、
同じ欠損値が見つかった場
合に Trueを返します。
欠損値を論理演算子で使用
すると、値は False (ゼロ)に
なります。

照合順序 最小値として表示 最小値として表示
スコープ すべてのデータ型の列は

null値を使用して存在しな
いデータを表します。

欠損値を保持できるのは固
定文字列と DOUBLE列の
みです。

ANSIから SASへの変換と SASから
ANSIへの変換について

ANSIモードから SASモードへ、またはその逆のデータ変換を行うと、データが失わ
れる可能性があります。したがって、存在しないデータを処理するデフォルトモー
ドを変更する前に、目的の結果を把握しておくことが重要です。
ANSIモードでは、DS2が SASデータセットから SAS数値欠損値を読み取るときに、
SAS欠損値を ANSI null値に変換します。その後、ANSI null値が SASデータセット
に書き込まれると、DS2は ANSI null値を SAS数値欠損値(.)に変換します。入力デ
ータセットからの SAS数値欠損値が.Aなどの特殊数値欠損値であった場合、ANSI 
nullとの間の変換中に.Aが失われ、SAS数値欠損値(.)が出力データセットに書き込
まれます。次の例では、列 xのデータ型は DOUBLEです。この例は、ANSIモード
で実行された DS2プログラムによって、SAS特殊数値欠損値(4行目の.A)が SAS数
値欠損値(.)に変換される方法を示しています。
/* assume input data set indata contains */
x
100
.
.A

/* Run the following program using ANSI mode */
proc ds2 mode=ansi;
data outdata;
   method run();
      set indata;
   end;
enddata;
run;
quit;

/* the output dataset outdata contains */
 x
 100
 .
 .

SASモードでは、DS2が ANSIデータソースからデータ型 CHAR(n)の ANSI null値を
読み取るときに、ANSI null値を SAS文字欠損値(完全に空白スペースで構成される
文字列)に変換します。 DS2が ANSIデータソースの CHAR(n)列に存在しない値を
書き込むと、その値は ANSI null値として書き込まれます。ただし、処理中、DS2
は存在しない値を SAS文字欠損値として扱います。次の例では、列 xのデータ型は
CHAR(5)です。この例は、SASモードで実行される DS2プログラムによって、ANSI 
null値(2行目)が空白スペース (' ')で構成される文字列として扱われることを示して
います。
/* assume input data set indata contains */
 x

 'abc  '
 null

/* run the following program using SAS mode */
proc ds2;
data db.outdata;
   method run();
      set db.indata;
   end;
enddata;
run;
quit;

/* the output dataset outdata contains */
 x
 'abc  '
 '     '

存在しないデータの読み取りおよび書き
込み時の ANSIモード動作の概要

次の表は、DS2が ANSIモードでデータソースから null値または SAS欠損値を読み
取るときに、クライアントアプリケーションに返される値を示しています。
表 7.2 ANSIモードでの存在しないデータ値の読み取り

列のデータ型 存在しないデータ値
アプリケーションに返され
る値

DOUBLE .、._、または.A-.Z1 null

DOUBLE null null

CHAR '␣'2 '␣'2

CHAR null null

1 値._ (ピリオドの後にアンダースコアが続く)または.A–.Zは、特殊数値欠損値を表します。SASが特殊欠損値を出力する場合、文字またはアンダースコアのみを出力します。
2 値'␣'は単一引用符の間の空白スペースで、ANSIモードでは空白スペースであり、存在しないデータではありません。

次の表は、存在しないデータ値が ANSIモードでデータソースに書き込まれたときに
格納される値を示しています。

表 7.3 ANSIモードでの存在しないデータの格納

列のデータ型
存在しないデータ
値

SASデータセット
に格納される値

ANSI SQL nullサポ
ートデータソース
に格納される値

DOUBLE null . null

CHAR null '␣'1 null

1 値'␣'は単一引用符の間の空白スペースで、(ANSIモードでは)存在しないデータではなく、空白スペースであることを示しています。

ANSIモードでは、空白スペースは常に空白スペースとして解釈され、存在しないデ
ータとして解釈されることはありません。また、ANSIモードでは、入力時または割
り当て時に SAS欠損値が ANSI null値に変換されます。

存在しないデータの読み取りおよび書き
込み時の SASモード動作の概要

次の表は、存在しないデータ値が SASモードでどのように読み取られるかを示して
います。
表 7.4 SASモードでの存在しないデータ値の読み取り

列のデータ型 存在しないデータ値
アプリケーションに返され
る値

DOUBLE .、._、または.A-.Z1 .、._、または.A-.Z1

DOUBLE null .

CHAR '␣'2 '␣'2

CHAR null '␣'2, 3

1 値._ (ピリオドの後にアンダースコアが続く)または.A–.Zは、特殊数値欠損値を表します。SASが特殊欠損値を出力する場合、文字またはアンダースコアのみを出力します。
2 値'␣'は単一引用符の間の空白スペースで、SASモードでは存在しない文字データを表します。
3 SETステートメントで固定幅文字列の nullが検出されると、その文字列は文字列の長さ分の空白文字で埋められます。

次の表は、欠損しているデータ値が SASモードでデータ送信先にどのように書き込
まれるかを示しています。

表 7.5 SASモードでの存在しないデータ値の書き込み

列のデータ型
存在しないデータ
値

SASデータセット
に格納される値

ANSI SQL nullサポ
ートデータソース
に格納される値

DOUBLE2 .、._、または.A .、._、または.A-.Z null

CHAR2 '␣'1 '␣'1 null

1 値'␣'は単一引用符の間の空白スペースで、SASモードでは存在しないデータです。
2 出力が SASデータセットに書き込まれる場合、他のデータ型は DOUBLEまたは CHARデータ型に強制されます。null値は、その型変換中に欠損値(.)になります。

SASモードでは、DOUBLEまたは CHAR型の ANSI null値は、入力時または割り当
て時に SAS欠損値に変換されます。FLOATや VARCHARなどの他のデータ型の場
合、null値は入力時または割り当て時に保持されます。

注: 特殊数値欠損値に対する演算は、通常の SAS欠損値、つまり単一のピリオドを
生成します。たとえば、Y=.Aの場合、X=Y+1は X=.を生成します。ただし、他の演算
とは異なり、直接割り当てでは特殊数値欠損値が保持されます。たとえば、Y=.Aの
場合、割り当てに応じて次の出力が生成されます。

X=Y+Y produces X=.
X=Y   produces X=A.
X=Y+1 produces X=.

null値のテスト
DS2は、null値をテストする NULL関数を提供します。NULL関数には 1つの引数
があり、式にすることができます。式が nullの場合、関数は 1を返します。式が
nullでない場合、関数は 0を返します。
この例は、null値のテストを示しています。
if null(numCopies) then put 'Number of copies is unknown.'
   else put 'Number of copies is' numCopies;

詳細については、“NULL 関数” (SAS DS2言語リファレンス)を参照してください。

欠損値のテスト
DS2には、nullまたは欠損値をテストするためのMISSING関数が用意されていま
す。MISSING関数には 1つの引数があり、数値式または文字式にすることができま
す。式が nullまたは欠損の場合、関数は 1を返します。式が nullでも欠損でもない
場合、関数は 0を返します。
NMISS関数は、nullおよび SAS欠損数値の数を返します。NMISS関数が数値を必
要とし、複数の数値を使用できるのに対し、MISSINGは数値か文字のいずれかの値
を 1つだけ使用できます。
詳細については、“MISSING 関数” (SAS DS2言語リファレンス)および“NMISS 関数” 
(SAS DS2言語リファレンス)を参照してください。

型変換の定義
バイナリデータ型

BINARYおよび VARBINARYデータ型を指します。
文字データ型

CHAR、VARCHAR、NCHAR、NVARCHARデータ型を指します。
強制型変換可能なデータ型
文字データ型だけでなく、複数のデータ型に変換できるデータ型。

日付/時間データ型
DATE、TIME、および TIMESTAMPデータ型を指します。

強制型変換できないデータ型
文字データ型にのみ変換できるデータ型。

数値データ型
DECIMAL、DOUBLE(または FLOAT)、REAL、BIGINT、INT、NUMERIC、
SMALLINT、TINYINTデータ型を指します。

標準文字変換
式が文字データ型でない場合は、CHARデータ型に変換されます。

標準数値変換
式が強制可能な非数値データ型の場合、DOUBLEデータ型に変換されます。

型変換の概要
DS2が式を解決するには、式のオペランドは、同じ一般的なデータ型、数値、文字、
バイナリ、日付/時間である必要があります。DS2は、必要な場合、式のオペランド
と演算子に応じて、オペランドのデータ型を別のデータ型に変換します。このプロ
セスは型変換と呼ばれます。たとえば、連結演算子( || )は文字データ型を操作しま
す。文字列‘First’と数値整数 1の連結では、オペランド 1の INTEGERデータ型は、
連結が行われる前に CHARデータ型に変換されます。
オペランドのデータ型が同じ一般的なデータ型内で変換される場合、オペランドの
データ型はプロモートされます。SMALLINTおよび TINYINTのデータ型のオペラ
ンドは INTEGERにプロモートされ、REAL型のオペランドは DOUBLEにプロモート
されます。メソッドおよび関数式の引数を含む、SMALLINT、TINYINT、および REAL
のすべての操作に対して型プロモーションが実行されます。
数値および文字データ型は強制型変換できます。BINARY、VARBINARY、および日
付/時間データ型の DATE、TIME、および TIMESTAMPは強制型変換できず、いずれ
かの文字データ型にのみ変換できます。
DS2が式を評価するとき、オペランドのデータ型が正確に一致する場合、型の変換
やプロモーションは必要なく、式は解決されます。それ以外の場合、各オペランド
は、演算子に応じて、標準の数値変換または標準の文字変換を行う必要があります。
数値式または文字式の結果は、データ型の優先順位に基づきます。両方のオペラン
ドが同じ一般的なデータ型内で異なる型を持つ場合、式のデータ型は、優先順位の
高いオペランドのデータ型になります。ここで、1が最高の優先順位です。たとえ
ば、数値データ型の場合、DOUBLEのデータ型が最も優先されます。式に INTEGER
型のオペランドと DOUBLE型のオペランドがある場合、式のデータ型は DOUBLE
です。優先順位のリストは、該当する場合は、さまざまなタイプの式の後続のトピ
ックにあります。
すべての型変換を示す表については、付録 1, “式オペランドの DS2型変換” (319ペ
ージ)を参照してください。

単項式の型変換
+1や-444などの単項式では、標準の数値変換がオペランドに適用されます。次の表
は、単項式のデータ型を示しています。

表 8.1 単項式のデータ型変換

式のタイプ 式のデータ型
単項プラス オペランドと同じ、または変換されたオペ

ランドの DOUBLE

単項マイナス オペランドと同じ
または変換されたオペランドの DOUBLE

単項否定 INTEGER

論理式の型変換
a & bや(a ~= start) | (f = finish)などの論理式では、標準の数値変換が各オペランドに適
用されます。次の表は、式のデータ型を決定するために使用される優先順位を示し
ています。1が最高の優先順位で、3が最低です。式のデータ型は、優先順位の高い
オペランドのデータ型です。
表 8.2 論理式のデータ型変換

優先順位
いずれかのオペランドのデー
タ型 式のデータ型

1 DOUBLE DOUBLE

2 BIGINT BIGINT

3 他のすべての数値データ型 INTEGER

算術式の型変換
a<>bや a + (b * c)などの算術式では、標準の数値変換が各オペランドに適用されま
す。
次の表は、加算、減算、乗算、除算演算子の算術式のデータ型を決定するために使
用される優先順位を示しています。1が最高の優先順位で、3が最低です。式のデー
タ型は、優先順位の高いオペランドのデータ型です。

表 8.3 加算、減算、乗算、除算式の型変換

優先順位
いずれかのオペランドのデー
タ型 式のデータ型

1 DOUBLE DOUBLE

2 DECIMAL、NUMERIC DECIMAL、NUMERIC

3 BIGINT BIGINT

4 他のすべての数値データ型 INTEGER

次の表は、最小、最大、およびべき乗演算子を使用する算術式のデータ型を示して
います。
表 8.4 最小、最大、およびべき乗演算子式のデータ型変換

演算子 演算子のデータ型 式のデータ型
最小または最大 DOUBLE DOUBLE

最小または最大 DECIMAL、NUMERIC DECIMAL、NUMERIC

最小または最大 BIGINT BIGINT

** すべての数値データ型 DOUBLE

関係式の型変換
x <= y や i > 4などの関係式では、適用される標準変換はオペランドのデータ型によっ
て異なります。次の表に示すように、式のデータ型は常に INTEGERです。
表 8.5 IN式以外の関係式のデータ型変換

データ型解決の順
序

いずれかのオペラ
ンドのデータ型 標準変換 式のデータ型

1 任意の数値データ
型

数値 INTEGER

2 CHAR、NCHAR 文字 INTEGER

データ型解決の順
序

いずれかのオペラ
ンドのデータ型 標準変換 式のデータ型

3 DATE、TIME、
TIMESTAMP

なし、データ型が一
致する必要がある

INTEGER

4 NVARCHAR、
VARCHAR

なし、エラーが返さ
れる

適用できません

表 8.6 IN式のデータ型変換

優先順位
いずれかのオペランドのデ
ータ型 式のデータ型

1 非数値 DOUBLE

2 DOUBLE DOUBLE

3 DECIMAL、NUMERIC DECIMAL

4 BIGINT以外のすべての型 BIGINT

連結式の型変換
a || bや x !! yなどの連結式では、標準の文字変換が各オペランドに適用されます。
次の表は、式のデータ型を決定するために使用される優先順位を示しています。1
が最高の優先順位で、2が最低です。式のデータ型は、優先順位の高いオペランド
のデータ型です。
表 8.7 連結式のデータ型変換

優先順位
いずれかのオペランドのデー
タ型 式のデータ型

1 どちらかが NCHAR型の場合 NCHAR

2 CHAR CHAR

型変換とあいまいなメソッド呼び出し
呼び出されたメソッドと同じ名前のメソッドが複数見つかった場合、DS2はメソッ
ドのオーバーロードを解決して、呼び出すメソッドを決定しようとします。ただし、
DS2が解決できない特定の状況があります。
この例では、BIGINTパラメーターを持つメソッドと CHARパラメーターを持つメソ
ッドの 2つの ambigメソッドが作成されます。メソッドが INITメソッドで呼び出
されると、プログラムは INTEGER引数を渡します。DS2は amig(bigint a)と
ambig(char(1) a)のどちらを呼び出すべきか判断できないため、これによりあいまいな
メソッド呼び出しエラーが生成されます。

data _null_;
   dcl integer inp;
   dcl char(100) res;
   method ambig(bigint a) returns char(10);
      return 'bigint'; 
   end;
   method ambig(char(1) a) returns char(10); 
      return 'char(1)'; 
   end;

   method init();
      inp = 2;
      res = ambig(inp);
      put inp= res=;
   end;
   enddata;
run;

エラーを回避するには、INTEGERパラメーターを持つメソッド ambigを追加しま
す。

data _null_;
   dcl integer inp;
   dcl char(100) res;
   method ambig(bigint a) returns char(10);
      return 'bigint';
   end;
   method ambig(char(1) a) returns char(10);
      return 'char(1)';
   end;

   /* Define ambig with a int parameter */
   /* fixes the problem by routing "integer" to "bigint" */

   method ambig(int a) returns char(10);
      dcl bigint l;
      l = a;
      return ambig(l);
   end;

   method init();
      inp = 2;
      /* call ambig(int) is no longer ambiguous */
      res = ambig(inp);  
      put inp= res=;
   end;
   enddata;
run;

DOUBLE引数を指定してメソッド ambigを呼び出そうとすると、DOUBLEが既存の
メソッド宣言の 1つと正確に一致しないため、あいまいなメソッド呼び出しエラー
が発生することに注意してください。かわりに、あいまいさを回避するために、新
しいメソッド定義(これは DOUBLEパラメーターを受け取るもの)を作成する必要が
あります。

式とは
式は、データ値を生成するプログラミング要素です。式には、一連の命令を形成し、
値に解決されるオペランドとオプションの演算子が含まれます。
オペランドは、単一の定数または変数にすることも、式にすることもできます。演
算子は、オペランドの計算、比較、または連結を要求する記号です。
DS2式の例を次に示します。
3
"col1"
a = b * c

s || 1 || z
a >= b**(c - 8)
system.put(a*5,hex.)
x'ff00effc'

式のタイプ
式の概要

式の基本的なタイプは、一次式です。演算子を使用して式(一次または複合)を結合す
る式は、複合式です。式は、関数式やメソッド式などの DS2コンストラクトを呼び
出すことができます。式は、“SYSTEM式”や THIS式など、関数または変数のスコー
プに影響を与える可能性があります。 IN式は、式の結果がリストに含まれているか
どうかに基づいて真理値を返します。
式が単純か複合か、コンストラクトを呼び出すか、スコープがグローバルかどうか
に関係なく、すべての式には値とデータ型があります。式のデータ型は、変数や関
数などの式が取る可能性のある値を制限します。DS2は複合式に遭遇すると、ルー
ルに従って、式の各部分を評価する順序を決定します。

一次式
最も単純な形式では、一次式は、識別子、数値、文字列、2進数および 16進数の定
数、リテラル値、日付と時間の値、null値です。
a
"var_a"
5.33
'Company'
date '2007-08-24'

次の表は、基本的な一次式、そのデータ型、および例を示しています。
表 9.1 一次式のデータ型と例

式のタイプ 簡単な説明 データ型 例
2進定数 2進定数と 16進定数 VARBINARY x'FE'

b'01000011'

文字定数 文字列 CHAR、VARCHAR 'New Report'

a='Stock'; 
b='Report';

式のタイプ 簡単な説明 データ型 例
1

各国文字定数 各国文字列 NCHAR、NVARCHAR n'New Report'

a=n'Stock';
b=n'Report';
1

SYSTEMTHIS System、THIS 式の解決された型 system.put(x,5.)
this.s

日付/時間定数 日付と時間の値 DATE、TIME、
TIMESTAMP2

date'2023-01-01'

time'20:44:59'

timestamp'2023-02-07 
07:00:00.7569'

識別子 さまざまな言語要素
の名前を提供する

宣言された型または
デフォルトの型。デ
フォルトは
DOUBLE。

a

"part1"

整数定数 整数 INTEGER、BIGINT 123

新規 パッケージメソッド
をインスタンス化す
る

式の解決された型 a=_new_ package-name();
1

小数定数 浮動小数点数 DOUBLE、FLOAT、
REAL

5.

4.3

Null定数 Null式 なし NULL

かっこ 評価の優先順位を上
げるためにかっこで
囲まれた演算子とオ
ペランド

かっこで囲まれた式
の解決された型。

(x)
(a + b) - c
1

欠損定数 SAS欠損数値式 DOUBLE .
.A

数値定数 正確な数値 DECIMAL、NUMERIC 99n
123456789012345678901234567890.12N

1 この例では、一次式は式または割り当てステートメントに含まれています。式の例が強調表示されています。
2 TO_DOUBLE関数を複合式で使用すると、SASの日付、時間、または日時の値をエンコードする DOUBLE値を生成できます。

複合式
複合式は、次の式のように、式と演算子を組み合わせてより拡張的な式を作成しま
す。
a + b * -c - 5
a = b = 5
x | y & a < c * d
x**y**z - 9 >= f
z >< c + e <> u**y * 10
x || c + 7
a in (1,2,3)

複合式の評価は、表 9.8 (102ページ)に示されている演算子の優先順位と、一次式の
データ型に基づいています。計算を行う前に、オペランドのデータ型が同じ一般的
なデータ型(数値、文字、バイナリ、日付/時間)である必要があります。データ型が
同じ場合、処理は続行されます。同じでない場合、オペランドのデータ型は、演算
子とオペランドのデータ型に基づいて変換されます。
演算子の優先順位と一次式のデータ型がどのように評価されるかを理解するため
に、式 a + b / -c - 5を考えてみましょう。この式には 4つの演算子があります。
n 二項+演算子
n 二項/演算子
n 単項-演算子
n 二項-演算子
演算子の優先順位によって、演算の評価順序が決まります。表 9.8 (102ページ)を見
ると、単項-演算子の優先順位は 2、二項/演算子の優先順位は 3、二項+演算子と二
項-演算子はどちらも優先順位が 4です。2つの演算子の優先順位が同じ場合、結合
規則によって評価の順序が決まります。二項+演算子と二項-演算子には、左から右
への結合があります。したがって、前述の式は評価のために次のようにグループ化
されます。
((a + (b / (-c))) – 5)

型変換も演算の評価に影響します。二項算術演算では、両方のオペランドを同じ型
にする必要があります。オペランドが同じ型でない場合、DS2コンパイラは、算術
式の優先規則の型変換を使用して、一方のオペランドを他方のオペランドと同じ型
に変換します。
表 9.2 算術演算のデータ型変換

優先順位 いずれかのオペランドのデータ型 式のデータ型
1 DOUBLE、REAL DOUBLE

2 DECIMAL、NUMERIC DECIMAL、NUMERIC

3 BIGINT BIGINT

優先順位 いずれかのオペランドのデータ型 式のデータ型
4 他のすべての数値データ型 INTEGER

次のように仮定します。
n aは、値が 1.35の DOUBLE変数です。
n bは、値が 2の INTEGER変数です。
n cは、値が 3の INTEGER変数です。
式の評価方法は次のとおりです。
1 cは INTEGER値 3であるため、-cは INTEGER値-3に評価されます。
2 bは INTEGER値 2であり、-cは INTEGER値-3に評価されるため、式 b/-cは 2/-3
になり、INTEGER値 0に評価されます。

3 aは DOUBLE値 1.35であり、式 b/-cは INTEGER値 0に評価されるため、
INTEGER値 0は DOUBLE値 0.0に変換されます。式 a + b/-cは 1.35 + 0.0にな
り、これは DOUBLE値 1.35に評価されます。

4 式 a + b /-cが DOUBLE値 1.35に評価され、5が INTEGER値であるため、
INTEGER値 5は DOUBLE値 5.0に変換されます。式 a + b/-c - 5は 1.35 - 5.0に
なり、これは DOUBLE値-3.65に評価されます。

算術演算の非数値オペランドは、DOUBLEに強制変換されます。CHAR + INTの式
の型は DOUBLEです。これは、CHARオペランドが DOUBLEに強制変換されるた
めです。DOUBLE + INTは、DOUBLEの優先順位が INTEGERよりも高いため、式
のデータ型が DOUBLEになります。
データ型変換の詳細については、8章, “DS2型変換” (69ページ)を参照してくださ
い。
別の例として、a = b = 5という式を考えてみましょう。最初の等号(=)は割り当て演
算子です。2番目の等号は論理等値演算子です。bが 5以外の値の場合、b=5は 0と
評価されます。したがって、aには値 0が割り当てられます。詳細については、“複
数の等号を持つ式の使用” (SAS DS2言語リファレンス)を参照してください。

DOT演算子式
ドット演算子式は、パッケージの外部にあるコードからパッケージデータにアクセ
スします。
DOT演算子式の構文は次のとおりです。
package-variable.[package-attribute | package-method(arguments)]

package-variable1.package-variable2….package-variableN.[package-attribute | 
package-method(arguments)]

. (dot)
パッケージ属性にアクセスするか、左側のオペランドによって参照されるパッケ
ージ変数から、右側のオペランドとして指定されたパッケージメソッドを呼び出
します。

package-attribute
左側のオペランドによって参照されるパッケージの属性を指定します。パッケ
ージ属性は、パッケージ内で定義されるグローバル変数です。このグローバル変
数は、パッケージのスカラー変数または配列変数にすることができます。ドット
表記は、変数配列属性、標準一時配列属性、および動的一時配列属性へのアクセ
スをサポートしています。

package-variable
アクセスする属性またはメソッドを持つパッケージのインスタンス化を指定し
ます。
注 これはスカラーパッケージ変数または配列パッケージ変数の要素にするこ
とができます。詳細については、“DECLARE PACKAGE ステートメント” (SAS 
DS2言語リファレンス)を参照してください。

package-method(arguments)
左側のオペランドによって参照されるパッケージ変数のメソッドを指定します。

ドット演算子の左側のオペランドは、スカラーパッケージ変数または配列パッケー
ジ変数の要素である必要があります。左側のオペランドが nullの場合(つまり、パッ
ケージ変数がパッケージインスタンスを参照していない場合)、ドット演算により致
命的な欄ライムエラーが発生します。
ドット演算子の右側のオペランドは、左側のオペランドによって参照されるパッケ
ージの属性またはメソッドでなければなりません。ドット演算子の右側のオペラン
ドがスカラーパッケージ変数または配列パッケージ変数の要素である属性である場
合、ドット演算子の結果自体を後続のドット演算の左側のオペランドとして使用で
きます。パッケージ変数を左側のオペランドとして指定する式は、ネストされたド
ット表記式です。
ネストされたドット表記は、パッケージの階層内のパッケージデータへのアクセス
をサポートします。ネストされたドット演算子の左側の各オペランドがパッケージ
変数またはパッケージ配列変数の要素に解決される限り、無制限の数のネストされ
たドット演算がサポートされます。
詳細については、“DS2パッケージ内のドット表記” (166ページ)を参照してくださ
い。

関数式
関数式は、DS2プログラム内の関数を呼び出します。関数を呼び出すには、次の構
文を使用します。
function-name ([argument [, …argument ]])

関数には引数が必要な場合があります。関数式に引数が含まれている場合、引数の
データ型は、必要に応じて、関数の引数の順序とデータ型である関数シグネチャの

データ型に変換されます。関数が引数を取るか取らないかに関係なく、関数呼び出
しにはかっこが必要です。たとえば、TIME関数は引数を取りません。
t = time();

関数式は、関数によって返される値に解決されます。前述の関数式では、time()は現
在時刻に解決されます。
メソッドと関数は似ています。関数はグローバルスコープを持ちます。メソッドは
プログラミングブロックであり、ローカルスコープを持ちます。
関数の名前がメソッド名と同じ場合、DS2はそのメソッドを呼び出します。メソッ
ドと同じ名前の関数は、SYSTEM式を使用してのみ呼び出すことができます。詳細
については、“SYSTEM式” (96ページ)を参照してください。
DS2関数のリストについては、“DS2関数” (SAS DS2言語リファレンス)を参照してく
ださい。

IF式
IF式の概要
条件付き IF式は、条件式の評価が真(ゼロ以外の値)か偽(ゼロ)かに基づいて、2つの
値を選択するために使用されます。
IF式を呼び出すには、次の構文を使用します。
IF expression-1THEN expression-2ELSE expression-3

expression-1がゼロ以外の値の場合、IF式の結果は expression-2の値になります。
それ以外の場合、IF式の結果は expression-3の値になります。次に例を示します。
 m=(if missing(u) then 0 else u);

IF式は、他の式を使用できる場所であればどこでも使用できます。IF式の優先順位
は、算術演算子および論理演算子よりも低くなります。したがって、このような混
合式ではかっこが必要です。
r = 25.5 + (if sum < 15 then -a else b*2);

かっこがないと、プラス(+)演算子が最初に評価され、サブ式 25.5 + ifから解析エラ
ーが発生します。

ネストされた IF式
IF式をネストして、多方向の決定のために多くの値から選択することができます。
IF condition-expression-1THENresult-expression-1

ELSE IF condition-expression-2THEN result-expression-2
…

ELSE IFcondition-expression-nTHEN result-expression-n

ELSEresult-expression-default

条件式は順番に評価されます。ネストされた IF式チェーンの結果は、真(ゼロ以外の
値)と評価される最初の condition-expressionに関連付けられた result-expressionで
す。すべての condition-expressionsが偽(ゼロ)と評価された場合、IF式の結果は
result-expression-defaultになります。次に例を示します。
grade = if score >= 90 then 'A'
   else if score >= 80 then 'B'
   else if score >= 70 then 'C'
   else if score >= 60 then 'D'
   else if score >= 0  then 'F'
   else NULL;

注: 相互に排他的な条件が長く連続する場合は、IF-THEN/ELSEステートメントを連
続で使用するかわりに SELECTステートメントを使用します。ネストされた IF-
THEN/ELSEステートメントが多数あると、内部エラーが発生する可能性がありま
す。対照的に、SELECTステートメントは 1回だけ評価されるため、エラーを発生さ
せずにパフォーマンスを向上させることができます。
たとえば、次のプログラムのように、プログラムに多数の ELSE IFコンストラクト
が含まれているとします。

if (e1) then …
else if (e2) then …
…
else if (en) then …
else …

かわりに、この SELECTステートメントを使用します。
select;
when (e1): …
when (e2): …
…
when (en): …
otherwise: …

IF式のデータ型
IF式のデータ型は、最初の結果式である expression-2の型を調べることによって決
定されます。
IF expression-1THEN expression-2ELSE expression-3

expression-2が数値データ型でない場合、IF式には expression-2の型が割り当てら
れます。
expression-2が数値データ型の場合、IF式には、expression-2と expression-3の広
い方の数値データ型が割り当てられます。たとえば、expression-2が SMALLINTで、
expression-3が DOUBLEの場合、IF式には DOUBLE型が割り当てられます。
expression-2が数値データ型で、expression-3が数値データ型でない場合、IF式に
は expression-2の型が割り当てられます。

ネストされた IF式チェーンの最初の結果式が数値データ型の場合、すべての結果式
が調べられ、ネストされた IF式チェーンの型として割り当てる最も広い数値データ
型が検索されます。次の例では、tは TINYINT、bは BIGNINT、dは DECIMAL(10,5)
です。ELSEの 0には BIGINT型が割り当てられます。したがって、ネストされた IF
式チェーンには、TINYINT、BIGINT、および DECIMAL(10,5)の中で最も広い数値型
である decimal(10,5)型が割り当てられます。
r = if n < 0 then t
   else if n = 0 then b
   else if n > 0 then d
   else 0;

注: ELSE値に 0ではなく 0.0が使用された場合、ELSEの結果には、BIGINT型では
なく DOUBLE型が割り当てられます。DOUBLE型の ELSE式の場合、結果式の最も
広い数値型は DOUBLE型になります。したがって、ネストされた IF式チェーンに
は、decimal(10,5)型ではなく、DOUBLE型が割り当てられます。

IF式の遅延評価
IF式は、結果式に遅延評価を使用します。
IF expression-1THEN expression-2ELSE expression-3

たとえば、このコードを使用してゼロ除算をチェックできます。
a = if c ne 0 then b/c else null;

式 expression-1は常に評価されますが、expression-2または expression-3のいずれ
かのみが評価されます。IF式の結果として選択されなかった式は評価されません。
したがって、expression-1がゼロ以外の値(真)の場合、expression-2のみが評価され
ます。expression-1がゼロ(偽)の場合、expression-3のみが評価されます。
遅延評価は、ネストされた IF式チェーンの結果式にも適用されます。
IF condition-expression-1 THEN result-expression-1
   ELSE IF condition-expression-2 THEN result-expression-2
   ...
   ELSE IF condition-expression-n THEN result-expression-2
   ELSE result-expression-default

選択された結果式は、評価される唯一の結果式です。遅延評価は条件式にも適用さ
れます。条件式は、条件が真(ゼロ以外)と評価されるか、すべての条件が評価される
まで、順番に評価されます。IF式に n個の条件式があり、i番目の条件が最初の非ゼ
ロ条件である場合、最初の 1から iまでの条件のみが評価されます。i+1から nまで
の条件は評価されません。

IN式
IN式は、式が定数リストに含まれているかどうかを判別します。“定数リスト” (46ペ
ージ)を参照してください。

IN式の構文は次のとおりです。
search-expression[not-operator]IN constant-list

検索式が定数リスト内のいずれかの定数と等しい場合、検索式は定数リストと一致
します。NOT演算子(またはそのエイリアス~および^)が IN演算子の前に指定され
ている場合、IN式の結果は、検索式が一致するかどうかの論理否定になります(論理
NOT式の規則 (104ページ)を参照)。
IN式の結果の型は常に INTEGERであり、その値は 1 (TRUE)、0 (FALSE)、または
null (不明)のいずれかです。次の表は、検索一致がどのように決定されるかを示して
います。
表 9.3 IN式の結果基準

TRUE 検索式は、定数リスト内の 1つ以上の定数と等しいかを比較します。
FALSE 検索式は、定数リスト内のすべての定数と等しくないかを比較します。
UNKNOWN 前述のいずれも当てはまらず、少なくとも 1つの同等性のテストが不明

でした。通常、これは検索式が nullの値に解決される場合に発生しま
す。

検索式が定数リスト内の定数と等しいかどうかを判断する場合、DS2は通常の等式
(=)と同じ規則を使用します。比較されるいずれかの項目の値が不明(null値で表さ
れる)の場合、比較の結果は不明です。
SASモードに関する特別な考慮事項:

n データ型が CHARまたは DOUBLEで、検索式が SAS欠損値と等しい場合、定数
リストの欠損値と一致します。

n 定数リストに null値が指定されている場合、SAS欠損値に変換されます。
n 定数リスト内の 1つ以上の定数の型と一致するように検索式を DOUBLEに変換
する必要があり、検索式の値が nullである場合、null値を DOUBLEに変換する
と、SAS欠損値が発生します。この場合、定数リストに欠損値が含まれている
と、IN式の結果は TRUEになります。

ANSIモードでは、定数リストに欠損値を指定しても意味がありません。ANSIモー
ドでは、定数リストの欠損値値が nullに変換され、不明な値として扱われるためで
す。
DS2が nullと SAS欠損値を処理する方法の詳細については、“存在しないデータの
DS2モード” (61ページ)を参照してください。
次の表は、SASモードと ANSIモードのどちらが有効かによって、さまざまなケース
での IN式の処理を示しています。これらの例では、次の変数宣言を想定していま
す。

dcl char(10) my_char; 
dcl double my_double; 
dcl integer my_int; 

表 9.4 IN式の例

モード 入力値 IN式 結果 説明
両方 my_char = '3.14' my_char in (3.14, 'def', 'ghi') 1 定数リストの 3.14は文字

列'3.14'に変換されます。
SAS my_double = .; my_double in (., 2, 3) 1 検索式の欠損値は、定数リ

ストの欠損値と一致しま
す。

SAS my_double = .; my_double in (1, 2, 3) 0 検索式の欠損値が非欠損値
と一致しません。

ANSI my_double = null; my_double in (1, 2, 3) null 検索式が不明の場合、定数
リストに関係なく IN式の
値が不明です。

ANSI my_int = -1; my_int in (1, 2, 'not a number') null 文字列'not a number'を
BIGINT型に変換しようと
すると失敗し、null値にな
ります。一致するものがな
く、不明な比較が少なくと
も 1つあるため、IN式全体
の値は不明です。

SAS my_int = null; my_int in (3, 3.14, .) 1 小数値を表すには、定数リ
ストを DOUBLEにプロモ
ートする必要があります。
したがって、検索引数も
DOUBLEに変換する必要が
あります。結果は欠損値で
あり、リスト内の定数と一
致します。

LIKE式
LIKE式の概要
LIKE式は、文字列がパターンマッチングの指定と一致するかどうかを判別します。
LIKE式の構文は次のとおりです。
expression[ NOT]LIKE pattern-matching-expression[ESCAPE character-expression]

式は、任意の文字列またはバイナリ文字列のデータ型にすることができます。

expressionが pattern-matching-expressionで指定されたパターンと一致する場合、
値 1 (真)が返されます。それ以外の場合は、値 0 (偽)が返されます。
NOT LIKEは、LIKEの逆の値を返します。たとえば、x like yが真の場合、x not like y
は偽です。
ESCAPE引数は、通常パターンマッチングに使用されるパーセント(%)文字とアンダ
ースコア(_)文字のリテラルインスタンスを検索するために使用されます。

検索パターン
パターンは、3つのクラスの文字で構成されます。
表 9.5 パターンマッチング文字

文字 説明
アンダースコア (_) 任意の 1文字に一致
パーセント記号(%) 0個以上の文字の任意のシーケンスに一致

注: 末尾の空白の影響に注意してください。値を一致
させるには、TRIM関数を使用して末尾の空白を削除
する必要がある場合があります。

その他の文字 その文字に一致

リテラル%および_の検索
%および_文字は LIKE式のコンテキストで特別な意味を持つため、入力文字列でこ
れらの文字リテラルを検索するには、ESCAPE引数を使用する必要があります。
これらの例では、値 app、a_%、a__、bbaa1、ba_1を使用しています。
n 条件 like 'a_%'は、app、a_%、および a__に一致します。検索パターンのアンダー
スコア(_)は、任意の 1文字(アンダースコアを含む)と一致し、検索パターンのパ
ーセント(%)は、'%'と'_'を含む 0個以上の文字と一致します。

n 条件 like 'a_^%' escape '^'は、a_%のみに一致します。エスケープ文字(^)は、パタ
ーンがリテラル'%'を検索することを指定します。

n 条件 like 'a_%' escape '_'は、どの値とも一致しません。エスケープ文字(_)は、パタ
ーンが'a'の後に続くリテラル'%'を検索することを指定します。これは、これらの
値のいずれにも適用されません。

大文字と小文字が混在する文字列の検索
DS2 LIKE式では大文字と小文字が区別されます。大文字と小文字が混在する文字
列を検索するには、次の例に示すように UPCASE関数を使用します。
upcase(str) like 'SM%'

LIKE式の例
次の表は、Smith, Smooth, Smothers, Smart, Smuggleという文字列を検索したときに
一致する結果の例を示しています。
表 9.6 LIKE式の例

LIKE式の例 マッチング結果
str like 'Sm%' Smith, Smooth, Smothers, Smart, Smuggle

str like '%th' Smith, Smooth

str LIKE 'S__gg%' Smuggle

str like 'S_o' (一致なし)

str like 'S_o%' Smooth, Smothers

str like 'S%th' Smith, Smooth

str not like 'Z' Smith, Smooth, Smothers, Smart, Smuggle

メソッド式
メソッド式は、METHODステートメントによって定義されたメソッドを呼び出しま
す。
メソッドを呼び出すには、次の構文を使用します。
method-name ([ expression[, … expression]])

メソッドは、メソッド名とシグネチャに基づいて呼び出されます。DS2は、最初に
メソッド名を識別します。メソッド名が関数名と同じ場合、DS2はそのメソッドを
呼び出します。メソッドと同じ名前の関数は、SYSTEM式を使用して呼び出すこと
ができます。詳細については、“SYSTEM式” (96ページ)を参照してください。

DS2ではオーバーロードされたメソッドが許可されているため、DS2は、メソッド
シグネチャの引数の数と引数のデータ型(メソッドの引数の順序とデータ型)が最も
一致する引数を持つメソッドを呼び出します。最適な一致とは、メソッドパラメー
ターの数が引数の数と等しく、指定された引数リストに対して正確なパラメーター
の型がこれほど多く一致するメソッドシグネチャが他にないようなものです。最適
な一致が見つからない場合、エラーが発生します。

ヒント DS2メソッドは、最大 1000個の引数をサポートできます。1000を
超える引数を持つ DS2メソッドは、コンパイルエラーを生成する可能性があ
ります。

実行するメソッドが識別されると、DS2は必要に応じて、引数のデータ型を対応す
るメソッドパラメーターのデータ型に変換します。その後、メソッドが実行されま
す。
メソッド式は、メソッドによって返される値に解決されます。
次の例では、メソッド CONCATおよび ADDが定義され、INIT()メソッドで呼び出さ
れます。INIT()メソッドで強調表示されている式はメソッド式です。
method concat(char(100) x, char(100) y) returns char(200);
   return trim(x)|| y;
end;

method add(double x, double y) returns double;
   return x + y;
end;

method init();
   dcl char(200) r;
   r = concat('abc', 'def');
   d = add(100,101);
end;

次の例では、Dメソッドがオーバーロードされたメソッドです。DS2は、メソッド
式の引数をメソッドシグネチャと比較して、最適な一致を見つける必要があります。
method d(double x, double y) returns double;
   return x + y;
end;

method d(int x, int y) returns int;
   return x + y;
end;

method init();
   dcl double r;
   dcl int i;
   r = d(1.2345, 5.6789);
   i = d(99, 100);
end;

最初のメソッドは、シグネチャが DOUBLEデータ型の値を必要とする Dメソッドを
呼び出します。2番目のメソッドは、シグネチャが INTEGERデータ型の値を必要と
する Dメソッドを呼び出します。

この最後の例は、DS2がメソッド式の値のデータ型が INTEGERか DOUBLEかを判
断できないことを示しています。あいまいであるため、DS2はエラーを発行します。
method d(int x, double y) returns double;
   return x + y;
end;

method d(double x, int y) returns double;
   return x + y;
end;

method run();
   d = d(100, 102);
end;

詳細については、“METHOD ステートメント” (SAS DS2言語リファレンス)を参照し
てください。

OF演算子
概要と基本構文
OF演算子を使用すると、変数リストと変数配列を関数呼び出しの引数として指定で
きます。変数グループを引数として関数にサブミットすることは、複合式における
OF演算子の唯一の目的です。
OF演算子の構文は次のとおりです。
OFvariable-array[*] | variable-list

variable-array[*]は、同種データ型のグローバル変数をグループ化したものです。添
字演算子のアスタリスクは必須であり、変数配列内のすべての要素が関数に渡され
ることを指定します。
variable-listは、省略形に一致する 1つ以上のグローバル変数を選択する省略形の仕
様です。DS2は、いくつかのタイプの省略変数リストをサポートしています。詳細
については、“省略変数リストの種類” (30ページ)を参照してください。
OF演算子は、変数配列または変数リスト内の各要素を変数名のカンマ区切りリスト
に変換し、これが関数呼び出しの引数になります。次に例を示します。
n OF式 OF x1-x5は引数リスト x1, x2, x3, x4, x5に変換されます。
n VARARRAY double a[6] u v w x y z; として定義されたか、OF演算子によって OF a[*]
として参照された配列は、引数リスト: u, v, w, x, y, zに変換されます。

OF演算子は、関数呼び出しで他の引数とインターリーブできます。たとえば、関数
呼び出し
MAX(v1, v2, OF x1-x5, OF a[*], v3); 

は次の変数のリストに変換されます。
MAX(v1, v2, x1, x2, x3, x4, x5, u, v, w, x, y, z, v3);

関数呼び出しの引数には、任意のタイプの式を使用できます。個々の引数は、OF演
算、添字演算、またはより複雑な式にすることができます。たとえば、次の関数呼
び出しは有効です。
MAX(OF a[*], a[1], y[3], a[1]+y[3]*100)

この関数呼び出しの引数は、次を含むように展開されます。
n 変数配列 aのすべての要素。
n 配列 aの要素 1の値。式の中の添字 1は、グループ内の変数へのインデックスで
す。

n yという名前の配列の要素 3の値。添字の 3は、グループ内の変数へのインデッ
クスです。

n 算術式 a[1]+y[3]*100の結果。
OF演算子には制限があります。
n DS2関数の呼び出しでのみ使用できます。OF演算子は、ユーザー定義メソッド
またはパッケージメソッドの呼び出しでは使用できません。

n さまざまな数のパラメーターを受け取る関数で使用できます。
n パラメーターの数が OFリストの要素の数と一致する関数で使用できます。

注: 関数に渡されるパラメーターには、OFリストなし、1つの OFリスト、いく
つかの OFリスト、他の式を含むいくつかの OFリスト、などを含めることがで
きます。すべての OFリストが展開された後の引数の総数は、関数に固定数の引
数がある場合に関数が取る引数の数と一致する必要があります。

n WHERE句で指定された関数では使用できません。
n DIF関数、LAG関数とは併用できません。
n OF variable-array [*]構文は、変数配列のみをサポートします。OF演算子は、一
時配列または varlistでは使用できません。

n OF variable-array[*]構文は配列インデックスとして使用できません。
n 添字演算を OF演算子のオペランドにすることはできません。添字演算の前に

OF演算子がある場合、DS2は次のエラーを返します。Parse encountered constant 
when expecting '*'. これはサポートされていないものの例です。
OF a[1]
OF y[3]

この例では、yは varlistです。変数情報カテゴリの関数のみが、varlistの要素を
引数として受け取ることができます。

OF演算子の拡張
2023.07以降、OF演算子の構文が拡張され、より柔軟になりました。
n OF演算子は、引数リスト内で 1回指定できます。この構文は引き続きサポート
されていますが、引数リスト内の各変数配列または変数リストの前に OF演算子

を付ける必要はなくなりました。引数リスト内のすべての変数配列と変数リス
トを展開するには、単一の OF演算子で十分です。

n OFを 1回指定し、それが引数リストの最初の引数である場合、引数リスト内の
ほとんどのグローバルスカラー変数はカンマではなくスペースで区切ることが
できます。ローカル変数、配列変数、または複雑な式である関数の引数は、カン
マを使用して OF演算とは区別する必要があります。

表 9.7に、元の構文形式と拡張された構文形式の違いを示します。両方の構文形式
がサポートされています。
表 9.7 有効な OF式の例

元の構文 拡張された構文
x= SUM (OF x - - z, OF a - - c); x= SUM (OF x - - z a - - c);

vsetmissing(z, of a[*], y) vsetmissing(of z a[*] y)

vsetmissing(z, of s:, y) vsetmissing(of z s: y)

vsetmissing(OF x, OF y, OF z, OF a[*]) vsetmissing(OF x y z a[*])

vsetmissing(z, of a[*], y, of t:, u) vsetmissing(of z a[*] y t: u)

OF演算子に対するその他の拡張は次のとおりです。
n 型変数リストは OF演算子でサポートされています。これは、現在サポートされ
ているものの例です。
(OF double)

n OF variable-array[*]構文は、型変数リスト指定を持つ変数配列をサポートするよ
うになりました。たとえば、次の使用法がサポートされるようになりました。
vararray int x[*] int;

n OF variable-array[*]構文は、型変数リスト指定を含む別の配列の範囲の DIMに
基づいた上限を持つ変数配列をサポートします。たとえば、次の使用法がサポー
トされるようになりました。
vararray int x[*] int;
vararray int  y[dim(x)];

n 宣言されていない変数があり、次のいずれかの場合、OF演算子の使用はサポー
トされなくなりました。
o OF演算子の変数リストオペランドは型変数リストです。
o OF演算子の vararray[*]オペランドは型変数リストで定義されます
o OF演算子の vararray[*]オペランドは、型変数リストで定義された別の

vararrayの DIMに基づく範囲を使用して宣言されます。
OF演算子を成功させるには、プログラム内のすべての変数を型変数リストで宣
言する必要があります。宣言されていない変数が OF演算で参照されない場合で
も、宣言されていない変数が存在すると、型変数リストで OF演算を使用できな
くなります。

許可されていない変数リスト要求のログ出力を次に示します。
20         proc ds2;
21           data _null_;
22             method init();
23               dcl double total;
24               x = 10.0; /* undeclared variable */
25               y = 20.0; /* undeclared variable */
26               total = sum(OF double);
27             end;
28           enddata;
29         run;
29       !      quit;
ERROR: Compilation error.
ERROR: Line 26: OF operator with undeclared variables and type 
variable list specification is not supported.
NOTE: PROC DS2 has set option NOEXEC and will continue to prepare 
statements.

許可されていない変数配列要求のログ出力を次に示します。
1        proc ds2;
2         data _null_;
3         vararray double x[*] double;
4
5         method init();
6         a = 1.0;
7         b = 2.0;
8
9         VSETMISSING(OF x[*]);
10         end;
11         enddata;
12         run;
ERROR: Compilation error.
ERROR: Line 9: OF operator with undeclared variables and type variable list
specification is not supported.

DS2と DATAステップの演算子の違い
DS2 OF演算子の機能は、DATAステップ演算子の機能と似ています。ただし、DS2 
OF演算子と DATAステップの OF演算子で異なる結果が出る場合があります。こ
れは、変数リストが各言語でどのように処理されるかによるものです。DATAステッ
プは、関数呼び出しの時点までの変数のみを取得します。DS2は、プログラム全体
で一致するすべての変数を取得します。DATAステッププログラムの SUM関数が
変数 A1と A4のみを使用する例を次に示します。DS2プログラムの SUM関数は、
変数 A1、A4、および A2を使用します。

/* DATA step program */
/* sums A1 and A4 with a result of 4 */
data _null_;
A1 = 1.0; 
A4 = 3.0; 
y = 1; 
lab: 

if (y = 2) then 
X = SUM(OF A: ); 
A2 = 2.0; 
if (y = 1) then do; 
y = 2; 
goto lab; 
end; 
put X=; 
;
run;

/* DS2 program */
/* sums A1, A4, and A2 with a result of 6*/
proc ds2;
data X (overwrite=yes); 
dcl double x A1 A2 A4 y; 
method run(); 
A1 = 1.0; 
A4 = 3.0; 
y = 1; 
lab: 
if (y = 2) then 
X = SUM(OF A: ); 
A2 = 2.0; 
if (y = 1) then do; 
y = 2; 
goto lab; 
end; 
put X=; 
end; 
enddata;
run;
quit;

パッケージメソッド式
パッケージメソッド式は、パッケージで定義されているメソッドをインスタンス化
します。
パッケージメソッド式を呼び出すには、次の構文を使用します。
package-variable.method(method-arguments)

パッケージ変数はスカラー変数または配列パッケージ変数の要素にすることができ
ます。配列パッケージ変数は、複数のパッケージインスタンスをグループ化できる
一時配列であり、各パッケージインスタンスは配列要素としてアクセスされます。
詳細については、DECLARE PACKAGEステートメントを参照してください。配列パ
ッケージ変数のパッケージインスタンスは、次のようにパッケージメソッド式で参
照されます。
package-array\[index\].method(method-arguments)

注: 構文規則の角かっこは、オプションの引数を示します。角かっこの前のエスケー
プ文字(\)は、構文に角かっこが必要であることを示します。配列の境界は、角かっ
こ([ ])で囲む必要があります。

パッケージメソッド式は、メソッド式と同様の方法で実行されます。つまり、DS2
がパッケージとメソッドが存在すると判断した後、メソッドシグネチャの最適な一
致が判断され、必要に応じて引数のデータ型が変換され、メソッドが実行されます。

ヒント DS2メソッドは、最大 1000個の引数をサポートできます。1000を
超える引数を持つ DS2メソッドは、コンパイルエラーを生成する可能性があ
ります。

次の例では、強調表示された式は、スカラーパッケージ変数をパッケージ変数とし
て使用するパッケージメソッド式です。
declare package myadd a1() a2();
a1.sale(3,4);
a1.add(1,2);
a2.bonus(5,12);
a2.add(10,20);

DECLARE PACKAGEステートメントは、a1および a2という名前の 2つのパッケー
ジインスタンスを宣言し、インスタンス化します。最初の 2つのパッケージメソッ
ド式は、MYADDパッケージからインスタンス化された A1パッケージの SALEメソ
ッドと ADDメソッドを呼び出します。最後の 2つのパッケージメソッド式は、A2
パッケージの BONUSメソッドと ADDメソッドを呼び出します。
配列パッケージ変数の使用例については、"パッケージ配列の宣言と使用”を参照して
ください。

注: DS2パッケージメソッド式を FedSQL SELECTステートメントの関数として呼
び出すことができます。詳細については、“式での DS2パッケージの使用” (SAS 
FedSQL言語リファレンス)を参照してください。

パッケージの詳細については、“PACKAGE ステートメント” (SAS DS2言語リファレン
ス)および 13章, “DS2パッケージ” (157ページ)を参照してください。

SYSTEM式
メソッドと関数が同じ名前の場合、メソッド呼び出しが関数呼び出しよりも優先さ
れます。関数は、SYSTEM式を使用してのみ呼び出すことができます。
SYSTEM式を呼び出すには、次の構文を使用します。
SYSTEM.function-expression

SYSTEM式は、関数式の先頭にドット表記 system.を追加します。たとえば、SUMが
関数の名前であると同時にメソッドの名前でもある場合、SUM関数は、SYSTEM式
system.sum(a,b,c)を使用してのみ呼び出すことができます。

THIS式
THIS式は、DS2プログラム内の任意の場所からグローバルスカラー変数を同時に宣
言して使用する代替方法を提供します。THIS式は、標準の変数ルックアップを回避
するために使用されます。THIS式では、DS2はグローバルスコープで識別子のスカ
ラー変数宣言を検索します。そのような宣言がない場合、DS2はグローバルスコー
プで識別子を DOUBLE型で宣言します。グローバル変数は、DS2プログラム内のす
べてのプログラミングブロックで参照できます。
THIS式を呼び出すには、次の構文を使用します。
THIS.variable-name

THIS式は、変数の先頭にドット表記 THIS.を追加します。

注: DS2は、THIS.variable-nameを variable-nameとしてのみ保存します。グローバ
ルスカラー変数と同じ名前のローカル変数があり、DS2が変数に関する診断メッセ
ージを発行する場合、どの変数が問題であるかを区別できません。たとえば、DS2
が xが宣言されていないという警告メッセージを発行した場合、そのメッセージが
グローバル変数 THIS.xを参照しているのか、ローカル変数 xを参照しているのかわ
かりません。

次の例では、変数 sは、THIS式を使用してグローバル変数になります。
method init();
  this.s = sum(a,b,c,d,e);
end;
method run();
   t = put(this.s 5.4);
end; 

THIS式は、同じ名前のローカル変数によって隠されているグローバル変数にアクセ
スする方法を提供します。次に例を示します。

proc ds2;
data;
   declare double x;    /* declare global x */

   method run();
      declare double x;  /* declare local x */

      /* Two variables exist with same name, "x". */
      /* Identifier "x" refers to local x in      */
      /* scope of run method.  Global x is hidden */
      /* by local x.                              */

      this.x = 1.0;    /* assign 1.0 to global x   */
      x = 0.0;         /* assign 0.0 to local x    */

      output;          /* global x with 1.0 output */
    end;
enddata;

run;
quit;

SELECT式
SELECT式の概要
SELECT式は、他の式の値に基づいて複数の式から選択するために使用されます。
SELECT式を呼び出すには、次の構文を使用します。
SELECT [(select-expression)]

WHEN (when-expression) […WHEN (when-expression)]result-expression
[…WHEN (when-expression) […WHEN (when-expression)] result-expression]

OTHERWISE [default-result-expression]

END

SELECT式は、一致する式が見つかるまで、各WHEN式を順番に評価します。次に、
関連付けられた result-expressionが SELECT式の結果として評価されます。
SELECT式は、他の式を使用できる場所であればどこでも使用できます。次に例を示
します。
r = 25.5 + select (t) when (1) -a when (3) b*2 end;

選択式を含む SELECT式
選択式が存在する場合は、それが評価されます。次に、WHEN式が順番に評価され
ます。SELECT式の結果は、選択式と同じ値に評価される最初のWHEN式の結果式
です。すべての WHEN式が選択式とは異なる値に評価される場合、SELECT式の結
果は、存在する場合のデフォルトの結果式です。それ以外の場合は、欠損値または
null値です。次に例を示します。
s = select (t)
   when (1) x*10
   when (3) x
   when (5) x*100
   when (0) 0
   otherwise .
end;

tが 5の場合、SELECT式は'x*100'に評価されます。

選択式のない SELECT式
選択式が存在しない場合は、WHEN式が順番に評価されます。SELECT式の結果は、
真(ゼロ以外の値)と評価される最初のWHEN式の結果式です。すべての WHEN式
が偽(ゼロ)と評価された場合、選択式の結果は、存在する場合のデフォルトの結果式
です。それ以外の場合は、欠損値または null値です。次に例を示します。
 grade = select
   when (score >= 90) 'A'
   when (score >= 80) 'B'
   when (score >= 70) 'C'
   when (score >= 60) 'D'
   when (score >= 0 ) 'F'
 end;

scoreが 76の場合、真と評価される最初の when-expressionは score>=70です。
select-expressionは'C'に評価されます。

オプションのそれ以外の式
それ以外のデフォルトの結果式が提供されない場合、DS2は、WHEN式が選択され
ていないときに選択するデフォルトの結果値を提供します。SASモードで SELECT
式の型が DOUBLEまたは CHARの場合、デフォルトの結果値は欠損値(.)です。どち
らのモードでも、他のすべてのデータ型の場合、デフォルトの結果値は NULLです。

複数のWhen式を使用した結果式
複数のWHEN式を 1つの結果式に関連付けることができます。WHEN式が連続し
てリストされ、その後に 1つの結果式が続きます。結果式に関連付けられたWHEN
式のいずれかが最初に一致するWHEN式である場合、SELECT式の結果は結果式で
す。
たとえば、次の SELECT式は、変数 tの値が'A'、'a'、'P'、'p'のいずれかである場合、
'airplane'と評価されます。
s = select (t)
   when ('A') 
   when ('a') 
   when ('P')
   when ('p') 'airplane'
   when ('C') 
   when ('c') 'car'
   when ('T') 
   when ('t') 'train'
   otherwise  'walk'
 end;

SELECT式のデータ型
SELECT式のデータ型は最初の結果式の型を調べることによって決定されます。最
初の結果式が数値データ型でない場合、SELECT式には最初の結果式の型が割り当て
られます。
最初の結果式が数値データ型の場合、すべての結果式が調べられ、SELECT式の型と
して割り当てる最も広い数値データ型が検索されます。
次の例では、tは TINYINT、bは BIGINT、dは DECIMAL(10,5)、sは CHAR(10)で
す。選択式には、TINYINT、BIGINT、DECIMAL(10,5)、CHAR(10)の中で最も広い数
値型である DECIMAL(10,5)型が割り当てられます。
r = select (t)
   when (1) t
   when (2) b
   when (3) d
   otherwise s
end;

注: sが最初の結果式に割り当てられていた場合、SELECT式の型は CHAR(10)になり
ます。最初の結果式が非数値型の場合、非数値型が SELECT式の型として割り当て
られます。

SELECT式の遅延評価
SELECT式は、WHEN式に遅延評価を使用します。WHEN式は、一致するWHEN式
が見つかるまで、またはすべての式が評価されるまで順番に評価されます。SELECT
式に n個のWHEN式があり、i番目のWHEN式が選択されている場合、最初の 1か
ら iまでのWHEN式のみが評価されます。i+1から nまでの when式は評価されま
せん。
遅延評価は、SELECT式の結果式にも適用されます。選択された結果式は、評価され
る唯一の結果式です。
次の例では、n[i]が 0に等しい場合、最初の 2つのWHEN式(n[i] < 0と n[i] = 0)のみが
評価され、2番目の結果式(y*10-r2)のみが評価されます。
m(select
    when (n[i] < 0) y*100-r1 
    when (n[i] = 0) y*10-r2
    when (n[i] > 0) y*10
 end);

添字演算子
添字演算子を使用すると、DS2配列または varlistの要素にアクセスできます。
添字演算の構文は、配列または varlist識別子と、それに続く配列または varlist内の
各次元のインデックス式で構成されます。構文の形式は次のとおりです。
identifier \[index-expression[,…index-expression] \]

注: 構文規則の角かっこは、オプションの構文を示します。角かっこの前のエスケー
プ文字(\)は、構文に角かっこが必要であることを示します。添字演算子のインデッ
クスは、角かっこ([ ])で囲む必要があります。

index-expressionのデータ型は INTEGERです。指定された index-expressionが整数
でない場合、INTEGERに強制変換されます。
デフォルトでは、インデックスは 1ベースです。varlistの下限は 1です。DECLARE
または VARARRAYステートメントで下限が指定されていない場合、配列の下限は 1
です。ただし、配列には別の下限を指定できます。たとえば、次の配列の下限
は-10、上限は 5です。したがって、有効なインデックスの範囲は-10から 5です。

declare double ta[-10:5];
vararray double va[-10:5];

添字演算子の例は次のとおりです。

a[i]
s[j * 2, k-3]
this.c[2, vwind, a[i]]

各添字演算は、配列または varlistの 1つの要素にアクセスします。多次元配列の場
合、添字演算では、配列の範囲ごとに index-expressionをカンマで区切って指定す
る必要があります。たとえば、次のログ出力を参照してください。

80   proc ds2;
81   data _null_;
82     method init();
83       dcl double x[2,2];
84       dcl int i j;
85       x := (100 200 300 400);
86       do i = 1 to dim(x, 1);
87         do j = 1 to dim(x, 1);
88           put x[i,j]=;
89         end;
90       end;
91     end;
92   enddata;
93   run;
93 !      quit;
x[1,1]=100
x[1,2]=200
x[2,1]=300
x[2,2]=400

インデックス式が配列または varlistの範囲制限を超えている場合、DS2はエラーを
発行します。
配列を使用した添字演算は、配列のデータ型のスカラー変数を受け入れる任意の式
で使用できます。varlistを使用した添字演算は、varlistの要素を受け入れる任意の
式で使用できます。現在、これは V*変数情報関数に限定されています。

式の演算子
複合式での演算子の優先順位

演算子は、加算、比較、論理否定など、オペランドに対して実行される演算の種類
を表します。式に複数の演算子とオペランドが含まれている場合、DS2は演算子の
優先順位を使用して式を解決します。演算は、優先順位の高い順に実行されます。
最高の優先順位は 1で、最低の優先順位は 9です。優先レベル内では、べき乗、最
小、および最大演算子を除いて、演算子は左から右に関連付けられます。べき乗、
最小、および最大演算子は、右から左に関連付けられます。
表 9.8 (102ページ)の優先順位を使用すると、式 5+a**b*3では、a**bが最初に計算
され、次に 3が乗算され、その結果が 5に加算されます。

ヒント DS2では、x < y < zは x < yおよび y < zように評価されます。

次の表に、演算子とその優先順位を示します。
表 9.8 DS2複合式における演算子とその優先順位

優先順位 演算子名 記号 結合
1 かっこ ( ) 左から右へ
1 ドット(.) . 左から右へ
1 添字 identifier[index] 左から右へ
1 SELECT式 左から右へ
2 単項+および– 右から左へ
2 論理 NOT NOTまたは^または~ 左から右へ
2 べき乗、最大、最

小
**、<>、>< 左から右へ

優先順位 演算子名 記号 結合
3 乗算、除算 *、/ 左から右へ
4 プラス、マイナス +、– 左から右へ
5 連結 ||または!! 左から右へ
5 削除 .. 左から右へ
6 IN、LIKE 左から右へ
7 等値または非等

値
=または eq、^=または
~=

右から左へ

7 より大(または小)
か等しい、より大
(または小)

>=、<=、>、<1 左から右へ

8 論理 AND ANDまたは& 左から右へ
9 論理 OR OR、|、または! 左から右へ
10 IF式 右から左へ
なし 変数割り当て = なし
なし 配列割り当て := なし
なし 名前のない

Varlist
[list-of-variable-names] なし

なし _NEW_ なし
なし OF なし
なし メソッド式 なし
1 DS2では、x < y < zは x < yおよび y < zように評価されます。

演算子別の式の値
次の表は、演算子に基づく式の解決値を示しています。

表 9.9 演算子別の式の値

演算子 値
論理式 1

NOTまたは^または~ オペランドがゼロ以外の場合、結果は 0になります。オペ
ランドがゼロまたは欠損している場合、結果は 1になりま
す。オペランドが nullの場合、結果は nullになります。

ORまたは|または! n 2つのオペランドの論理 OR

n 一方のオペランドがゼロまたは欠損しており、もう一方
のオペランドが nullの場合は null

n 両方のオペランドが nullの場合は null

n いずれかのオペランドがゼロ以外の場合は(もう一方の
オペランドが nullまたは欠損している場合でも) 1

n 両方のオペランドがゼロまたは欠損している場合は 0

ANDまたは& n 2つのオペランドの論理 AND

n 一方のオペランドがゼロ以外で、もう一方のオペランド
が nullの場合は null

n 両方のオペランドが nullの場合は null

n 両方のオペランドがゼロ以外の場合は 1

n いずれかのオペランドがゼロまたは欠損している場合は
(もう一方のオペランドが nullであっても ) 0

算術式
単項プラス 式オペランドと同じ値を持つ
単項マイナス オペランドの算術否定
+ オペランドの算術和
– オペランドの算術差
* オペランドの算術積
/ オペランドの算術商
** 左のオペランドを右のオペランドでべき乗したもの
>< 左右のオペランドの最小値
<> 左右のオペランドの最大値

演算子 値
任意の算術演算子 いずれかまたは両方の演算子が nullの場合は null

関係式
< 左のオペランドが右のオペランドより小さい場合は 1、そ

れ以外の場合は 0

> 左のオペランドが右のオペランドより大きい場合は 1、そ
れ以外の場合は 0

<= 左のオペランドが右のオペランド以下の場合は 1、それ以
外の場合は 0

>= 左のオペランドが右のオペランド以上の場合は 1、それ以
外の場合は 0

=または eq 左のオペランドが右のオペランドと等しい場合は 1、それ
以外の場合は 0

^= 左のオペランドが右のオペランドと等しくない場合は 1、
それ以外の場合は 0

任意の論理演算子 いずれかまたは両方の演算子が nullの場合は null

連結式
||または!!2 左のオペランドが右のオペランドとマージされて新しい値

を形成する
.. 連結する前に各引数を削除する
IF式
IF 論理式がゼロでも nullでもない場合は THEN式の値、それ

以外の場合は ELSE式の値。詳細については、“IF式” (83ペ
ージ)を参照してください。

IN式
IN 定数リストに比較式が含まれる場合は 1。詳細について

は、“IN式” (85ページ)を参照してください。
LIKE式
LIKE 式が指定されたパターンに一致する場合は 1、それ以外の

場合は 0 (ゼロ)。詳細については、“LIKE式” (87ページ)を
参照してください。

演算子 値
SELECT式
SELECT SELECT式と一致する最初のWHEN式の結果式の値。詳細

については、“SELECT式” (98ページ)を参照してください。
その他の式
. パッケージ属性にアクセスするか、パッケージの外部のコ

ードからパッケージメソッドを呼び出します。詳細につい
ては、“DOT演算子式” (81ページ)を参照してください。

= 値または式を変数に割り当てます。
:= 配列に値を割り当てます。
identifier[index-expression]配列または varlistの要素にアクセスします。
[list-of-variable-names] 名前のない varlistを作成します。
_NEW_ パッケージのインスタンスを構築します。詳細について

は、“_NEW_ 演算子” (SAS DS2言語リファレンス)を参照して
ください。

OF 変数リストまたは変数配列を関数の引数のリストに変換し
ます。

メソッド式 システムメソッドまたはユーザー定義メソッドを呼び出し
ます。詳細については、“DS2メソッドについて” (SAS DS2
言語リファレンス)を参照してください。

1 DS2は、TRUE、FALSE、UNKNOWNの 3つの真理値をサポートしています。ゼロ以外のオペランドは TRUEにマッピングされます。ゼロまたは欠損オペランドは FALSEにマッピングされます。nullオペランドは UNKNOWNにマッピングされます。
2 ||連結演算子はスペースを削除しません。TRIM関数を使用して、末尾のスペースを削除できます。ただし、..演算子は、TRIMとそれに続く連結と同じ機能を実行します。TRIM関数

(TRIM(a) || TRIM(b))を使用するよりも、..演算子(a .. b)を使用する方が高速です。

DS2での短絡評価
場合によっては、両方のオペランドを評価せずに論理 ANDまたは OR式の値を決定
できます。たとえば、オペランド Aが偽の場合、式(A AND B)の値は、オペランド B
の値に関係なく偽になります。最初のオペランドのみに基づいて論理式全体の値を
決定できる場合、プログラミング言語は評価を"短絡"し、2番目のオペランドを評価
するために必要な計算処理を回避できます。

オプション LOGICALEXPR=OPTIMIZEDを設定することにより、DS2で短絡評価を
使用できます。LOGICALEXPR=オプションは、DS2_OPTIONSステートメントの引
数、DS2プロシジャオプション、ならびに DS2 runDS2および runModelアクショ
ンのパラメーターとして使用できます。
DS2は、論理 ANDおよび OR式のオペランドを左から右に評価します。
LOGICALEXPR=OPTIMIZEDが有効な場合、DS2は論理演算子を次のように評価し
ます。
n ANDの場合、最初のオペランドがゼロまたは欠損している場合、2番目のオペラ
ンドは評価されず、式全体が falseになります。最初のオペランドがゼロ以外ま
たは nullの場合、DS2は 2番目のオペランドを評価する必要があります。

n ORの場合、最初のオペランドがゼロ以外の場合、2番目のオペランドは評価さ
れず、式全体が TRUEになります。最初のオペランドがゼロ、欠損、または null
の場合、DS2は 2番目のオペランドを評価する必要があります。

LOGICALEXPR=OPTIMIZEDを使用すると、数学的に無効な計算の実行を回避できま
す。次のプログラムを考えてみましょう。プログラムでは、式 1/sqrt(a)は、aが非
負(平方根による)で、sqrt(a)が非ゼロ(除算による)である場合にのみ計算できます。
LOGICALEXPR=OPTIMIZEDが設定されている場合、AND式の 2番目のオペランド
が評価されないため、プログラムはエラーなしで完了します。

proc ds2 errorstop;
   ds2_options logicalexpr=optimized;
   data _null_;
      declare double a;
      method init();
         a=0;
         if (a>0 AND 1/sqrt(a) < .5) then
            put 'criteria satisfied';
         else
            put 'unacceptable value';
         put 'execution continues ...';
      end;
   enddata;
run; quit;

次のような行がログに書き込まれます。

unacceptable value
execution continues ...
NOTE: Execution succeeded.No rows affected.

LOGICALEXPR=OPTIMIZEDを使用して、計算コストの高い計算の実行を回避するこ
ともできます。次のコードスニペットを考えてみましょう。ユーザー定義パッケー
ジ credit_evalの scoreメソッドには、複雑なロジックが含まれています。このビジ
ネスケースでは、顧客がすでにローンの債務不履行に陥っている場合、そのメソッ
ドの呼び出しは避ける必要があります。

dcl package credit_eval ce;
reject_loan_request = (has_defaulted OR ce.score() < 500);
…

AND演算子と OR演算子、およびそれらの代替演算子"&"、"|"、"!"には短絡評価が
用意されています。

DS2の日付、時間、およびタイムスタン
プ
DS2の日付、時間、およびタイムスタンプの概要

DS2は、他のデータソースで使用されている SQLスタイルの日付と時間の規則をサ
ポートしています。データソースが SASデータセットでない場合、DS2は DATE、
TIME、および TIMESTAMPのデータ型を持つ日付と時間を処理できます。
DATE、TIME、および TIMESTAMPのデータ型の日付と時間の値は、SASの日付、
時間、または日時の値に変換できます。数値列が SASデータセットから読み取られ、
数値列に SASの日付、時間、または日時の出力形式が関連付けられている場合、そ
の列は DS2型の DATE、TIME、または TIMESTAMPデータ型に変換されます。

DS2は、任意の日付または時間の値を SAS日付、時間、および日時の値に変換し、
認識可能な日付または時間の値に戻す日付と時間関数を提供します。詳細について
は、“日付、時間、および日時関数” (114ページ)を参照してください。
ANSI SQLでサポートされている日付と時間の間隔は、DS2ではサポートされていま
せん。

日付、時間、およびタイムスタンプ変数の宣言
次の例のように、DECLAREステートメントで DATE、TIME、または TIMESTAMPデ
ータ型を使用して、日付、時間、またはタイムスタンプ変数を宣言します。
dcl date dt;
dcl time tm;
dcl timestamp tmstmp;

注: 時間変数またはタイムスタンプ変数を宣言するときに精度を使用する場合、時間
またはタイムスタンプの値は、DATAステートメントによって生成されるまで、指定
された精度に丸められません。内部的には、時間またはタイムスタンプの定数値は
単に時間またはタイムスタンプ変数にコピーされます。

注: SAS以外のデータソースで TIME値と TIMESTAMP値を操作していて、精度を指
定しない場合、デフォルトの精度は DS2のデフォルトの精度である TIMEは 6、
TIMESTAMPは 6になります。

DS2の日付と時間のデータ型の詳細については、5章, “DS2データ型” (47ページ)を
参照してください。

DS2の日付、時間、およびタイムスタンプの値
日付、時間、またはタイムスタンプの変数を宣言すると、その変数の値は、次の構
文を持つ DS2の日付、時間、またはタイムスタンプの定数のみにすることができま
す。
DATE'yyyy-mm-dd'

TIME'hh:nn:ss[.fraction]'

TIMESTAMP'yyyy-mm-dd hh:nn:ss[.fraction]'

ここでは、次のとおりです。
n yyyyは 4桁の年です
n mmは 01 - 12の 2桁の月です
n ddは 01 - 31の 2桁の日です
n hhは 2桁の軍用時間です
n mmは 2桁の分です

n ssは 2桁の秒です
n fractionは 1から 9桁(0から 9)で、オプションであり、秒の端数を表します
DATE、TIME、または TIMESTAMPキーワードの後の値の文字列部分は、単一引用
符で囲む必要があります。
日付定数にはハイフンが必要で、日付文字列の長さは 10にする必要があります。
時間定数にはコロンが必要です。秒の端数が存在しない場合、時間文字列は 8文字
の長さでピリオドを除く必要があります。端数なしでピリオドが存在する場合、
DS2はエラーを発行します。秒の端数が存在する場合、端数の長さは最大 9桁、時
間文字列の長さは最大 18文字(ピリオドを含む)です。
タイムスタンプ定数では、日付のハイフンと時間のコロンが必要です。秒の端数が
存在しない場合、タイムスタンプ文字列は 19文字の長さでピリオドを除く必要があ
ります。秒の端数が存在する場合、端数の長さは最大 9桁、タイムスタンプ文字列
の長さは最大 29文字です。
DS2の日付、時間、およびタイムスタンプ定数の例を次に示します。
date '2017-01-31'
time '20:44:59'
timestamp '2017-02-07 07:00:00.7569'

DS2の日付と時間の操作
DATE、TIME、および TIMESTAMP値に対して実行できる操作は、次のステートメ
ントのように関係演算子<、>、<=、>=、=、^=、INを使用する操作のみです。
if tm in(time'10:22:31', time'12:55:01') then 
   if tm < time'13:30:00' then put 'Early afternoon';
   else put 'Time not available';

DS2は、DATE、TIME、および TIMESTAMPのデータ型を持つ値の日付と時間の間
隔を計算しません。

SASの日付、時間、および日時の値
SAS日付値は、1960年 1月 1日から指定日付までの日数です。1960年 1月 1日よ
り前の日付は負の数、それより後の日付は正の数です。たとえば、1960年 1月 1日
の SAS日付値は 0、1959年 1月 1日の場合は-365、2008年 1月 1日の場合は
17532です。
SAS時間値は、現在日の午前 0時からの秒数です。SAS時間値は 0 - 86400です。
SAS日時値は、1960年 1月 1日から特定の日付の特定の時、分、秒までの秒数で
す。

数値列が SASデータセットから読み取られ、数値列に SASの日付、時間、または日
時の出力形式が関連付けられている場合、その列は DS2型の DATE、TIME、または
TIMESTAMP列に変換されます。SASデータセットの数値列に出力形式がないか、
SASの日付、時間、日時出力形式以外の出力形式である場合、列は DOUBLE型とし
て処理されます。
DOUBLEが DS2型の DATE、TIME、または TIMESTAMPに変換されると、日付と時
間に関するすべての計算が SAS日付値、SAS時間値、または SAS日時値として実行
されます。詳細については、“日付、時間、および日時関数” (114ページ)を参照して
ください。計算が完了したら、SASの日付、時間、および日時の値を認識可能な日
付と時間の出力形式にフォーマットできる他の関数があります。

SASの日付、時間、および日時の値を DS2
の日付、時間、またはタイムスタンプの
値に変換する

SASの日付、時間、および日時の値は、TO_DATE、TO_TIME、および TO_TIMESTAMP
関数を使用して、DS2の日付、時間、およびタイムスタンプの値に変換できます。
これらの関数の引数は、SASの日付、時間、または日時の値を表し、DOUBLE型を
持つ任意の値または式です。次に、PUTステートメントまたは DECLAREステート
メントの出力形式を使用して、日付、時間、またはタイムスタンプの値をフォーマ
ットできます。
次に例を示します。
data _null_; 
   dcl date ds2d having format YYMMDD10.;
   dcl time ds2t having format TIME18.9;
   dcl timestamp ds2dt having format DATETIME28.9;
   dcl double d t ts;
   method init();
      d = 20854;
      ds2d = to_date(d);
      ds2t= to_time(d);
      ds2dt= to_timestamp(d);
      put ds2d ds2t ds2dt;
   end;
enddata;

次の行が SASログに書き込まれます。

2017-02-04
 5:47:34.000000000
01JAN1960:05:47:34.000000000

詳細については、“TO_DATE 関数” (SAS DS2言語リファレンス)、“TO_TIME 関数” (SAS 
DS2言語リファレンス)、および“TO_TIMESTAMP 関数” (SAS DS2言語リファレンス)
を参照してください。

DS2の日付、時間、およびタイムスタン
プの値を SASの日付、時間、または日時
の値に変換する

DS2の DATE、TIME、および TIMESTAMP値は、TO_DOUBLE関数を使用して SAS
の日付、時間、および日時の値に変換できます。
n TO_DOUBLE関数は、DS2 DATEデータ型の値を、SAS日付値である DS2 

DOUBLEデータ型の値に変換します。
n TO_DOUBLE関数は、DS2 TIMEデータ型の値を、SAS時間値である DS2 

DOUBLEデータ型の値に変換します。
n TO_DOUBLE関数は、DS2 TIMESTAMPデータ型の値を、SAS日時値である DS2 

DOUBLEデータ型の値に変換します。
その後、SASの日付、時間、または日時出力形式を使用して、DS2 DOUBLE値を目
的の表現で表示できます。SASの日付、時間、または日時の値の型は、出力形式の
型と一致する必要があることに注意してください。
次の DS2プログラムは、DS2の日付、時間、およびタイムスタンプの値を SASの日
付、時刻、および日時の値に変換する方法を示しています。
data _null_;
  method run();
    dcl date ds2Date;
    dcl time ds2Time;
    dcl timestamp ds2TimeStamp;
    dcl double sasDate having format date.;
    dcl double sasTime having format time.;
    dcl double sasDateTime having format datetime.;
    dcl varchar(100) fmtDate fmtTime fmtDateTime;

    ds2Date = date'2017-06-13';
    ds2Time = time'10:54:34.012';
    ds2TimeStamp = timestamp'2017-06-13 10:54:34.012';

    put ds2Date=;    
    put ds2Time=;
    put ds2TimeStamp=;

    sasDate = to_double(ds2Date);
    sasTime = to_double(ds2Time);
    sasDateTime = to_double(ds2TimeStamp);

    put sasDate= sasDate:best16.7;
    put sasTime= sasTime:best16.7;
    put sasDateTime= sasDateTime:best16.7;

    fmtDate = put(sasDate, monname.);
    fmtTime = put(sasTime, hour.);
    fmtDateTime = put(sasDateTime, dateampm.);

    put fmtDate=;
    put fmtTime=;
    put fmtDateTime=;
  end;
enddata;

次の出力が SASログに書き込まれます。

ds2Date=2017-06-13
ds2Time=10:54:34.012000000
ds2TimeStamp=2017-06-13 10:54:34.012000000
sasDate=13JUN17 20983
sasTime=10:54:34 39274.012
sasDateTime=13JUN17:10:54:34 1812970474.012
fmtDate=     June
fmtTime=11
fmtDateTime=13JUN17:10:54:34 AM

この例では、DS2の DATE、TIME、および TIMESTAMP列の値が SAS日付値に変換
され、出力値が DS2 DOUBLE型の列に格納されます。DOUBLE列には、日付、時
間、および日時の出力形式が定義されています。
DS2の DATE、TIME、および TIMESTAMP変数に、SASの日付、時間、および日時
の値を割り当てることはできません。それらのデータ型は異なります。SASの日付
または時間の値を DS2の DATE、TIME、または TIMESTAMP変数に割り当てようと
すると、DS2は無効なデータ型変換エラーを発行します。
詳細については、“PUT 関数” (SAS DS2言語リファレンス)を参照してください。出力
形式の完全なリストについては、“日付、時間、および日時の出力形式” (116ページ)
を参照してください。

日付、時間、および日時関数
日付と時間の計算を実行するために、DS2の日付と時刻関数は次のことを行います。
n 日付と時間を SASの日付、時間、または日時値として取得または変換する
n SASの日付、時間、または日時の値を認識可能な日付または時間にフォーマット
する

n SAS日時値から日付または時間を抽出する
次の表に、日付と時間の関数とその機能を示します。これらの関数の具体的な情報
については、“DS2関数” (SAS DS2言語リファレンス)を参照してください。

表 10.1 日付と時間を取得、あるいは SASの日付、時間、または日時値に変換する
関数

関数 説明
DATE SAS日付値として現在の日付を返す
TODAY SAS日付値として現在の日付を返す
DATEJUL ユリウス暦の日付を SAS日付値に変換する
DHMS 日、時、分、秒の値から SAS日時値を返す
HMS 時、分および秒の値から SAS時間値を返す
MDY 月、日および年の値から SAS日付値を返す
TIME 現在時刻を SAS時間値として返す
YYQ 年と四半期の値から SAS日付値を返す

表 10.2 SAS日付または日時値を認識可能な日付または時間としてフォーマットす
る関数

関数 説明
DAY SAS日付値として月の日を返す
HOUR SAS時間または日時値から時間を返す
JULDATE SAS日付値からユリウス暦の日付を返す
JULDATE7 SAS日付値から 7桁のユリウス暦の日付を返す
MINUTE SAS時間または日時値から分を返す
MONTH SAS日付値から月を表す数字を返す
QTR SAS日付値から年の四半期を返す
SECOND SAS時間または日時値から秒を返す
WEEKDAY SAS日付値から、曜日に対応する整数を返す
YEAR SAS日付値から年を返す

表 10.3 SAS日時値から日付と時間を抽出する関数

関数 説明
DATEPART SAS日時値から日付を抽出し、その日付を SAS日付値とし

て返す
TIMEPART SAS日時値から時間を抽出し、その時間を SAS日時値とし

て返す

日付、時間、および日時の出力形式
DS2出力形式は、認識可能な日付と時間として SASの日付、時間、および日時の値
を書き込みます。PUT関数を使用して、SASの日付、時間、または日時の値をフォ
ーマットします。PUT関数で出力形式を指定する構文は次のとおりです。
PUT(SAS-value, format.);

PUT関数の第 1引数は、SAS日付値、SAS日時値、または SAS時間値です。第 2引
数は出力形式です。
PUT関数がどのように実行されるかを示す単純な DS2プログラムは次のとおりで
す。

proc ds2;
data _null_;
   declare double d;
   declare varchar(15) x;
   method run();
      d=SAS-date-value;
      x=put (d,format-name.);
      put x=;
   end;
enddata;
run;
quit;

入力変数は DOUBLEです。出力変数は文字変数です。PUT関数は文字変数を操作
します。
表 10.4 (117ページ)は、さまざまな DS2の出力形式の種類を使用して、2017年 6
月 13日(2017-06-13 10:54:34.012)を表す SASの日付、日時、および時間の値をフ
ォーマットした結果を示しています。SASの日付、時間、および日時の値がどのよ
うに取得されたかについては、前述の“DS2の日付、時間、およびタイムスタンプの
値を SASの日付、時間、または日時の値に変換する” (113ページ)を参照してくださ
い。

注: SASの日付、時間、および日時出力形式を使用する場合、SAS入力値はその出力
形式の種類に適している必要があります。つまり、SAS日付値は日付出力形式で指

定します。SAS日時値は日時出力形式で指定します。SAS時間値は時間出力形式で
指定します。SAS日付値は、四半期および年の出力形式で使用されます。適切な値
を指定しないと、DS2は不適切な応答を返します。

表 10.4 DS2の日付と時間の出力形式の例

出力形式
の種類 出力形式名 SAS入力値 出力形式指定 結果
日付 DATEw. 20983 デフォルト 13JUN17

20983 DATE9. 13JUN2017

DAYw. 20983 デフォルト 13

DDMMYYw. 20983 デフォルト 13/06/17

20983 DDMMYY10. 13/06/2017

DDMMYYxw. 20983 DDMMYYB. 13 06 17

20983 DDMMYYB10. 13 06 2017

20983 DDMMYYC. 13:06:17

20983 DDMMYYC10. 13:06:2017

20983 DDMMYYD. 13-06-17

20983 DDMMYYD10. 13-06-2017

20983 DDMMYYN6. 130617

20983 DDMMYYN8. 13062017

20983 DDMMYYP. 13.06.17

20983 DDMMYYP10. 13.06.2017

20983 DDMMYYS. 13/06/17

20983 DDMMYYS10. 13/06/2017

DOWNAMEw. 20983 デフォルト Tuesday

JULIANw. 20983 デフォルト 17164

MMDDYYw. 20983 デフォルト 06/13/17

MMDDYYw. 20983 MMDDYY10. 06/13/2017

MMDDYYxw. 20983 MMDDYYP. 06.13.17

出力形式
の種類 出力形式名 SAS入力値 出力形式指定 結果

20983 MMDDYYP10. 06.13.2017

20983 MMDDYYS. 06/13/17

20983 MMDDYYS10. 06/13/2017

MMYYw. 20983 デフォルト 06M2017

MMYYxw. 20983 MMYYC. 06:2017

20983 MMYYD. 06-2017

20983 MMYYN. 062017

20983 MMYYP. 06.2017

20983 MMYYS. 06/2017

MONNAMEw. 20983 デフォルト June

MONTHw. 20983 デフォルト 6

MONYYw. 20983 デフォルト JUN17

WEEKDATEw. 20983 デフォルト Tuesday, June 13, 
20171

WEEKDATXw. 20983 デフォルト Tuesday, 13 June 
20171

WEEKDAY2. 20983 デフォルト 3

四半期 QTRw. 20983 デフォルト 2

QTRRw. 20983 デフォルト II

日時 DATEAMPMw.
d

1812970474.
012

デフォルト 13JUN17:10:54:34 
AM1

DATETIMEw. 1812970474.
012

デフォルト 13JUN17:10:54:341

DTDATEw. 1812970474.
012

デフォルト 13JUN17

DTMONYYw. 1812970474.
012

デフォルト JUN17

DTWKDATXw. 1812970474.
012

デフォルト Tuesday, 13 June 
20171

出力形式
の種類 出力形式名 SAS入力値 出力形式指定 結果

DTYEARw. 1812970474.
012

デフォルト 2017

DTYYQCw. 1812970474.
012

デフォルト 17:2

時間 HOURw.d 39274.012 デフォルト 11

TIMEw.d 39274.012 デフォルト 10:54:34

TIMEAMPMw.
d

39274.012 デフォルト 10:54:34 AM

TODw.d 39274.012 デフォルト 10:54:34

年 YEARw. 20983 デフォルト 2017

YYMMw. 20983 デフォルト 2017M06

YYMMxw. 20983 YYMMC. 2017:06

20983 YYMMD. 2017-06

20983 YYMMN. 201706

20983 YYMMP. 2017.06

20983 YYMMS. 2017/06

20983 デフォルト 17-06-13

20983 YYMMDDB. 17 06 13

20983 YYMMDDC. 17:06:13

20983 YYMMDDD. 17-06-13

20983 YYMMDDN. 20170613

20983 YYMMDDP. 17.06.13

20983 YYMMDDS. 17/06/13

YYMONw. 20983 デフォルト 2017JUN

年/四半期 YYQw. 20983 デフォルト 2017Q2

YYQxw. 20983 YYQC. 2017:2

20983 YYQD. 2017-2

出力形式
の種類 出力形式名 SAS入力値 出力形式指定 結果

20983 YYQP. 2017.2

20983 YYQS. 2017/2

20983 YYQN. 20172

YYQRw. 20983 デフォルト 2017QII

YYQRxw. 20983 YYQRC. 2017:II

20999 YYQRD. 2017-II

20983 YYQRP. 2017.II

20983 YYQRS. 2017/II

20983 YYQRN. 2017II

YYQZw. 20983 デフォルト 1702

1 この例では、出力変数は VARCHAR(15)ではなく VARCHAR(29)として定義されています。

DS2配列の概要
DS2では、配列は同種データの名前付き集計コレクションです。DS2には、変数配
列と一時配列の 2種類の配列があります。一時配列には、固定範囲または動的範囲
を設定できます。
変数配列と標準一時配列のサイズは固定です。固定サイズの配列は、ランタイム条
件に依存しない同様のデータをグループ化するための優れた方法を提供します。た
だし、配列に必要な要素数がランタイム条件に依存する場合、固定配列は、最悪の
ケースのランタイム条件で必要な要素数を考慮して静的にサイズ設定する必要があ
ります。通常、最悪のケースは平均的なケースではありません。そのため、実際に
必要なサイズを大幅に超えるサイズで配列が宣言されることが多く、メモリの使用
率が低下します。
動的範囲制限ありの配列は、実行中に必要に応じて配列のサイズを拡大または縮小
できるようにすることで、メモリを節約できます。DS2は、一時配列に対してのみ
動的範囲をサポートします。
次の表は、変数配列、標準一時配列、および動的範囲を持つ一時配列の違いをまと
めたものです。

変数配列 標準一時配列 動的一時配列
PDVの変数への参照の集
合

一時要素の集合 一時要素の集合

VARARRAYステートメン
トで作成される

DECLAREステートメント
で作成される

DECLAREステートメント
で作成される

符号付き整数値でインデッ
クス付けされるか、DS2が
配列内の変数をカウントし
て添え字を決定する。これ
は、VARARRAYステートメ
ントにリストする必要があ
る。

符号付き整数値によるイン
デックス

下限 1で作成され、0要素
で初期化される

DS2プログラムの実行中、
配列内の要素の数は静的

DS2プログラムの実行中、
配列内の要素の数は静的

配列は RESIZEARRAYステ
ートメントを使用して実行
時にサイズ調整される

範囲の数は>=1にすること
ができる

範囲の数は>=1にすること
ができる

1次元

グローバルスコープで宣言
する必要がある

ローカルまたはグローバル
スコープで宣言できる

ローカルまたはグローバル
スコープで宣言できる

変数配列 標準一時配列 動的一時配列
他の言語で見られるポイン
ターまたは参照の配列に似
ている

他の言語で見られる配列に
似ている

他の言語で見られる配列に
似ている

DATAステップで、ARRAY
ステートメントで作成され
る

DATAステップで、
_TEMPORARY_引数を指定
した ARRAYステートメン
トで作成される

対応する DATAステップな
し

配列の割り当ては、すべての配列タイプで同じです。

変数配列
変数配列の概要

変数配列は、入力データで類似した名前または目的を持つ一連の変数の処理を簡素
化する方法です。変数配列の要素は、PDV内の変数を参照します。変数配列は、DS2
プログラムの期間中のみ存在します。ただし、参照された変数の内容は、1つ以上
の結果セットに保持される場合があります。
VARARRAYステートメントを使用して、名前、データ型、および配列範囲の数とサ
イズを指定します。たとえば、次の VARARRAYステートメントは、PDV内の 3つの
double変数(a1、a2、a3)を参照する 3要素の変数配列を指定します。
vararray double a[3];

前の例の VARARRAYステートメントは、未作成の double変数(a1、a2、a3)のいず
れかを作成します。変数配列要素 a[1]は変数 a1を参照し、a[2]は変数 a2を参照し、
a[3]は変数 a3を参照します。
変数配列が作成されると、変数配列要素は、PDV内の参照変数に保存されているデ
ータの読み取りまたは変更に使用できる 2番目の識別子セットとして機能します。
VARARRAYステートメントで HAVING句を使用して、ラベル、出力形式、および入
力形式属性を変数配列に関連付けることができます。

注: 文字変数配列に含まれるすべての文字変数は、同じ長さとエンコーディングであ
る必要があります。

詳細については、“VARARRAY ステートメント” (SAS DS2言語リファレンス)を参照し
てください。

変数配列宣言
変数配列宣言はスカラー宣言に似ています。データ型と名前に加えて、配列範囲の
数とサイズも指定できます。複数の範囲(または次元)は、カンマ区切り文字を使用し
て指定されます。
配列範囲には 2つの形式があります。
符号付き整数ペア
符号付き整数ペアの形式は、配列の各次元の下限と上限[l:h]を指定します。ここ
で、lは指定された範囲の最小インデックスを表し、hは指定された範囲の最大
インデックスを表します。下限の指定 lはオプションです。次元の下限が指定さ
れていない場合、下限はデフォルトで 1になります。
上限 hが 1より小さい場合、エラーが返されます。1つの整数のみで配列範囲を
指定すると、その整数が上限として解釈されます。デフォルトの最下限は 1で
す。
vararray double a[5];は、1から 5までのインデックスが付けられた 5要素を持つ
double型の配列 aを宣言します。vararray char b[5,10];は、1番目の次元に 5要
素、2番目の次元に 10要素で、合計 50の要素を持つ 2次元文字配列 bを宣言し
ます。vararray int c[3] x y z;は、3要素を持つ配列 cを宣言します。配列には、下
限 1と上限 3でインデックスが付けられます。
配列の上限は、以前に宣言された配列の次元の要素数に基づいてサイズを変更す
ることもできます。上限には DIM関数呼び出しを使用します。DIM関数は、配
列の上限を指定するために使用できる唯一の関数です。DIM関数は、次元の下限
を指定するためには使用できません。

* (アスタリスク)
*形式は、下限が 1で、上限が変数リストに含まれる変数の数である 1次元配列
を指定します。

詳細については、“DS2変数リスト” (28ページ)を参照してください。変数配列を宣
言する方法と複数の範囲を指定する方法の詳細については、“VARARRAY ステートメ
ント” (SAS DS2言語リファレンス)を参照してください。

VARARRAYステートメントでの変数の定義
VARARRAYステートメントは、変数リスト内の未定義の変数を定義します。一部の
変数リストタイプは、既存の変数のみを参照するため、新しい変数の定義にはなり
ません。次の表は、新しい変数を定義できる変数リストタイプと、既存の変数のみ
を参照する変数リストタイプを示しています。

変数リスト 例 変数展開 説明
名前 x y z 即時 新しい変数を定義できる

124 11章 / DS2配列

http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p0zxzvxwfy019bn1jc4x3hzakwzk.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p0zxzvxwfy019bn1jc4x3hzakwzk.htm&locale=ja


変数リスト 例 変数展開 説明
番号付き範囲 x1–x5 即時 新しい変数を定義できる
名前範囲 sales_jan--sales_mar 遅延 既存の変数のみを参照する
名前接頭辞 sales: 遅延 既存の変数のみを参照する
型 smallint 遅延 既存の変数のみを参照する
特殊名 _all_ 遅延 既存の変数のみを参照する

DS2プログラム内のすべての変数を検査しなくても、名前および番号付き範囲の変
数リストを展開できます。したがって、これらのタイプの変数リストは、プログラ
ム内で VARARRAYステートメントが検出されるとすぐに展開されます。
他のタイプの変数リストでは、DS2プログラムで定義されているすべての変数を検
査する必要があります。これらの変数リストの展開は、DS2プログラム内のすべて
のステートメントが検査され、すべての変数が定義されるまで遅延します。
この遅延により、予期しないエラー状態が発生する可能性があります。たとえば、
次のプログラムを考えてみましょう。

注: このセクションのコード例の DS2ステートメントは、PROC DS2を使用して
SASに送信されます。

proc ds2;
data;
   vararray int x[1] x:; /* 1  */

   method run(); 
     x1 = 5.0;           /* 2  */
   end;
enddata;
run;
quit;

1 接頭辞変数リスト x:は既存の変数のみを参照するため、VARARRAYステートメン
トは変数を作成しません。接頭辞変数リスト x:の展開は、プログラム内のすべて
のステートメントが検査されるまで遅延します。

2 割り当てステートメントでは、変数 x1は未定義です。したがって、割り当てス
テートメントは、右辺の値の型(DOUBLE)を変数 x1に割り当てます。

すべてのプログラムステートメントが検査された後、接頭辞変数リスト x:は、接頭
辞 xを持つ唯一の既存の変数である x1に展開されます。DOUBLE型の変数 x1は
INTEGER型の変数配列 xと互換性がないため、DS2コンパイラはコンパイルエラー
を発行します。
エラー状態を取り除く 1つの方法は、VARARRAYステートメント vararray int x[1] x:;
を vararray int x[1] x1;または vararray int x[1];に変更することです。改訂されたステー
トメントは、変数 x1を INTEGER型として定義します。

DIM変数配列範囲を使用した遅延変数定義
変数配列の次元の上限が別の配列の次元に基づいている場合、プログラム内のすべ
てのステートメントが検査されるまで、配列の変数の定義を遅延させることができ
ます。次に例を示します。

proc ds2;
data;
   vararray double x[*] x:;  /* 1  */
   vararray int out[dim(x)]; /* 2  */

   method init();
   out1 = 0.0;               /* 3  */
   end;

   method run();
      set in;
   end;
enddata;
run;
quit;

1 接頭辞変数リスト x:の展開は、プログラム内のすべてのステートメントが検査さ
れるまで遅延します。したがって、変数配列 xのサイズは、すべてのステートメ
ントが検査されるまでわかりません。

2 out変数配列には、デフォルトの変数リスト out1-outnがあります。ここで、n
は、変数配列に指定された要素の数です。out の要素数の決定は、配列 xのサイ
ズが判明するまで遅延します(これは、すべてのステートメントが検査されたとき
に発生します)。

3 割り当てステートメントでは、変数 out1は未定義です。したがって、割り当て
ステートメントは、右辺の値の型(DOUBLE)を変数 out1に割り当てます。

前述の DS2プログラムのすべてのステートメントが処理されると、次のことが起こ
ります。テーブル inには 3つの double変数 x1 x2 x3があるとします。
n 接頭辞変数リスト x:は x1 x2 x3に展開されます。
n 変数配列 xのサイズは 3に決定されます。
n 変数配列 outのサイズは 3 (dim(x))に決定されます。
n デフォルトの変数リスト out1-out3は out1 out2 out3に展開されます。
n 変数 out2および out3は、以前に定義されていなかったため、INTEGER型として
定義されています。out1は、割り当てステートメント out1 = 0.0;によって
DOUBLE型として定義されていることに注意してください。

DOUBLE型の変数 out1は INTEGER型の変数配列 outと互換性がないため、DS2コ
ンパイラはコンパイルエラーを発行します。
エラー状態を取り除く 1つの方法は、配列の割り当て out1 = 0.0;を out[1]=0.0;に変更
することです。この変更により、out配列参照を使用して out1データ値が更新され、

割り当てステートメントが DOUBLE型を変数 out1に割り当てないようになりま
す。

変数配列を関数の引数として指定する
OF演算子を使用すると、変数配列を一部の関数の引数として指定できます。OF演
算子は、変数配列を変数名のカンマ区切りリストに変換し、これが関数の引数にな
ります。

注: OF演算子の使用にはルールと制限があります。詳細については、“OF演算子” 
(91ページ)を参照してください。

変数配列を引数として SUM関数にサブミットする例を次に示します。

proc ds2;
data _null_;
   vararray double a[4];
   method init();
      a:=(1,2,3,4);
   end;

   method run();
      dcl double x y z;
         y=99.0;
        z=100.0;
        x=sum(z, of a[*], y);
        put x=;

      x= sum(of a:, z);
      put x=;
   end;
enddata;
run;
quit;

例のコード 11.1 変数配列の使用例のログ出力

x=209
x=110

一時配列
一時配列の概要

一時配列の要素は、PDVに配置されていないため、どの結果テーブルにも表示され
ないという点で一時的です。一時データ要素の値は、次の反復の開始時に欠損にリ
セットされるのではなく、反復間で自動的に保持されます。一時配列は、DS2プロ
グラムの期間中のみ存在します。
DECLAREステートメントを使用して、一時配列を指定します。DS2一時配列には、
固定範囲または動的範囲を設定できます。標準配列には固定範囲があります。標準
一時配列の範囲はコンパイル時に設定され、実行時に変更することはできません。
標準配列の要素数も実行時に固定されます。動的範囲を持つ配列の要素数は実行時
に設定されるため、ランタイム条件に応じて配列の要素数を調整できます。
いずれかのタイプの配列の DECLAREステートメントで HAVING句を使用して、ラ
ベル、出力形式、および入力形式属性をいずれかのタイプの一時配列に関連付ける
ことができます。

標準一時配列宣言
標準一時配列宣言は、変数配列宣言に似ています。データ型と名前に加えて、配列
範囲の数とサイズを指定します。複数の範囲(または次元)は、カンマ区切り文字を使
用して指定されます。
PDVの外部に 3つの一時的な double値を保存する 3要素の一時配列を指定する
DECLAREステートメントの例を次に示します。
declare double a[3];

符号付き整数ペアの形式は、配列の各次元の下限と上限[l:h]を指定します。ここで、
lは指定された範囲の最小インデックスを表し、hは指定された範囲の最大インデッ
クスを表します。下限の指定 lはオプションです。次元の下限が指定されていない
場合、下限はデフォルトで 1になります。
上限 hが下限より小さい場合、エラーが返されます。1つの整数のみで配列範囲を
指定すると、その整数が上限として解釈されます。デフォルトの最下限は 1です。
配列の上限は、以前に宣言された配列の次元の要素数に基づいてサイズを変更する
こともできます。上限には DIM関数呼び出しを使用します。DIM関数は、配列の上
限を指定するために使用できる唯一の関数です。DIM関数は、次元の下限を指定す
るためには使用できません。

動的一時配列宣言
DECLAREステートメントで動的範囲を持つ一時配列を宣言するには、配列の名前と
データ型を指定し、配列範囲としてアスタリスク([ * ])を指定します。この例では、
CHAR(n)型の動的一時配列 dを宣言しています。
declare char(10) d[*];

アスタリスクは、1次元、下限 1、初期上限 0の配列を定義し、その結果、0要素を
含むように初期化された 1次元配列になります。配列の上限は、実行時に
RESIZEARRAYステートメントを使用して変更できます。動的範囲制限ありの一時
配列の次元数と下限は変更できません。動的範囲制限ありの配列は常に、下限が 1
の 1次元配列です。
動的範囲を持つ一時配列の要素を設定する方法の例を次に示します。
RESIZEARRAY d[100];

RESIZEARRAYステートメントは、動的一時配列 dの要素数を 100に設定します。動
的配列 dには、CHAR(10)型の要素が 100個あります。
動的範囲を持つ一時配列の要素数は、整数定数または整数式として指定できます。
たとえば、次の式は有効です。
RESIZEARRAY d[top+100];

この式の使用例については、“動的範囲を持つ一時配列の使用例” (131ページ)を参照
してください。

動的一時配列の操作
RESIZEARRAYステートメントは、配列の末尾に要素を追加および削除します。
RESIZEARRAYステートメントを使用して配列のサイズを複数回変更すると、古い要
素が削除されると配列の値が削除され、新しい要素が追加されると配列の値が初期
化されます。動的範囲制限ありの一時配列を作成し、そのサイズを複数回変更する
DS2プログラムの次のログ出力について考えてみましょう。

1          proc ds2;
2          data _null_;
3            method init();
4              dcl double a[*];
5          
6              put 'Resizing array a from 0 elements to 10 elements';
7              resizearray a[10];
8              put a[*]=;
9          
10             put 'Assigning values to 10 elements of array a';
11             a := (1 2 3 4 5 6 7 8 9 10);
12             put a[*]=;
13         
14             put 'Resizing array a from 10 elements to 5 elements';
15             resizearray a[5];
16             put a[*]=;
17         
18             put 'Resizing array a from 5 elements to 15 elements';
19             resizearray a[15];
20             put a[*]=;
21           end;
22         enddata;
23         run;
Resizing array a from 0 elements to 10 elements
a[1]=. a[2]=. a[3]=. a[4]=. a[5]=. a[6]=. a[7]=. a[8]=. a[9]=. a[10]=. 
Assigning values to 10 elements of array a
a[1]=1 a[2]=2 a[3]=3 a[4]=4 a[5]=5 a[6]=6 a[7]=7 a[8]=8 a[9]=9 a[10]=10
Resizing array a from 10 elements to 5 elements
a[1]=1 a[2]=2 a[3]=3 a[4]=4 a[5]=5  /* 1 */
Resizing array a from 5 elements to 15 elements
a[1]=1 a[2]=2 a[3]=3 a[4]=4 a[5]=5 a[6]=. a[7]=. a[8]=. a[9]=. a[10]=. a[11]=. a[12]=. a[13]=. a[14]=. a[15]=. /* 2  
*/
NOTE: Execution succeeded. No rows affected.
24         quit;

1 RESIZEARRAYステートメントによって要素数が 10から 5配列要素に減らされ
ると、最後の 5つの配列値が削除されます。

2 RESIZEARRAYステートメントによって追加された新しい要素には、新しい値が
割り当てられるまで null値または欠損値が含まれます。

グローバル配列の場合、RESIZEARRAYステートメントは、グローバル配列を宣言し
たパッケージ、スレッド、またはデータブロック内のメソッドから配列の範囲のサ
イズを変更する場合に呼び出すことができます。ローカル配列の場合、
RESIZEARRAYステートメントは、ローカル配列を宣言したメソッドから配列の範囲
のサイズを変更する場合に呼び出すことができます。RESIZEARRAYステートメン
トの詳細については、“RESIZEARRAY ステートメント” (SAS DS2言語リファレンス)
を参照してください。
動的範囲を持つ一時配列の配列割り当ては、変数配列および標準一時配列の場合と
同じです。動的範囲を持つ一時配列は、他の DS2配列を使用できる任意のステート
メント(Array割り当てステートメントを含む)で、配列引数として、および PUTステ
ートメントで使用できます。

動的範囲を持つ一時配列の使用例
スタックは、動的配列のサイズ変更によってメリットが得られるデータ構造のタイ
プの例です。スタックを使用すると、コレクション内の要素の数を増減できますが、
要素を同じ順序に保つことができます。
標準の DS2一時配列を使用してスタックを実装する DS2パッケージは、スタックに
よって格納できる可能性のある最大の要素のコレクションに合わせてサイズ設定さ
れた DS2配列を静的に宣言する必要があります。たとえば、次の実装では、最悪の
ケースの 100,000要素に対して DS2配列のサイズを静的に設定します。実行中に
スタックに最大 10個の要素しか格納されなかったとしても、100,000個の要素を格
納するのに十分なスペースが静的に割り当てられます。

proc ds2;
package stack / overwrite=yes;
  /* Statically size array values for worst case of 100,000 values */
  dcl double values[100000];
  dcl int top;

/* Stack constructor method */
/* Initialize the stack as empty */
  method stack();
    top = 0;                 
  end;

 /* Add a value to the stack */
 /* Insert value to top of stack */
  method push(double value);
    top = top + 1;
    values[top] = value;
  end;

/* Remove and return value from the stack */  
  method pop() returns double;
    dcl double value;
    if (top < 1) then do;
      value = null; 
    end;
    else do;
      value = values[top];
      top = top - 1; 
    end;
    return value;
  end;
endpackage;
run;
quit;

動的範囲を持つ一時配列を使用する DS2スタックパッケージは、メモリをより効率
的に使用できます。次の例では、実行中にスタックに最大 10個の要素しか格納され
ていない場合、100個の要素のスペースのみがスタックによって割り当てられます。

proc ds2;

/* Declare array values without a specified size */
package stack / overwrite=yes;
  dcl double values[*];
  dcl int top;

/* Stack constructor method */
/* Initialize the stack as empty */
   method stack();
    top = 0;
  end;

/* Add a value to the stack. */
  method push(double value);

  /* Use the RESIZEARRAY statement to expand the array as needed.
   * Memory allocation is typically an expensive operation, so blocks 
   * of 100 elements are allocated rather than allocating 1 element
   * at a time. */
   if (dim(values) <= top) then RESIZEARRAY values[top + 100];

/* Insert the value on top of stack */
    top = top + 1;           
    values[top] = value;
  end;

 /* Remove and return value from top of stack */
  method pop() returns double;
    dcl double value;
    if (top < 1) then do;
      value = null;
    end;
    else do;
      value = values[top];
      top = top - 1;
    end;
    return value;
  end;
endpackage;

/* use methods from package stack in a data program */
data _null_;
  method init();
    dcl package stack s();
    dcl double value;

    s.push(1);
    s.push(2);
    s.push(3);

    value = s.pop(); put value=;
    value = s.pop(); put value=;

    s.push(4);
    s.push(5);

    value = s.pop(); put value=;

    value = s.pop(); put value=;
    value = s.pop(); put value=; 
/* array still has 100 elements after the pops */
  end;
enddata;
run;
 quit;

例のコード 11.2 スタックプログラムからの出力

value=3

value=2

value=5

value=4

value=1

一時配列の詳細については、“DECLARE ステートメント” (SAS DS2言語リファレン
ス)を参照してください。

HAVING句を使用した配列の宣言
変数配列または一時配列の宣言ステートメントには、HAVING句を含めることがで
きます。HAVING句は、ラベル、出力形式、および入力形式属性を配列に関連付け
ます。配列が変数配列の場合、HAVING句は、変数配列によって参照される変数に
も関連付けられます。
変数配列によって参照される変数にいつ HAVING句を適用するかについての決定
は、変数参照を含む変数リストがいつ展開されるかによって異なります。通常、
VARARRAYステートメントが処理されると、名前および番号付き範囲変数リストが
展開されます。したがって、HAVING句は、その時点でこれらのリストによって参
照されるすべての変数に適用されます。プログラム内のすべてのステートメントが
検査され、プログラム内のすべての変数が定義された後、名前範囲、名前接頭辞、
型、特殊名の変数リストが展開されます。したがって、これらの変数リストによっ
て参照されるすべての変数の HAVING句は、DS2プログラム内のすべてのステート
メントが検査された後に適用されます。
次のプログラムを考えてみましょう。

proc ds2;
data;
  declare double x1 x2 having format 5.0;  /* 1  */
  vararray double x[3] having format 5.2;  /* 2  */
  declare double x3 having format ROMAN5.; /* 3  */
enddata;
run;
quit;

1 DECLAREステートメントが処理されます。変数 x1と x2が定義されています。
HAVING句 format 5.0は変数 x1および x2に関連付けられています。

変数 x1 x2
出力形式 5.0 5.0

2 VARARRAYステートメントが処理されます。デフォルトの変数リスト x1-x3は
x1 x2 x3に展開されます。HAVING句 format 5.2 は変数 x1 x2 x3に関連付けら
れています。
変数 x1 x2 x3
出力形式 5.2 5.2 5.2

3 DECLAREステートメントが処理されます。変数 x3が定義されています。
HAVING句 format ROMAN5.は変数 x3に関連付けられています。
変数 x1 x2 x3
出力形式 5.2 5.2 ROMAN5.

ここで、変数リストの型が、変数リストの処理が遅延する型に変更された場合に何
が起こるかを考えてみましょう。

proc ds2;
data;
  declare double x1 x2 having format 5.0;    /* 1  */
  vararray double x[*] x: having format 5.2; /* 2 , 4  */
  declare double x3 having format ROMAN5.;   /* 3  */
enddata;
run;
quit;

1 DECLAREステートメントが処理されます。変数 x1と x2が定義されています。
HAVING句 format 5.0は変数 x1および x2に関連付けられています。
変数 x1 x2
出力形式 5.0 5.0

2 VARARRAYステートメントの処理が遅延するか、処理を開始します。接頭辞変数
リスト x:は、プログラム内のすべてのステートメントが検査され、すべての変数
が定義されるまで展開できません。
変数 x1 x2
出力形式 5.0 5.0

3 DECLAREステートメントが処理されます。変数 x3が定義されています。
HAVING句 format ROMAN5.は変数 x3に関連付けられています。
変数 x1 x2 x3
出力形式 5.0 5.0 ROMAN5.

4 VARARRAYステートメントは処理を完了します。接頭辞変数リスト x:は x1 x2 
x3に展開されます。HAVING句 format 5.2は変数 x1 x2 x3に関連付けられてい
ます。
変数 x1 x2 x3
出力形式 5.2 5.2 5.2

配列割り当て
配列割り当ての概要

DS2は、:=演算子による配列割り当てをサポートしています。配列または定数リス
トを割り当てるための構文は次のとおりです。
array:=array;

array:=(constant list);

配列割り当てでは、arrayは一時配列または変数配列のいずれかになります。

別の配列からの配列割り当て
ある配列を別の配列に割り当てる場合、2つの配列のデータ型は互換性がある(同じ
または変換可能)必要があります。次元数と各次元の要素の総数は同じである必要は
ありません。
次のステートメントに示すように、配列 yから配列 xへの割り当てを考えてみまし
ょう。
x:=y;

割り当て中に、配列 yの各要素が配列 xの各要素に割り当てられます。たとえば、
次のようになります。x[1]=y[1];...x[n]=y[n];

x[i] = y[i]を評価するための基本的なアルゴリズムは次のとおりです。最初に y[i]が
調べられ、欠損(SASモード)か null (ANSIモード)かが確認されます。
n y[i]が欠損または nullの場合、x[i]には欠損または nullが割り当てられます。欠
損または nullを配列要素に割り当てる決定は、配列のデータ型と、プログラムが
SASまたは ANSIモードで実行されているかどうかによって異なります。詳細に
ついては、7章, “DS2による nullおよび SAS欠損値の処理方法” (61ページ)を参
照してください。

n y[i]が欠損でも nullでもない場合、x[i]と y[i]の型が調べられます。
o y[i]の型が x[i]の型と異なる場合、y[i]は x[i]の型に変換されます。

n y[i]の変換が成功した場合、変換の結果が x[i]に割り当てられます。
n y[i]の変換が失敗した場合、プログラムが実行されているモードに応じて、
欠損または nullが x[i]に割り当てられます。

o y[i]の型が x[i]の型と同じ場合、y[i]は x[i]に割り当てられます。

配列 xと配列 yの要素数が同じでない場合、配列 yから配列 xにできるだけ多くの
要素が割り当てられ、配列 xの残りの要素には nullまたは欠損が割り当てられます。
配列 xの長さは割り当てによって変更されません。
次の例では、配列 xに 10の要素があり、配列 yに 7つの要素があり、配列 yが配列
xに割り当てられています。したがって、配列 yの 7つの要素は配列 xの最初の 7
つの要素に割り当てられ、配列 xの最後の 3つの要素には欠損が割り当てられます。
proc ds2;
data _null_;
   method init();
      declare double x[10];
      declare double y[7];

      x := (0 0 0 0 0 0 0 0 0 0);
      y := (1 2 3 4 5 6 7);
      x := y;
      put x[*]=;
   end;
enddata;
run;
quit;

次の行が SASログに書き込まれます。

x[1]=1 x[2]=2 x[3]=3 x[4]=4 x[5]=5 x[6]=6 x[7]=7 x[8]=. x[9]=. x[10]=.

Double欠損値の特殊なケース
y[i]が SAS特殊欠損値(.Zなど)を持つ DOUBLEである場合、DS2は次のルールに従
って割り当て中に SAS欠損値を保持しようとします。
n x[i]が DOUBLEの場合、y[i]の SAS欠損値が x[i]に割り当てられます。
n x[i]が文字列の場合、y[i]の欠損文字表現(たとえば、.Zの場合は Z)が x[i]に割り当
てられます。

この特殊なケースの処理は、DS2が SASモードで、y[i]の型が DOUBLEの場合にの
み発生します。
null値と欠損値の詳細については、7章, “DS2による nullおよび SAS欠損値の処理
方法” (61ページ)を参照してください。配列割り当ての例については、“例: 配列” 
(SAS DS2言語リファレンス)を参照してください。

変数配列による配列割り当て
変数配列の要素は、PDV内の変数を参照します。ARRAY割り当てステートメント
x := yでは、yの要素の値が要素ごとに xの要素に割り当てられます。xまたは yが
変数配列(vararray)の場合、割り当ては常に x[i]=y[i]です。

変数配列への配列割り当ては、変数配列内の要素(参照)を変更しません。かわりに、
配列の要素によって参照される変数のデータが変更されます。同様に、変数配列か
らの配列割り当てでは、変数配列の要素によって参照される変数内のデータが割り
当てに使用されます。

定数リストからの配列割り当て
定数リストから配列に割り当てるには、定数リスト内の定数が配列のデータ型と互
換性がある(同じか変換可能である)必要があります。定数リストと配列は、同じ次元
数または各次元の要素数が同じである必要はありません。
この Array割り当てステートメントを想定します。
x := (c1 c2 c3 ... cn);

定数リストから配列 xへの配列割り当て時に、次の展開形式に示すように、定数の
各要素が配列 xの各要素に割り当てられます。
x[1] = c1;
x[2] = c2;
...
x[i] = ci;
...
x[n] = cn;

x[i] = ciを評価するための基本的なアルゴリズムは次のとおりです。最初に ciが調
べられ、欠損か nullかが確認されます。欠損または nullを配列要素に割り当てる決
定は、配列のデータ型と、プログラムが SASと ANSIモードのどちらで実行されて
いるかによって決まります。詳細については、7章, “DS2による nullおよび SAS欠
損値の処理方法” (61ページ)を参照してください。
n ciが欠損または nullの場合、x[i]には欠損(SASモード)または null (ANSIモード)
が割り当てられます。

n ciが欠損でも nullでもない場合、x[i]と ciの型が調べられます。
o ciの型が x[i]の型と異なる場合、ciは x[i]の型に変換されます。

n ciの変換が成功した場合、変換の結果が x[i]に割り当てられます。
n ciの変換が失敗した場合、プログラムが実行されているモードに応じて、

SAS欠損または ANSI nullが x[i]に割り当てられます。
o ciの型が x[i]の型と同じ場合、ciは x[i]に割り当てられます。

定数リストと配列 xの要素数が同じでない場合、定数リストから配列 xにできるだ
け多くの定数が割り当てられ、配列 xの残りの要素には nullまたは欠損値が割り当
てられます。配列 xの長さは割り当てによって変更されません。
次の例では、7つの double要素を持つ配列 xに、5つの定数を持つ定数リストが割
り当てられます。定数リスト内の 5つの定数は、配列 xの最初の 5つの要素に割り
当てられます。プログラムが SASモードで実行されている場合、SAS数値欠損値が
配列 xの最後の 2つの要素に割り当てられます。SAS数値欠損値も、型変換の結果
として要素 4に割り当てられます。定数.Zは SASの特殊数値欠損値であり、その値
は出力に保持されます。
proc ds2;

data _null_;
   method init();
     declare double x[7];
     x := (1 '2' 3.3 '' .Z);
     put x[*]=;
   end;
enddata;
run;
quit;

次の行が SASログに書き込まれます。

x[1]=1 x[2]=2 x[3]=3.3 x[4]=. x[5]=Z x[6]=. x[7]=.

注: 定数リスト内の要素の型は、すべての要素の型が割り当てられた配列の型に変換
可能である限り、異種であってもかまいません。

次に、SASモードで実行されているプログラムの定数リストから配列を割り当てる
別の例を示します。
declare char(2) a[2, 3];
   ...
a := (('aa' 'bb' 'cc')('dd' 'ee' ''));

前述の割り当てステートメントの後の配列 aの要素は、次のようになります。
'aa' 'bb' 'cc'
'dd' 'ee' ''

aは文字配列であるため、空の定数は SAS文字欠損値として出力されます。
同じ要求が ANSIモードでどのように処理されるかについては、“存在しないデータ
の読み取りおよび書き込み時の ANSIモード動作の概要” (65ページ)を参照してく
ださい。

配列引数
配列割り当ての概要

DS2配列は、引数として DS2メソッドに渡すことができます。DS2配列は、常にメ
ソッドへの参照によって渡されます。DS2配列は値によって渡すことはできませ
ん。つまり、配列のコピーを引数としてメソッドに渡すことはできません。DS2配
列引数は、配列パラメーターと一致させるために、配列パラメーターで指定された
ものと同じ型である必要があります。配列引数は、別の型への暗黙的な型変換をサ
ポートしていません。DS2配列は、範囲制限ありの配列パラメーター(a[8]など)また
は範囲制限なしの配列パラメーター(a[*]など)として渡されます。

配列パラメーターの定義
DS2メソッドは、配列パラメーターを使用して定義できます。配列のタイプ(一時ま
たは変数)は、パラメーター定義で指定されます。次の表は、さまざまなタイプのパ
ラメーターを定義するための構文を示しています。

パラメーターのタイ
プ パラメーターの構文 例
スカラーパラメータ
ー

data-type parameter-name double x 

一時配列パラメータ
ー

data-type parameter-name [bounds] double x[5]

変数配列パラメータ
ー

VARARRAYdata-type parameter-name 
[bounds]

vararray double x[2,4]

配列引数のデータ型とタイプは、配列パラメーター定義で指定された型と種類と正
確に一致する必要があります。DS2は、配列引数を別の種類またはデータ型に変換
しません。たとえば、パラメーターが倍精度の一時配列として定義されている場合、
引数は倍精度の一時配列でなければなりません。倍精度の変数配列または整数の一
時配列が、倍精度の一時配列パラメーターの引数として渡されると、エラーが発生
します。
次の DS2プログラムは、配列パラメーターを持つメソッドの定義と、配列引数を使
用したメソッドの呼び出しを示しています。
proc ds2;
data _null_;
   declare  double x i;
   declare  double y[4];
   vararray double z[4];
 
   method m(double u, double v[4], vararray double w[4]);
      do i = 1 to 4;
         v[i] = u;
         w[i] = u; 
      end;
   end;
 
   method init();
      m(x, y, z);      /* call method m */
      put y[*]=;
      put z[*]=;
   end;
 enddata;
run;
quit;

範囲制限ありの配列パラメーター
範囲制限ありの配列パラメーターは、配列引数の要素にアクセスするための明示的
な範囲情報を提供します。次に例を示します。
method m(double a[4]); 

method m(vararray double a[5:10,3:6]); 

範囲制限ありの配列パラメーターは、配列引数の次元に関係なく、要素数が同じ任
意の DS2配列引数と一致します。たとえば、範囲制限ありの配列パラメーター
a[2,4]は、配列 a1、a2、および a3にそれぞれ 8つの要素があるため、配列引数
a1[8]、a2[11:12,11:14]、および a3[2,2,2]と一致します。配列引数の次元が配列パ
ラメーターと異なる場合、配列パラメーターの要素は配列引数の対応する要素にマ
ッピングされます。このマッピングは、行優先順を使用して、配列内の要素の位置
に基づいています。たとえば、配列パラメーター a1[2,1]は 8要素配列の 5番目の要
素にアクセスするため、a1[5]、a2[12,11]、および a3[2,2,1]にマッピングされます。

範囲制限なしの配列パラメーター
範囲制限なしの配列パラメーターは、対応する配列引数の明示的な範囲情報を提供
しません。次に例を示します。
method m(double a[*]);

配列範囲のアスタリスク(*)は、パラメーター aが範囲制限なしの配列パラメーター
であることを指定します。
範囲制限なしの配列パラメーターは、配列引数の要素数や次元に関係なく、任意の
DS2配列引数と一致します。DS2メソッドでは、対応する配列引数が多次元配列で
あっても、配列パラメーターは 1次元配列として扱われます。範囲制限なしの配列
パラメーターは、行優先順を使用して多次元配列にマッピングされます。2x3配列
a[2, 3]を考えてみましょう。

1 2 3
4 5 6

配列 aが範囲制限なしの配列パラメーター bとしてメソッドに渡される場合、bは
6要素の 1次元配列としてアクセスされます。

1 2 3 4 5 6

配列パラメーター bの要素にアクセスすると、配列 aの要素にアクセスすることに
注意してください。これは、配列 aが参照によって DS2メソッドに渡されるためで
す。次に例を示します。
method m(double b[*]);
   b[1] = 10;          /* assigns 10 to a[1, 1] */
   b[2] = 20;          /* assigns 20 to a[1, 2] */
   b[3] = 30;          /* assigns 30 to a[1, 3] */

   b[4] = 40;          /* assigns 40 to a[2, 1] */
   b[5] = 50;          /* assigns 50 to a[2, 2] */
   b[6] = 60;          /* assigns 60 to a[2, 3] */
end;
 
method init();
   declare double a[2, 3];
   m(a);
end;

a[i]の形式の配列式(aは範囲制限なしの配列パラメーター)では、インデックス iの範
囲制限チェックが実行時に実行されます。配列パラメーターのインデックスが配列
引数の範囲制限を超えている場合、DS2はエラーを発行し、配列式は NULLまたは
欠損と評価されます。

注: 範囲制限なしの配列パラメーターは、範囲制限ありの配列パラメーターの引数と
して使用できません。

配列の次元をクエリする方法
次の関数を使用して、配列に関する次元情報を取得できます。各関数の詳細につい
ては、“DS2関数” (SAS DS2言語リファレンス)を参照してください。
DIM(a)
配列 aの最初の次元の要素数を返します

DIM(a, n)
配列 aの次元 nの要素数を返します

LBOUND(a)
配列 aの最初の次元の下限を返します

LBOUND(a, n)
配列 aの次元 nの下限を返します

HBOUND(a)
配列 aの最初の次元の上限を返します

HBOUND(a, n)
配列 aの次元 nの上限を返します

NDIMS(a)
配列 aの次元数を返します

どのクエリ関数でも、配列引数 aは一時配列または変数配列にすることができ、次
元引数 nは整数値に評価される式にする必要があります。次の例は、これらのクエ
リ関数を示しています。
do i = 1 to dim(a1);
   put a1[i];
end;

numelems = 0;
do i = 1 to ndims(a2);
   numelems = numelems + dim(a2, i);
end;
 
do i = lbound(a2, 1) to hbound(a2, 1);
   do j = lbound(a2, 2) to hbound(a2, 2);
      do k = lbound(a2, 3) to hbound(a2, 3);
         sum = sum + a2[i,j,k];
      end;
   end;
end;

配列の次元外の次元値を使用して配列関数が呼び出された場合、ランタイムエラー
が発生し、関数は NULL整数値を返します。

配列の内容を書き込む方法
DS2 PUTステートメントを使用して、配列の個々の要素または配列のすべての要素
を書き込むことができます。
個々の配列要素を書き込む構文は次のとおりです。
PUT array-name[element]<=>;

配列のすべての要素を書き込む構文は次のとおりです。
PUT array-name[*]<=>;

PUTステートメントは、一時配列と変数配列の要素を書き込むことができます。配
列のすべての要素が array-name[*]構文で書き込まれる場合、配列のすべての要素は
同じ出力形式で書き込まれます。つまり、array-name[*]で異なる配列要素に対して異
なる出力形式を指定することはできません。
次の例は、配列の内容の PUTステートメントの出力を示しています。
proc ds2;
data _null_;
   vararray varchar(10) x[10];
   declare  double      y[2,2,2];
   method init();
     x[1] = 'a';
     do i = 2 to dim(x);
       x[i] = x[i-1] || x[1];
     end;
     put 'X:' x[*];

     y := (10 20 30 40 50 60 70 80);
     put 'Y:' y[*]=;
   end;
enddata;
run;
quit;

次の行が SASログに書き込まれます。

X: a aa aaa aaaa aaaaa aaaaaa aaaaaaa aaaaaaaa aaaaaaaaa aaaaaaaaaa
Y: y[1,1,1]=10 y[1,1,2]=20 y[1,2,1]=30 y[1,2,2]=40 y[2,1,1]=50 y[2,1,2]=60 y[2,2,1]=70 y[2,2,2]=80

詳細については、“PUT ステートメント” (SAS DS2言語リファレンス)を参照してくだ
さい。

DS2 varlistの概要
varlistは、異種変数をグループ化するためのデータ構造です。varlistは、DS2ビル
トインデータ型のグローバルなスカラー変数への一連の参照で構成されます。参照
される変数は、次のいずれかにすることはできません。
n ローカル変数
n 配列または別の varlist

n ビルトインまたはユーザー定義のパッケージ型の変数
varlistは、ローカルメソッド、パッケージメソッド、および varlistをパラメーター
として受け取る DS2ビルトイン関数に渡すことができます。パッケージは、ユーザ
ー定義パッケージと DS2事前定義パッケージのいずれかです。
DS2は、名前付き varlistと名前のない varlistをサポートしています。名前付き
varlistは、VARLISTステートメントで作成されます。名前のない varlistは、使用時
に作成されます。それ以外は、機能は同じです。

配列と同様に、varlistは DS2プログラムの実行中だけ存在します。ただし、参照さ
れた変数の内容は 1つ以上の結果セットに保存される場合があります。

varlistと配列の類似点
varlistには、変数配列と同様の機能があります。varlistには次の特徴があります。
n 名前付き varlistは、VARLISTステートメントを使用して作成されます。参照す
る変数の名前は角かっこ( [ ] )で指定します。(変数配列は VARARRAYステートメ
ントで作成され、配列範囲は角かっこ内に指定されます。)

n グローバルスコープで宣言する必要がある
n 要素には添字演算子を使用してアクセスします。たとえば、式 a[2]は、aという
名前の varlistまたは配列の 2番目の要素を指定します。

n ドット表記を使用して(配列構文を使用して)アクセスできます。次に例を示しま
す。
pkgvar.a
pkgvar.a[2]

最初のコードは、aという名前の varlistを参照します。2番目は、varlistaの 2
番目の要素を参照します。詳細については、“DS2パッケージ内のドット表記” 
(166ページ)を参照してください。

n THIS式で使用できます。次に例を示します。
this.a

varlistと配列の違い
n varlistには、異種の型の変数が含まれています。
n varlistは常に 1次元であり、下限は 1です。上限は、VARLISTステートメントの
変数リストまたは[ list-of-variables ]演算で指定された変数名の数をカウントす
ることによって決定されます。配列の下限は柔軟で、配列は多次元にすることが
できます。

n varlistと配列は、データ構造でデータ値を設定および取得する方法が異なりま
す。
o varlist要素のデータ値は、添字演算子と VSETVALUE関数を使用して設定さ
れます。配列要素のデータ値は、割り当てステートメントで添字演算子を使
用して設定できます。たとえば、vaが vararrayで、vlが varlistの場合、次
のステートメントは、変数配列 vaと varlist vlの最初の要素をデータ値 100
に設定します。

va[1] = 100;
VSETVALUE(vl[1], 100);

o varlist要素のデータ値は、VGETVALUE関数で添字演算子を使用して取得する
必要があります。配列要素のデータ値は、添字演算子のみを使用して取得で
きます。たとえば、vaが vararrayで vlが varlistの場合、次のステートメン
トは、vararray vaと varlist vlの最初の要素を取得し、データ値を変数 yに割
り当てます。

y = va[1];
VGETVALUE(vl[1], y);

詳細については、“VSETVALUE 関数” (SAS DS2言語リファレンス)および
“VGETVALUE 関数” (SAS DS2言語リファレンス)を参照してください。

n 変数配列では、添字演算子を使用して取得した要素値を、同じステートメント内
の後続の演算で使用できます。たとえば、次のように、取得した値を算術演算で
使用し、その結果を引数としてメソッド呼び出しに渡すことができます。
compute(va[1] + 100);

varlistでは、添字演算子と VGETVALUE関数を使用して取得した要素値は、同じ
ステートメント内の後続の演算では使用できません。取得した値に対するその
他の演算は、後続のステートメントで実行する必要があります。次に例を示しま
す。
vgetvalue(vl[1], var);
compute(var + 100);

n OF演算子を使用して varlistを指定することはできません。OF演算子で変数の
リストを指定するには、省略変数リスト構文 (29ページ)を使用する必要がありま
す。.

例: 名前付き varlistの作成
名前付き varlistを作成する例を次に示します。varlistは変数情報関数 VNAMEお
よび VTYPEに引数として渡され、varlist内の変数に関する nameと typeのメタデー
タを取得します。

proc ds2;
data _null_;
   vararray double x[5];
   dcl bigint xyz;
   dcl date xanadu;

   varlist t1 [x:];                       /* 1 */

   method printNames(varlist v);          /* 2 */
      dcl varchar(100) name type;
      dcl bigint i;
      do i = 1 to dim(v);                 /* 3 */
          name = vname(v[i]);

          type = vtype(v[i]);
          put name= type=;
      end;
   end;

   method init();
      printNames(t1);                   /* 4 */
   end;
enddata; 
run;
quit;

1 VARLISTステートメントは、varlist t1の作成を指定し、名前接頭辞 x:を使用して
使用可能なグローバル変数から選択するように指定します。

2 varlistを入力として受け取る printNamesメソッドが定義されています。
varlistは、メソッドの本体で vと呼ばれます。このメソッドは、ローカル変数
name、type、および iも定義します。

3 メソッドの次に、DOステートメントを DIM関数とともに使用して、varlist vの
変数を反復処理します。変数ごとに、VNAME関数と VTYPE関数を使用してメタ
データが取得され、ローカル変数 nameと typeに割り当てられます。PUTステ
ートメントは、変数 nameと typeの値を SASログに出力します。

4 varlist t1は、INIT()メソッドのメソッド引数として printNamesメソッドに渡さ
れます。

SASログに書き込まれる結果は次のとおりです。

name=x1 type=double
name=x2 type=double
name=x3 type=double
name=x4 type=double
name=x5 type=double
name=xyz type=bigint
name=xanadu type=date

詳細については、“VARLIST ステートメント” (SAS DS2言語リファレンス)を参照して
ください。

名前のない varlist
名前のない varlistは、VARLISTステートメントで varlistを明示的に宣言せずに、関
数呼び出しまたはメソッド呼び出しの引数として渡されます。名前のない varlist
は、[list-of-variable-names] 演算子を用いて、使用時に作成されます。
名前のない varlistを DS2プログラムで使用する方法の例を次に示します。

proc ds2;
data _null_;
  dcl double a b x y z;

  method init();

    dcl package hash h([a b], [x y z]);
  end;
enddata;
run; 
quit;

このプログラムは、コンストラクターメソッドで 2つの varlistをハッシュパッケー
ジに指定します。最初の varlistには、変数 aと bが含まれます。2番目の varlistに
は、変数 x、y、zが含まれます。
前述のプログラムは、次のプログラムと同じ機能を持っています。

proc ds2;
data _null_;
  dcl double a b x y z;
  varlist keyvars [a b];
  varlist datavars [x y z];

  method init();
    dcl package hash h(keyvars, datavars);
  end;
enddata;
run;
quit;

このプログラムは、VARLISTステートメントを使用して 2つの varlistを定義しま
す。変数 aと bを含むグループ化には、keyvarsという名前が割り当てられていま
す。変数 x、y、zを含むグループ化には、datavarsという名前が割り当てられます。
どちらのプログラムも同じ varlistを作成します。ハッシュパッケージは、両方のプ
ログラムでまったく同じデータ構造を受け取ります。違いは、2番目のプログラム
が varlistを変数として宣言し、varlist変数をハッシュパッケージのコンストラクタ
ーメソッドに渡すことです。最初のプログラムでは、varlist変数が宣言されていま
せん。
名前のない変数リスト構文が式で使用されると、名前のない varlistの変数を参照す
る名前のないデータ構造がグローバルスコープで作成されます。名前のない varlist
は、名前のない varlistが指定されたステートメント以外のステートメントからはア
クセスできません。したがって、他の式で再利用することはできません。
名前のない varlistをパラメーターとして受け取る DS2事前定義パッケージの他の
メソッドの例は、事前定義 DS2ハッシュパッケージの KEYSメソッドと、DS2事前
定義 SQLSTMTパッケージの BINDPARAMETERSメソッドです。詳細については、
“KEYS メソッド” (SAS DS2言語リファレンス)および“BINDPARAMETERS メソッド” 
(SAS DS2言語リファレンス)を参照してください。

例: varlistのデータ値の設定
varlistの要素の値を設定するには、添字演算子を VSETVALUE関数とともに使用し
ます。

VSETVALUE関数を使用し、名前のない varlist内の要素のデータ値を設定する“例: 名
前付き varlistの作成”の修正バージョンを次に示します。

proc ds2;
data _null_;
  vararray double x[5]; 
  dcl bigint xyz;
  dcl date xanadu;

  method addValues(varlist v);           /* 1 */
    dcl varchar(100) type;
    dcl integer i;

    do i = 1 to dim(v);                
      type = vtype(v[i]);

      select(type);                    
        when('double') vsetvalue(v[i], 100 * i);
        when('bigint') vsetvalue(v[i], 100000000000000);
        when('date')   vsetvalue(v[i], date'2023-03-27');
      end;

    end;
  end;

  method printValues(varlist v);         /* 2 */
    dcl varchar(100) name type value;
    dcl integer i;

    do i = 1 to dim(v);
      name = vname(v[i]);
      type = vtype(v[i]);
      vgetvalue(v[i], value);         

      put 'printValues ' name= type= value=;
    end;
  end;

  method init();            /* 3 */
    addValues([x:]);
    printValues([x:]);
  end;

enddata;
run;
quit;

1 varlistを入力として受け取る addValueメソッドが定義されています。
addValueメソッドは、次のタスクを実行します。
n ローカル変数の typeと iを定義します。
n varlistの変数を反復処理して、各変数の型メタデータを取得します。
n 変数ごとに取得した型の値を使用して、変数の型に従って各変数にデータ値
を割り当てます。VSETVALUE関数は、変数のデータ値を設定します。

2 varlistを入力として受け取る printValuesメソッドが定義されています。この
メソッドは、ローカル変数 name、type、value、iを定義し、varlistを反復処理し
て、VGETNAME、VGETTYPE、および VGETVALUE関数を使用して変数の値を取
得します。ローカル変数は PUTステートメントに渡されます。

3 INIT()メソッドでは、名前接頭辞の省略形 x:が[list-of-variable-names]演算子で指
定され、メソッド引数として addValuesメソッドと printValuesメソッドに渡さ
れます。[x:]演算子は、名前が文字 xで始まるすべてのグローバル変数を指定す
る名前のない varlistを作成し、addValuesメソッドと printValuesメソッドで処
理します。

ログに書き込まれる結果は次のとおりです。

printValues  name=x1 type=double value=100
printValues  name=x2 type=double value=200
printValues  name=x3 type=double value=300
printValues  name=x4 type=double value=400
printValues  name=x5 type=double value=500
printValues  name=xyz type=bigint value=100000000000000
printValues  name=xanadu type=date value=2023-03-27

例: varlistのデータ値の取得
varlistのデータ値の取得の概要

varlist内の要素のデータ値を取得するには、添字演算子を VGETVALUE関数ととも
に使用します。
このセクションでは、添字演算子と VGETVALUE関数の使用例を 2つ示します。1
つは名前付き varlistを使用し、もう 1つは名前のない varlistを使用します。
プログラム例では次のことを行います。
n 入力データセットを読み取る
n VGETVALUE関数を使用して、入力データの各行の変数のデータ値を取得する
n データ値を処理して数値欠損値を発見および分析する
n 2つの列を追加して出力データセットを作成する
出力データセットの追加の新しい列は次のとおりです。
n sum: 現在の行の非欠損数値の合計が含まれる
n mean: 現在の行の非欠損数値の平均が含まれる

入力データ
プログラムは、indataという名前の入力データセットを使用します。入力データセ
ットは次のとおりです。

Obs     v1     v2     v3     v4     v5

 1     100    100    100    100    100
 2       1      2      3      4      5
 3      18      4      9     16     24

どちらのプログラムも、次の入力データセットを変更せずに処理できます。

       first_    last_
Obs     name     name        score1    score2    score3    score4    score5    score6

 1     Mary      Brown         97        95        92        88        99        76  
 2     James     Smith         88        76        89        90        73        92  
 3     Linda     Johnson       77        73        68        95        82        78  
 4     Robert    Williams      56        99        82        76        90        88 

名前付き varlistの例
このプログラムは、名前付き varlistを使用して、入力テーブルの値を取得します。

proc ds2;
data outdata / overwrite=yes;

    varlist v [ _all_ ];                  /* 1 */
   
    dcl double sum mean;
   
    method run();
        dcl double value nvalues;
        dcl int i rc;

        set indata;
      
        sum = 0;                          /* 2 */
        nvalues = 0;
        do i = 1 to dim(v);
          if (vname(v[i]) ne 'sum' and vname(v[i]) ne 'mean') then do;
              rc = vgetvalue(v[i], value);
              if (rc = 0 and not missing(value)) then do;
                  nvalues = nvalues + 1;
                  sum = sum + value;

              end;
            end;
        end;

            if (0 < nvalues) then do;      /* 3 */
            mean = sum / nvalues;
        end;
    end;
enddata;
run; quit;

1 VARLISTステートメントは、このデータブロックのすべてのグローバル変数を参
照する vという名前の varlistを作成します。_all_変数リストの指定には、SETス
テートメントで指定されたデータセットから読み取られる変数と、このデータブ
ロックのグローバルスコープで宣言された sum変数と mean変数が含まれます。

2 これらのステートメントは、VGETVALUE関数を使用して各変数の値を取得し、
現在の入力行からすべての非欠損数値の合計を計算します。

3 これらのステートメントは、現在の入力行からすべての非欠損数値の平均を計算
します。

名前のない varlistの例
名前のない varlistは、varlistパラメーターが必要なメソッド呼び出し式への引数と
してのみ有効です。この例では、名前付き varlistの例の DATAプログラムを変更し
て、varlistパラメーターが必要なメソッドを持つようにします。

proc ds2;
data outdata / overwrite=yes;
   
    dcl double sum mean;

    method compute(varlist v);
       dcl double value nvalues;
       dcl int i rc;

    /*
     * Compute the same of all the non-missing numeric values
     * from the current input row.
     */
        sum = 0;
        nvalues = 0;
        do i = 1 to dim(v);
            if (vname(v[i]) ne 'sum' and vname(v[i]) ne 'mean') then do;
                rc = vgetvalue(v[i], value);
                if (rc = 0 and not missing(value)) then do;
                    nvalues = nvalues + 1;
                    sum = sum + value;
                end;
            end;
        end;

        /* 
         * Compute the mean of all the non-missing numeric values from
         * the current input row.
         */
        if (0 < nvalues) then do;
            mean = sum / nvalues;
        end;
    end;
    
    method run();
       set indata;
    /* Create a varlist that references all global
     * variables of this data block. The _all_ variable list
     * specification includes the variables input from the data set
     * specified on the SET statement as well as the sum and mean
     * variables declared in the global scope of this data block.
     */
       compute([_all_]);
    end;
enddata;
run;
quit;

proc print data=outdata; run;

OutDataデータセット
最初の indataデータセットを処理した後に両方のプログラムが返す結果は次のとお
りです。

2番目の indataデータセットを処理した後に両方のプログラムが返す結果は次のと
おりです。

演算子の優先順位
DS2プログラムでの添字演算子とと[list-of-variable-names]演算子の優先順位につ
いては、“複合式での演算子の優先順位” (102ページ)を参照してください。

DS2パッケージの概要
DS2パッケージは、DS2プログラムで使用できるメソッドと変数のコレクションで
す。DS2パッケージは一連の関連タスクをサポートし、再利用できるように設計さ
れています。
パッケージには、次の 2種類があります。
ユーザー定義パッケージ
これらは、任意の目的で属性にデータを格納したり、メソッドにコードを格納し
たりするために使用できるパッケージです。
詳細については、“ユーザー定義パッケージ” (177ページ)を参照してください。

事前定義パッケージ
これらは、DS2で事前定義されているパッケージです。
詳細については、“事前定義された DS2パッケージ” (180ページ)を参照してくだ
さい。

パッケージは、DS2プログラムの別のパッケージ、スレッド、またはデータブロッ
クのコードで使用できます。これらのプログラミング要素については、“ブロックス
テートメント” (SAS DS2言語リファレンス)を参照してください。

ヒント DS2パッケージメソッドを FedSQL SELECTまたは FedSQL CALL
ステートメントの関数として呼び出すこともできます。詳細については、“式
での DS2パッケージの使用” (SAS FedSQL言語リファレンス)および“パッケー
ジメソッド式” (95ページ)を参照してください。

注: DS2スレッドによって作成されたパッケージインスタンスは、他の DS2スレッ
ドからはアクセスできず、パッケージインスタンスの属性データはスレッド間で共
有されません。スレッドブロックに、パッケージをインスタンス化するグローバル
DECLAREステートメントがある場合、DS2プログラムによって生成された各 DS2
スレッドは、独自のパッケージインスタンスをインスタンス化します。たとえば、
DS2プログラムが 10個の DS2スレッドを生成する場合、メモリには 10個のパッケ
ージインスタンスがありますが、各スレッドはインスタンス化した 1つのパッケー
ジインスタンスにしかアクセスできません。

データプログラムでのパッケージの定義
と使用

パッケージは、パッケージのインスタンスを作成するために使用されるテンプレー
トです。パッケージは、パッケージのすべてのインスタンスに共通する一連の属性
とメソッドを定義します。パッケージを定義したら、パッケージ定義を使用して、
パッケージのインスタンスを必要な数だけ作成できます。各パッケージインスタン
スには、パッケージによって定義された属性の独自のコピーがあります。パッケー
ジはオブジェクト指向プログラミング言語におけるクラスに似ており、パッケージ
インスタンスはオブジェクトに似ています。パッケージは、“PACKAGE ステートメ
ント” (SAS DS2言語リファレンス)を使用して作成されます。
パッケージの属性にアクセスしたりメソッドを呼び出したりする前に、パッケージ
のインスタンスを構築し、新しく構築されたパッケージインスタンスを参照する変
数を作成する必要があります。パッケージ変数は、“DECLARE PACKAGE ステートメ
ント” (SAS DS2言語リファレンス)を使用して作成されます。
パッケージインスタンスの属性とメソッドには、ドット演算子を使用して、パッケ
ージの外部にあるコードからアクセスします。詳細については、“DS2パッケージ内
のドット表記” (166ページ)を参照してください。パッケージインスタンスは、
DECLARE PACKAGEステートメントまたは_NEW_演算子のいずれかを使用して作成
されます。
パッケージの作成と使用に関する基本情報については、"入門ガイド"、"例による学
習: サンプルプログラムの使用"を参照してください。

メソッドでのパッケージの使用
メソッドでパッケージを使用するには、いくつかの方法があります。このセクショ
ンでは、パッケージインスタンスをメソッドから返す方法、またはメソッドの引数
として使用する方法について説明します。メソッドはパッケージ変数を返すことは
できません。

注: このセクションの例では、DS2プロシジャを使用して DS2プログラムを SAS 
Compute Serverにサブミットします。

メソッドからパッケージインスタンスを返す
RETURNステートメントを使用して、メソッドからパッケージインスタンスを返す
ことができます。次に例を示します。
proc ds2;
package mypkg/overwrite=yes;
   method m(double x) returns double;
      return x+99;
   end;
endpackage;

data _null_;
   dcl package mypkg p; 
   dcl double x;

   method r() returns package mypkg;
      return _new_ mypkg();
   end;

   method init();
      p = r();
      x = p.m(100);
      put x=;
   end;

   method term();
      x = p.m(200);
      put x=;
    end;

enddata;
run;
quit;

この例では、pという名前のパッケージ mypkgのパッケージ変数がデータブロック
で宣言されています。ただし、DECLARE PACKAGEステートメントではコンストラ
クターを呼び出すためのかっこが指定されていないため、パッケージはまだインス
タンス化されていません。メソッド rは、_NEW_演算子を使用してパッケージ
mypkgのインスタンスを作成し、そのインスタンスを呼び出しプロセスに返します。
INITメソッドはメソッド rを呼び出し、インスタンスをパッケージ変数 pに割り当
てます。次に、INIT()メソッドはインスタンスの mメソッドを使用して、変数 xに
値を割り当てます。パッケージ変数 pはデータブロックのグローバル部分で宣言さ
れているため、グローバルスコープを持ち、ここで TERMメソッドに示されている
ように、他のメソッドで再び使用できます。
パッケージ変数が INIT()メソッドに対してローカルに宣言されていた場合、TERMメ
ソッドでは使用できませんでした。
DELETEメソッドは、パッケージ変数によって参照されるパッケージインスタンスを
破棄し、そのパッケージ変数と、破棄されたパッケージインスタンスを参照する他
のすべてのパッケージ変数に NULLを割り当てます。次に例を示します。

method init();
   p = r();
   x = p.m(100);
   put x=;
   p.delete();
end;

p.delete()が実行されると、pによって参照されるパッケージインスタンスが破棄さ
れ、pに NULL値が割り当てられます。
この効果は、pが INITメソッドに対してローカルであることを宣言することによっ
て、より簡単に実現できます。

パッケージを引数としてメソッドに渡す
DS2では、メソッドからパッケージインスタンスを返すだけでなく、パッケージイ
ンスタンスをメソッドの引数としてメソッドに渡すことができます。
この例では、グローバル変数 p2によって参照されるパッケージインスタンスがメソ
ッド tpに渡され、ローカル変数 p2によって参照されます。メソッド tpの実行中、
グローバル変数 p2とローカル変数 p2の両方が同じパッケージインスタンスを参照
します。メソッド tpでは、新しいパッケージインスタンスが作成され、ローカル変
数 p2によって参照されるパッケージインスタンス内のパッケージ変数 pに割り当
てられます。
proc ds2;
package mypkg/overwrite=yes;
   method m(double x) returns double;
      return x+99;
    end;
  endpackage;

 package mypkg2/overwrite=yes;
   dcl package mypkg p;
 endpackage;

data _null_;
   dcl package mypkg2 p2();
   dcl double x; 

   method tp(package mypkg2 p2);
       p2.p=_new_ mypkg();
   end;

   method init();
      dcl package mypkg p;
      tp(p2);
      p=p2.p;
      x = p.m(100);
      put x=;
    end;
enddata;
run;
quit;

メソッドでのパッケージの使用 161



パッケージ、スコープ、および有効期間
パッケージ変数のスコープとパッケージインスタン
スの有効期間

パッケージ変数は、パッケージインスタンスへの 1つ以上の参照をメモリに格納し
ます。DECLARE PACKAGEステートメントは、パッケージ変数を作成し、オプショ
ンでパッケージインスタンスを構築します。DECLARE PACKAGEステートメントが
パッケージインスタンスを構築する場合、パッケージインスタンスはパッケージ変
数に割り当てられ、パッケージ変数を使用してインスタンスのデータメンバーにア
クセスできます。DECLARE PACKAGEステートメントがパッケージインスタンスを
構築しない場合、パッケージ変数は nullに初期化されます(何も参照しません)。null
を参照するパッケージ変数を使用してデータメンバーにアクセスしたり、パッケー
ジメソッドを呼び出したりしようとすると、致命的なエラーが発生します。割り当
てステートメントで_NEW_演算子を使用して、パッケージインスタンスを既存のパ
ッケージ変数に割り当てることができます。_NEW_演算子は、新しいパッケージイ
ンスタンスを構築します。
他の変数と同様に、プログラムの実行中にさまざまな値をパッケージインスタンス
に割り当てることができます。
パッケージインスタンスの有効期間は、インスタンスが作成されるスコープによっ
て異なります。インスタンスが作成されたスコープを実行が終了すると、パッケー
ジインスタンスは自動的に削除されます。たとえば、メソッドで作成されたパッケ
ージ変数は、メソッドのローカルスコープで作成されます。その結果、スカラー変
数と配列変数はメソッドに対してローカルとなり、メソッドが完了すると削除され
ます。
複数のパッケージ変数が単一のパッケージインスタンスを参照できます。
パッケージインスタンスは、DS2パッケージ変数がそれを参照している限り、メモ
リ内に残ります。パッケージインスタンスの有効期間は、インスタンス自体が作成
されたスコープを超えて延長できます。たとえば、ローカルスコープのパッケージ
変数を使用してメソッド内でインスタンス化されたパッケージインスタンスの有効
期間は、グローバルスコープのパッケージ変数にも割り当てられている場合に延長
されます。メソッドの実行が終了すると、ローカルパッケージ変数はスコープ外に
渡されて破棄されますが、グローバルパッケージ変数は引き続きメソッドで作成さ
れたインスタンスを参照します。パッケージインスタンスを参照するパッケージ変
数がなくなると、パッケージインスタンスは自動的に破棄されます。

例: パッケージインスタンスの有効期間
この例は、パッケージインスタンスがどのように作成および破棄されるかを示して
います。

proc ds2; 
 package mypkg/overwrite=yes; /* 1 */
    dcl varchar(50) name;
  
    /* constructor - automatically invoked when a package instance is
     * created. */
    method mypkg(varchar(50) name);
      this.name = name;
      put 'creating' this.name;
    end;
  
    /* destructor - automatically invoked when a package instance is
     * destroyed. */
    method delete();
      put 'destroying' this.name;
    end;
endpackage;
  
 data _null_;
   dcl package mypkg var1('pkga'); /* 2 */
  
   method init();
     dcl package mypkg var2('pkgb'); /* 3 */
     dcl package mypkg var3('pkgc'); 
  
     var1 = var2;                    /* 4 */
   end;                              /* 5 */
  
   method term();
     var1 = _new_ mypkg('pkgd');     /* 6 */
   end;                                 
  
 enddata;                            /* 7 */
run;
quit;

1 この例では、mypkgという名前のパッケージを作成します。パッケージ mypkg
は、パッケージ属性 name、mypkgという名前のカスタムコンストラクターメソ
ッド、および delete()という名前のカスタムコンストラクターメソッドを定義し
ます。mypkgパッケージは、デフォルトのコンストラクターメソッドが実行され
た後に実行され、コンストラクターパラメーターで指定された値を name属性に
割り当ててから、name属性の値を"creating"という単語を前に付けてログに出力
します。deleteメソッドは、name属性の値を SASログに出力し、自動デストラ
クターメソッドを呼び出す前に、"destroying"という単語を前に付けます。

2 4つのパッケージ変数が定義され、インスタンス化されます。最初に、パッケー
ジインスタンス pkgaが作成され、パッケージ変数 var1によって参照されます。

3 INITメソッドでローカル変数を作成した後の結果は次のとおりです。
n パッケージインスタンス pkgbが作成され、パッケージ変数 var2によって参
照されます。

n パッケージインスタンス pkgcが作成され、パッケージ変数 var3によって参
照されます。

4 パッケージ変数 var2がパッケージ変数 var1に割り当てられると、結果は次のよ
うになります。
n パッケージインスタンス pkgaはパッケージ変数によって参照されていない
ため、破棄されます。

n パッケージインスタンス pkgbは、パッケージ変数 var1およびパッケージ変
数 var2によって参照されます。

n パッケージインスタンス pkgcは、パッケージ変数 var3によって参照されま
す。

5 INITメソッドが戻ると、メソッドのローカルパッケージ変数(var2および var3)が
削除されます。
n パッケージインスタンス pkgbは、パッケージ変数 var1によって参照されま
す。

n パッケージインスタンス pkgcはパッケージ変数によって参照されていない
ため、破棄されます。

6 新しいパッケージインスタンス pkgdがパッケージ変数 var1に割り当てられる
と、結果は次のようになります。
n パッケージインスタンス pkgbはパッケージ変数によって参照されていない
ため、破棄されます。

n パッケージインスタンス pkgdは、パッケージ変数 var1によって参照されま
す。

7 プログラムが終了すると、プログラムのグローバル変数(var1)が削除されます。
n パッケージインスタンス pkgdはパッケージ変数によって参照されていない
ため、破棄されます。

次の行が SASログに書き込まれます。

creating pkga
creating pkgb
creating pkgc
destroying pkga
destroying pkgc
creating pkgd
destroying pkgb
destroying pkgd

属性とメソッド
PRIVATEアクセス修飾子は、パッケージ内での内部使用を目的とした属性またはメ
ソッドに使用できます。これにより、ユーザーが直接取得または設定することを意

図した属性のみを公開することで、プログラムの複雑さを管理できます。プライベ
ート属性は、メソッドを呼び出した結果の状態情報を保存するのに役立ちます。ユ
ーザーが直接触れた場合、それ以降の結果が無効になる可能性があります。
nextNumberメソッドで属性min、max、または sumを直接変更しないプライベー
ト属性を使用する例を次に示します。
proc ds2;
package stats / overwrite=yes; 
   /*-- put scratch space in private attributes --*/
   dcl private double min max sum ini;
   /*-- put shared logic in a private method --*/
   private method p_update( double v );  
      if missing(v) then return; 
        if missing(ini) then do;
           min=v;
            max=v;
            sum=v;
            ini = 1;
        end;
      else do;
         if v < min then min = v; 
         if v > max then max = v; 
         sum = sum + v; 
        end;
      return;
   end;

   method nextNumber( double v ); 
        put 'in the double method';
      p_update( v );
   end;

   method nextNumber( char(20) c );
      dcl double v;
        v = c;
        put 'in the char method';
      p_update( v );
   end;

   method nextNumber( int i );
      dcl double v;
        v = i;
        put 'in the int method';
      p_update( v );
   end;

   method getStats( in_out double min, in_out double max, in_out double sum ); 
      min = this.min; 
      max = this.max; 
      sum = this.sum;
   end;
endpackage;

data;
dcl package stats st();
dcl double min max sum;
method run();

   st.getStats( min, max, sum );
   output;   
   st.nextNumber( 5 );
   st.nextNumber( '4.0' );
   st.nextNumber( 2.0 );
   st.getStats( min, max, sum );
   output;
end;
enddata;
run;
quit;

パッケージのコンストラクターとデストラクター
コンストラクターとデストラクターは、パッケージインスタンスの構築と破棄の際
に使用される特別なパッケージメソッドです。コンストラクターメソッドは、新し
く構築されたパッケージインスタンスを初期化します。デストラクターメソッド
は、クリーンアップを実行するか、パッケージインスタンスが破棄される前にパッ
ケージインスタンスによって保持されているリソースを解放するか、またはその両
方を行います。パッケージのコンストラクターメソッドはクラスと同じ名前を持
ち、パッケージのデストラクターは DELETEメソッドです。コンストラクターとデ
ストラクターには戻り値の型がなく、値を返しません。
DS2は、パッケージのインスタンスが構築されるときに、パッケージのコンストラ
クターを自動的に呼び出します。パッケージインスタンスがパッケージ変数によっ
て参照されなくなり、破棄されると、DS2は自動的にパッケージのデストラクター
を呼び出します。DECLARE PACKAGEステートメントでパッケージ変数を作成して
も、DS2はパッケージのコンストラクターを呼び出さないことに注意してください。
パッケージのコンストラクターは、パッケージインスタンスが、コンストラクター
引数を指定した DECLARE PACKAGEステートメントまたは_NEW_演算子を使用し
て構築された場合にのみ呼び出されます。
詳細については、このドキュメントの言語リファレンスセクションで、該当する
DECLARE PACKAGEステートメントと_NEW_演算子を参照してください。

DS2パッケージ内のドット表記
ドット表記のサポートと使用

ドット演算子を使用してパッケージの外部にあるコードから、パッケージインスタ
ンスの属性にアクセスし、そのメソッドを呼び出します。パッケージの属性には、
パッケージのグローバルスカラー変数と配列変数が含まれます。

パッケージの属性には、パッケージインスタンスを参照するパッケージ変数を使用
して、パッケージの外部にあるコードからアクセスします。表 13.1で説明されてい
る構文を参照してください。
表 13.1 パッケージの属性にアクセスするための基本的な構文

属性 構文
スカラー変数 package-variable.scalar-attribute

配列変数 package-variable.array-attribute

配列変数の要素 package-variable.array-attribute\[index\]
1

1 構文規則の角かっこは、オプションの引数を示します。角かっこの前のエスケープ文字(\)は、構文に角かっこが必要であることを示します。配列の境界は、角かっこ([ ])で囲む必要があります。

前述の例では、式のドット演算子は、package-variableによって参照されるパッケ
ージインスタンスから指定された attributeにアクセスします。
パッケージのメソッドは、次の構文を使用して、パッケージの外部にあるコードに
よって呼び出されます。
package-variable.method(arguments)

この場合、ドット演算子を含む式は、package-variableによって参照されるパッケ
ージインスタンスから指定されたmethodを呼び出します。package-variableによ
って参照されるパッケージインスタンスは、暗黙的な THIS引数をメソッド呼び出し
に提供しますが、argumentsはメソッド呼び出しに明示的な引数を提供します。
ネストされたドット表記を使用して、パッケージの階層内でネストされているパッ
ケージの属性とメソッドにアクセスできます。パッケージ階層内でネストされたパ
ッケージの属性またはメソッドにアクセスするには、次の形式でドット表記をネス
トした式を使用します。
package-variable1.package-variable2 ... .package-variableN.[attribute|method(arguments)]

ドット演算子の右側のオペランドがパッケージ変数である属性であると、ドット演
算の結果を後続のドット演算の左側のオペランドとして使用できます。ドット演算
を連結すると、パッケージの階層内でネストされたパッケージの属性へのアクセス
がサポートされます。
ネストされたドット演算では:

n 最初のオペランドはパッケージ変数である必要があり、最後のオペランドは、ネ
ストされたドット演算のチェーンによって参照されるパッケージインスタンス
の属性やメソッドである必要があります。

n 間にあるドット演算の結果は、後続のドット演算の左側のオペランドとして使用
されます。そのため、間にある各ドット演算の右側のオペランドはパッケージ変
数、つまり左側の演算子によって参照されるパッケージインスタンスの属性であ
る必要があります。

ネストされたドット表記を使用する次の簡単なプログラム例について考えてみまし
ょう。

proc ds2;

/* pkga defines a method that assigns a value to package attribute x */
  package pkga / inline;
   dcl varchar(1024) x;
   method pkga();
    x = 'apple';
   end;
  endpackage;
  
/* pkga is instantiated as an attribute of pkgb as pa */ 
  package pkgb / inline;
   dcl package pkga pa();
  endpackage;
   
/* pkgb is instantiated as an attribute of pkgc as pb */
  package pkgc / inline;
   dcl package pkgb pb();
  endpackage;
   
/* pkgc is instantiated in the INIT method as pc. 
Then, x is called as an attribute of the package that is referenced 
by package variable pc.pb.pa. */
  data _null_;
   method init();
    dcl package pkgc pc();
    put pc.pb.pa.x;
   end;
  enddata;
  run; 
  quit; 

pc.pb.pa.xは、3つのドット演算で構成される複合式です。
表 13.2 複合式でのドット演算の説明

処理順序 ドット演算 1 説明
1 pc.pb pbはパッケージ変数であり、pcの属性

です。
2 pc.pb.pa paはパッケージ変数であり、pc.pbの

属性です。
3 pc.pb.pa.x xは、pc.pb.paの属性です。
1 ドット式の網掛け部分は、ドット式の左側の演算子です。網掛けでない部分は、右側の演算子です。

ドット表記を使用すると、PUTステートメントは次のようになります。
n INITメソッドのローカル変数 pcで参照されるパッケージ pkgcのインスタンス
を取得します。

n パッケージ属性 pbで参照されるパッケージ pkgbのインスタンスを、パッケージ
pkgcインスタンスから取得します。

n パッケージ pkga のインスタンスを取得します。このインスタンスは、パッケー
ジ pkgbのインスタンスのパッケージ属性 paで参照されています。

n パッケージ属性 xの値を、パッケージ pkgaのインスタンスから取得します。
n 値を SASログに書き込みます。
各ドット演算の左側のオペランドがパッケージ変数であるか、パッケージ変数に解
決されるドット演算である限り、式で使用できるドット演算の数に制限はありませ
ん。DS2式でのドット演算子の評価方法については、“式の演算子 ” (102ページ)を
参照してください。ドット表記で PUTステートメントを使用する方法の詳細につ
いては、“DOT演算子式” (81ページ)を参照してください。
ドット表記は、ユーザー定義の DS2パッケージのパッケージ属性とパッケージメソ
ッド、および事前定義パッケージのパッケージメソッドへのアクセスをサポートし
ます。
変数を指定できる場所にあるパッケージ属性にアクセスすために、ドット表記を使
用できますが、次の場合は除きます。
n パッケージスカラー属性を DS2メソッド呼び出しの IN_OUTパラメーターへの
引数として指定する。

n ユーザー定義の DS2メソッド呼び出しの配列パラメーターにパッケージからの
配列属性を指定する。

n SUMステートメントの左側の変数を指定する。
n Array割り当てステートメント(:=) 内。
使用例については、“例: ドット表記を使用したスカラー属性およびメソッドへのア
クセス”、“例: ドット表記を使用した配列属性へのアクセス”、“例: ドット表記を使用
してパッケージ階層内のデータにアクセスする” (172ページ)を参照してください。

例: ドット表記を使用したスカラー属性およびメソ
ッドへのアクセス

この例では、CIRCLEという名前のパッケージを作成し、パッケージ変数を使用して
パッケージを参照します。

proc ds2;
package circle / inline;             /* 1 */
  dcl double diameter;

  method radius() returns double;
    return diameter/2;
  end;

  method area() returns double;
    return 2 * 3.14159 * radius();
  end;
endpackage;

data;
  dcl package circle c();          /* 2 */
  dcl double area;                 /* 3 */

method init();

    c.diameter = 10;                 /* 4 */
    area = c.area();
  end;
enddata;
run;
quit;

1 circleという名前のパッケージは、PACKAGEステートメントを使用して作成され
ます。パッケージ circleは、1つの属性 diameterと、2つのメソッド radius()およ
び areaを定義します。パッケージ circle内では、グローバル変数 diameterとメソ
ッド radius()が名前で参照されます。

2 データブロックでは、DECLARE PACKAGEステートメントがパッケージ変数を宣
言し、パッケージ circleのインスタンスを作成します。DECLARE PACKAGEステ
ートメントは、必要なコンストラクターパラメーター値がかっこ内に指定されて
いる場合、パッケージ変数に加えてパッケージインスタンスを作成します。この
場合、circleパッケージのコンストラクターにはパラメーター値は必要ありませ
ん。パッケージ変数 Cに続く空のかっこは、コンストラクターをトリガーしてイ
ンスタンスを作成します。

3 DECLAREステートメントは、パッケージインスタンスで参照できるように、パ
ッケージ内のパッケージ属性 diameterのスカラー変数を定義します。

4 INIT()メソッドのコードは、ドット表記を使用して、パッケージ変数 c によって
参照される circleパッケージインスタンスからデータメンバーにアクセスしま
す。式 c.diameterは、パッケージインスタンスのパッケージ属性 diameterの値を
設定します。式 c.area()は、パッケージインスタンスからメソッド areaを呼び出
します。

例: ドット表記を使用した配列属性へのアクセス
この例は、ドット表記を使用してパッケージの配列属性にアクセスする方法を示し
ています。

proc ds2;
package dynamic_record / inline;                 /* 1 */
  dcl double sales[*];
  dcl double profit[*];

  method resize(int n);
    resizearray sales[n];
    resizearray profit[n];
  end;
endpackage;

data _null_;                                     /* 2 */
  method compute_totals(package dynamic_record record);
    dcl double total_sales total_profit;
    dcl int i;

    total_sales = 0;
    do i = 1 to dim(record.sales);
      total_sales = total_sales + record.sales[i];

    end;

    total_profit = 0;
    do i = 1 to dim(record.profit);
      total_profit = total_profit + record.profit[i];
    end;

    put '            ----------    ----------';
    put '   TOTAL ' total_sales dollar12.2 ' ' total_profit dollar12.2;
  end;
  
  method init();                                 /* 3 */
    dcl package dynamic_record record();
    dcl int quarter;

    /* Resize record to hold sales and profit for 4 quarters. */
    record.resize(4);

    record.sales[1] = 1001.50;
    record.sales[2] = 155.25;
    record.sales[3] = 405.00;
    record.sales[4] = 2000.50;

    record.profit[1] = 24.00;
    record.profit[2] = 75.00;
    record.profit[3] = 26.00;
    record.profit[4] = 100.00;

    put ' QUARTER         SALES        PROFIT';    /* 4 */
    do quarter=1 to 4;
        put quarter 8. ' '
            record.sales[quarter] dollar12.2 ' '
            record.profit[quarter] dollar12.2;
    end;

    compute_totals(record);
  end;
enddata;
run;
quit;

1 このコードは、2つの動的一時配列 salesと profitを宣言する dynamic_recordとい
う名前のパッケージを定義します。このコードでは、resizeという名前のメソッ
ドも定義して、配列内の要素の数を簡単に変更できるようにしています。パッケ
ージメソッドへのすべての呼び出しには、暗黙的パラメーター(パッケージメソッ
ドの呼び出しに使用されるパッケージのインスタンス)が含まれます。resizeパ
ッケージメソッドが呼び出されると、このメソッドは、整数パラメーター nで指
定された要素数を持つように、dynamic_recordパッケージインスタンスの sales
配列と profit配列のサイズを変更します。

2 DATAプログラムは、dynamic_recordパッケージインスタンスの sales変数の合計
と profit変数の合計を計算して出力する compute_totalsという名前のメソッドを
定義します。パラメーター recordは、dynamic_recordパッケージインスタンスを
参照します。このメソッドは、インスタンスの sales配列の要素の値をローカル
変数 total_salesに加算し、インスタンスの profits配列の要素をローカル変数
total_profitsに加算することによって、 dynamic_recordパッケージインスタンスの

総売上高と総利益を計算します。ドット表記は、パラメーター recordによって参
照される dynamic_recordパッケージインスタンス内の sales配列と profits配列
の要素にアクセスするために使用されます。

3 dynamic_recordパッケージのインスタンスは、INITメソッドでインスタンス化さ
れ、パッケージ変数 recordに割り当てられます。この DS2プログラムによって
dynamic_recordパッケージのインスタンスが構築されるのは、このときだけで
す。新しく構築された dynamic_recordパッケージインスタンスの sales配列と
profit配列の要素は 0です。INITメソッドは、パッケージ変数 recordを指定して
dynamic_recordパッケージの resizeメソッドを呼び出し、dynamic_recordパッケ
ージインスタンス内の配列のサイズを変更して、それぞれが 4つの要素を持つよ
うにします。次に、INITメソッドは、パッケージ変数 recordによって参照され
る dynamic_recordパッケージインスタンス内の sales配列と profit配列の 4つの
要素のそれぞれに値を割り当てます

4 INITメソッドは、配列の要素をループし、その値を出力します。最後に、INIT
メソッドはパッケージ変数 recordを指定して compute_totalsメソッドを呼び出
し、dynamic_recordパッケージインスタンスの各配列の合計を出力します。

SASログに書き込まれる結果は次のとおりです。

 QUARTER         SALES        PROFIT
       1     $1,001.50        $24.00
       2       $155.25        $75.00
       3       $405.00        $26.00
       4     $2,000.50       $100.00
            ----------    ----------
   TOTAL     $3,562.25       $225.00

例: ドット表記を使用してパッケージ階層内のデー
タにアクセスする

パッケージの階層に存在するスカラー属性とメソッドにドット表記を使用してアク
セスする方法を示すプログラム例を次に示します。

proc ds2;
package ContactAddress / inline; /* 1 */
  declare varchar(100) street;
  declare varchar(100) city;
  declare varchar(100) state;
  declare varchar(100) zipcode;
endpackage;

package ContactInfo / inline;
  declare package ContactAddress address();
  declare varchar(100) phone;
endpackage;

package CustomerHistory / inline;
  declare int numberOfOrders;
  declare int numberOfReturns;
endpackage;

package Customer / inline;
  declare varchar(100) name;
  declare package ContactInfo work();
  declare package ContactInfo home();
  declare package CustomerHistory history();
  declare double discount;
endpackage;

data _null_;                                     /* 2 */
  method createTestCustomer(package Customer person);
    person.name = 'Jane Doe';
    person.work.address.street = '100 SAS Campus Dr';
    person.work.address.city = 'Cary';
    person.work.address.state = 'NC';
    person.work.address.zipcode = '27513';
    person.work.phone = '(919) 677-8000';
    person.home.address.street = '123 Main St';
    person.home.address.city = 'Apex';
    person.home.address.state = 'NC';
    person.home.address.zipcode = '27502';
    person.home.phone = '(919) 555-5555';
    person.history.numberOfOrders = 4;
    person.history.numberOfReturns = 0;
  end;

  method computeDiscount(package Customer person);
    person.discount = .00;

    /* If customer is located in North Carolina, discount 5%. */
    if (person.home.address.state = 'NC') then
      person.discount = .05;
    
    /* If customer has at least 10 orders, discount 10%. */
    if (10 <= person.history.numberOfOrders) then
      person.discount = .10;
  end;

  method printDiscount(package Customer person);
    declare varchar(100) first_name;
    declare varchar(100) percentage;
    first_name = substr(person.name, 1, find(person.name, ' ')-1);
    percentage = person.discount * 100 || '%';
    put first_name 'receives a discount of' percentage;
  end;

  method init();                                  /* 3 */
    declare package Customer person();
    createTestCustomer(person);
    computeDiscount(person);
    printDiscount(person);
  end;
enddata;
run;
quit;

1 プログラム例では、ContactAddress、ContactInfo、CustomerHistory、Customerの
4つのパッケージが定義されています。パッケージ Customerは、2つの
ContactInfoパッケージ属性を持つものとして定義されます。1つは顧客の職場
の連絡先情報を格納するためのもので、2つ目は顧客の自宅の連絡先情報を格納
するためのものです。パッケージ Customerには、顧客の注文と返品の履歴を保
存するための CustomerHistoryパッケージ属性もあり、nameと discountという 2
つの追加属性を定義します。パッケージ ContactInfoは、連絡先の住所情報を格
納するための ContactAddressパッケージ属性を持つものとして定義されていま
す。

2 データブロックは、次の 3つのメソッドを作成します。
n メソッド createTestCustomerは、単一の引数(パッケージ Customerのインスタ
ンス)を取り、Jane Doeという名前の顧客に関する情報を、パッケージ
Customerをルートとするパッケージ階層内のパッケージの属性に割り当て
ます。ドット表記を使用して、パッケージ Customerから Jane Doeの情報が
格納される属性までパッケージ階層を走査します。メソッド
createTestCustomerおよび computeDiscountの呼び出し後にメモリに格納さ
れるパッケージデータの図については、図 13.1 (175ページ)を参照してくだ
さい。

n メソッド computeDiscountは、パッケージ Customerとパッケージ変数 person
をメソッド引数として指定し、IF式を指定して、場所と注文履歴に基づいて
顧客が価格割引を受ける資格があるかどうかを判断します。式
person.discountは、Customerパッケージのパッケージ属性 discountを参照し
ます。式 person.home.address.stateは、ContactAddressパッケージのパッケー
ジ属性 Stateを参照します。式 person.history.numberOfOrdersは、
CustomerHistoryパッケージのパッケージ属性 numberOfOrdersを参照しま
す。

n メソッド printDiscountは、パッケージ Customerとパッケージ変数 personを
メソッド引数として指定し、式を使用して値を定義し、ローカル変数
first_nameと percentageに割り当てます。変数 first_nameの値は、式
person.nameをスカラー引数として SUBSTRおよび FIND関数に渡すことに
よって作成されます。変数 percentageの値は、式 person.discountを DS2数式
のスカラー引数として組み込み、結果の値にパーセント記号を追加すること
によって作成されます。

3 INIT()メソッドは、personという名前の単一の Customerパッケージをインスタン
ス化し、createTestCustomer(person)、computeDiscount(person)、および
computeDiscount(person)メソッドを呼び出します。

プログラムからの出力を次に示します。

Jane receives a discount of 5%

この図は、パッケージ階層内のパッケージ間の関係を示しています。

図 13.1 パッケージ Customerによってインスタンス化されるパッケージ階層

配列パッケージ変数
配列パッケージ変数は、パッケージインスタンスを参照する配列です。配列パッケ
ージ変数は、大規模なデータを扱う場合に便利です。配列パッケージ変数を使用す
ると、1つの変数を通じてパッケージの多数のインスタンスを参照できます。配列
パッケージ変数の各要素はパッケージのインスタンスを参照し、スカラーパッケー
ジ変数と同等です。
次は、ドット演算式で配列パッケージ変数がどのように参照されるかの例です。
array-package-variable\[index\].[attribute|method(arguments)]

array-package-variableは配列の名前です。Indexは、配列内で参照するパッケージ
インスタンスを含む要素の位置を示します。配列式のインデックスには、定数、変
数、または別の式を指定できます。式内のドット演算子は、指定された attributeに
アクセスするか、配列パッケージ変数によって参照されるパッケージインスタンス
から指定されたmethodを呼び出します。
ネストされたドット表記を使用する場合、ドット演算の左側のオペランドがパッケ
ージ変数または配列パッケージ変数の要素であり、右側のオペランドが左側のオペ
ランドのパッケージの属性またはメソッドである限り、スカラーパッケージ変数と
配列パッケージ変数要素は交換可能です。
有効なネストされたドット演算の例を次に示します。

x[i].y[j].z[1]                          1
a.b[i+100].c[min(x, y, z) - 1].d.e.m()  2

1 変数 xは配列パッケージ変数です。属性 yは x[i]の配列パッケージ変数です。属
性 zは x[i].y[j]の配列属性です。

2 変数 a、d、および eはパッケージ変数です。変数 bと cは配列パッケージ変数で
す。メソッド mは、a.b[i+100].c[min(x, y, z) - 1].d.eによって参照されるパッケージ
インスタンスで呼び出されます。

詳細については、"配列パッケージ変数の宣言と使用"を参照してください。

ドット表記を使用した PUTステートメントの機能
PUTステートメントでドット表記を使用する場合、PUTステートメントを使用して
ローカルパッケージ変数を出力する場合との次の違いに注意する必要があります。
n ドット表記を使用してパッケージ属性を出力する場合、PUTステートメントは、
変数の DECLAREステートメントの HAVING句で変数に指定されている出力形
式を無視します。ローカルパッケージ変数を出力する場合、PUTステートメント
は、デフォルトで、DECLAREステートメントの HAVING句で指定された出力形
式を使用します。パッケージ属性に出力形式を適用するには、PUTステートメン
トや PUT関数で出力形式を指定する必要があります。
この動作を示す例を次に示します。
proc ds2;
/* define package item */
 package item / inline;
 dcl varchar(1024) name having format $upcase.;
 dcl double price having format dollar10.2;
   method item(varchar(1024) name1, double price1);
   name = name1;
   price = price1;
   put 'Package attributes printed locally';
   put name price; 
   end;
   endpackage;

/* Instantiate the package and declare variables representing package attributes */
data _null_;
method init();
   dcl package item e1('hammer', 10.00);
   put 'Default output of dot-notated package attributes';
   put e1.name e1.price; 
   put 'Dot-notated package attributes specified with a format';
   put e1.name $upcase. e1.price dollar10.2;
   end;
enddata;
run;
quit;

次の出力が SASログに書き込まれます。
Package attributes printed locally
HAMMER     $10.00
Default output of dot-notated package attributes
hammer 10
Dot-notated package attributes specified with a format
HAMMER    $10.00

n PUTステートメントを使用してパッケージから配列属性を出力するとき、ドット
表記を* (アスタリスク)表記と組み合わせることはサポートされません。

n

同種のデータ配列で使用できるのと同じ PUTステートメント機能が、配列パッケー
ジ変数でも使用できます。PUTステートメントは、配列パッケージ変数の要素から
属性の値を出力するためにサポートされています。たとえば、次のものがサポート
されています。
put array-package-variable[i].attribute;

パッケージ属性の出力の詳細については、“PUT ステートメント” (SAS DS2言語リフ
ァレンス)を参照してください。

ユーザー定義パッケージ
ユーザー定義パッケージの概要

作成したメソッドをユーザー定義パッケージに格納できます。これらのパッケージ
は、メソッドのライブラリと考えることができます。(PACKAGEステートメントを
使用して)パッケージにメソッドを格納したら、DECLARE PACKAGEステートメント
または_NEW_演算子を使用してパッケージのインスタンスを作成することにより、
メソッドにアクセスできます。
コンストラクター引数を指定した DECLARE PACKAGEステートメントは、パッケー
ジ変数を作成し、プログラムのコンパイル時にパッケージインスタンスを構築しま
す。この方法は、"コンパイル時のインスタンス化"と呼ばれることがよくあります。
declare package complex c();

または、コンパイル時にパッケージ変数を宣言し、プログラムの実行中に_NEW_演
算子を使用してパッケージ インスタンスを作成することもできます。この方法は、"
実行時のインスタンス化"と呼ばれることがよくあります。
declare package complex c;

 c = _new_ complex();

詳細については、“PACKAGE ステートメント” (SAS DS2言語リファレンス)、 
“DECLARE PACKAGE ステートメント” (SAS DS2言語リファレンス)、および“_NEW_ 
演算子” (SAS DS2言語リファレンス)を参照してください。

注: ユーザー定義パッケージを使用して、パッケージ名をオーバーロードすることに
より、事前定義された DS2パッケージを非表示にすることはできません。ビルトイ
ンパッケージ名へのパッケージ参照は、ユーザー定義パッケージではなくビルトイ
ンパッケージに解決されます。

MATHという非常に単純なユーザー定義パッケージの例を次に示します。2つの数
値を加算するメソッドが含まれています。
package math;
   method add(double x, double y) returns double;
      return x+y;

   end;
endpackage;

次の例では、前の例で作成したMATHパッケージの ADDメソッドを使用して、2つ
の数値を加算します。最初に、MATHパッケージが宣言され、インスタンス化され
ます。次に、ADDメソッドが呼び出され、結果が SUMに割り当てられます。
data _null_;
   dcl double sum;
   method init();
      dcl package math f();
      sum = f.add(2,3);
         put 'sum= ' sum;
   end;
enddata;

単純なリストパッケージの実装
DS2パッケージの構文を使用すると、複合構造を作成し、その構造にアクション(メ
ソッド)を割り当てることができます。DS2パッケージは、別のパッケージから属性
またはメソッドを継承できませんが、他のパッケージへの参照を含めることができ
ます。したがって、合成(has-a)関係をサポートできます。
単純だが具体的なリストの例を次に示します。これは、MYELEMENTという特定の
要素を保持するMYLISTというリストパッケージから始まります。小さなデータプ
ログラムは、MYLISTおよびMIELEMENTパッケージを実行します。
proc ds2;
/*-- Define a simple list element --*/
package myElement/overwrite = yes;
   dcl package myElement next;
   dcl int d;

   /*-- Custom constructor with parameter --*/
   method myElement( int d );
      this.d = d; /*-- Use “this” to differentiate d’s --*/
   end;

   method print();
      put d=;
   end;

   /*-- A default delete method is implicitly defined --*/
endpackage;

/*-- Define a package to hold the elements --*/
package myList/overwrite=yes;
   dcl package myElement front;
   dcl package myElement back;

   method add( package myElement element );
      if null( front ) then do; /*-- first element --*/
         front = element;

         back = element;
      end;
      else do; /*-- link in new element --*/
         back.next = element;
         back = element;
      end;
   end;

   method printFront();
      if ^null( front ) then front.print();
      else put 'Front is NULL';
   end;

   method printBack();
      if ^null( back ) then back.print();
      else put 'Back is NULL';
   end;

/*-- A custom destructor – called when references go to zero --*/
   method delete();
      dcl package myElement cur next;
      cur = front;

      /*-- Explicitly empty the list --*/
      do while ( ^null( cur ) );
         next = cur.get_next();
         cur.print();
         cur.delete();
         cur = next;
      end;
   end;
endpackage;
run;

data _null_;
   method run();
      dcl int i;
      dcl package myList ml(); /*-- Instantiated --*/
      dcl package myElement me; /*-- Reference only --*/
      do i = 1 to 10;
          me = _new_ myElement(i); /*-- Constructor with parms --*/
          ml.add(me);
      end;
      ml.printFront();
      ml.printBack();
   end;
enddata;
run;
quit;

MYELEMENTパッケージが、内部構造の最初の要素として、自身への参照である
nextを宣言する方法に注意してください。これは、リンクされた構造をメモリ内に
構築するための基盤です。パッケージ自体を参照する機能です。次に、リストが保
持する内容(この場合は dという単一の整数)を宣言します。2つのメソッドが定義
されています。最初の myElement(int d) はパッケージと同じ名前です。これは、パッ
ケージのコンストラクターであることを意味します。コンストラクターを定義する

必要はありませんが、定義すると、この場合の dの値などの初期化パラメーターを
取ることができます。2番目のメソッド PRINTは、単に dで PUTを呼び出して、要
素の値を SASログに表示します。ユーザー作成の DELETEメソッドが含まれていな
い場合、デフォルトのデストラクターがコンパイラによって提供されます。
次に、MYLISTパッケージには、タイプMYELEMENTの 2つの要素である frontと
backが含まれています。それらと ADDメソッドを使用すると、単純なリスト構造
が得られます。たとえば、MYLISTパッケージには PRINTFRONTおよび
PRINTBACKメソッドが含まれており、これらのメソッドはMYELEMENTパッケー
ジの PRINTメソッドを呼び出します。最後に、あるパッケージが他のパッケージへ
の参照を保持している場合に推奨されるカスタム DELETEメソッドがあります。他
のパッケージへの参照は、参照の循環につながる可能性があります。実行が RUNメ
ソッドを離れると、mlによって保持されている MYLISTのインスタンスで DELETE
が暗黙的に呼び出され、リスト内のMYELEMENTのすべてのインスタンスで
DELETEが明示的に呼び出されます。

事前定義された DS2パッケージ
事前定義された DS2パッケージの概要

SASには、DS2言語で使用する次の事前定義パッケージが用意されています。
データグリッド
ネイティブデータグリッドオブジェクトの作成を有効にします。
詳細については、“データグリッドパッケージの使用” (181ページ)を参照してく
ださい。

FCMP
DS2言語内からの FCMP関数およびサブルーチンの呼び出しをサポートします。
詳細については、“FCMPパッケージの使用” (191ページ)を参照してください。

注: FCMPパッケージは、CASサーバーではサポートされていません。

ハッシュおよびハッシュ反復子
一意のルックアップキーに基づいてデータを迅速かつ効率的に格納、検索、およ
び取得できます。ハッシュパッケージのキーとデータは変数です。キーおよび
データ値には、定数値、テーブルからの値、または式で計算された値を直接割り
当てることができます。
詳細については、“ハッシュパッケージの使用” (194ページ)および“ハッシュ反復
子パッケージの使用” (209ページ)を参照してください。

HTTP
HTTP Webサービスにアクセスするための HTTPクライアントを構築します。
詳細については、“HTTPパッケージの使用” (209ページ)を参照してください。

JSON
JSONテキストの作成と解析を可能にします。

ロガー
SASログ機能への基本的なインターフェイス(開く、書き込み、およびレベルクエ
リ)を提供します。
詳細については、“ロガーパッケージの使用” (218ページ)を参照してください。

行列
強力で柔軟な行列プログラミング機能を提供します。SAS/IML機能の DS2レベ
ルの実装を提供します。
詳細については、“MATRIXパッケージの使用” (221ページ)を参照してください。

PCRXFINDと PCRXREPLACE
指定された文字列内の部分文字列を検索したり、部分文字列を置き換えたりする
方法を提供します。

SQLSTMT
FedSQLステートメントを DBMSに渡して実行し、DBMSから返された結果セッ
トにアクセスする方法を提供します。
詳細については、“SQLSTMTパッケージの使用” (232ページ)を参照してくださ
い。

注: SQLSTMTパッケージは、CASサーバーではサポートされていません。

TZ
ローカルおよび国際的な時刻と日付の値を処理する方法を提供します。
詳細については、“TZパッケージの使用” (237ページ)を参照してください。

データグリッドパッケージの使用
データグリッドパッケージの概要
DS2パッケージのデータグリッドを使用すると、ネイティブデータグリッドオブジ
ェクトを作成できます。パッケージデータグリッドには、オブジェクトをインスタ
ンス化し、列と行を追加、管理、および操作できるようにする一連のメソッドが含
まれています。データグリッドパッケージの使用方法については、“DS2データグリ
ッドパッケージのメソッド、演算子、およびステートメント” (SAS DS2言語リファレ
ンス)を参照してください。

データグリッドについて
データグリッドは、値のテーブルです。すべての列は、10進数、文字、整数(32ビ
ット)、長整数(64ビット)、ブール型のいずれかの値を定義します。各行には、1つ
のデータレコードが含まれます。

注: このデータグリッドパッケージのドキュメントでは、倍精度型の列にキーワード
decimalが使用されています。

たとえば、データグリッドには、すべての顧客の保険契約のレコードを含めること
ができます。このテーブルは、表 13.3に示されているテーブルのようになります。
表 13.3 保険契約テーブル

PolicyHolder PolicyNumber YearlyPremium

Smyth, Joe 453975R398 439.50

Smyth, Joe 987348P210 132.90

Dupree, Marcel 983092B228 334.00

Dupree, Marcel 274933P412 219.25

または、データグリッドを使用して、この保険契約情報を図 13.2で表されるように
格納することもできます。ここで、表 13.3に表示されるデータは、データグリッド
を使用してテーブルに再構築されます。特定の保険契約番号と保険料は、各保険契
約者レコードのデータグリッドにまとめられます。
図 13.2 2つのデータグリッドを使用した保険契約テーブル

データグリッドの JSON文字列について
データグリッドは、JavaScript Object Notation (JSON)文字列で表されます。データ
グリッドの JSON文字列には、次の基本的な形式があります。
[{"metadata":[column-definitions]},{"data":[column-data]}]

列の定義は、カンマで区切られた名前と値のペアです。
{"column1-name":"data-type"},{"column2-name":"data-type"},...

データグリッドの各行のデータは、各値の間にカンマを入れて角かっこで指定しま
す。
[column1-data,column2-data...]

たとえば、図 13.2に示されているデータグリッドがシリアル化されている場合、保
険契約テーブルは表 13.4に示されているように表示されます。
表 13.4 シリアル化された保険契約テーブル

PolicyHolder Policies

Smyth, Joe [{"metadata":[{"POLICYNUMBER":"string"},
{"YEARLYPREMIUM":"decimal"}]}, {"data":[["453975R398",439.50],
["987348P210",132.90]]}]

Dupree, 
Marcel

[{"metadata":[{"POLICYNUMBER":"string"},{"YEARLYPREMIUM":"decimal"}]}, {"data":[["983092B228",324.00],["274933P412",219.35]]}]

データグリッド JSON表現の詳細と例については、“データグリッドの JSON表現に
ついて” (187ページ)を参照してください。

データグリッドパッケージの宣言とインスタンス化
まず、データグリッドパッケージを宣言してインスタンス化する必要があります。
データグリッドパッケージのインスタンスを構築する方法は 2つあります。
n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。

declare package datagrid dg;
dg = _new_ datagrid();

n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package datagrid dg();

詳細については、“DECLARE PACKAGE ステートメント: データグリッドパッケージ” 
(SAS DS2言語リファレンス)、および“_NEW_ 演算子: データグリッドパッケージ” 
(SAS DS2言語リファレンス)を参照してください。

事前定義された DS2パッケージ 183

http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n1stcoa7nnyi0ln1nljhri5h5gll.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n1stcoa7nnyi0ln1nljhri5h5gll.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n07yro647nux1nn1wr7mgdzdqkje.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n07yro647nux1nn1wr7mgdzdqkje.htm&locale=ja


例: データグリッドパッケージの使用
この DS2の例では、データグリッドパッケージの逆シリアル化メソッドを使用し
て、JSON文字列をデータグリッドに変換し、グリッドから値を取得します。

proc ds2;
data _null_;
dcl package datagrid Joe_Smyth();
dcl double myDouble;
dcl varchar(64000) str;

method run();
  put '--------------------';
  str = '[{"metadata":[{"POLICYNUMBER":"string"}, 
          {"YEARLYPREMIUM":"decimal"}]}, 
          {"data":[["453975R398",439.50]
, ["987348P210",132.90]]}]';
  Joe_Smyth.deserialize(str);
  str = Joe_Smyth.getValue('PolicyNumber', 2);
  put 'Policy Number: ' str;
  put '--------------------';
end;
enddata;
run;

これにより、次の出力が生成されます。

Policy Number: 453975R398

比較演算子
次の演算子のいずれかを、filteredGet関数やmatchCount関数などの値を比較する
関数で使用できます。

演算子 説明
EQ、=、== 等しい
NE、!=、^=、<> 等しくない
GT、> より大
LT、< より小
LE、<= より小か等しい

演算子 説明
GE、>= より大か等しい

データグリッド、列名、およびインデックス値の指
定について
すべてのデータグリッド引数は、DATAGRID型である必要があります。
メソッドで列が必要な場合、その列は、col、cmpCol、colIndexなどの引数名を使用
してメソッドシグネチャに表示されます。列名を指定するには、文字列値を使用し
ます。列インデックスを指定するには、数値を使用します。データグリッドの列に
は、1から始まる番号が付けられます。

注: データグリッドパッケージは、列名に含まれる先頭および末尾のスペースを削除
します。

行処理について
データグリッドの行には、1から始まる番号が付けられます。
一部のメソッドは、定義された行範囲をオプションの引数として受け入れます。こ
れらの引数は、 rowStartIndexおよび rowEndIndexとしてメソッドシグネチャに表
示されます。行は、rowStartIndexから始まり、rowEndIndexまで処理されます。
rowStartIndexの 0 (ゼロ)値は、最初から開始することを示します(これは 1と同じで
す)。rowEndIndexの 0 (ゼロ)は、すべての行を使用することを示します。
ここでは、データグリッドの行処理を説明する例を示します。次のテーブルで考え
てみましょう。

Col_1 Col_2

1 A1 A2

2 B1 B2

3 C1 C2

4 D1 D2

5 E1 E2

Col_1 Col_2

6 F1 F2

7 G1 G2

8 H1 H2

値(rowStartIndex = 5、rowEndIndex = 7)が与えられた場合、影響を受ける行は次の
とおりです。

Col_1 Col_2

5 E1 E2

6 F1 F2

7 G1 G2

データグリッドの null値と欠損値の処理方法につい
て
7章, “DS2による nullおよび SAS欠損値の処理方法” (61ページ)で説明したよう
に、CHAR列の null値に対して検索または比較操作を実行すると、DS2は常に null
値を返します。これにより、filteredGetや columnCountMatchなど、null値また
は欠損値を検索または比較するデータグリッドメソッドを使用すると、予期しない
結果が生じる可能性があります。これを回避するために、データグリッドパッケー
ジの動作が少し異なります。
次の表は、null値、欠損値、および文字列文字を相互に比較するときに使用される
データグリッドロジックを示しています。各交点のセル値は、データグリッドが比
較を有効と見なして要求された結果を返すか(1)、無効と見なすか(0)、または null値
を返すかを表します。
次の表は、等式演算に適用されるデータグリッドロジックを示しています。

= null値 欠損値 文字列文字
null値 1 null null

欠損値 null 1 0

文字列文字 null 0 1

186 13章 / DS2パッケージ



不等式演算の場合、データグリッドロジックは等式演算の逆で動作します。次の表
は、不等式演算に適用されるデータグリッドロジックを示しています。

^= null値 欠損値 文字列文字
null値 null 1 1

欠損値 1 0 1

文字列文字 1 1 0

いずれの場合も、nullが検索または比較操作に含まれる場合、nullは nullとのみ等
しくなります。逆に、nullは null以外のすべてのものと等しくありません。

データグリッドの JSON表現について
データグリッドパッケージの構文規則は、標準の DS2構文規則とは異なることに注
意してください。データグリッドパッケージの構文規則は次のとおりです。
< >
山かっこはオプションのコンテンツを示します。

[ ]
角かっこは、値の配列を定義するために必要な構文エントリを示します。

{ }
中かっこは、オブジェクトを定義するために必要な構文エントリを示します。

斜体のテキスト
斜体のテキストは、指定する引数または値を示します。

...
省略記号は、省略記号に続く引数または引数のグループを何度でも繰り返すこと
ができることを示します。省略記号とそれに続く引数が山かっこで囲まれてい
る場合、それらはオプションです。

ここでは、データグリッドの JavaScript Object Notation (JSON)構造を示します。
 [<{"metadata":[{"column-1":"data-type"} <...,{"column-n":"data-type"}>]} , >
   {"data":[ < [row-1-data] <... , [row-n-data] > ]}]

n 最初の部分はメタデータと呼ばれます。テーブル内の各列を名前と型で指定し
ます。割り当てられる型は、ブール、10進数、整数、文字列のいずれかです。

注: メタデータが定義されている場合は、少なくとも 1つの列を定義する必要が
あります。

n 次の部分には、メタデータ部分で指定された列の順序と型に対応するデータの行
が含まれています。

n データグリッドに行データが含まれていても、メタデータが含まれていない可能
性があります。これは、ヘッドレスデータグリッドとして知られています。ヘッ
ドレスデータグリッドに関する次の情報に注意してください。

事前定義された DS2パッケージ 187



o 最初の非 nullデータ行によって、文字列または数値を含む列のデータ型が決
まります。

o 数値列型は次のように割り当てられます。
n 列のすべての値が整数の場合、その列には longの数値型が割り当てられ
ます。

n 列の少なくとも 1つの値が 10進数の場合、その列には decimalの数値型
が割り当てられます。

次に例を示します。
JSON表現:

[{"data":[[null,null,null,null,null],["Str_01",true,null,15,1],
["Str_02",false,null,77,2],["Str_03",false,null,93,3.03],
["Str_04",true,null,15,4.04],["Str_05",false,null,null,5.05]]}]

結果のデータグリッド:

row[0000]: string(mv) boolean(mv) decimal(mv) long(m1v) decimal(mv) 
row[0001]: string(Str_01) boolean(true) decimal(mv) long(15) decimal(1.00)
row[0002]: string(Str_02) boolean(false) decimal(mv) long(77) decimal(2.00)
row[0003]: string(Str_03) boolean(false) decimal(mv) long(93) decimal(3.03)
row[0004]: string(Str_04) boolean(true) decimal(mv) long(15) decimal(4.04)
row[0005]: string(Str_05) boolean(false) decimal(mv) long(mv) decimal(5.05)

列 5の少なくとも 1つのセルに 10進数値が含まれているため、行 1と 2の整数
値は 10進数値に変換されます。

データグリッドの JSON表現の例
n メタデータとデータを含むテーブルを次に示します。

[{
    "metadata" : [
          { "aStringColumn" : "string" },
          { "anIntColumn" : "integer" },
    { "aFloatColumn" : "decimal" },
    { "aBooleanColumn" : "boolean" }]
},{
"data" : [[ "one", 1, 1.11, true ],
          [ "two", 2, 2.22, false ],
          [ "three", 3, 3.33, true ],
          [ "four", 4, 4.44, false ],
          [ "five", 5, 5.55, true ],
    [ null, null, null, null ]]
}]

n データのないテーブルを次に示します。
[{
    "metadata" : [
          { "aStringColumn" : "string" },
          { "anIntColumn" : "integer" },
    { "aFloatColumn" : "decimal" },
    { "aBooleanColumn" : "boolean" }]

    },{
    "data" : [] 
}]

n メタデータのないテーブル(ヘッドレスデータグリッド)を次に示します。
[{
    "data" : [["Be1","Bd1","Bc1","Ba1"],  
             ["Be2","Bd2","Bc2","Ba2"],  
             ["Be3","Bd3","Bc3","Ba3"]]
}]

n null値テーブルを次に示します。
null

データグリッドパッケージのエラー報告

無効なパラメーター
getInt、setValue、getStringなど、値を取得または設定するメソッドに無効なパラ
メーターが指定された場合、データグリッドパッケージはエラーをログに記録しま
す。
SAS Micro Analytic Serviceでデータグリッドパッケージを使用している場合、環境
変数を設定することで、これらのデータグリッドの問題を SAS Micro Analytic 
Serviceの実行エラーとしてプロモートできます。詳細については、“Promoting 
DS2 Execution Issues to a SAS Micro Analytic Execution Failure” (SAS Micro 
Analytic Service: Programming and Administration Guide)を参照してください。
エラーがログに記録される状況の例を次に示します。
表 13.5 データグリッドメソッドのエラーをログに記録する条件

オブジェクト 条件 例
列 指定されたインデックス値が

無効。たとえば、0より小さい。
インデックス値が 0より小さい
ため、ここでエラーが発生しま
す。
dgA.getValue(‑1,2);

指定されたインデックス値が
データグリッドの列数より大
きい。

dgAには 10列が含まれている
ため、ここでエラーが発生しま
す。
dgA.getValue(11,2);

指定された列名が存在しない。 dgAには Dateという名前の列
が含まれていないため、ここで
エラーが発生します。
dgA.getValue('Date',2);

オブジェクト 条件 例
行 指定されたインデックス値が

無効。たとえば、0より小さい。
インデックス値が 0より小さい
ため、ここでエラーが発生しま
す。
dgA.getValue('Name',‑1);

指定されたインデックス値が
データグリッドの行数より大
きい。

dgAには 10行が含まれている
ため、ここでエラーが発生しま
す。
dgA.getValue(1,12);

演算子 必要な演算子がない。 ColAと 200の間の演算子がな
いため、ここでエラーが発生し
ます。
dgA.filteredget('ColA',200,'ColB',5,
10);

無効またはサポートされてい
ない演算子が指定されている。

ここでは、除算演算子(/)が無効
なため、エラーが発生します。
dgA.filteredGet('ColA','/',200,'ColB'
,5,10);

範囲外のインデックス
デフォルトでは、データグリッドパッケージで範囲外のエラーが発生すると、ERROR
ではなく NOTEとしてログに記録されます。この状態の重要度を ERRORにアップ
グレードできます。これを行うには、SAS_MAS_DATAGRID_INDEX_OOB_IS_ERROR
環境変数を定義し、値を Trueに設定します。
環境変数の設定については、コンピューティング環境のドキュメントを参照してく
ださい。たとえば、SAS Compute Service、SAS Compute Server、SAS Cloud 
Analytic Services、または SAS Micro Analytic Serviceなどです。
n SAS Viya Platform: Programming Run-Time Servers

n “SAS Cloud Analytic Services: リファレンス” (SAS Viyaプラットフォーム: SAS 
Cloud Analytic Services)

n “Configuring Supplemental SAS Micro Analytic Service Environment 
Variables” (SAS Micro Analytic Service: Programming and Administration Guide)

FCMPパッケージの使用
FCMPパッケージの概要
注: FCMPパッケージは、CASサーバーではサポートされていません。

DS2言語は、FCMPパッケージを介して FCMPプロシジャで使用可能または作成さ
れた関数およびサブルーチンの呼び出しをサポートします。
PACKAGEステートメントで LANGUAGE=FCMPおよび TABLE=オプションを使用
して、FCMPパッケージを作成します。パッケージを作成したら、FCMPパッケー
ジのインスタンスを宣言します。FCMPパッケージのインスタンスを構築する方法
は 2つあります。
n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。

 declare package fcmp banking;
 banking = _new_ fcmp();

n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package fcmp banking();

詳細については、“DECLARE PACKAGE ステートメント: FCMPパッケージ” (SAS DS2
言語リファレンス)および“PACKAGE ステートメント” (SAS DS2言語リファレンス)
を参照してください。

FCMPパッケージの機能
これらは、DS2言語で FCMPパッケージを使用する機能です。
n スカラー DOUBLE、CHAR、および NCHARパラメーターを使用して FCMP関数
またはサブルーチンを呼び出し、両方のパラメーターを返します。
DS2言語は自動型変換を行うため、ほぼすべての型がサポートされます。例とし
て、TINYINTから DOUBLEへの変換があります。詳細については、“型変換の概
要” (70ページ)を参照してください。

n スカラー DOUBLE OUTARGSパラメーターを使用して FCMP関数またはサブル
ーチンを呼び出します。FCMPプロシジャの DOUBLE OUTARGSパラメーター
は、METHODステートメントでは IN_OUTパラメーターとして扱われます。
IN_OUTパラメーターの詳細については、“METHOD ステートメント” (SAS DS2
言語リファレンス)を参照してください。

注: DS2は、OUTARGSパラメーターを介して、式ではなく DOUBLE変数を渡す
必要があります。

FCMPパッケージを使用する際の考慮事項と制限事
項
n FCMPパッケージは VARARGS関数呼び出しをサポートしていないため、FCMP
プロシジャの VARARGSインターフェイスを使用できません。

n FCMPプロシージャと FCMPパッケージの間で情報が渡されるときに発生する
エラーは、常に正しく報告されるとは限りません。たとえば、PACKAGEステー
トメントで誤ったテーブル名を指定すると、ログファイルにエラーが書き込まれ
ます。ただし、操作が失敗したことは示されません。

n FCMPパッケージはセッションエンコーディングを想定しており、現在、同じ関
数呼び出し内の異なるパラメーター、または複数の関数呼び出し間での同じパラ
メーターに対して異なるエンコーディングを許可するメカニズムはありません。

n 接続文字列で FCMPライブラリが配置されているカタログを定義している限り、
任意の FCMPライブラリにアクセスできます。

n 次の FCMP関数は、DS2プログラムから呼び出された場合、DS2プログラムで
設定されたシード値を認識しません。
COMPCOST
COMPGED
RAND
STREAMINIT

DATAステップでは、RANDからのすべての乱数はデフォルトで同じストリーム
から取得されますが、CALL STREAMルーチンを使用して複数の独立したストリ
ームを作成できます。
PROC DS2では、予期しない機能の結果として、FCMPパッケージ内で呼び出さ
れる RANDの乱数は、別のストリームから取得されます。再現可能な乱数を取得
するには、RANDを呼び出す各 FCMP関数内で CALL STREAMINITルーチンを使
用します。独立したストリームを取得するには、RANDを呼び出す各 FCMP関数
内で CALL STREAMルーチンも使用します。そうすると、DS2からの結果は
DATAステップおよび将来のリリースと一致します。
次に例を示します。
title1 'Generate 3 Uniform and 3 Normal Pseudorandom Variables' ;
title2 'With and Without FCMP' ;

%let seed = 1234; %put seed=&seed;
%let nobs = 10;   %put nobs=&nobs;

* To get reproducible streams in FCMP, call STREAMINIT in each FCMP function;
* To get independent streams in FCMP, call STREAM in each FCMP function;
proc fcmp outlib=work.funcs.rand;

   function rand_uniform( stream );
      call streaminit( &seed );
      call stream (stream );
      u = rand('uniform' );

      return (u);
   endsub;

   function rand_normal( stream, mean, std );
      call streaminit( &seed );
      call stream (stream );
      z = rand('normal', mean, std );
      return (z);
   endsub;

run;
quit;

title2 'PROC DS2 Step With and Without FCMP';

proc ds2;
   package pkg / overwrite=yes language='fcmp' table='work.funcs'; 
   run;
   data ds2(overwrite=yes);
      dcl package pkg randpkg();
      declare double obs u1 u2 u3 z1 z2 z3;
      method run();
         declare double junk;
         junk = streaminit( &seed );
         do obs = 1 to &nobs;
            junk = stream( 1 );
            u1 = rand( 'uniform' );
            u2 = randpkg.rand_uniform( 2 );
            u3 = randpkg.rand_uniform( 3 );
            junk = stream( 4 );
            z1 = rand( 'normal', 100, 20 );
            z2 = randpkg.rand_normal( 5, 100, 20 );
            z3 = randpkg.rand_normal( 6, 100, 20 );
            output;
         end;
      end;
   enddata;
run; 
quit; 

proc print data=ds2;
run;

次のテーブルが生成されます。

アウトプット 13.1 乱数

ハッシュパッケージの使用
ハッシュパッケージの概要
ハッシュパッケージは、迅速なデータの格納と取得のための効率的で便利なメカニ
ズムを提供します。ハッシュパッケージは、一意のルックアップキーに基づいてデ
ータを格納および取得します。一意のルックアップキーの数とテーブルのサイズに
よっては、ハッシュパッケージのルックアップは、標準形式のルックアップや配列
よりも大幅に高速になる場合があります。
DS2ハッシュパッケージを使用する前に、ハッシュパッケージのインスタンスを定
義および構築(インスタンス化)する必要があります。
ハッシュパッケージインスタンスを定義して作成すると、次のような多くのタスク
を実行できます。
n データの格納と取得。
n データの置換と削除。
n ハッシュパッケージ内のデータを含むテーブルを生成する。
たとえば、一意の患者番号と体重に対応する数値検査結果を含む大きなテーブルと、
患者番号を含む小さなテーブル(大きなテーブルのサブセット)があるとします。一
意の患者番号をキーとして、体重値をデータとして使用して、大きなテーブルをハ
ッシュパッケージにロードできます。次に、患者番号を使用して小さなテーブルを
反復処理し、体重が特定の値を超えている現在の患者をハッシュパッケージ内で検
索し、そのデータを別のテーブルに出力できます。

ハッシュパッケージインスタンスの定義と作成
ハッシュパッケージのインスタンスを作成するには、構築するハッシュインスタン
スに関するキー、データ、およびオプションの初期化データを指定します。ハッシ
ュパッケージインスタンスは、構築時に完全に定義することも、構築時と後続の一
連のメソッド呼び出しで定義することもできます。
次の例では、ハッシュインスタンス h1と h2に同じインスタンス定義があります。
ハッシュインスタンス h1は構築時に完全に定義されますが、h2は構築時および一
連のメソッド呼び出しによって定義されます。
declare package hash h1([key], [data1 data2 data3],
   0, 'testdata', '', '', '', 'multidata');
 
declare package hash h2();
   method init();
      h2.keys([key]);
      h2.data([data1 data2 data3]);
      h2.dataset('testdata');
      h2.multidata();
      h2.defineDone();
   end;

詳細については、“コンストラクターを使用したハッシュインスタンスの定義” (195
ページ)および“メソッド呼び出しを使用したハッシュインスタンスの定義” (196ペ
ージ)を参照してください。

コンストラクターを使用したハッシュインスタンス
の定義
コンストラクターは、ハッシュパッケージをインスタンス化し、ハッシュパッケー
ジデータを初期化するために使用できるメソッドです。
コンストラクターを使用してハッシュパッケージインスタンスを作成するには、3
つの異なる方法があります。
n 部分的に定義されたハッシュインスタンスを作成します。
DECLARE PACKAGE HASH instance(hashexp, {datasource' | '\{sql-text\}'},
'ordered', 'duplicate', 'suminc', 'multidata');

キー変数とデータ変数は、メソッド呼び出しによって定義されます。初期化デー
タを提供するオプションのパラメーターは、前述の DECLARE PACKAGEステー
トメント、_NEW_演算子、メソッド呼び出し、またはこれらの組み合わせのいず
れかで指定できます。DEFINEDONEメソッドを 1回呼び出すだけで、定義が完
了します。

注: sql-textは CASサーバーではサポートされていません。

n 指定されたキー変数とデータ変数を使用して、完全に定義されたハッシュインス
タンスを作成します。
DECLARE PACKAGE HASH instance(\[keys\], \[data\]

[, hashexp, {'datasource' |
’\{sql-text\}'}, 'ordered', 'duplicate', 'suminc', 'multidata']);

キー変数とデータ変数は DECLARE PACKAGEステートメントで定義されます。
これは、完全に定義されたインスタンスを作成する必要があることを示していま
す。後続のメソッド呼び出しで追加の初期化データを指定することはできませ
ん。

n 指定されたキー変数のみを使用して、完全に定義されたハッシュインスタンス(キ
ーのみのハッシュインスタンス)を作成します。データ変数はありません。
DECLARE PACKAGE HASH instance(\[keys\][, hashexp, {'datasource' | '\{sql-

text\}'},
'ordered', 'duplicate', 'suminc', 'multidata']);

キー変数とデータ変数は DECLARE PACKAGEステートメントで定義されます。
これは、完全に定義されたインスタンスを作成する必要があることを示していま
す。後続のメソッド呼び出しで追加の初期化データを指定することはできませ
ん。

オプションパラメーターの詳細については、“ハッシュパッケージの初期化データの
提供” (198ページ)を参照してください。メソッド呼び出しを使用したオプションパ
ラメーターの定義の詳細については、“メソッド呼び出しを使用したハッシュインス
タンスの定義” (196ページ)を参照してください。

注: ハッシュインスタンスに渡されるすべての変数は、グローバル変数である必要が
あります。

メソッド呼び出しを使用したハッシュインスタンス
の定義
インスタンスの構築中にハッシュインスタンスが部分的に定義されている場合は、
次のメソッドを呼び出すことでインスタンスをさらに定義できます。

KEYS
DEFINEKEY
DATA
DEFINEDATA
DATASET
DUPLICATE
HASHEXP
ORDERED
MULTIDATA
SUMINC
DEFINEDONE

これらのメソッドの詳細については、“DS2ハッシュおよびハッシュ反復子パッケー
ジの属性、メソッド、演算子、ステートメント” (SAS DS2言語リファレンス)を参照
してください。

注: DEFINEDONEメソッドの呼び出しによってハッシュインスタンスの指定が完
了した後、前述のいずれかのメソッドを呼び出すと、エラーが発生します。

メソッド呼び出しを使用して定義されたハッシュインスタンス hの例を次に示しま
す。
data _null_; 
    declare package hash h(0, 'testdata');
    method init();
       h.keys([key]);
       h.data([data1 data2 data3]);
       h.ordered('descending');
       h.duplicate('error');
       h.defineDone();
    end;
enddata;

キー変数とデータ変数の定義
ハッシュパッケージは、一意のルックアップキーを使用してデータを保存および取
得します。キーとデータは、ドット表記メソッド呼び出しを使用してハッシュパッ
ケージを初期化するために使用する変数です。
キー変数とデータ変数は、3つの方法のいずれかで定義できます。
n 変数メソッドの DEFINEDATAおよび DEFINEKEYを使用します。
n 変数リストメソッドの DATAおよび KEYSを使用します。
n DECLARE PACKAGEステートメントで指定されたキーおよびデータ変数リスト
を使用します。

構築時にハッシュパッケージのインスタンスが完全に定義されていない場合、つま
り構築時にキー変数とデータ変数が指定されていない場合は、DEFINEDONEメソッ
ドを呼び出して、ハッシュインスタンスの初期化を完了する必要があります。
次に例を示します。
/* Keys and data defined using the implicit variable method */
declare package hash h();
h.definekey('k');
h.definedata('d');
h.definedone();

/* Keys and data defined using the variable list methods */
declare package hash h();
h.keys([k]);
h.data([d]);
h.definedone();

/* Keys and data defined using the variable list constructors */

declare package hash h([k],[d]);

キー変数は、DS2ビルトイン型(文字、数値、または日時)にする必要があります。デ
ータ変数は、DS2ビルトイン型、あるいはビルトインまたはユーザー定義のパッケ
ージ型のいずれかです。
詳細については、“暗黙の変数と変数リストのメソッド” (199ページ)を参照してくだ
さい。

ハッシュパッケージの初期化データの提供
キーとデータに加えて、ハッシュパッケージを初期化するときに、次のオプション
パラメーターを指定できます。
n ハッシュテーブルのサイズが 2nである内部テーブルサイズ(hashexp)

n ロードするテーブルの名前(datasource)、またはロードするデータを選択するた
めの FedSQLクエリ

注: FedSQLクエリの使用は、CASサーバーではサポートされていません。
n データがキーと値の順序で返されるかどうか、またはどのように返されるか

(ordered)

n テーブルをロードするときに重複キーを無視するかどうか(duplicate)

n ハッシュパッケージキーの要約カウントを維持する変数の名前(suminc)

n キーごとに複数のデータアイテムを許可するかどうか(multidata)

初期化データは、DECLARE PACKAGEステートメント、_NEW_演算子、メソッド呼
び出し、またはこれらの方法の組み合わせで指定できます。

注: DECLARE PACKAGEステートメントまたは_NEW_演算子でコンストラクターを
使用してハッシュパッケージデータを初期化する場合、オプションパラメーターを
hashexp、datasource、ordered、duplicate、suminc、multidataの順序で指定する
必要があります。これらの位置コンストラクターパラメーターはすべて、1組のか
っこで囲み、カンマで区切り、hashexpパラメーターを除いて、単一引用符で囲む
必要があります。オプションパラメーターは位置指定であるため、指定する最後の
パラメーターまで、各パラメーターのプレースホルダーを提供する必要があります。
使用する必要があるプレースホルダーは、パラメーターによって異なります。プレ
ースホルダーの詳細については、“DECLARE PACKAGE ステートメント: ハッシュパ
ッケージ” (SAS DS2言語リファレンス)または“_NEW_ 演算子: ハッシュパッケージ” 
(SAS DS2言語リファレンス)を参照してください。次の例では、昇順を指定して重複
を置き換えるには、hashexpパラメーターのプレースホルダーとして -1、datasource
パラメーターのプレースホルダーとして空の単一引用符(' ')、orderedパラメーター
に'a'、 duplicateに'replace'を使用する必要があります。duplicateパラメーターは最
後に指定されたものであるため、sumincおよびmultidataパラメーターにプレース
ホルダーは必要ありません。

declare package hash variable-name(8,'', 'a', 'replace');

詳細については、“コンストラクターを使用したハッシュインスタンスの定義” (195
ページ)、“メソッド呼び出しを使用したハッシュインスタンスの定義” (196ページ)、
および“_NEW_演算子を使用してハッシュインスタンスを作成する” (199ページ)を
参照してください。

_NEW_演算子を使用してハッシュインスタンスを作
成する
DECLARE PACKAGEステートメントを使用してハッシュ変数とハッシュインスタン
スを作成するかわりに、DECLARE PACKAGEステートメントを使用してハッシュ変
数を作成し、_NEW_演算子を使用してハッシュインスタンスを作成することができ
ます。DECLARE PACKAGEステートメントを使用して、ハッシュパッケージ変数を
宣言します。次に、_NEW_演算子を使用してハッシュパッケージのインスタンスを
インスタンス化し、新しくインスタンス化されたハッシュインスタンスを参照する
ようにハッシュ変数を設定します。このシナリオでは、DECLARE PACKAGEステー
トメントを使用して初期化データを提供することはできません。ハッシュインスタ
ンスが完全に定義された状態で構築されていない場合は、_NEW_演算子と後続のメ
ソッド呼び出しを使用して、ハッシュインスタンスの初期化データを提供できます。
次の例の DECLARE PACKAGEステートメントは、変数MYHASHがハッシュパッケ
ージ型であることをコンパイラに伝えます。この時点では、変数MYHASHのみを宣
言しています。ハッシュインスタンスを参照する可能性がありますが、現在は何も
参照していないため、nullパッケージ参照です。ハッシュ変数パッケージは一度だ
け宣言する必要があります。_NEW_演算子は、ハッシュパッケージのインスタンス
を作成し、それを変数MYHASHに割り当てます。
declare package hash myhash();
myhash = _new_ hash(8, 'mytable', 'yes', 'replace', 'sumnum', 'y');

前述のステートメントは、次のコードと同等です。
declare package hash myhash(8, 'mytable', 'yes', 'replace', 'sumnum', 'y');

詳細については、“_NEW_ 演算子: ハッシュパッケージ” (SAS DS2言語リファレンス)
を参照してください。

暗黙の変数と変数リストのメソッド
ハッシュインスタンスを定義するときは、一連のキー変数とデータ変数を指定しま
す。ハッシュインスタンスが完全に定義された後、キー変数とデータ変数は、後続
の操作中に暗黙的または明示的に読み書きできます。

注: ハッシュインスタンスに渡されるすべての変数は、グローバル変数である必要が
あります。

キー変数とデータ変数をハッシュインスタンスに渡すには、暗黙の変数メソッドと
変数リストメソッドの 2つの方法があります。
暗黙変数メソッドは、DATAステップのハッシュオブジェクトインターフェイスに似
ています。暗黙変数メソッドを使用すると、一連の DEFINEKEYおよび

事前定義された DS2パッケージ 199

http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p1cnwa3vwltaqgn1fnowe2jqxl8y.htm&locale=ja


DEFINEDATAメソッド呼び出しと単一の DEFINEDONEメソッド呼び出しによっ
て、キー変数とデータ変数が定義されます。次に、キー変数とデータ変数の定義の
セットは、他のハッシュパッケージメソッドの実行中に暗黙的な引数として使用さ
れます。
次の例では、2つのキー変数 k1と k2、および 2つのデータ変数 d1と d2を持つハ
ッシュテーブルを定義します。FINDメソッドは、暗黙のキー変数の値を読み取り、
ハッシュテーブルでキー値を検索します。キー値が見つかった場合、DS2は対応す
るデータ値をハッシュインスタンスに定義された暗黙的なデータ変数に書き込みま
す。
declare package hash h();
   h.definekey('k1');
   h.definekey('k2');
   h.definedata('d1');
   h.definedata('d2');
   h.definedone();
   
   /* No explicit arguments specify what key values to find */
   /* or what to do with the data values if keys are found. */
   /* Implicitly uses key variables k1 and k2 and           */
   /* data variables d1 and d2.                             */
   h.find();

変数リストメソッドでは、ハッシュメソッドが呼び出されるときに、変数リストで
対象の変数を明示的な引数として指定します。
この例では、変数リストメソッドを使用しており、暗黙変数メソッドを使用する前
述の例と同じです。
declare package hash h();
rc=h.keys([k1 k2]);
rc=h.data([d1 d2]);
rc=h.definedone();
h.find([k1 k2], [d1 d2]);

変数リストメソッドは、暗黙のキー変数とデータ変数以外の変数を柔軟に使用でき
るようにします。次の FINDメソッドは、 xおよび yに格納されている値を検索し
ます。値が見つかった場合、それらは uおよび vに書き込まれます。
 h.find([x y], [u v]);

変数リストの詳細については、“DS2変数リスト” (28ページ)を参照してください。
すべてのハッシュ暗黙変数メソッドは、キーとデータの両方を持つハッシュインス
タンスと、キーのみのハッシュインスタンスで機能します。
変数リストメソッドには、キーのみのハッシュインスタンスでのみ機能するものと、
キーとデータの両方を持つハッシュインスタンスでのみ機能するものがあります。
キーのみのハッシュインスタンスが、キーとデータを持つハッシュインスタンスに
対してのみ機能する変数リストメソッドを呼び出すと、ランタイムエラーが発生し
ます。逆の場合も同様です。キーのみのハッシュインスタンスで機能するメソッド
の詳細については、“DS2ハッシュおよびハッシュ反復子パッケージの属性、メソッ
ド、演算子、ステートメント” (SAS DS2言語リファレンス)の各メソッドを参照して
ください。

一意でないキーとデータのペア
デフォルトでは、ハッシュパッケージ内のすべてのキーは一意です。これは、キー
ごとに 1セットのデータ変数が存在することを意味します。状況によっては、ハッ
シュパッケージに重複キーを含めること、つまり、複数のデータ変数のセットを 1
つのキーに関連付けることが必要になる場合があります。
たとえば、キーが患者 IDで、データが来院日であるとします。患者が複数回来院す
る場合、複数の来院日が患者 IDに関連付けられます。MULTIDATAパラメーターま
たはメソッドを YESに設定してハッシュパッケージを作成すると、データ変数の複
数のセットがキーに関連付けられます。
テーブルに重複キーが含まれている場合、デフォルトでは、最初のインスタンスは
ハッシュパッケージに保存され、後続のインスタンスは無視されます。最後のイン
スタンスをハッシュパッケージに格納するには、DUPLICATEパラメーターまたはメ
ソッドを使用します。DUPLICATEパラメーターまたはメソッドは、重複キーがある
場合にも SASログにエラーを書き込みます。
ただし、MULTIDATAパラメーターまたはメソッドを使用する場合、ハッシュパッケ
ージではキーごとに複数の値を保存できます。ハッシュパッケージは、キーに関連
付けられたリストに複数の値を保持します。このリストは、HAS_NEXTや
FIND_NEXTなどのいくつかのメソッドを使用して走査および操作できます。
複数のデータアイテムリストを走査するには、現在のリストアイテムを知っている
必要があります。まず、特定のキーに対して FINDメソッドを呼び出します。FIND
メソッドは、現在のリストアイテムを設定します。次に、キーに複数のデータ値が
あるかどうかを判断するために、HAS_NEXTメソッドを呼び出します。キーに別の
データ値があることを確認したら、FIND_NEXTメソッドを使用してその値を取得で
きます。FIND_NEXTメソッドは、現在のリストアイテムをリスト内の次のアイテム
に設定し、そのアイテムに対応するデータ変数を設定します。
特定のキーのリストを順方向に移動するだけでなく、HAS_PREVおよび FIND_PREV
メソッドを同様の方法で使用して、リストを逆方向にループできます。

注: 複数データアイテムリストのアイテムは、挿入した順序で維持されます。

MULTIDATAおよび DUPLICATEパラメーターの詳細については、“DECLARE 
PACKAGE ステートメント: ハッシュパッケージ” (SAS DS2言語リファレンス)または
“_NEW_ 演算子: ハッシュパッケージ” (SAS DS2言語リファレンス)を参照してくだ
さい。 MULTIDATAおよび DUPLICATEメソッドの詳細については、“MULTIDATA 
メソッド” (SAS DS2言語リファレンス)および“DUPLICATE メソッド” (SAS DS2言語
リファレンス)を参照してください。

キー要約の維持
SUMINCパラメーターまたはメソッドを使用して、ハッシュパッケージキーの要約
カウントを維持できます。SUMINCは、レコードが FIND、CHECK、または REFメ
ソッドによって使用されるたびに、レコードに要約値を保存できるように、各レコ

ードに内部ストレージを割り当てるようにハッシュパッケージに指示します。
SUMINC値は、FIND、CHECK、または REFメソッドの後にハッシュパラメーター
キーの要約カウントを維持するためにも使用されます。
SUMINCには、合計増分、つまり、キーへの各参照のキー要約に追加する量を保持
する変数が指定されます。SUMINC値は、0より大きい、小さい、または等しい場
合があります。SUMINC値は、ADDメソッドの要約を初期化するためにも使用され
ます。ADDメソッドが発生するたびに、SUMINC値のキーが初期化されます。
SUMメソッドは、キーに 1つのデータアイテムのみ存在する場合、指定されたキー
に対する要約値を取得します。
複数のアイテムが存在する場合、SUMDUPメソッドは、キー要約の現在の値を取得
します。
キー要約は、DATASOURCEパラメーターまたは DATAメソッドと組み合わせて使用
できます。DEFINEDONEメソッドまたは DECLARE PACKAGEステートメントを使
用してテーブルがハッシュパッケージに読み込まれると、すべてのキー要約が
SUMINC値に設定され、後続のすべての FIND、CHECK、または ADDメソッドによ
って、対応するキー要約が変更されます。
SUMINCパラメーターの詳細については、“DECLARE PACKAGE ステートメント: ハ
ッシュパッケージ” (SAS DS2言語リファレンス)を参照してください。SUMINCおよ
び SUMDUPメソッドの詳細については、“SUMINC メソッド” (SAS DS2言語リファ
レンス)および“SUMDUP メソッド” (SAS DS2言語リファレンス)を参照してくださ
い。

データの格納と取得
ハッシュパッケージのキー変数とデータ変数を初期化したら、ADDメソッドを使用
してハッシュパッケージにデータを格納できます。または、DATASOURCEパラメー
ターまたは DATASETメソッドを使用して、テーブルをハッシュパッケージにロード
することもできます。DATASOURCEパラメーターまたは DATASETメソッドを使用
し、テーブルに同じキー値を持つ複数の行が含まれている場合、SASはデフォルト
でハッシュテーブルの最初の行を保持し、後続の行を無視します。最後のインスタ
ンスをハッシュパッケージに格納するか、重複キーがある場合にログにエラーを送
信するには、DUPLICATEパラメーターまたはメソッドを使用します。各キーの重複
値を許可するには、MULTIDATAパラメーターまたはメソッドを使用します。
その後、FINDメソッドを使用して、ハッシュパッケージからデータを検索および取
得できます。各キーに複数のデータアイテムが存在する場合は、FIND_NEXTおよび
FIND_PREVメソッドを使用してデータを検索および取得します。
詳細については、“ADD メソッド: ハッシュパッケージ” (SAS DS2言語リファレンス)、
“FIND メソッド” (SAS DS2言語リファレンス)、“FIND_NEXT メソッド” (SAS DS2言語
リファレンス)、“FIND_PREV メソッド” (SAS DS2言語リファレンス)を参照してくだ
さい。
REFメソッドを使用して、FINDメソッドと ADDメソッドを統合できます。次の例
では、このコードの量を減らすことができます。
rc = h.find();
   if (rc != 0) then
      rc = h.add();

これを単一のメソッド呼び出しにします。
rc = h.ref();

詳細については、“REF メソッド” (SAS DS2言語リファレンス)を参照してください。

注: また、ハッシュ反復子パッケージを使用して、ハッシュパッケージデータを一度
に 1データアイテムずつ正順および逆順で取得することもできます。ハッシュ反復
子パッケージの詳細については、“ハッシュ反復子パッケージの使用” (209ページ)を
参照してください。

データの置換と削除
次のいずれかの方法を使用して、ハッシュパッケージ内のデータを削除または置換
できます。
n 指定したキーからデータアイテムを削除するには、REMOVEメソッドを使用しま
す。

n すべてのデータアイテムを削除するには、REMOVEALLメソッドを使用します。
n 複数のデータアイテムを持つキーのデータアイテムを削除するには、

REMOVEDUPメソッドを使用します。
n すべてのデータアイテムを置き換えるには、REPLACEメソッドを使用します。
n 現在のデータアイテムのみを置き換えるには、REPLACEDUPメソッドを使用し
ます。

注: 関連付けられたハッシュ反復子がキーを指している場合、REMOVEメソッドは
キーまたはデータをハッシュパッケージから削除しません。エラーメッセージがロ
グに発行されます。

詳細については、“REMOVE メソッド” (SAS DS2言語リファレンス)、“REMOVEALL メ
ソッド” (SAS DS2言語リファレンス)、“REMOVEDUP メソッド” (SAS DS2言語リファ
レンス)、“REPLACE メソッド” (SAS DS2言語リファレンス)、“REPLACEDUP メソッ
ド” (SAS DS2言語リファレンス)を参照してください。

ハッシュパッケージデータをテーブルに保存する
OUTPUTメソッドを使用して、指定したハッシュパッケージ内のデータを含むテー
ブルを作成できます。
次の例では、最初のテーブルプログラムが data1テーブルを生成します。2番目の
テーブルプログラムは、1つのキー kと 2つのデータ d1および d2を含むハッシュ
パッケージ hを作成します。次に、data1テーブルの各行が読み取られ、キーとデ
ータがハッシュパッケージ hに追加されます。最後に、ハッシュパッケージ hに格
納されているデータ値が out1テーブルに書き込まれます。3番目のテーブルプロ
グラムは、out1テーブルの内容を書き込みます。

  /* Generate and output 5 rows for table data1. */
  data data1(overwrite=yes); 
    declare double k d1 d2;
    method init(); 
      declare int i;
      do i = 1 to 5;
        k = i; d1 = i*10; d2 = i*2; output;
      end; 
    end; 
  enddata; 
  run;
  
  data _null_;
    declare double k d1 d2;
    declare package hash h(0, '', 'descending'); 

    /* Define key and data variables for hash h. */
    method init(); 
      h.defineKey('k');
      h.defineData('d1');
      h.defineData('d2'); 
      h.defineDone();
    end;
 
    /* Read rows from table data1.
     * Add key and data values from rows to hash h. */
    method run();
      set data1;
      h.add();
    end;

    /* Add additional key and data values to hash h.
     * Output hash h to table out1. */
    method term();
      k = 11; d1 = 110; d2 = 22; h.add(); 
      k = 12; d1 = 120; d2 = 24; h.add();
      k = 13; d1 = 130; d2 = 26; h.add(); 
      k = 14; d1 = 140; d2 = 28; h.add(); 
      h.output('out1');
    end; 

  enddata; 
  run;
  
  /* Outputs rows from table out1. */
  data; 
    method run(); 
      set out1; 
    end; 
  enddata;
  run;

アウトプット 13.2 テーブル OUT1からの出力行

ハッシュパッケージキーは、出力テーブルの一部として格納されないことに注意し
てください。出力テーブルにキーを含める場合は、DEFINEDATAメソッドでキーを
データとして定義する必要があります。前の例では、DEFINEDATAメソッドは次の
ように記述されます。
h.defineKey('k');
h.defineData('k');
h.defineData('d1');
h.defineData('d2');

前述の変更により、次の行が出力されます。

アウトプット 13.3 キーとデータを含むテーブル OUT1からの出力行

FedSQLクエリとハッシュインスタンスを使用して
実行時に動的に行を取得する
注: FedSQLクエリの使用は、CASサーバーではサポートされていません。

ハッシュインスタンスを使用すると、テーブルから取得する行の決定を実行時まで
遅らせることができます。実行時に、ハッシュインスタンスが作成され、選択され
た行がロードされます。行は、データソースに指定された FedSQLクエリに基づい
て選択されます。ハッシュ反復子を使用して行をループし、行データにアクセスで
きます。
次の例では、SELECT * FROM test WHERE a=1 FedSQLクエリによって行が選択されて
います。このクエリは、それらをハッシュパッケージにロードする executeメソッ
ドに渡されます。ハッシュ反復子は行をループし、選択された行を SASログと
resultテーブルに書き込みます。
data test;
   a=1;  b=2;output;
   a=11; b=22;output;
   a=1;  b=3;output;
   a=22; b=44;output;
run;

proc ds2;
package pkg /overwrite=yes;
   dcl double a b; 

   method execute(char(200) sql);
      dcl package hash  h();
      dcl package hiter hi(h);
      h.keys([a]);
      h.data([a b]);
      h.multidata('yes');
      h.dataset(sql);
      h.definedone();

      put 'SELECTED ROWS:';
      rc = hi.first();
      do while(rc = 0);
         put a= b=;
         rc = hi.next();
      end;

      h.output('result');
   end; 
endpackage;
run; quit;

proc ds2;
data _null_;
   method init();
      declare package pkg p1();
      p1.execute('{SELECT * FROM test WHERE a=1}');
   end;

enddata;
run; quit;

proc print data=test;
   title2 'TEST TABLE';
run; quit;

proc print data=result;
   title2 'RESULT TABLE';
run; quit;

次の行が SASログに書き込まれます。

SELECTED ROWS:
a=1 b=2
a=1 b=3

入力テーブルと出力テーブルは次のとおりです。

アウトプット 13.4 入力テーブル

アウトプット 13.5 出力テーブル

ハッシュパッケージ属性の使用
ハッシュパッケージで使用できる属性は 2つあります。NUM_ITEMSはハッシュパ
ッケージ内のアイテムの数を返し、ITEM_SIZEはアイテムのサイズ (バイト単位)を返
します。
次の例では、ハッシュパッケージ内のアイテムの数を取得します。
num_items = myhash.num_items;  

次の例では、ハッシュパッケージ内のアイテムのサイズを取得します。
item_size = myhash.item_size; 

ITEM_SIZEおよび NUM_ITEMS属性で、ハッシュパッケージが使用しているメモリ
量を把握できます。ITEM_SIZE属性は、ハッシュパッケージに必要な初期オーバー
ヘッドを反映したものではなく、必要な内部調整も考慮されていません。したがっ
て、ITEM_SIZEを使用しても正確なメモリ使用量はわかりませんが、適切な概算値
が得られます。詳細については、“ITEM_SIZE 属性” (SAS DS2言語リファレンス)およ
び“NUM_ITEMS 属性” (SAS DS2言語リファレンス)を参照してください。

ハッシュ反復子パッケージの使用
ハッシュ反復子パッケージを使用して、一意のルックアップキーに基づいてデータ
を格納および検索します。ハッシュ反復子パッケージを使用すると、ハッシュパッ
ケージデータを順方向または逆方向のキー順で取得できます。
DECLARE PACKAGEステートメントを使用して、ハッシュ反復子パッケージを宣言
します。新しいハッシュ反復子パッケージを宣言した後、ハッシュパッケージ名を
パラメーターとして、_NEW_演算子を使用してパッケージをインスタンス化します。
次に例を示します。
declare package hiter myiter;
myiter = _new_ hiter('h');

DECLARE PACKAGEステートメントは、変数MYITERがハッシュ反復子型であるこ
とをコンパイラに伝えます。この時点では、変数MYITERのみを宣言しています。
ハッシュ反復子インスタンスを参照する可能性がありますが、現在は何も参照して
いないため、nullパッケージ参照です。ハッシュ反復子パッケージ変数は一度だけ
宣言する必要があります。_NEW_演算子は、ハッシュ反復子パッケージのインスタ
ンスを構築し、それを変数MYITERに割り当てます。ハッシュパッケージ Hは、コ
ンストラクターパラメーターとして渡されます。
DECLARE PACKAGEと_NEW_演算子を使用してハッシュ反復子パッケージを宣言お
よびインスタンス化する 2ステップのプロセスのかわりに、コンストラクターメソ
ッドとして DECLARE PACKAGEステートメントを使用することにより、パッケージ
を 1ステップで宣言およびインスタンス化できます。これは、DECLARE PACKAGE
ステートメントのみを使用した同じ例です。
declare package hiter myiter('h');

詳細については、“DECLARE PACKAGE ステートメント: ハッシュ反復子パッケージ” 
(SAS DS2言語リファレンス)、および“_NEW_ 演算子: ハッシュ反復子パッケージ” 
(SAS DS2言語リファレンス)を参照してください。

注: ハッシュ反復子パッケージを作成する前に、ハッシュパッケージを宣言してイン
スタンス化する必要があります。

HTTPパッケージの使用
HTTPパッケージの概要
HTTPパッケージを使用して、HTTP Webサーバーにアクセスするための HTTPクラ
イアントを構築します。
一般的なタスクは次のとおりです。

1 HTTPパッケージを宣言してインスタンス化します。
2 HTTP GET、HEAD、または POSTメソッドを作成します。
追加の HTTPパッケージメソッドを使用して、ヘッダー情報を追加し、要求デー
タを送信できます。

3 HTTP GET、HEAD、または POSTメソッドを実行します。
4 Webサーバーから応答情報を取得します。
n 完全なエンティティとして、またはストリーミングによる応答本文
n 応答のコンテンツの種類
n 応答ヘッダー

HTTPパッケージを使用すると、次のタスクも実行できます。
n HTTP応答からステータスコードを取得します。
n ソケットのタイムアウト値を設定します。
n クライアント HTTPインスタンスが接続する URLまたはプロキシ URLを指定し
ます。

n URLまたはプロキシ URLで必要な場合は、ユーザー名とパスワードを指定しま
す。

n 要求本文のエンコード時または応答本文のデコード時に使用する文字セットを
指定します。

n 要求に OpenAuthorization (OAuth)アクセストークンを指定するか、SAS環境で
トークンを検索します。

n SASログ機能を使用して、HTTPクライアントとサーバー間の HTTPトラフィッ
クをログに記録します。

HTTPパッケージの宣言とインスタンス化
まず、HTTPパッケージを宣言してインスタンス化する必要があります。HTTPパッ
ケージのインスタンスを構築する方法は 2つあります。
n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。

declare package http httpclt;
httpclt = _new_ http();

n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package http httpclt();

詳細については、“DECLARE PACKAGE ステートメント: HTTPパッケージ” (SAS DS2
言語リファレンス)および“_NEW_ 演算子: HTTPパッケージ” (SAS DS2言語リファレ
ンス)を参照してください。

ヒント Webサービスアプリケーションでは、HTTPトラフィックを同期的
に処理するために必要な HTTPクライアントが 1つだけの場合があります。
または、アプリケーションで必要な場合は、複数の HTTPクライアントをイ
ンスタンス化して、データを非同期的に要求および処理することができます。

HTTP GET、HEAD、または POSTメソッドの作成
1 CREATEGETMETHOD、CREATEHEADMETHOD、または CREATEPOSTMETHOD
メソッドを使用して、GET、HEAD、および POSTメソッドを作成します。
詳細については、“CREATEGETMETHOD メソッド” (SAS DS2言語リファレンス)、
“CREATEHEADMETHOD メソッド” (SAS DS2言語リファレンス)、および
“CREATEPOSTMETHOD メソッド” (SAS DS2言語リファレンス)を参照してくだ
さい。

2 (オプション) HTTP GETメソッドにヘッダーを追加するには、
ADDREQUESTHEADERメソッドを使用します。
詳細については、“ADDREQUESTHEADER メソッド” (SAS DS2言語リファレンス)
を参照してください。

3 (オプション) 要求本文を HTTPメソッドに追加するには、
SETREQUESTBODYASBINARYまたは SETREQUESTBODYASSTRINGメソッドを
使用します。
詳細については、“SETREQUESTBODYASBINARY メソッド” (SAS DS2言語リファ
レンス)および“SETREQUESTBODYASSTRING メソッド” (SAS DS2言語リファレ
ンス)を参照してください。

4 (オプション) HTTPメソッドで要求コンテンツの種類を指定するには、
SETREQUESTCONTENTTYPEメソッドを使用します。
詳細については、“SETREQUESTCONTENTTYPE メソッド” (SAS DS2言語リファレ
ンス)を参照してください。

HTTP GET、HEAD、および POSTメソッドの実行
HTTP GET、HEAD、および POSTメソッドを実行すると、要求が HTTP Webサーバ
ーに送信されます。

注: HTTP GET、HEAD、および POSTメソッドを実行する前に、HTTP GET、HEAD、
または POSTメソッドを作成する必要があります。詳細については、“HTTP GET、
HEAD、または POSTメソッドの作成” (211ページ)を参照してください。

ほとんどの HTTPメソッドでは、EXECUTEMETHODメソッドを使用して要求を
Webサーバーに送信します。詳細については、“EXECUTEMETHOD メソッド” (SAS 
DS2言語リファレンス)を参照してください。

EXECUTEMETHODメソッドは、応答本文のストリーミングをサポートしていませ
ん。応答本文をストリーミングする場合は、別の実行メソッド
EXECUTEMETHODSTREAMが必要です。詳細については、“HTTPリソースの取得” 
(213ページ)を参照してください。
別の要求を送信するには、GET、HEAD、または POST要求の作成から始まり、
EXECUTEMETHODまたは EXECUTEMETHODSTREAMメソッドで終わるプロセス
を繰り返します。
このプログラム例は、HTTPパッケージ(クライアント)をインスタンス化し、HTTP 
GETメソッドを作成し、GETメソッドを実行して HTTP Webサービスに要求を送信
し、HTTP Webサービスからの応答から本文情報を文字列として取得します。
data _null_;
  method run();
    /* instantiate the package */
    declare package http h();
    declare varchar(1024) character set utf8 body;
    declare int rc;
 
    /* create a GET */
    h.createGetMethod('http://api.worldbank.org/countries/fr/');
    /* execute the GET */
    h.executeMethod();
    /* retrieve the response body as a string */
    h.getResponseBodyAsString(body, rc);
    put body;
  end;

次の行が SASログに書き込まれます。

<?xml version="1.0" encoding="utf-8"?>
<wb:countries page="1" pages="1" per_page="50" total="1"
xmlns:wb="http://www.worldbank.org">
  <wb:country id="FRA">

<wb:iso2Code>FR</wb:iso2Code>
    <wb:name>France</wb:name>
    <wb:region id="ECS">Europe
&amp; Central Asia (all income levels)</wb:region>
    <wb:adminregion id="" />

<wb:incomeLevel id="OEC">High income: OECD</wb:incomeLevel>
    <wb:lendingType id="LNX">Not
classified</wb:lendingType>
    <wb:capitalCity>Paris</wb:capitalCity>

<wb:longitude>2.35097</wb:longitude>
    <wb:latitude>48.8566</wb:latitude>

</wb:country>
</wb:countries>

HTTPリソースの取得
次の HTTPパッケージメソッドを使用して、HTTPリソースからヘッダー、応答本
文、および応答本文の種類を取得できます。応答本文を取得する場合は、1つのエ
ンティティとして取得するか、応答本文をストリーミングできます。

タスク HTTPパッケージメソッド
ヘッダーを取得す
る

“GETRESPONSEHEADERSASSTRING メソッド” (SAS DS2言語リ
ファレンス)

応答本文を 1つの
エンティティとし
て取得する 1

“GETRESPONSEBODYASBINARY メソッド” (SAS DS2言語リフ
ァレンス)

“GETRESPONSEBODYASSTRING メソッド” (SAS DS2言語リフ
ァレンス)

応答本文をストリ
ーミングする 1

“STREAMRESPONSEBODYASBINARY メソッド” (SAS DS2言語
リファレンス)

“STREAMRESPONSEBODYASSTRING メソッド” (SAS DS2言語
リファレンス)

応答本文のコンテ
ンツの種類を取得
する

“GETRESPONSECONTENTTYPE メソッド” (SAS DS2言語リファ
レンス)

1 応答本文を 1つのエンティティとして取得する場合は、最初に GETメソッドを作成し、そのメソッドを EXECUTEMETHODメソッドで実行する必要があります。応答本文をストリーミングする場合は、最初に GETメソッドを作成し、そのメソッドを EXECUTESTREAMMETHODメソッドで実行する必要があります。

HTTPパッケージを使用する際の考慮事項
n HTTPパッケージは、GET、HEAD、および POST HTTPメソッドのみをサポート
します。

n 各クライアントは要求を送信し、応答を同期的に処理します。アプリケーション
は、複数のクライアントを作成して、HTTPリソースに対してアクションを非同
期に実行できます。

n HTTPパッケージは、要求のデータと応答のデータを DS2文字列値またはバイナ
リ値としてメモリに格納します。データをファイルとしてディスクに格納する
場合は、HTTPプロシジャの使用を検討してください。詳細については、“HTTP 
プロシジャ” (Base SASプロシジャガイド)を参照してください。

n HTTPパッケージは、認証が必要なセキュア HTTPエンドポイントに要求を送信
できます。HTTPエンドポイントがクライアント認証を必要とする場合、サポー
トされている認証メカニズムのリストでクライアントに応答します。HTTPパッ

ケージは現在、3つの最も一般的な認証メカニズムのうちの 2つ、基本とネゴシ
エートをサポートしています。基本認証自体は資格情報の機密性を提供しない
ため、トランスポート層セキュリティ(TLS)を介してデータが暗号化されている場
合にのみ使用する必要があります。SASが TLSを検証する方法の詳細について
は、Encryption in SASを参照してください。ネゴシエート認証は Kerberosをサ
ポートし、Windowsでは NT LAN Manager (NTLM)をサポートします。

オープン認証
Open Authorization (OAuth)は、インターネット上でのトークンベースの認証と権
限のためのオープンスタンダードです。OAuthは特に HTTPで動作するように設計
されており、基本的にリソース所有者の承認を得て、権限サーバーによってサード
パーティクライアントにアクセストークンを発行できるようにします。そして、サ
ードパーティクライアントはアクセストークンを使用して、リソースサーバーによ
ってホストされている保護されたリソースにアクセスします。
HTTPパッケージの場合、SETAUTHTOKENメソッドを使用してトークンを提供する
か、ADDSASOAUTHTOKENメソッドを使用して、SASが OAuthアクセストークン
の環境を検索できるようにすることができます。どちらの場合も、OAuthアクセス
トークンは Authorizationヘッダーフィールドの Bearer値として要求ヘッダーに
追加されます。
詳細については、“SETOAUTHTOKEN メソッド” (SAS DS2言語リファレンス)および
“ADDSASOAUTHTOKEN メソッド” (SAS DS2言語リファレンス)を参照してくださ
い。

HTTPトラフィックのログ記録
HTTPパッケージは、SASログ機能によるログをサポートしています。
App.TableServices.d2pkg.HTTPロガーは、HTTPクライアントと Webサーバーと
の間で送受信されるエラー、ヘッダー、およびデータをログに記録します。
詳細については、“HTTPパッケージロガー” (326ページ)を参照してください。

JSONパッケージの使用
JSONパッケージの概要
Java Script Object Notation (JSON)は、人間が判読できるデータ交換用に設計され
た、テキストベースのオープンスタンダードデータ形式です。JSONは JavaScriptプ
ログラミング言語のサブセットに基づいており、JavaScript構文を使用してデータ
オブジェクトを記述します。

JSONパッケージは、JSONテキストを作成および解析するためのインターフェイス
を提供します。JSONパッケージのWriteメソッドは書き込み要求をメモリに蓄積
し、テキストを取得できます。JSONパッケージパーサーを使用すると、テキストを
読み取って解析できます。

JSONパッケージの宣言とインスタンス化
JSONパッケージのインスタンスを構築する方法は 2つあります。
n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。

 declare package json myjsonpkg;
 myjsonpkg = _new_ json();

n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package json myjsonpkg();

詳細については、“DECLARE PACKAGE ステートメント: JSONパッケージ” (SAS DS2
言語リファレンス)、および “_NEW_ 演算子: JSONパッケージ” (SAS DS2言語リファ
レンス)を参照してください。

JSONテキストの書き込み
JSONテキストを作成するには、CREATEWRITERメソッドを使用して JSONライタ
ーインスタンスを作成します。
JSON出力は、次の 2種類のデータ構造コンテナーで構成されます。
JSONオブジェクトコンテナー({ })
左中かっこ({)で始まり、右中かっこ(})で終わります。オブジェクトコンテナー
は、名前と値のペアとして書き込まれる名前と値のペアを収集します。値は、サ
ポートされている JSONデータ型、オブジェクト、または配列のいずれかです。
各名前の後にはコロンと値が続きます。名前と値のペアはカンマで区切られま
す。
WRITEOBJOPENおよびWRITECLOSEメソッドを使用して、オブジェクトコンテ
ナーを作成します。

JSON配列コンテナー([])
左角かっこ([)で始まり、右角かっこ(])で終わります。配列コンテナーは、名前の
ない値のリストとして書き込まれる値のリストを収集します。値は、サポートさ
れている JSONデータ型、オブジェクト、または配列のいずれかです。値はカン
マで区切ります。
WRITEARRAYOPENおよびWRITECLOSEメソッドを使用して、配列コンテナー
を作成します。

トップレベルのコンテナーには、任意の数のコンテナーを含めることができます。
同様に、コンテナーはコンテナーを任意の深さにネストできます。コンテナーをネ
ストするときは、現在のコンテナーのデータ構造要件を守るように注意してくださ
い。

n オブジェクトには、名前と値のペアのリストが必要です。値自体はオブジェクト
または配列にすることができます。

n 配列には、名前と値のペアのような構造上の要件はなく、単なる値、オブジェク
ト、または配列のリストです。

JSONパッケージメソッドを使用すると、文字値と数値、null値、ブールの真偽値を
書き込むことができます。
JSONテキスト出力の例を次に示します。
{"SASJSONExport":"1.0","SASTableData+CLASS":[{"Name":"Joyce","Sex":
"F","Age":11,"Height":51.3,"Weight":50.5},{"Name":"Thomas","Sex":"M",
"Age":11,"Height":57.5,"Weight":85}]}

WRITERGETTEXTメソッドを使用して、ライターによって生成された JSONテキス
トを取得できます。
テキストの書き込みが終了したら、DESTROYWRITERメソッドを呼び出してライタ
ーインスタンスを削除します。
次の例では、JSONテキストを作成し、SASログに書き込みます。
data _null_;
    method init();
        dcl package json j();
        dcl double dblVal;
        dcl int rc;
        dcl nvarchar(30) jsontxt;

        rc = j.createWriter();
             if rc=0 then rc = j.writeArrayOpen();
        if rc=0 then rc = j.writeString( '     Hello World!     ' );
        if rc=0 then rc = j.writeClose();
        j.writerGetText( rc, jsontxt );
        put rc= jsontxt=;
    end;
enddata;
run;

次の行が SASログに書き込まれます。

rc=0 jsontxt=[" Hello World! "]

JSONテキストの書き込みを可能にするメソッドの詳細については、“DS2 JSONパッ
ケージのメソッド、演算子、およびステートメント” (SAS DS2言語リファレンス)を
参照してください。

JSONテキストの解析
JSONテキストを解析または読み取るには、CREATEPARSERメソッドを使用して
JSONライターインスタンスを作成します。CREATEPARSERメソッド、
SETPARSERINPUTメソッド、またはその両方を使用して、入力 JSONテキストをパ
ーサーに提供できます。

GETNEXTTOKENメソッドは、JSONテキストから次の検証 JSON言語要素を返すた
めに使用されます。GETNEXTOKENメソッドは、トークンタイプ、解析フラグ、行
番号、列番号も返すことができます。JSONパッケージの IS*メソッド(ISLEFTBRACE
など)を使用すると、次のトークンタイプをクエリできます。
ブール真
ブール偽
浮動小数点数
整数
ラベル
左中かっこ( { )
左角かっこ( [ )
null
数値
部分
右中かっこ( } )
右角かっこ( ] )
文字列

テキストの解析が終了したら、DESTROYPARSERメソッドを呼び出してパーサーイ
ンスタンスを削除します。
次の例では、パーサーを作成し、CREATEPARSERメソッドを使用して入力 JSONテ
キストを提供します。
data _null_;
    method init();
        dcl package json j();
        dcl int rc tokenType parseFlags;
        dcl bigint lineNum colNum;
        dcl nvarchar(128) token abc t1;
        abc = 'xyz';
        t1 = '{"abc" : 1 }';
        rc = j.createParser( t1 );
             if (rc ne 0) then goto TestError;

        * obj open;
        j.getNextToken( rc, token, tokenType, parseFlags, lineNum, colNum );
        if ( rc ne 0 ) then goto TestError;

        * obj label; 
        j.getNextToken( rc, token, tokenType, parseFlags, lineNum, colNum );
        if ( rc ne 0 ) then goto TestError;

        * obj value; 
        j.getNextToken( rc, token, tokenType, parseFlags, lineNum, colNum );
        if ( rc  ne 0) then goto TestError;

        * obj close; 
        j.getNextToken( rc, token, tokenType, parseFlags, lineNum, colNum );
        if ( rc ne 0 ) then goto TestError;

     Exit:

        rc = j.destroyParser();
        return;

    TestError:
        put 'Test ended abnormally.';
        goto Exit;

    end;
enddata;
run;

JSONテキストの解析を可能にするメソッドの詳細については、“DS2 JSONパッケー
ジのメソッド、演算子、およびステートメント” (SAS DS2言語リファレンス)を参照
してください。

JSONパッケージロガー
JSONパッケージは、SASログ機能によるログをサポートしています。
DS2 JSONパッケージは、内部 DS2 JSONパッケージの DEBUGおよび TRACE情報
を収集するためのロガー App.tk.DS2PKG.JSONを提供します。JSONパッケージを
使用する DS2プログラムに TRACEステートメントを簡単に追加できない場合は、
App.tk.DS2PKG.JSONロガーレベルを DEBUGまたは TRACEに設定します。
詳細については、“JSONパッケージロガー” (326ページ)を参照してください。

ロガーパッケージの使用
ロガーパッケージの概要
SASログ機能では、ロガーはメッセージカテゴリを識別する名前付きエンティティ
です。ロガーの属性は、レベルと、メッセージカテゴリのログイベントを処理する
1つ以上のアペンダーで構成されます。レベルは、このメッセージカテゴリに対し
て処理されるしきい値、または最低イベントレベルを示します。
ロガーパッケージを使用して、SASログ機能とのインターフェイスを取ります。新
しいロガーパッケージを宣言した後、指定されたログレベルでロガーにメッセージ
を送信できます。

注: SAS Enterprise Guideセッションからロガーパッケージを使用することはでき
ません。

DS2ロガーと SASログ機能の詳細については、付録 2, “DS2ロガー” (323ページ)お
よび SAS Viyaログ機能: 構成とリファレンスを参照してください。

ロガーパッケージの宣言とインスタンス化
ロガーパッケージのインスタンスを構築する方法は 2つあります。
n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。

 declare package logger logpkg;
 logpkg = _new_ logger();

n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package logger logpkg();

詳細については、“DECLARE PACKAGE ステートメント: ロガーパッケージ” (SAS DS2
言語リファレンス)、および “_NEW_ 演算子: ロガーパッケージ” (SAS DS2言語リファ
レンス)を参照してください。

未フォーマットメッセージとフォーマット済みメッ
セージ
SASログ機能に直接書き込まれる生(未フォーマット)のメッセージを指定できます。
$形式マーカーを使用して、フォーマット済みメッセージを作成することもできま
す。メッセージ形式の各$s形式マーカーは、指定した対応する引数の内容に置き換
えられます。
詳細については、“LOG メソッド: ロガーパッケージ” (SAS DS2言語リファレンス)を
参照してください。

メッセージをテーブルにログ記録する
1 メッセージを格納する空のテーブルを作成します。
次に例を示します。
libname logs 'library-path';
data logs.edmlog;
  length seqno  8;
  length date   8; format date DATETIME19.;
  length msg    $ 256;
  stop;
run;

2 XMLログ構成ファイルを使用して、ロガーとアペンダーを作成します。
App.Program.EDMという名前の App.Programロガーを作成する例を次に示し
ます。ロガーが書き込むアペンダーは EDMApenderです。アペンダーは、library-
pathの SASデータセット名 edmlogに書き込むように構成されています。この
XMLログ構成ファイルを edm-l4s.xmlとして保存します。

事前定義された DS2パッケージ 219

http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p0rmd437k1uf13n1f8ju8ytefwcu.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p0rmd437k1uf13n1f8ju8ytefwcu.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n1hd9iiweigqyun1c9rf4dsk3ytb.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n1hd9iiweigqyun1c9rf4dsk3ytb.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=n12g87ixybensqn1btjel9a1avq3.htm&locale=ja


<?xml version="1.0"?>
<logging:configuration xmlns:logging="http://www.sas.com/xml/logging/1.0/">
 
   <appender name="EDMAppender" class="DBAppender">
      <param name="ConnectionString" value="
         DRIVER=base;CATALOG=base;
         schema=(name=base;primarypath='library-path')"/>
      <param name="MaxBufferedEvents" value="1000"/>
      <param name="TableName" value="edmlog"/>
      <param name="Column" value="sn"/> 
      <param name="Column" value="d"/> 
      <param name="Column" value="m"/>
   </appender>
 
   <appender name="null" class="ConsoleAppender">
      <filter class="DenyAllFilter"/>
   </appender>
 
   <logger name="App.Program.EDM">
      <appender-ref ref="EDMAppender"/>
   </logger>
 
   <root>
      <appender-ref ref="null"/>
   </root>
</logging:configuration>

3 ステップ 2で作成した XML構成ファイルの名前に LOGCONFIGLOCシステムオ
プションを設定して、SASを起動します。
次に例を示します。
options logconfigloc edm-l4s.xml 

4 ステップ 2の XML構成ファイルで作成したロガーに関連付けられた DS2ロガ
ーパッケージインスタンスを作成します。
次に例を示します。
data _null_;
  dcl package logger l('App.Program.EDM');
  method init();
     dcl double i;
     i = l.islevelactive(4);  put i=;
     l.log('T', 'Hello World! Trace');
     l.log('D', 'Hello World! Debug');
     l.log('I', 'Hello World! Info');
     l.log('W', 'Hello World! Warning');
     l.log('E', 'Hello World! Error');
  end;
enddata;
run;

edmlogデータセットの内容を出力できます。
libname logs 'library-path';
proc print data=logs.edmlog; run;

データセットの内容はこのようになります。

Obs    seqno                   date            msg
 
          1      285      25APR2013:17:11:42    Hello World! Trace  
          2      286      25APR2013:17:11:42    Hello World! Debug  
          3      287      25APR2013:17:11:42    Hello World! Info   
          4      288      25APR2013:17:11:42    Hello World! Warning
          5      289      25APR2013:17:11:42    Hello World! Error

SASログ機能の詳細については、SAS Viyaログ機能: 構成とリファレンスを参照して
ください。

MATRIXパッケージの使用
MATRIXパッケージの概要
行列は、数値または文字値の 2次元配列です。行列の次元は、行と列の数によって
定義されます。n x p行列の要素は、n行 p列に配置されます。
行列パッケージは、SAS/IML機能の DS2レベルの実装を提供します。行列パッケー
ジメソッドを使用して、逆行列計算などの複雑なタスクを実行できます。算術演算、
関係演算、および論理演算を実行できます。ある演算は要素ごとに、他の演算はデ
ータ行列全体で実行することができます。
配列または外部データを使用して、行列パッケージにデータをロードできます。行
列全体を一度に、または行ごとに配列に書き込むことで、データを生成できます。
その後、配列を結果テーブルに書き込むことができます。
これらのメソッドの詳細については、“DS2行列パッケージのメソッド、演算子、お
よびステートメント” (SAS DS2言語リファレンス)を参照してください。

MATRIXパッケージの宣言とインスタンス化
行列パッケージは、DS2 DECLARE PACKAGEステートメントを使用して行列パッケ
ージを宣言およびインスタンス化することによって作成されます。次に例を示しま
す。
declare package matrix m(2, 2);

このステートメントは、2x2行列を作成し、そのインスタンスを変数mに格納しま
す。

注: 行列は、null値または欠損値ではなく、ゼロで埋められます。

詳細については、“DECLARE PACKAGE ステートメント: 行列パッケージ” (SAS DS2言
語リファレンス)を参照してください。

行列データ入力
データの初期化または行列パッケージへのロードには、2つの方法があります。
n ゼロ値で初期化する

o declare package matrix package-variable(rows,columns);

o package-variable=_new_ matrix(rows,columns);

n 配列値で初期化する
o /* loads an array using the array name */

package-variable=_new_ matrix(array-name, rows, columns);

o /* loads an array using the IN method */

package-variable=_new_ matrix(rows,columns);
 package-variable.in(array-name);

o /* loads row-by-row using an input table*/
/* a variable array, and the SET statement */ 

package-variable=_new_ matrix(rows,columns);
. . .
set table-name; 
package-variable.in(variable-array-name, i);

これらの例では、ゼロ値で初期化される 2x2行列を作成します。
declare package matrix m(2,2);

m=_new_ matrix(2, 2);

これらの例では、配列 aを行列mにロードします。
/* simple array */
method init();
   dcl double a[3, 3];
   dcl package matrix m;

   a :=(1, 2, -1, 2, 1, 0, -1, 1, 2);
   m=_new_ matrix(a, 3, 3);
end;

/* variable array */
vararray double a[3,3];
dcl package matrix m;

method init();
   a := (1, 2, -1, 2, 1, 0, -1, 1, 2);
   m = _new_ matrix(a, 3, 3);
end;

これらの例では、INメソッドを使用して配列 vaを行列mにロードします。
/* simple array */

dcl double va[3,3];
dcl package matrix m;
m = _new matrix(3,3);
m.in(va);

/* variable array */
vararray double va[3,3];
dcl package matrix m;

m.in(va); */

この例では、SETステートメントを使用してテーブル xを変数配列 aに読み込みま
す。その変数配列は、行列mをロードするために使用されます。
vararray double a[4];
dcl package matrix m;

/* Create an empty matrix to hold the input values */
method init();
   m = _new_ [this] matrix(4, 4);
   i = 1;
end;

/* Read and initialize each row of matrix from vararray a */
method run();
    set x;
    m.in(a, i);
    i + 1;
end;

詳細については、“_NEW_ 演算子: 行列パッケージ” (SAS DS2言語リファレンス)およ
び“IN メソッド” (SAS DS2言語リファレンス)を参照してください。

注: _NEW_演算子はコードストリームの一部です。宣言では使用されません。

行列データ出力
行列全体を一度に、または行ごとに配列に書き込むことで、行列データを生成でき
ます。
TOARRAYまたは TOVARARRAYメソッドを使用して、一度に行列を配列に書き込み
ます。この例では、配列 aが行列mにロードされています。次に、行列mの転置が
計算され、配列 cに書き込まれます。
dcl double a[3,3];
dcl double b[3,3];

method run();
   dcl package matrix m r;
     dcl double i j;

   a := (1,2,3,4,5,6,7,8,9);

   m = _new_ matrix(a, 3, 3);
   r = m.trans();
   r.toarray(b);

   do i = 1 to 3;
      do j = 1 to 3;
         put b[i,j];
      end;
   end;
end;

詳細については、“TOARRAY メソッド” (SAS DS2言語リファレンス)および
“TOVARARRAY メソッド” (SAS DS2言語リファレンス)を参照してください。
変数配列は、OUTメソッドと OUTPUTステートメントを使用してデータを書き込む
ために使用できます。OUTメソッドは、行列の行データを変数配列に書き込みま
す。OUTPUTステートメントは、結果テーブルにデータを書き込みます。行列は、
変数配列を使用して一度に 1行ずつ結果テーブルに書き込まれます。詳細について
は、“OUT メソッド” (SAS DS2言語リファレンス)および“データのロードと書き込み” 
(SAS DS2言語リファレンス)を参照してください。

行列演算
次の表は、一般的な DS2ドット構文メソッド呼び出しを使用して行列に対して実行
できる演算をまとめたものです。

表 13.6 行列演算

演算の種類
メソッド
名 実行される演算 注記

二項算術 ADD 加算 n 加算演算または減算演算で使用される行
列の配列次元は、互換性がある必要はあり
ません。ただし、最初の行列の列数が 2番
目の行列の行数と同じであるか、2番目の
行列が 1x1行列である必要があります。

n 乗算演算で使用される行列の配列次元に
は互換性が必要です。最初の行列の列数
は、2番目の行列の行数と等しい必要があ
ります。

MULT 乗算
SUB 減算

単項 ABS 絶対値 n 単項演算は、1つの行列に対して実行され
ます。

n 行列は正方または特異である必要があり
ます。逆行列および行列式の演算の場合、
行列は正方にする必要があります。

COPY 行列のコピー
DET 正方行列の行列式
EXP 指数値
FLOOR 各行列値の整数部

演算の種類
メソッド
名 実行される演算 注記
INVERSE 逆行列
LOG 自然対数
SQRT 平方根
TRANS 転置

二項関係 EQ 等しい n 二項関係演算の結果は、最初の行列の[i, j]
th要素が 2番目の行列の[i, j]th要素とどの
ように比較されるかを示す行列になりま
す。結果の値は、比較が偽の場合は 0、比
較が真の場合は 1です。

n 行列のサイズは一致する必要があります。
一致しない場合は、スカラー比較を使用で
きます。

GT より大
GE より大か等しい
LT より小
LE より小か等しい
NE 等しくない

二項論理 AND

OR

and比較
or比較

n 二項論理演算の結果は、最初の行列の[i, j]
th要素が 2番目の行列の[i, j]th要素とどの
ように比較されるかを示す行列になりま
す。結果の値は、比較が偽の場合は 0、比
較が真の場合は 1です。

ALL関係 ALL_EQ ALL等しい n ALL関係演算は、最初の行列の[i, j]th要素
が 2番目の行列の[i, j]th要素との比較を満
たすかどうかを示すスカラー結果を生成
します。スカラー結果は 0または 1です。
結果が 1になるためには、すべての [i, j]要
素比較が真でなければなりません。それ
以外の場合、結果は 0になります。

n 行列のサイズは一致する必要があります。
一致しない場合は、スカラー比較を使用で
きます。

ALL_GT ALLより大
ALL_GE ALLより大か等しい
ALL_LT ALLより小
ALL_LE ALLより小か等しい
ALL_NE ALL等しくない

ANY関係 ANY_EQ ANY等しい n ANY関係演算は、最初の行列の[i, j]th要素
が 2番目の行列の[i, j]th要素との比較を満
たすかどうかを示すスカラー結果を生成
します。スカラー結果は 0または 1です。
[i, j]要素の比較のいずれかが真の場合、結
果は 1になります。それ以外の場合、結果
は 0になります。

ANY_GT ANYより大
ANY_GE ANYより大か等しい
ANY_LT ANYより小

演算の種類
メソッド
名 実行される演算 注記
ANY_LE

n 行列のサイズは一致する必要があります。
一致しない場合は、スカラー比較を使用で
きます。

ANYより小か等しい
ANY_NE ANY等しくない

ALL論理 ALL_AND ALL AND n ALL論理演算は、最初の行列の[i, j]th要素
が 2番目の行列の[i, j]th要素との比較を満
たすかどうかを示すスカラー結果を生成
します。スカラー結果は 0または 1です。
結果が 1になるためには、すべての [i, j]要
素比較が真でなければなりません。それ
以外の場合、結果は 0になります。

n 行列のサイズは一致する必要があります。
一致しない場合は、スカラー比較を使用で
きます。

ALL_OR ALL OR

ANY論理 ANY_AND ANY AND n ANY関係演算は、最初の行列の[i, j]th要素
が 2番目の行列の[i, j]th要素との比較を満
たすかどうかを示すスカラー結果を生成
します。スカラー結果は 0または 1です。
[i, j]要素の比較のいずれかが真の場合、結
果は 1になります。それ以外の場合、結果
は 0になります。

n 行列のサイズは一致する必要があります。
一致しない場合は、スカラー比較を使用で
きます。

ANY_OR ANY OR

要素ごと EDIV 要素ごとの除算 n 要素ごとの演算では、同じ次元の行列、行
の次元が一般行列の行の次元と一致する
ベクトル、列の次元が一般行列の列の次元
と一致するベクトル、または 1x1行列を使
用して一般行列に演算を適用し、各[i,j]要
素に対してスカラー演算を効率的に実行
することができます。

n 要素ごとの演算は、2つの引数行列に対す
る要素ごとの演算から結果行列を生成し
ます。

EMAX 要素ごとの最大値
EMIN 要素ごとの最小値
EMOD 要素の除算の要素ごとの剰余
EMULT 要素ごとの乗算
EPOW 要素ごとのべき乗

行列パッケージを使用する際の考慮事項
n コンストラクターを使用して行列を作成する場合(たとえば、 m= _new_ 

matrix(2,2);または declare package matrix m(2, 2);)、行列は欠損値ではなくゼロで埋
められます。

n nullまたは欠損値を含む行列はあまり役に立ちません。行列に nullまたは欠損
値が含まれている場合、その行列に対する一部の演算は異常と見なされる可能性
があります。たとえば、nullまたは欠損値で行列を乗算すると、ランタイムエラ
ーが発生します。ただし、nullまたは欠損値を含む行列を加算または減算して
も、エラーは発生しません。行列要素をゼロで除算しても、浮動小数点例外は発
生しません。結果は欠損値です。

n ループを使用して行列データを入力または出力する場合、ループの処理中に行列
に対する演算を避ける必要があります。その理由は、ループが完了するまで行列
が部分的にしか埋められないためです。

n 行列のサイズを追跡するのは必ずしも容易ではありません。行列演算の次元が
一貫していることを確認してください。

n 行列がスレッドプログラムで宣言されると、各スレッドプログラムは独自の個別
の行列インスタンスを持ちます。DS2行列パッケージは、並列行列演算を実行す
るためのノードまたはスレッド間のデータパーティション分割をサポートして
いません。かわりに、各スレッドは、行列の独自のインスタンスに対して行列演
算を実行します。

行列から DS2配列への値の移動
次の例は、TOARRAYメソッドの使用方法を示しています。DS2プログラムで使用す
るために、行列から DS2配列に値を移動できる唯一の方法は、このメソッドを使用
することです。

dcl double a[3, 3];
dcl double c[3, 3];
declare package matrix m;

a := (1, 2, 3, 4, 5, 6, 7, 8, 9);
m=_new_ matrix(a, 3, 3);
m.toarray(c);

do i=1 to 3;
   do j=1 to 3;
      put c[i, j];
   end;
end;

PCRXFINDパッケージの使用
PCRXFINDパッケージの概要
PCRXFINDパッケージを使用すると、特定のテキスト文字列が別のテキスト文字列
内にあるかどうかを確認できます。このパッケージは、軽量のテキスト分析に使用
され、テキスト一致から詳細を取得するための使いやすい関数が含まれています。

これらの詳細には、テキストブロック内の部分文字列の開始位置または終了位置の
検索、または文字列からの一致の特定部分の取得が含まれます。

PCRXFINDパッケージの宣言とインスタンス化
DECLARE PACKAGEステートメントを使用して、PCRXFINDパッケージを宣言しま
す。新しい PCRXFINDパッケージを宣言した後、Perl互換の正規表現を解析できま
す。この表現は、パッケージが宣言されたときに渡すことができます。パッケージ
宣言中に表現を渡すことはオプションです。次に、ロードされた表現を使用して、
一致の開始と終了のインデックス、および以前に解析された正規表現内の各かっこ
で囲まれたキャプチャグループに関連するインデックスを提供することにより、一
致操作を実行できます。
PCRXFINDパッケージのインスタンスを構築する方法は 3つあります。
n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package pcrxfind variable(<regular-expression>);

n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。
 declare package pcrxfind variable;
 variable = _new_ pcrxfind(<regular-expression>);

n 表現を指定せずに DECLARE PACKAGEステートメントを使用します。
 declare package pcrxfind variable;
 variable = _new_ pcrxfind();

詳細については、“DECLARE PACKAGE ステートメント: PCRXFINDパッケージ” (SAS 
DS2言語リファレンス)、および “_NEW_ 演算子: PCRXFINDパッケージ” (SAS DS2言
語リファレンス)を参照してください。

例
PCRXFINDパッケージを使用して住所を調べ、いくつかの情報を抽出する例を次に
示します。指定された正規表現と一致しない場合、住所は無効であると報告されま
す。この例では、GETGROUP、GETGROUPLENGTH、および GETMATCHメソッド
だけでなく、大文字と小文字を区別しない一致も示しています。

/* Generate a set of valid and invalid addresses */
data sampleaddr/overwrite=yes;
   dcl varchar(256) address;
   method init();
      address = '123 Pineforest ln Cary NC 27513'; output;
      address = 'Extra text before the match 5123 Oakwood rd Cary NC 27513';
      output;
      address = '24 Durham NC 27519'; output;
      address = '0 INVALID BLVD NC 27605'; output;
      address = 'APT. 731 1008 TRADE AVE RALEIGH NC 27605'; output;
      address = 'HWY 55 123 Thistlecrest ln Cary NC 27511'; output;
      address = 'This one is invalid due to length of the street... 
         423 XYYYSDDDDDTHHHHHHZZZJJJXQTYTRRRRRRRASDAOOIIKKKK CT JASDGTJ

         UW 5290889';
       output;
   end;
enddata; run;

data _null_;
   /* 
    * Instantiate the address validator with a regular expression that does 
    * a case insensitive match on addresses. This will be used to extract 
    * each component of the address or reject the address as invalid. 
    */
   dcl package pcrxfind addrValidator
      ('/(apt\.?[ ]?[#]?\d+ |hwy \d+ )?(\d{1,5}) (\w+) (
      rd|ln|cir|st|blvd|ave|ct) (\w+) ([A-Z]{2}) (\d+)/i' );

   method run();
      dcl int valid havePrefix streetLength cityLength;
      dcl char(2) state;
      dcl varchar(256) city prefix zipText street streetType streetNum outaddr 
         streetAddress;
      dcl double zip;
      set sampleaddr;

      /* Perform the match */
      valid = addrValidator.match( address );

      /* Confirm that a match was found before proceeding */
      if valid > 0 then do;
         /*-- Grab each component of the match --*/
         addrValidator.getGroup( state, 6 );
         addrValidator.getGroup( city, 5 );
         cityLength = addrValidator.getGroupLength( 5 );
         addrValidator.getGroup( streetType, 4 );
         addrValidator.getGroup( street, 3 );
         streetLength = addrValidator.getGroupLength( 3 );
         addrValidator.getGroup( streetNum, 2 );

         /* Assemble the street address */
         streetAddress = cat( streetNum, ' ', street, ' ', streetType );
         havePrefix = addrValidator.getGroup( prefix, 1 );
         /* Convert text zip to numeric */
         addrValidator.getGroup( zipText, 7 );
         zip = inputn( zipText, 5 );
         
         /*
          * Make sure that the city and street didn't match over 25 chars. 
          * Filters potentially false results.
          */ 
         if (cityLength > 25) || (streetLength > 25) then do;
            addrValidator.getMatch( outaddr );
            put 'City or street likely invalid due to length: ' outaddr;
         end;
         else do;
            /* output the details for this address */
            put;
            put 'Valid address:';

            if havePrefix = 0 then put 'prefix: ' prefix;
            put streetAddress=;
            put city=;
            put state=;
            put zip=;
            put;
         end;
      end;
      /* If valid is -1, the expression was run and no match was found */
      else if valid = -1 then do;
         outaddr = trim( address );
         put 'Invalid address: ' outaddr;
      end;
   end;
enddata;

run;

次の行がログに出力されます。
Valid address:
streetAddress=123 Pineforest ln
city=Cary
state=NC
zip=27513

Invalid address:  Extra text before the match 5123 Oakwood rd Cary NC 27513
Invalid address:  24 Durham NC 27519
Invalid address:  0 INVALID BLVD NC 27605

Valid address:
prefix:  APT. 731
streetAddress=1008 TRADE AVE
city=RALEIGH
state=NC
zip=27605

Valid address:
prefix:  HWY 55
streetAddress=123 Thistlecrest ln
city=Cary
state=NC
zip=27511

Invalid address:  This one is invalid due to the length of the street...         423
XYYYSDDDDDTHHHHHHZZZJJJXQTYTRRRRRRRASDAOOIIKKKK CT JASDGTJ         UW 5290889

PCRXREPLACEパッケージの使用
PCRXREPLACEパッケージの概要
PCRXREPLACEパッケージを使用すると、テキストに対して置換または代入操作を
実行できます。たとえば、機密性の高い名前や電話番号を REDACTEDという単語に
置き換えたり、プログラムでテキストを更新したりすることができます。
REPLACEパッケージは UTF-8および Latin1テキストをネイティブに処理できるた
め、これらの形式をトランスコードする必要はありません。

PCRXREPLACEパッケージの宣言とインスタンス化
DECLARE PACKAGEステートメントを使用して、PCRXREPLACEパッケージを宣言
します。正規表現を含めて、PCRXREPLACEパッケージを表現および置換テキスト
に関連付けることができます。パッケージ宣言中に正規表現を含めることはオプシ
ョンです。新しい PCRXREPLACEパッケージを宣言した後、そのパッケージを使用
して文字列の置換操作を実行できます。
PCRXREPLACEのインスタンスを構築する方法は 3つあります。
n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package pcrxreplace variable('/textToReplace/replacementText');

n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。
 declare package pcrxreplace variable;
 variable = _new_ pcrxreplace('/textToReplace/replacementText');

n 表現を指定せずに DECLARE PACKAGEステートメントを使用します。
 declare package pcrxreplace variable;
 variable = _new_ pcrxreplace();

詳細については、“DECLARE PACKAGE ステートメント: PCRXREPLACEパッケージ” 
(SAS DS2言語リファレンス)、および “_NEW_ 演算子: PCRXREPLACEパッケージ” 
(SAS DS2言語リファレンス)を参照してください。

例
データセット内の姓と名を入れ替える例を次に示します。

data names(overwrite=yes);
   dcl varchar(32) name;
      method init();
         /* create a dataset to work on */

         name = 'Jones, Fred'; output;
         name = 'Kavich, Kate'; output;
         name = 'Turley, Ron'; output;
         name = 'Dulix, Yolanda'; output;
      end;
enddata;
run;

data _null_;
   /* Variable declaration */
   dcl package pcrxReplace nameSwap; 

   method init();
      nameSwap = _new_ pcrxReplace( 's/(\w+), (\w+)/$2 $1/g' );
   end;

   /* Method that switches the first and last names in 
      a dataset formatted as follows: 'lastname, firstname' */
   method run();
      dcl varchar(32) fname lname replacementText outputText;
      dcl int rc;
      set names;
      outputText = nameSwap.apply(name);
      put name=;
    end;

enddata;
run;

次の行がログに出力されます。
name=Fred Jones
name=Kate Kavich
name=Ron Turley
name=Yolanda Dulix

SQLSTMTパッケージの使用
SQLSTMTパッケージの概要
注: SQLSTMTパッケージは、CASサーバーではサポートされていません。

SQLSTMTパッケージは、FedSQLステートメントを DBMSに渡して実行し、DBMS
から返された結果セットにアクセスする方法を提供します。FedSQLステートメン
トは、テーブルから行を作成、選択、変更、挿入、削除できます。FedSQLステー
トメントがテーブルから行を選択する場合、SQLSTMTパッケージは、結果セットで
返された行を問い合わせるためのメソッドを提供します。
SQLSTMTインスタンスが作成されると、FedSQLステートメントが FedSQL言語プ
ロセッサに送信され、次に FedSQL言語プロセッサがステートメントを DBMSに送

信して、準備されてインスタンスに格納されます。その後、インスタンスを使用し
て、FedSQLステートメントを複数回効率的に実行できます。実行時までのステー
トメント準備の遅延により、FedSQLステートメントは、DS2プログラムの実行中
に動的に構築およびカスタマイズできます。
Hadoop分散: Hadoopディストリビューションを使用している場合、SQLSTMTパ
ッケージを使用するには Hive 0.13以降が必要です。
次に、SQLSTMTを使用して Teradataテーブルに値を挿入する簡単な例を示しま
す。
dcl package sqlstmt s('insert into td.testdata (x, y, z) values (?, ?, ?)',
    [x y z]);
 
   do i=1 to 5;
      x=i;
      y=i*1.1;
      z=i*10.01;
      s.execute(); 
   end;
end;

次の行がテーブルに挿入されます。
1 1.1 10.01
2 2.2 20.02
3 3.3 30.03
4 4.4 40.04
5 5.5 50.05

その他の例については、“DS2サンプルプログラム” (SAS DS2言語リファレンス)を参
照してください。

SQLSTMTパッケージの宣言とインスタンス化
DECLARE PACKAGEステートメントを使用して、SQLSTMTパッケージを宣言しま
す。パッケージが宣言されると、パッケージのインスタンスを参照できる変数が作
成されます。コンストラクター引数がパッケージ変数宣言で提供されている場合、
パッケージインスタンスが構築され、構築されたパッケージインスタンスを参照す
るようにパッケージ変数が設定されます。
SQLSTMTパッケージのインスタンスを構築する方法は 3つあります。
n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。次の 2つの構文形式があります。

DECLARE PACKAGE SQLSTMT variable[('sql-txt' [, \[parameter-variable-list\])];

DECLARE PACKAGE SQLSTMT variable [('sql-txt' [, connection-string])];

n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。次の
2つの構文形式があります。

DECLARE PACKAGE SQLSTMT variable;

variable = _NEW_ SQLSTMT('sql-txt' [, \[parameter-variable-list\]);

DECLARE PACKAGE SQLSTMT variable;

variable = _NEW_ SQLSTMT ('sql-txt' [, connection-string]);

注: DECLARE PACKAGEステートメントは、_NEW_演算子が実行されるまで、
SQLSTMTパッケージインスタンスを構築しません。_NEW_演算子が実行され
るまで、SQLステートメントの準備は行われません。

n SQLテキストを指定せずに DECLARE PACKAGEステートメントを使用します。
DECLARE PACKAGE SQLSTMT variable( );

variable = _NEW_ SQLSTMT ( );

_NEW_演算子を使用すると、sql-textは、式から生成された文字列値または変数に格
納された文字列値にすることができます。
DECLAREステートメントのかっこ内に構築用の引数が含まれている場合(SQLSTMT
パッケージでは引数の省略も有効)、パッケージインスタンスが割り当てられます。
かっこが含まれていない場合、変数は作成されますが、パッケージインスタンスは
割り当てられません。
複数のパッケージ変数を作成でき、単一の DECLARE PACKAGEステートメントで複
数のパッケージインスタンスを構築できます。各パッケージインスタンスは、パッ
ケージの完全に別個のコピーを表します。

FedSQLステートメントのパラメーター値の指定
FedSQLステートメントにパラメーターが含まれている場合、FedSQLステートメン
トを実行するには、パラメーターと置き換える値を取得する必要があります。置換
値は次のいずれかです。
n コンストラクターの DECLARE PACKAGEステートメントまたは_NEW_演算子で
指定された変数の現在の値
詳細については、“DECLARE PACKAGE ステートメント: SQLSTMTパッケージ” 
(SAS DS2言語リファレンス)および“_NEW_ 演算子: SQLSTMTパッケージ” (SAS 
DS2言語リファレンス)を参照してください。

n BINDPARAMETERSメソッドで指定された変数の現在の値
BINDPARAMETERSメソッドを使用して FedSQLステートメントを実行すると、
ステートメントの実行時にバインドされた変数の値が読み取られ、ステートメン
トのパラメーターの値として使用されます。バインドされた変数の型が対応す
るパラメーターの型と異なる場合、バインドされた変数の値はパラメーターの型
に変換されます。詳細については、“BINDPARAMETERS メソッド” (SAS DS2言語
リファレンス)を参照してください。

注: SQLSTMTパッケージを使用して DS2パッケージのMETHODステートメン
トを実行する場合は、REGISTEROUTPARAMETERメソッドを使用して、FedSQL
ステートメントの出力パラメーターを DS2パッケージのMETHODステートメ
ントの IN_OUTパラメーターにマッピングします。パラメーターデータは、バイ
ンドされた変数で排他的に指定する必要があります。SETtypeメソッドと
REGISTEROUTPARAMETERメソッドが呼び出されると、ランタイムエラーが発

生します。詳細については、“REGISTEROUTPARAMETER メソッド” (SAS DS2言
語リファレンス)を参照してください。

n SETtypeメソッドで指定された値
これらのメソッドの詳細については、“DS2 SQLSTMTパッケージのメソッド、演
算子、およびステートメント” (SAS DS2言語リファレンス)を参照してください。

パラメーター値は、バインドされた変数または SETtypeメソッドで排他的に指定す
る必要があります。

注: FedSQL言語の識別子のルールは、識別子の DS2ルールではなく、SQLSTMTパ
ッケージで使用される変数に適用されます。これは、FedSQLが DS2ではなく SQL
ステートメントを含む文字列を解析するために発生します。

接続文字列の指定
接続文字列は、データへの接続方法を定義します。接続文字列は、サブミットする
クエリ言語と、1つまたは複数のデータソースに接続するために必要な情報を識別
します。
SQLSTMTパッケージを宣言してインスタンス化するときに、接続文字列を指定でき
ます。接続文字列パラメーターは、主に SAS Federation Serverで使用するために
設計されています。完全に指定された接続文字列の作成の詳細については、SAS 
Federation Server: 管理者ガイドを参照してください。
接続文字列が提供されていない場合、SQLSTMTパッケージインスタンスは、現在割
り当てられている librefの属性を使用して、HPDS2または DS2プロシジャによって
生成された接続文字列を使用します。

FedSQLステートメントの実行
EXECUTEメソッドは、FedSQLステートメントを実行し、ステータスインジケータ
ーを返します。正常に実行されるとゼロ、エラーが発生した場合は 1、データがな
い場合(NODATA)は 2が返されます。NODATA条件は、SQL UPDATEまたは
DELETEステートメントがどの行にも影響を与えない場合に存在します。
FedSQLステートメントが実行されると、バインドされた変数の値が読み取られ、
ステートメントのパラメーターの値として使用されます。
SQLSTMTインスタンスは、1つの結果セットのみを維持します。前回の実行の結果
セットがある場合は、FedSQLステートメントが実行される前に解放されます。
FedSQLステートメントは、実行時に動的に実行されます。ステートメントは実行
時に準備されるため、DS2プログラムの実行中に動的に構築およびカスタマイズで
きます。

結果セットデータへのアクセス
FETCHメソッドは、結果セットから次のデータ行を返します。ステータスインジケ
ーターが返されます。正常に実行されるとゼロ、エラーが発生した場合は 1、デー
タがない場合(NODATA)は 2が返されます。フェッチされる次の行が結果セットの
終了後にある場合、NODATA条件が存在します。
変数が BINDRESULTSメソッドまたは FETCHメソッドで結果セットの列にバイン
ドされている場合、各結果セットの列のフェッチされたデータは、その列にバイン
ドされている変数に配置されます。変数が結果セットの列にバインドされていない
場合、フェッチされたデータは GETtypeメソッドで返すことができます。

注: 文字データの場合、結果セットの列データがすべて取得されるまで、GETtypeメ
ソッドを繰り返し呼び出すことができます。数値データの場合、GETtypeメソッド
を 1回だけ呼び出して、結果セットの列データを返すことができます。後続のメソ
ッド呼び出しの結果、rcステータスインジケーターの値は 2 (NODATA)になります。
詳細については、GETtypeメソッドを“DS2 SQLSTMTパッケージのメソッド、演算
子、およびステートメント” (SAS DS2言語リファレンス)で参照してください。

SQLSTMTインスタンスは、1つの結果セットのみを維持します。CLOSERESULTS
メソッドは、FedSQLステートメントが実行または削除されると、結果セットを自
動的に解放します。
FedSQLステートメントの実行前に FETCHメソッドが呼び出されると、ランタイム
エラーが発生します。

SQLSTMTパッケージと SQLEXEC関数の比較
次の表は、SQLSTMTパッケージと SQLEXEC関数を比較したものです。

SQLSTMTパッケージ SQLEXEC関数
FedSQLステートメントが複数回
実行される場合に適用可能

FedSQLステートメントが 1回だけ実行される場
合に適用可能

実行時に FedSQLステートメント
を動的に割り当て、準備、実行、
解放する

実行時に FedSQLステートメントを動的に割り
当て、準備、実行、解放する

パラメーターの受け渡しをサポー
トする

パラメーターの受け渡しをサポートしない

結果セットを生成する 結果セットを生成しない
ランタイム SELECTクエリ生成を
サポートする

SELECTステートメントでは使用できない

SQLSTMTパッケージ SQLEXEC関数
Java Database Connectivity 
(JDBC) PreparedStatementクラ
スと同様

SQL EXECUTE IMMEDIATEステートメントまた
は JDBC Statement.executeUpdate(String)メソ
ッドと同様

TZパッケージの使用
TZパッケージの概要
DS2は、他のデータソースで使用されている SQLスタイルの日付と時間の規則をサ
ポートしています。データソースが SASデータセットでない場合、DS2は DATE、
TIME、および TIMESTAMPのデータ型を持つ日付と時間を処理できます。SASの日
付、時間、および日時の値は、TO_DATE、TO_TIME、および TO_TIMESTAMP関数
を使用して、DS2の日付、時間、およびタイムスタンプの値に変換できます。ただ
し、これらの関数にはタイムゾーンが組み込まれていません。
TZパッケージを使用すると、ローカルおよび国際的な時刻と日付の値を処理できま
す。

TZパッケージの宣言とインスタンス化
TZパッケージのインスタンスを構築する方法は 2つあります。
n DECLARE PACKAGEステートメントを_NEW_演算子と一緒に使用します。

 declare package tz tzpkg;
 tzpkg = _new_ tz();

n DECLARE PACKAGEステートメントをそのコンストラクター構文と一緒に使用
します。
declare package tz tzpkg();

詳細については、“DECLARE PACKAGE ステートメント: TZパッケージ” (SAS DS2言
語リファレンス)、および “_NEW_ 演算子: TZパッケージ” (SAS DS2言語リファレン
ス)を参照してください。

時間とタイムゾーンの情報を返す
TZパッケージを使用して、次の値を返すことができます。
n 現在のローカル時間
n 現在の協定世界時(UTC)時間

n 現在のタイムゾーン ID

n 現在のタイムゾーン名
n 指定されたローカル時間での UTCからのタイムゾーンのタイムゾーンオフセッ
ト

n 指定された UTC時間での UTCからのタイムゾーンのタイムゾーンオフセット
TZパッケージを使用して世界時計を取得する方法の例を次に示します。
data _null_ ; 
   method init();
  
     declare package tz tzone() ;
     dcl double tokyo_time london_time new_york_time utc_time ; 
  
     tokyo_time = tzone.getLocalTime('Asia/Tokyo') ; 
     london_time = tzone.getLocalTime('Europe/London') ;
     new_york_time = tzone.getLocalTime('America/New_York') ; 
  
     utc_time = tzone.getUTCTime(); 
  
     put utc_time = datetime. ;
     put tokyo_time = datetime. ; 
     put london_time = datetime. ;
     put new_york_time = datetime. ;
  end ;
enddata ; 
run;

次の行が SASログに書き込まれます。

utc_time=18NOV14:13:53:11
tokyo_time=18NOV14:22:53:11
london_time=18NOV14:13:53:11
new_york_time=18NOV14:08:53:11

これらのアクションを実行するためのメソッドの詳細については、“DS2 TZパッケ
ージのメソッド、演算子、およびステートメント” (SAS DS2言語リファレンス)を参
照してください。タイムゾーン IDと名前の詳細については、“タイムゾーン IDとタ
イムゾーン名” (SAS各国語サポート(NLS): リファレンスガイド)を参照してくださ
い。

タイムゾーンオフセットを返す
TZパッケージを使用して、指定されたローカル時間または指定された UTC時間の
UTCからのタイムゾーンオフセットを返すことができます。
タイムゾーンオフセットは、タイムゾーンと UTCの時間差と分差を+|–hhmmまた
は+|–hh:mmの形式で示します。
'asia/tokyo'と'America/New_York'からのタイムゾーンオフセットを返す例を次に
示します。
data _null_ ; 
  method init();

238 13章 / DS2パッケージ

http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p1oa4jiyneqvqcn1180nxji6v91l.htm&locale=ja
http://documentation.sas.com/?docsetId=ds2ref&docsetVersion=v_001&docsetTarget=p1oa4jiyneqvqcn1180nxji6v91l.htm&locale=ja
http://documentation.sas.com/?docsetId=nlsref&docsetVersion=v_001&docsetTarget=n1tj735aocxmw7n1kfoz1qpdvb9l.htm&locale=ja
http://documentation.sas.com/?docsetId=nlsref&docsetVersion=v_001&docsetTarget=n1tj735aocxmw7n1kfoz1qpdvb9l.htm&locale=ja


  
  declare package tz tzone('asia/tokyo') ;
  dcl double new_york local_time; 
  dcl char(40) cstr ;
  
  local_time = tzone.getOffset() ; 
  put local_time time.;

  new_york = tzone.getOffset('America/New_York') ; 
  put new_york time.;
   
   
  end;
  enddata ; 
run; 

次の行が SASログに書き込まれます。

9:00:00
-4:00:00

ローカル時間または UTC時間の変換
TZパッケージを使用して、ローカルを次のいずれかの時間形式に変換できます。
n ISO8601 (タイムゾーンオフセットありまたはなし)

n タイムゾーン IDを含む TIMESTAMP文字列
n UTC時間
さらに、UTC時間をローカル時間に変換することもできます。
次に例を示します。
data _null_ ; 
   method init();
  
      declare package tz tzone('asia/tokyo') ;
      dcl double local_time ;
      dcl char(35) local_time_iso local_time_utc local_time_tz; 
      
    
      local_time = tzone.tolocaltime(15550) ; 
      put local_time time.;

      local_time_iso = tzone.toiso8601(15500) ; 
      put local_time_iso ;
      
      local_time_utc = tzone.toutctime(15500) ; 
      put local_time_utc time.;

      local_time_tz =tzone.totimestampz(15500);
      put local_time_tz;

   end;
enddata ; 
run;  

次の行が SASログに書き込まれます。

13:19:10
1960-01-01T04:18:20.00+09:00
-4:41:40
1960-01-01 04:18:20.00 asia/tokyo

スレッド処理の概要
注: DS2が CASサーバー上でどのように実行されるかについての詳細は、“CASでの
DS2の実行方法” (297ページ)を参照してください。

通常、DS2コードは順次実行されます。つまり、あるプロセスが完了するまで実行
されてから、次のプロセスが開始されます。スレッド処理を使用して、複数のプロ
セスを同時に実行することができます。スレッド処理では、同時に実行されるコー
ドの各セクションは、スレッドで実行されていると言われます。DS2スレッドは、
複数のコアを備えたマシンでも超並列処理(MPP)データベース内でもうまく機能し
ます。
DS2プログラムは、入力データを処理し、出力データを生成します。DS2プログラ
ムは、プログラムまたはスレッドとして、2つの異なる方法で実行できます。DS2
プログラムをプログラムとして実行した場合の結果を次に示します。
n 入力データには、データベーステーブルの行と DS2プログラムスレッドの行の
両方を含めることができます。

n 出力データは、クライアントアプリケーションに返されるデータベーステーブル
または行のいずれかになります。

DS2プログラムをスレッドとして実行した場合の結果を次に示します。
n 入力データには、データベーステーブルの行のみを含めることができ、他のスレ
ッドの行は含めることができません。

n 出力データには、スレッドを開始した DS2プログラムに返される行が含まれま
す。

DS2コードをスレッドで実行できるようにするには、次の操作を実行します。
1 DS2コードを THREAD...ENDTHREADステートメントで囲んでスレッドを作成
します。

2 DECLARE THREADステートメントを使用して、DS2プログラム内にスレッドの
インスタンスを 1つ以上作成します。

3 SET FROMステートメントを使用して 1つまたは複数のスレッドを実行します。
この例では、THREADステートメントを使用して、非常に単純なスレッド Tを作成
します。
thread t;
   dcl int x;
   method init();
      dcl int i;
      do i = 1 to 3;
         x = i;
         output;
      end;
   end;
endthread;

この DS2プログラムでは、Tのインスタンスが宣言され、RUNメソッドの SET 
FROMステートメントを使用して 2つのスレッドが実行されます。2つのスレッド
のそれぞれが xに対して 3つの行を生成し、出力テーブルに合計 6つの行が生成さ
れます。
  data; 
    dcl thread t t_instance;
    method run(); 
      set from t_instance threads=2;
      put 'x= ' x ; 
    end;
  enddata; 

DS2プログラムを実行すると、SASログに次の出力が表示される場合があります。
スレッドの処理方法により、出力の順序が異なる可能性があります。

x=  1
x=  2
x=  3
x=  1
x=  2
x=  3

注: 1つの計算スレッドが I/Oスレッドに追いつくことができる場合、その 1つのス
レッドがすべての計算に使用されます。

注: 1つのリーダーがすべてのスレッドにフィードします。スレッドプログラムの
SETステートメントは、その SETステートメントの単一のリーダーを共有します。
入力テーブルの各行は、正確に 1つのスレッドに送信されます。

注: データプログラムとスレッドプログラムは型が一致している必要があります。
10進データ型の場合、精度とスケールの両方が一致する必要があります。文字列の
場合、列サイズ、文字セット、および文字列が可変長であるか固定長であるかは、
すべて一致する必要があります。

詳細については、“THREAD ステートメント” (SAS DS2言語リファレンス)を参照して
ください。

スレッドと DS2プログラム
シリアルプログラムと並列プログラムの概要

DS2プログラムは、複数のデータ行に対して同時に操作を実行できるため、大きな
データセットの処理に要する時間が短縮されます。DS2プログラムの構造に基づい
て、DS2コンパイラは、複数の行に対して同時に実行できる操作と、各行に順次適
用する必要がある操作を決定します。DS2プログラムは、シリアルプログラム、パ
ラレルプログラム、またはパラレル-シリアルプログラムに分類されます。
シリアルプログラム
行間のデータ依存関係を持つ操作が含まれます。したがって、行は順次処理する
必要があります。データプログラムはデータセットを処理し、結果セットを生成
します。

パラレルプログラム
行間のデータ依存関係を持つ操作は含まれません。したがって、複数のデータ行
を並列処理できます。各スレッドは、データセットのサブセットを処理し、結果
セットのサブセットを生成します。

パラレル-シリアルプログラム
行間にデータ依存関係のある操作と、データ依存関係のない操作が含まれていま
す。操作の処理は、パラレルステージとシリアルステージの 2つのステージに分
けられます。パラレルステージでは、各スレッドが入力データセットのサブセッ
トを処理し、中間データセットのサブセットを生成します。シリアルステージで
は、1つのスレッドが完全な中間データセットを処理し、完全な結果セットを生
成します。

注: 詳細については、“CASでの DS2の実行方法” (297ページ)を参照してください。

データ操作
DS2データ操作は、シリアル操作または並列操作のいずれかに分類されます。

シリアル操作
行間でデータの依存関係を持つ操作。
n シリアル操作は、各データ行に順次適用する必要があります。
n シリアル操作は、DS2データプログラム内のステートメントによって実装さ
れます。

並列操作
行間でデータの依存関係がない操作。
n 並列操作は、複数のデータ行に並列に適用できます。
n 並列操作は、DS2スレッドプログラム内のステートメントによって実装され
ます。

DS2コンパイラは、DS2プログラム内のステートメントの配置に基づいてステート
メントを分類します。DS2コンパイラは、スレッドプログラム内のすべてのステー
トメントが並列操作を実装し、データプログラム内のすべてのステートメントがシ
リアル操作を実装すると想定します。唯一の例外は、データの入出力ステートメン
トです。DS2コンパイラは、SET、SET FROM、または OUTPUTをデータ操作ステ
ートメントとは見なしません。

DS2プログラムの概要
DS2プログラムは、含まれるプログラムのタイプとデータ操作のタイプに基づいて、
シリアルプログラム、並列プログラム、または並列-シリアルプログラムのいずれか
に分類されます。
DS2シリアルプログラム
シリアル操作のみを含むプログラム。
n データプログラムのみがあります(スレッドプログラムなし)。
n データプログラムには、シリアルデータ操作が含まれます。

DS2並列プログラム
並列操作のみを含むプログラム。
n スレッドプログラムとデータプログラムがあります。
n スレッドプログラムには、並列データ操作が含まれています。
n データプログラムには、データ操作が含まれていません。つまり、データプ
ログラムには、SET FROMと OUTPUT以外のステートメントは含まれていま
せん。

DS2並列-シリアルプログラム
プログラムには、並列操作とシリアル操作の両方が含まれています。
n スレッドプログラム(パラレルステージ)とデータプログラム(シリアルステー
ジ)があります。

n スレッドプログラムには、並列データ操作が含まれています。
n データプログラムには、シリアルデータ操作が含まれます。つまり、データ
プログラムには、SET FROMと OUTPUT以外に少なくとも 1つのステートメ
ントが含まれています。

DS2スレッドで役立つ自動変数
DS2スレッド間で問題をサブセット化するために使用される自動変数がいくつかあ
ります。これらの自動変数は、PUTステートメントを使用してデバッグするときに
コンテキストを提供するのにも役立ちます。

変数 説明
_HOSTNAME_ DS2プログラムが実行されているワーカーノードまたはホス

トの名前。
_LOCALNTHREADS_ ローカルノードで実行されているプログラムの DS2スレッド

の総数。たとえば、THREADS=10の SET FROMステートメン
トを含む DS2プログラムが、4つのワーカーを持つ CASサーバ
ーで実行される場合、_LOCALNTHREADS_は 10になります。
単一ノード環境では、すべてのスレッドがローカルノードで実
行されるため、_LOCALNTHREADS_は_NTHREADS_と同じ値に
なります。

_LOCALTHREADID_ ローカルノードの現在の DS2スレッドに割り当てられた一意
の数値識別子。たとえば、THREADS=10の SET FROMステー
トメントを含む DS2プログラムが、4つのワーカーを持つ CAS
サーバーで実行される場合、_LOCALTHREADID_は 1から 10
までの値を返します。単一ノード環境では、
_LOCALTHREADID_は_THREADID_と同じ値になります。

_NTHREADS_ プログラムで実行されている DS2スレッドの総数。並列環境
では、_NTHREADS_は、DS2プログラムが実行されているすべ
てのノードにわたる DS2スレッドの総数です。たとえば、
THREADS=10の SET FROMステートメントを含む DS2プログ
ラムが、4つのワーカーを持つ CASサーバーで実行される場
合、_NTHREADS_は 40になります。

_THREADID_ 現在の DS2スレッドに割り当てられた一意の数値識別子。並
列環境では、_THREADID_は、DS2プログラムが実行されてい
るすべてのノードにあるすべての DS2スレッド間で重複しま
せん。たとえば、THREADS=10の SET FROMステートメント
を含む DS2プログラムが、4つのワーカーを持つ CASサーバー
で実行される場合、_THREADID_は 1から 40までの値を返しま
す。

_NTHREADS_と_LOCALNTHREADS_の両方が、THREADブロックで実行されている
スレッドの数を参照します。DATAブロックで実行されているスレッドはカウント
されません。

THREADID_の値は 1から_NTHREADS_までです。一方、_LOCALTHREADID_の値は
1から_LOCALNTHREADS_までです。DATAブロック内で実行されているスレッド
に対して、_THREADIDと_LOCALTHREADID_のどちらも 0になります。

15
DS2と FedSQLの使用

DS2から FedSQLステートメントを動的に実行する . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 247

DS2から FedSQLステートメントを動的
に実行する

注: DS2アクションまたは PROC DS2の SETステートメントを除いて、DS2プログ
ラム内での FedSQLクエリの使用は、現在 CASサーバーではサポートされていませ
ん。

DS2プログラム内から FedSQLステートメントを埋め込んで実行できます。次の場
合に FedSQLを DS2で使用できます。
n DS2パッケージメソッド式を FedSQL SELECTステートメントの関数として呼び
出すことができます。
詳細については、“式での DS2パッケージの使用” (SAS FedSQL言語リファレンス)
を参照してください。

n SQLSTMTパッケージを使用して、FedSQLステートメントを生成、準備、およ
び実行し、実行時にテーブルから行を作成、選択、変更、挿入、削除できます。
SQLSTMTパッケージは、複数回実行される FedSQLステートメント、パラメー
ターを含むステートメント、または結果セットを生成するステートメントで使用
することを目的としています。詳細については、“SQLSTMTパッケージの使用” 
(232ページ)を参照してください。

n また、SQLEXEC関数を使用して、FedSQLステートメントを生成、準備、および
実行し、実行時にテーブルから行を作成、選択、変更、挿入、削除できます。
SQLEXEC関数は、1回だけ実行され、パラメーターを持たず、結果セットを生成
しないステートメントで使用することを目的としています。
詳細については、“SQLEXEC 関数” (SAS DS2言語リファレンス)を参照してくださ
い。

n DECLARE PACKAGEステートメントまたは DATASETメソッドで FedSQL 
SELECTステートメントを使用して、実行時にデータをハッシュインスタンスに
ロードできます。
詳細については、“FedSQLクエリとハッシュインスタンスを使用して実行時に動
的に行を取得する” (206ページ)、“DATASET メソッド” (SAS DS2言語リファレン
ス)および“DECLARE PACKAGE ステートメント: ハッシュパッケージ” (SAS DS2
言語リファレンス)を参照してください。

n SETステートメントを使用して、FedSQL SELECTステートメントでデータを読
み込むことができます。
詳細については、“FedSQLが埋め込まれた SETステートメント” (253ページ)お
よび“SET ステートメント” (SAS DS2言語リファレンス)を参照してください。

16
DS2入力と出力

DS2入力と出力の概要 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 249

SETステートメントを使用したデータの読み取り . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 250
SETステートメントの概要 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 250
SETステートメントのコンパイル . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 251
SETステートメントの実行 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 251
FedSQLが埋め込まれた SETステートメント . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 253

ハッシュパッケージを使用したデータの読み取り . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 254

SQLSTMTパッケージと SQLEXEC関数を使用したデータの読み取りと書き込み . . . . 255

OUTPUTステートメントを使用したデータの書き込み . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 255

SAS外部のデータソースを使用する場合の出力テーブルの列の順序 . . . . . . . . . . . . . . . . 256

NLSトランスコーディングの失敗 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 256

DS2入力と出力の概要
次の方法を使用して、DS2データを読み取ることができます。
n DS2プログラムのパッケージおよびメソッド内で変数または配列の割り当てを
使用して、データを生成できます。

n SETステートメントを使用して、既存のテーブルを読み取ることができます。
詳細については、“SETステートメントを使用したデータの読み取り” (250ペー
ジ)および“SET ステートメント” (SAS DS2言語リファレンス)を参照してくださ
い。

n ハッシュパッケージを使用してデータを読み取ることができます。
詳細については、“ハッシュパッケージを使用したデータの読み取り” (254ペー
ジ)を参照してください。

n SELECTステートメントを実行して結果セットにアクセスすることにより、
SQLSTMTパッケージを使用してデータを取得できます。

注: SQLSTMTパッケージは、CASサーバーではサポートされていません。

詳細については、“SQLSTMTパッケージと SQLEXEC関数を使用したデータの読
み取りと書き込み” (255ページ)を参照してください。

次の方法を使用して、DS2データを書き込むことができます。
n OUTPUTステートメントを使用して DS2データを書き込みます。OUTPUTステ
ートメントは、結果テーブルに行を書き込みます。
詳細については、“OUTPUTステートメントを使用したデータの書き込み” (255
ページ)を参照してください。

n ハッシュパッケージの OUTPUTメソッドを使用できます。
詳細については、“OUTPUT メソッド” (SAS DS2言語リファレンス)および“ハッシ
ュパッケージデータをテーブルに保存する” (203ページ)を参照してください。

n SQLEXEC関数または SQLSTMTパッケージで FedSQL INSERTおよび UPDATE
ステートメントを使用できます。

注: SQLEXEC関数と SQLSTMTパッケージは、CASサーバーではサポートされて
いません。

“SQLSTMTパッケージと SQLEXEC関数を使用したデータの読み取りと書き込
み” (255ページ)、“SQLEXEC 関数” (SAS DS2言語リファレンス)、および“結果セッ
トデータへのアクセス” (236ページ)。

注: MongoDBデータソースには、テーブルを作成するための特別な要件がありま
す。データに応じて、ルートテーブル、親テーブル、および子テーブルを作成する
ことをお勧めします。DS2ではルートテーブルのみを作成できます。親テーブルと
子テーブルを作成するには、FedSQLを使用する必要があります。SAS/ACCESS for 
Relational Databases: リファレンスのMongoDBに関する情報を参照してくださ。

SETステートメントを使用したデータの
読み取り
SETステートメントの概要

SETステートメントは柔軟性があり、DS2プログラミングでさまざまな用途があり
ます。これらの使用法は、SETステートメントで使用するオプションとステートメ
ントによって決まります。
n DS2プログラムでさらに処理するために、既存のテーブルから行と列を読み取る

n テーブルの連結とインターリーブ、およびテーブルの 1対 1の読み取りの実行
詳細については、17章, “テーブルの結合” (259ページ)を参照してください。

SETステートメントが実行されるたびに、1行がプログラムデータベクトルに読み込
まれます。SETは、特に指定しない限り、入力テーブルからすべての列とすべての
行を読み取ります。SETステートメントには複数のテーブルを含めることができ、
DS2プログラムには、複数の SETステートメントを含めることができます。
詳細については、“SET ステートメント” (SAS DS2言語リファレンス)を参照してくだ
さい。

注: スレッドプログラムの SETステートメントは、その SETステートメントの単一
のリーダーを共有します。SETステートメントが実行されるたびに、指定された入
力テーブルから 1行が読み取られます。ただし、すべてのスレッドが単一のリーダ
ーを共有します。入力テーブルの各行は、正確に 1つのスレッドに送信されます。
INITメソッドで SETステートメントを使用する場合、最初のスレッドがすべての行
を読み取ります。他のスレッドはファイルの終わりに到達し、関連する RUNメソッ
ドに進まずに処理を終了します。したがって、スレッドプログラムの INITまたは
TERMメソッドでは、SETステートメントを使用しないことをお勧めします。

注: SETステートメントは、RUNメソッドの暗黙的なループ機能を利用できるよう
に、RUNメソッドでの使用が適しています。

SETステートメントのコンパイル
DS2コンパイラは、SETステートメントを評価するときに、各 table-referenceの列
情報を読み取ります。各テーブルの列ごとに、列の型属性と同じ型属性を持つグロ
ーバル変数が DS2プログラムに作成されます。このようにして、SETステートメン
トは関連する列変数のセットを作成します。グローバル変数がすでに存在し、その
型がテーブル列の型と一致しない場合、エラーが発生します。これは、2つのテー
ブルに同じ名前の列がある場合、それらの列の型に互換性がある必要があることを
意味します。

SETステートメントの実行
データプログラムでは、SETステートメントが実行されると、最初のテーブルから
最初の行が読み取られます。連続して実行すると、現在のテーブルが完全に読み取
られるまで行が読み取られます。次に、次のテーブルの最初の行が読み取られます。
行が読み取られるたびに、行の値が対応する列変数に割り当てられます。SETステ
ートメントによって作成された列変数は保持されます。SETステートメントの実行
は、すべてのテーブルのすべての行が完全に読み取られると終了します。
スレッドプログラムの場合、特定のスレッドに入る行の順序は、BYステートメント
を使用しない限り未定義です。その場合でも、スレッドのスレッドプログラムでは、

すべての行が他のテーブルからの行の前に受信されると想定する方法はありませ
ん。
読み取られているテーブルに表示されない列変数は、SASモードか ANSIモードかに
応じて、SAS欠損値または null値に設定されます。SETステートメントで宣言され
た変数は、次のように初期化されます。
n SASモードの場合:

o DOUBLEデータ型は、SAS数値欠損値(.)に初期化されます。
o 固定幅の CHARおよび NCHARデータ型は、SAS文字欠損値(空白で埋められ
た文字列)に初期化されます。

o VARCHARや NVARCHARなど、その他すべてのデータ型は ANSI nullに初期
化されます。

n ANSIモードでは、すべての型が ANSI nullに初期化されます。

注: 詳細については、7章, “DS2による nullおよび SAS欠損値の処理方法” (61ペー
ジ)を参照してください。

次に例を示します。このプログラムでは、2行目と 4行目に列変数modifiersが指定
されていません。OUTPUTメソッドは、最後の OUTPUTステートメントの後に値が
割り当てられているかどうかにかかわらず、すべての定義済み変数の値を使用しま
す。
proc ds2;
data test (overwrite=yes);
   dcl char string1 string2 modifiers having informat $char8. format $char8.;
   method init();
      string1='aBc'; string2='AbC'; modifiers='i'; output;
      string1='   abc'; string2='abc'; output;
      string1='   abc'; string2='abc';  modifiers='l'; output;
      string1=' abc'; string2='   abx'; output;
      string1=' abc'; string2='   abx'; modifiers='l';  output;
   end;
enddata;
run;

data test_out (overwrite=yes);
   dcl double result;
   method run();
      set test;
      result=compare(string1, string2, modifiers);
      put 'String 1= ' string1 'String 2= ' string2 'Modifier= ' modifiers 
         'Result= ' result;
   end;
enddata;
run;
quit;

次の行が SASログに書き込まれます。

String 1=  aBc      String 2=  AbC      Modifier=  i        Result=  0
String 1=     abc   String 2=  abc      Modifier=  i        Result=  -1
String 1=     abc   String 2=  abc      Modifier=  l        Result=  0
String 1=   abc     String 2=     abx   Modifier=  l        Result=  -3
String 1=   abc     String 2=     abx   Modifier=  l        Result=  -3

2行目と 4行目がmodifiers列変数に指定された値を持たないようにするには、変数
を欠損値または null値に設定する必要があります。
proc ds2;
data test (overwrite=yes);
   dcl char string1 string2 modifiers having informat $char8. format $char8.;
      method init();
         string1='aBc'; string2='AbC'; modifiers='i'; output;
         string1='   abc'; string2='abc'; modifiers=' '; output;
         string1='   abc'; string2='abc';  modifiers='l'; output;
         string1=' abc'; string2='   abx'; modifiers=' '; output;
         string1=' abc'; string2='   abx'; modifiers='l';  output;
      end;
enddata;
run;

data test_out (overwrite=yes);
   method run();
      set test;
      result=compare(string1, string2, modifiers);
      put 'String 1= ' string1 'String 2= ' string2 'Modifier= ' modifiers 
         'Result= ' result;
   end;
enddata;
run;
quit;

次の行が SASログに書き込まれます。

String 1=  aBc      String 2=  AbC      Modifier=  i        Result=  0
String 1=     abc   String 2=  abc      Modifier=           Result=  -1
String 1=     abc   String 2=  abc      Modifier=  l        Result=  0
String 1=   abc     String 2=     abx   Modifier=           Result=  2
String 1=   abc     String 2=     abx   Modifier=  l        Result=  -3

FedSQLが埋め込まれた SETステートメント
次の例のように、SETステートメントは FedSQLコードを使用してテーブルを読み
取ることができます。
set {select * from catalog_base.investment};

テーブル名と埋め込み FedSQLを SETステートメントにインターリーブすることが
できます。この例では、SETステートメントは同じテーブルを 2回読み取ります。
set {select * from catalog_base.investment} catalog_base.investment;

SETステートメントで使用される埋め込み FedSQLは、有効な FedSQLコードであ
る必要があり、テーブル行のセットに解決される必要があります。それ以外の場合
は、エラーが発生します。

注: 埋め込み SQLに含まれる内容に関係なく、埋め込み SQLから表示される行の順
序は保証されません。

set {select * from sql13a order by "X", "Y"};

一部の環境では、ORDER BY句によって課された順序が保持される場合があります
が、他の環境では保持されません。DS2は埋め込み SQLを追加のコードでラップし
ます。また、DS2は FIRSTまたは LAST情報を検出および生成できないため、ORDER 
BY句は受け入れられません。よりよい方法は、埋め込み SQLテキストの外側で BY
を使用してプログラムを作成することです。

set {select * from sql13a}; by "X" "Y";

CASサーバーでは、ORDER BY順序はデータプログラムまたはスレッドプログラム
のいずれにも保持されません。これは、埋め込み SQLの結果が FedSQLアクション
によって一時テーブルに保存されるために発生します。ORDER BY順序は、一時テ
ーブルが作成される前に FedSQLアクションによって適用されます。一時テーブル
が DS2アクションによって読み取られるとき、順序は暗示されません。

ハッシュパッケージを使用したデータの
読み取り

ハッシュパッケージを使用して、テーブルからデータを読み取ることができます。
DECLARE PACKAGEステートメント、_NEW_演算子、および DATASETメソッドは、
次のいずれかの方法を使用してデータソースを識別します。
n テーブルを識別する名前
n テーブル行のセットに解決される有効な FedSQL SELECTステートメント

注: FedSQLクエリを使用してデータを選択することは、CASサーバーではサポ
ートされていません。

注: DS2ハッシュテーブルから FCMPパッケージにデータをロードすることはでき
ません。

詳細については、“ハッシュパッケージの使用” (194ページ)を参照してください。

SQLSTMTパッケージと SQLEXEC関数を
使用したデータの読み取りと書き込み

注: SQLSTMTパッケージと SQLEXEC関数は、CASサーバーではサポートされてい
ません。

SQLSTMTパッケージと SQLEXEC関数を使用すると、DS2プログラムで FedSQLス
テートメントを動的に生成、準備、および実行して、テーブルの行を更新、挿入、
または削除できます。SQLSTMTパッケージまたは SQLEXEC関数のインスタンス
を使用すると、FedSQLステートメントの割り当て、準備、実行、解放が実行時に
行われます。
さらに、SQLSTMTパッケージは、実行された FedSQLステートメントによって生成
される結果セットを調べることができます。

注意
SQLEXEC関数および SQLSTMTパッケージで使用できるのは、FedSQLステートメ
ントのみです。DBMS固有の SQLは使用できません。 詳細については、SAS FedSQL
言語リファレンスを参照してください。

詳細については、“SQLSTMTパッケージの使用” (232ページ)および“SQLEXEC 関数” 
(SAS DS2言語リファレンス)を参照してください。

ヒント SQLSTMTパッケージでパラメーター化された FedSQLクエリを使
用することにより、KEY=を指定した DATAステップの SETステートメントの
機能を実現できます。

OUTPUTステートメントを使用したデー
タの書き込み

OUTPUTステートメントは、OUTPUTステートメントの実行時にグローバル変数に
含まれる行の値を使用して、出力行を作成します。OUTPUTステートメントは、DS2
プログラムの最後ではなく、すぐに現在の行をテーブルに書き込みます。OUTPUT
ステートメントでテーブル名が指定されていない場合、行は DATAステートメント
にリストされている 1つまたは複数のテーブルに書き込まれます。

DS2は、値が既存のテーブルから読み取られたかプログラムで作成さたかに関係な
く、コンパイラが DS2プログラム内で検出した順序で値を追跡します。
OUTPUTステートメントを指定しなかった場合、DS2は RUNメソッドの最後に暗
黙的に 1つ追加し、作成中の 1つまたは複数のテーブルに行を書き込みます。
DS2プログラムに明示的な OUTPUTステートメントを配置すると、自動出力がオー
バーライドされ、明示的な OUTPUTステートメントが実行された場合にのみテーブ
ルに行が追加されます。ただし、OUTPUTステートメントを使用して任意の 1つの
テーブルに行を書き込むと、RUNメソッドの最後に暗黙的な OUTPUTステートメ
ントは存在しなくなります。この状況では、DS2プログラムは、明示的な OUTPUT
が実行された場合にのみテーブルに行を書き込みます。OUTPUTステートメント
は、単独で使用することも、IF-THEN/ELSEまたは SELECTステートメントの一部と
して使用することも、DOループ処理で使用することもできます。

注: スレッドプログラムの OUTPUTステートメントにテーブル名を含めることはで
きません。各出力行は、スレッドを開始したデータプログラムに返されます。

SAS外部のデータソースを使用する場合
の出力テーブルの列の順序

一部のデータソースは、DS2からの出力テーブルの列の優先順序を 1つ選択します。
たとえば、Hiveでは、BYPARTITION列は常にテーブルの最後に移動します。これ
は、さまざまなデータソースがパフォーマンスを最適化しようとする場合によくあ
ることです。
DS2プログラムでの宣言の順序は、データソースの列の順序として使用されない場
合があります。たとえば、keep K1- - K4;を使用すると、CREATE TABLEステートメン
トで K1が K4の後に表示されるため、期待どおりに列を取得できないか、エラーが
発生する可能性があります。

NLSトランスコーディングの失敗
トランスコーディングは、文字データをあるエンコーディングから別のエンコーデ
ィングに変換するプロセスです。NLSトランスコーディングの失敗は、行の入力ま
たは出力操作中、または文字列の割り当て中に発生する可能性があります。デフォ
ルトでは、このランタイムエラーにより、行の処理が停止します。PROC FEDSQL
および PROC DS2で XCODE=オプションを使用して、デフォルトの動作を変更でき
ます。

このオプションを使用すると、エラーを無視して行の処理を続行することを選択で
きます。
詳細については、Base SASプロシジャガイドを参照してください。

テーブルの結合

データ結合の定義
DS2処理のコンテキストでは、データの結合には次の意味があります。
n 連結
n インターリーブ
n 1対 1の読み取り
n マッチマージ
テーブルの結合に使用される 2つのステートメントは、MERGEと SETです。

テーブルの結合: 基本概念
複数のテーブルに保存されている情報を結合する前
に知っておくべきこと

多くのアプリケーションでは、データを処理して意味のある結果を生成する前に、
入力データを特定の形式にする必要があります。通常、データは複数のソースから
取得され、さまざまな形式になっている場合があります。したがって、データを分
析したり、データからレポートを作成したりする前に、データを論理的に関連付け
て処理するための中間的な手順を踏まなければならないことがよくあります。
アプリケーションの要件はさまざまですが、データにアクセスし、結合し、処理す
るすべてのアプリケーションに共通する要素があります。出力をどのように表示す
るかを決定したら、次のタスクを実行する必要があります。
n 入力データがどのように関連しているかを判断します。
n 必要に応じて、データが適切に並べ替えられていることを確認します。
n 入力データを処理するための適切なアクセス方法を選択します。
n 適切なツールを選択してタスクを完了します。

データを関連付ける 4つの方法
データリレーションシップのカテゴリ
入力データの複数のソース間のリレーションシップは、各ソースに物理レベルまた
は論理レベルのいずれかで共通のデータが含まれている場合に存在します。たとえ
ば、従業員データと部門データは、共通の値を共有する従業員 ID列を介して関連付
けることができます。別のテーブルには、部分的な値が行番号によって別のテーブ
ルに論理的に関連付けられた数値シーケンス番号を含めることができます。
データ内の既存のリレーションシップを識別できる必要があります。この知識は、
入力データを処理して目的の結果を生成する方法を理解するために不可欠です。関
連するすべてのデータは、次の 4つのカテゴリのいずれかに分類されます。これら
のカテゴリは、テーブル間での行の関係によって特徴付けられます。
n 1対 1

n 1対多

n 多対 1

n 多対多
必要な結果を得るためには、これらの各方法でどのように行を結合するか、各方法
で共通列の重複値をどのように扱うか、各方法で共通列の欠損値や不一致値をどの
ように扱うかを理解する必要があります。一部の方法では、テーブルを並べ替えて
前処理する必要もあります。“テーブル結合方法” (263ページ)の各方法の説明を参照
してください。

1対 1リレーションシップ
1対 1リレーションシップでは、通常、選択した 1つ以上の列の値に基づいて、1つ
のテーブルの 1つの行が別のテーブルの 1つの行に関連付けられます。1対 1リレ
ーションシップとは、選択した列の各値が各テーブルで 1回しか出現しないことを
意味します。選択した複数の列を操作する場合、このリレーションシップは、値の
各組み合わせが各テーブルで 1回しか出現しないことを意味します。
次の例では、Salaryテーブルと Taxesテーブルの行が、EmployeeNumberの共通
の値によって関連付けられています。
図 17.1 1対 1リレーションシップ

SALARY

EmployeeNumber Salary

TAXES

TaxBracket

1234 55000 1111 0.18
3333 72000 1234 0.28
4876 32000 3333 0.32
5489 17000 4222 0.18

4876 0.24

EmployeeNumber

1対多および多対 1のリレーションシップ
入力テーブル間の 1対多または多対 1のリレーションシップは、1つのテーブルに
は、選択された列の特定の値を持つ行が最大で 1つあることを意味しますが、もう
1つの入力テーブルには、各値が複数回出現する可能性があります。選択した複数
の列を操作する場合、このリレーションシップは、値の各組み合わせが 1つのテー
ブルで 1回しか出現しないことを意味します。ただし、その組み合わせは他のテー
ブルで複数回出現する可能性があります。入力テーブルが処理される順序によっ
て、リレーションシップが 1対多か多対 1かが決まります。
次の例では、テーブル Oneと Twoの行は、列 Aの共通の値によって関連付けられ
ています。Aの値は、テーブル Oneでは一意ですが、テーブル Twoでは一意ではあ
りません。

図 17.2 1対多リレーションシップ
ONE TWO

1
3

5
3

6
4

1
1

2
3

0
99

1 4 88
1
2

5
1

77
66

2 2 55
3 4 44

A E FA B C

次の例では、テーブル One、Two、および Threeの行が、列 IDの共通の値によっ
て関連付けられています。IDの値は、テーブル Oneと Threeでは一意ですが、テ
ーブル Twoでは一意ではありません。IDの値 2と 3の場合、テーブル Oneと Two
の行の間には 1対多リレーションシップが存在し、テーブル Twoと Threeの行の間
には多対 1リレーションシップが存在します。
図 17.3 1対多および多対 1のリレーションシップ

TWO

1
2
2
3
3
3
4
5

28000
30000
40000
15000
20000
25000
35000
40000

ID Sales

1
2
3
4
5

15000
7000

15000
5000
8000

ID Quota

THREEONE

1
2
3
4
5

Joe Smith
Sally Smith
Cindy Long
Sue Brown
Mike Jones

ID Name

多対多リレーションシップ
多対多のカテゴリは、1つ以上の共通列の値に基づいて、各入力テーブルの複数の
行を関連付けることができることを意味します。
次の例では、BreakDownテーブルと Maintenanceテーブルの行が、Vehicle列の
共通の値によって関連付けられています。Vehicleの値は、どちらのテーブルでも一
意ではありません。Vehicleの値 AAAと CCCについて、これらのテーブルの行間に
多対多リレーションシップが存在します。

図 17.4 多対多リレーションシップ
BREAKDOWN

AAA
AAA
AAA
AAA
BBB
CCC
CCC

02MAR99
20MAY99
19JUN99
29NOV99
04JUL99

31MAY99
24DEC99

Vehicle BreakDownDate

MAINTENANCE

AAA
AAA
AAA
CCC
CCC
CCC
DDD
DDD
DDD

03JAN99
05APR99
10AUG99
28JAN99

16MAY99
07OCT99
24FEB99
22JUN99
19SEP99

Vehicle MaintenanceDate

テーブル結合方法の概要
テーブル結合方法
これらの方法を使用してテーブルを結合することができます。
n 連結
n インターリーブ
n 1対 1の読み取り
n マッチマージ

連結
次の図は、2つのテーブルを連結した結果を示しています。テーブルを連結すると、
あるテーブルの行が別のテーブルに追加されます。データプログラムは、すべての
行が処理されるまで Data1を順次読み取り、次に Data2を読み取ります。テーブル
Combinedには、連結の結果が含まれます。テーブルは、SETステートメントにリ
ストされている順序で処理されることに注意してください。
data concatenate;
   method run();
      set data1 data2;
   end;
enddata;
run;

注: スレッドプログラムの場合、特定のスレッドに入る行の順序は、BYステートメ
ントを使用しない限り未定義です。その場合でも、スレッドのスレッドプログラム
では、すべての行が他のテーブルからの行の前に受信されると想定する方法はあり
ません。

図 17.5 2つのテーブルの連結

詳細については、“DS2テーブルの結合: 方法” (270ページ)を参照してください。

インターリーブ
次のデータプログラムは、2つのテーブルをインターリーブします。インターリー
ブは、1つ以上の共通の列に基づいて、2つ以上のテーブルの行を散在させます。テ
ーブル Combinedは結果を示しています。
data interleave;
   method run();
      set data1 data2;
      by year;
   end;
enddata;
run;

図 17.6 2つのテーブルのインターリーブ

詳細については、“DS2テーブルの結合: 方法” (270ページ)を参照してください。

1対 1の読み取り
次の図は、1対 1の読み取りの結果を示しています。1対 1の読み取りでは、寄与す
る各テーブルのすべての列を含む行を作成することにより、2つ以上のテーブルの
行を結合します。行は、各テーブル内の相対的な位置に基づいて結合されます。つ
まり、あるテーブルの最初の行と別のテーブルの最初の行というように結合されま
す。最小のテーブルから最後の行を読み取った後、データプログラムは停止します。
テーブル Combinedは結果を示しています。
data one2one;
   method run();
      set data1;
      set data2;
   end;
enddata;
run;

次のデータプログラムでは、2つのテーブルが 1対 1で読み取られます。
図 17.7 1対 1の読み取り

詳細については、“DS2テーブルの結合: 方法” (270ページ)を参照してください。

マッチマージ
次の図は、マッチマージの結果を示しています。マッチマージは、1つ以上の共通
列の値に基づいて、2つ以上のテーブルの行を新しいテーブルの 1つの行に結合し
ます。次のデータプログラムでは、2つのテーブルが 1対 1で読み取られます。テ
ーブル Combinedは結果を示しています。
data one2one;
   method run();
      merge data1 data2;
      by year;
   end;
enddata;
run;

図 17.8 2つのテーブルのマッチマージ

詳細については、“DS2テーブルの結合: 方法” (270ページ)を参照してください。

テーブルの準備方法
テーブルを準備するためのガイドライン
テーブルを結合する前に、次のガイドラインに従って、必要な結果を生成してくだ
さい。
n テーブルの構造と内容を理解します。
n 一般的な問題の原因を調べます。
n 行が正しい順序になっていること、または正しい順序で取得できること(たとえ
ば、事前に並べ替えることによって )を確認してください。

n プログラムをテストします。

テーブルの構造と内容を理解する
データがどのように関連しているかを判断するには、テーブルの構造を調べてくだ
さい。テーブル構造を表示するには、DATASETSプロシジャまたは CONTENTSプロ
シジャを実行して記述子情報を表示します。記述子情報には、各テーブルの行数、
各列の名前と属性、および拡張属性(テーブルと列の拡張属性を含む)のアルファベッ
ト順リストが含まれます。行のサンプルを印刷するには、PRINTプロシジャまたは
REPORTプロシジャを使用します。
VTYPEや VLENGTHなどの関数を使用して、特定の記述子情報を表示することもで
きます。詳細については、“DS2関数” (SAS DS2言語リファレンス)を参照してくださ
い。

一般的な問題の原因を調べる
プログラムが正しく実行されない場合は、入力データに次のエラーがないか確認し
てください。
n 名前が同じで異なるデータを表す列

DS2は、指定された名前の列を 1つだけ新しいテーブルに含めます。名前が同じ
でデータが異なる列を持つ 2つのテーブルをマージする場合、最後に読み取られ
たテーブルの値が他のテーブルの値を上書きします。
エラーを修正するには、SETまたはMERGEステートメントで RENAME=テーブ
ルオプションを使用して、テーブルを結合する前に列の名前を変更します。また
は、DATASETSプロシジャを使用できます。

n データは同じだが属性が異なる共通の列
DS2がこれらの違いを処理する方法は、どの属性が異なるかによって異なりま
す。
o 型属性
型属性に互換性がない場合、DS2はデータプログラムの処理を停止し、列に
互換性がないことを示すエラーメッセージを発行します。
このエラーを修正するには、別の DS2データプログラムを使用して、必要に
応じて型を変更する必要があります。使用する DS2ステートメントは、列の
性質によって異なります。次の表には、列を結合するときの結果属性が含ま
れています。

表 17.1 型プロモーション
左側 右側 結果
BIGINT

BIGINT

BIGINT

BIGINT

INTEGER

SMALLINT

TINYINT

BIGINT

BIGINT

BIGINT

BIGINT

左側 右側 結果
BIGINT

INTEGER

INTEGER

INTEGER

INTEGER

BIGINT

INTEGER

SMALLINT

TINYINT

BIGINT

INTEGER

INTEGER

INTEGER

SMALLINT

SMALLINT

SMALLINT

SMALLINT

BIGINT

INTEGER

SMALLINT

TINYINT

BIGINT

INTEGER

SMALLINT

INTEGER

TINYINT

TINYINT

TINYINT

TINYINT

BIGINT

INTEGER

SMALLINT

TINYINT

BIGINT

INTEGER

INTEGER

TINYINT

DOUBLE

DOUBLE

INTEGER

REAL

BIGINT

DECIMAL (X,Y)

DOUBLE

FLOAT

DECIMAL

REAL

DOUBLE

DECIMAL (X,Z)

DOUBLE

DOUBLE

DOUBLE

DOUBLE

DOUBLE

DOUBLE

TIME1

TIMESTAMP

TIMESTAMP

DATE

TIME1

TIMESTAMP

DATE

DATE

TIME1

TIMESTAMP

ERROR

DATE

VARBINARY1

BINARY

BINARY

VARBINARY

BINARY

VARBINARY

VARBINARY

BINARY

VARBINARY

CHAR(N)

CHAR(N)2

VARCHAR(N)

VARCHAR(N)

CHAR(N)

VARCHAR(N)

NCHAR(M)

CHAR(N)

CHAR(N)2

VARCHAR(N)

VARCHAR(M)

CHAR(M)

CHAR(M)

VARCHAR(M)

CHAR(N)

VARCHAR(N)2

VARCHAR(N)

VARCHAR(MAX (N, M))

VARCHAR (MAX (N, M))

VARCHAR (MAX (N, M))

VARCHAR(M)

左側 右側 結果
NCHAR(M) NCHAR(M) NCHAR(M)

INTEGER CHARACTER ERROR

DATE DOUBLE ERROR

1 Hiveのサポートなし
2 DB2、Impala、ODBC、Oracleの場合

注: ベストプラクティスは、スレッドプログラムですべての変数を宣言するこ
とです。これにより、データソース間の型の不一致を回避できます。

o 長さ属性
長さ属性が異なる場合、DS2は最大長の列を含むテーブルから長さを取得し
ます。次の例では、MERGEステートメントにリストされているすべてのテー
ブルに列Mileageが含まれています。Quarter1では、列Mileageの長さは
4バイトです。Quarter2では 8バイト、Quarter3と Quarter4では 6バイ
トです。出力テーブル Yearlyでは、列Mileageの長さは 8バイトで、これは
Quarter2から派生した長さです。
data yearly;
   dcl char(4) quarter1 char(8) quarter2 char(6) quarter3 quarter4;
   method run();
      merge quarter1 quarter2 quarter3 quarter4;
      by Account;
   end;
enddata;
run;

o ラベル、出力形式、入力形式属性
これらの属性のいずれかが異なる場合、DS2は、その属性を持つ列を含む最
初のテーブルから属性を取得します。ただし、明示的に指定したラベル、出
力形式、または入力形式はデフォルトをオーバーライドします。すべてのテ
ーブルに明示的に指定された属性が含まれている場合、最初のテーブルで指
定された属性が他の属性をオーバーライドします。
VLABELなどの関数を使用して、特定の記述子情報を表示することもできま
す。詳細については、“DS2関数” (SAS DS2言語リファレンス)を参照してくだ
さい。

プログラムのテスト
テーブルを準備する最後のステップとして、プログラムをテストする必要がありま
す。プログラムのすべてのロジックをテストする行のサンプルを含む小さな一時テ
ーブルを作成します。ロジックに問題があり、予期しない出力が得られた場合は、
プログラムをデバッグできます。

DS2テーブルの結合: 方法
連結

定義
テーブルの連結とは、2つ以上のテーブルを次々に組み合わせて 1つのテーブルに
することです。新しいテーブルの行数は、元のテーブルの行数の合計です。行の順
序はシーケンシャルです。最初のテーブルのすべての行の後に、2番目のテーブル
のすべての行が続きます。
最も単純なケースでは、すべての入力テーブルに同じ列が含まれます。入力テーブ
ルに異なる列が含まれている場合、あるテーブルの行には、他のテーブルでのみ定
義されている列の欠損値があります。いずれの場合も、新しいテーブルの列は古い
テーブルの列と同じです。

構文
テーブルを連結するには、次の形式の SETステートメントを使用します。
SET table(s);

引数
table(s)
有効なテーブル名を指定します。

詳細については、“SET ステートメント” (SAS DS2言語リファレンス)を参照してくだ
さい。

連結中の DS2処理
コンパイルフェーズ

DS2は、SETステートメントで指定された各テーブルの記述子情報を読み取り、
すべてのテーブルのすべての列とデータプログラムによって作成された列を含
むプログラムデータベクトルを作成します。

実行 - ステップ 1
DS2は、最初のテーブルの最初の行をプログラムデータベクトルに読み込みま
す。最初の行を処理し、データプログラムの他のステートメントを実行します。
次に、プログラムデータベクトルの内容を新しいテーブルに書き込みます。

SETステートメントは、データプログラム中に値が計算または割り当てられる列
を除いて、プログラムデータベクトルの値を欠損値にリセットしません。データ
プログラムによって作成された列は、保持されない限り、データプログラムの各
反復の開始時に欠損に設定されます。テーブルから読み取られる変数はそうで
はありません。

実行 - ステップ 2
データプログラムでは、DS2は、ファイルの終わりインジケーターが見つかるま
で、最初のテーブルから一度に 1行ずつ読み取り続けます。次に、プログラムデ
ータベクトルの列の値が欠損に設定され、データプログラムは 2番目のテーブル
から行の読み取りを開始し、すべてのテーブルからすべての行を読み取るまで、
これを繰り返していきます。スレッドプログラムの場合、特定のスレッドに入る
行の順序は、BYステートメントを使用しない限り未定義です。その場合でも、
スレッドのスレッドプログラムでは、すべての行が他のテーブルからの行の前に
受信されると想定する方法はありません。

例 1: データプログラムを使用した連結
この例では、各テーブルに列 Commonと列 Numberが含まれており、行は
Commonの値の順に並べられています。通常、同じ列を持つテーブルを連結しま
す。この場合、各テーブルには、テーブルの結合による効果をより明確に示すため
の一意の列も含まれています。次のプログラムは、SETステートメントを使用して
テーブルを連結し、結果を出力します。
data animal(overwrite=yes);
   dcl varchar(10) common animal number;
   method init();
      common='a'; animal='Ant'; number='5'; output;
      common='b'; animal='Bird'; number=''; output;
      common='c'; animal='Cat'; number='17'; output;
      common='d'; animal='Dog'; number='9';  output;
      common='e'; animal='Eagle'; number=''; output;
      common='f'; animal='Frog';  number='76'; output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant number;
   method init();
      common='g'; plant='Grape'; number='69'; output;
      common='h'; plant='Hazelnut'; number='55'; output;
      common='i'; plant='Indigo'; number=''; output;
      common='j'; plant='Jicama';  number='14'; output;
      common='k'; plant='Kale'; number='5'; output;
      common='l'; plant='Lentil';  number='77'; output;
   end;
enddata; 
run;

/*  set concatenates */

data concatenate (overwrite=yes);
   method run();
      set animal plant; 
   end;
enddata;
run;

proc print data=concatenate;
run;
quit;

アウトプット 17.1 連結テーブル(SETステートメント)

結果のテーブル Concatenateには 12行あります。これは、結合されたテーブルの
行の合計です。プログラムデータベクトルには、すべてのテーブルのすべての列が
含まれます。あるテーブルでは見つかったが別のテーブルでは見つからなかった列
の値は、欠損に設定されます。

例 2: SQLを使用した連結
SQL言語を使用してテーブルを連結することもできます。この例では、SQLは両方
のテーブルの各行を読み取り、Combinedという名前の新しいテーブルを作成しま
す。YEAR1および YEAR2入力テーブルを次に示します。
次の SQLコードは、テーブル Combinedを作成して出力します。

proc sql;
   create table combined as
      select * from animal
            union all
         select * from plant;
         select * from combined;
quit;

proc print data=combined;
run;

出力は、SETステートメントを使用した場合の出力とまったく同じです。

インターリーブ
定義
インターリーブでは、SETステートメントと BYステートメントを使用して、複数の
テーブルを 1つの新しいテーブルに結合します。これは、BYグループ処理とも呼ば
れます。BYグループ処理は、1つ以上の共通列の値によってグループ化または順序
付けされた 1つ以上のテーブルの行を結合する方法です。SETステートメントの直
後に BYステートメントを指定すると、SETステートメントは入力テーブルの行を並
べ替え順にインターリーブします。並べ替え順序または並べ替えキーは、BYステー
トメントの列名によって指定されます。新しいテーブルの行数は、元のテーブルの
行数の合計です。

注: 文字データ型の列を持つテーブルがあると仮定します。DECLAREステートメン
トを使用した入力時に列を数値データ型に変更した場合、結果の列の並べ替え順は
数値ではありません。たとえば、次の文字列(CHAR)がテーブルに存在するとします: 
9、10、500。SETおよび BYステートメントを使用してテーブルを読み取るときに
sを数値列(DOUBLE)として宣言すると、データは英数字順、つまり 10、500、9で
出力として生成されます。SETステートメントは、文字列が数値データ型に変換さ
れる前に、行を英数字順に並べ替えます。

列を昇順ではなく降順で並べ替えるには、BYステートメントの列名の前にキーワー
ド DESCENDINGを使用します。

構文
BY変数を使用する場合、次の形式の SETステートメントを使用してテーブルをイン
ターリーブします。
SET table(s);
BY [DESCENDING]column […[DESCENDING] column];

引数
table
テーブル名を指定します。

DESCENDING
指定された列によってテーブルが降順で並べ替えられることを指定します。
DESCENDINGは、数値列の場合は最大から最小へ、文字列の場合はアルファベ
ットの逆順になります。

column
テーブルを並べ替えるための各列を指定します。これらの列は、現在のデータプ
ログラムの BY変数と呼ばれます。

並べ替え要件
BYステートメントを使用すると、内部的に DS2は並べ替えられた順序で行を要求
します。行がすでに並べ替えられている場合は、データの"再並べ替え"が必要にな
ることがあります。

インターリーブ中の DS2処理
コンパイルフェーズ
n DS2は、SETステートメントで指定された各テーブルの記述子情報を読み取
り、すべてのテーブルのすべての列とデータプログラムによって作成された
列を含むプログラムデータベクトルを作成します。

n DS2プログラムでは、SASは BY列ごとに FIRST.variableと LAST.variableの
2つの一時変数を作成することにより、各 BYグループの開始と終了を識別し
ます。これらの値は、行に次の特性があるかどうかを示します。
o BYグループの最初
o BYグループの最後
o BYグループの最初でも最後でもない
o BYグループに行が 1つしかない場合のように、最初と最後の両方
行が BYグループの最初の行である場合、値が変更された列、および BYステ
ートメント内の後続のすべての列に対して、SASは FIRST.variableの値を 1
に設定します。BYグループの他のすべての行では、FIRST.variableの値は 0
です。同様に、その行が BYグループの最後の行である場合、次の行で値が変
更される列、および BYステートメント内の後続のすべての列に対して、SAS
は LAST.variableの値を 1に設定します。BYグループの他のすべての行で
は、LAST.variableの値は 0です。テーブルの最後の行では、すべての
LAST.variable変数の値が 1に設定されます。これらの一時変数は DS2プロ
グラミングで使用できますが、出力テーブルには追加されません。
BYグループの最初の行を処理しているか、最後の行を処理しているかに基づ
いて、条件付きでアクションを実行できます。

注:  BYグループ処理を示す例については、“テーブルのインターリーブ ” (SAS 
DS2言語リファレンス)を参照してください。

注: SPD Engineデータセットの場合、ユーティリティファイルは、余分なス
ペースを必要とする特定の操作に使用されます。BYステートメントにはユ
ーティリティファイルが必要で、SAS UTILLOC=システムオプションはそのユ
ーティリティファイルにスペースを割り当てます。詳細については、SASシ
ステムオプション: リファレンスの SAS UTILLOC=システムオプションを参
照してください。

実行 - ステップ 1
DS2は、SETステートメントで指定された各テーブルの最初の行を比較して、ど
の BYグループを新しいテーブルに最初に出現させるかを決定します。選択した
テーブルの最初の BYグループからすべての行を読み取ります。この BYグルー
プが複数のテーブルに出現する場合、SETステートメントに出現する順序でテー
ブルから読み取ります。プログラムデータベクトル内の列の値は、DS2が新しい
テーブルの読み取りを開始するたびに、また BYグループが変更されるたびに、
欠損に設定されます。

実行 - ステップ 2
DS2は、各テーブルの次の行を比較して次の BYグループを決定し、この BYグ
ループの行を含む SETステートメントで選択されたテーブルから行の読み取り
を開始します。DS2は、すべてのテーブルからすべての行を読み取るまで続行し
ます。

例 1: 最も単純なケースでのインターリーブ
この例では、各テーブルに BY変数 Commonが含まれており、行は BY変数の値の
順に並べられています。次の例では、Animalおよび Plant入力テーブルを作成しま
す。
data animal(overwrite=yes);
   dcl varchar(10) common animal number;
   method init();
      common='a'; animal='Ant'; number='5'; output;
      common='b'; animal='Bird'; number=''; output;
      common='c'; animal='Cat'; number='17'; output;
      common='d'; animal='Dog'; number='9';  output;
      common='e'; animal='Eagle'; number=''; output;
      common='f'; animal='Frog';  number='76'; output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant number;
   method init();
      common='a'; plant=''; number='69'; output;
      common='b'; plant='Bamboo'; number='55'; output;
      common='c'; plant='Cabbage'; number=''; output;
      common='d'; plant='Daffodil';  number='14'; output;
      common='e'; plant='Eucalyptus; number='5'; output;
      common='f'; plant='Fig';  number='77'; output;
   end;

enddata; 
run;

次のプログラムは、SETステートメントと BYステートメントを使用してテーブルを
インターリーブし、結果を出力します。

/*  set with by interleaves */
data interleave (overwrite=yes);
   method run();
      set animal plant; by common; 
   end;
enddata;
run;

proc print data=interleave;
run;

アウトプット 17.2 インターリーブテーブル (SETステートメント)

結果のテーブル Interleaveは 12行あります。これは、結合されたテーブルの行の
合計です。新しいテーブルには、両方のテーブルのすべての列が含まれます。一方
のテーブルで見つかったが、もう一方のテーブルでは見つからなかった列の値は、
欠損に設定され、行は BY変数の値で並べられます。

例 2: BY変数の重複値によるインターリーブ
テーブルに BY変数の重複値が含まれている場合、行は元のテーブルに出現した順序
で新しいテーブルに書き込まれます。この例には、BY変数 Commonの重複値が含
まれています。次のプログラムでは、Animalおよび Plant入力テーブルを作成しま
す。
data animal(overwrite=yes);
   dcl varchar(10) common animal number;
   method init();

      common='a'; animal='Ant'; number='5'; output;
      common='a'; animal='Bird'; number=''; output;
      common='b'; animal='Cat'; number='17'; output;
      common='c'; animal='Dog'; number='9';  output;
      common='d'; animal='Eagle'; number=''; output;
      common='e'; animal='Frog';  number='76'; output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant number;
   method init();
      common='a'; plant='Grape'; number='69'; output;
      common='b'; plant='Hazelnut'; number='55'; output;
      common='c'; plant='Indigo'; number=''; output;
      common='c'; plant='Jicama';  number='14'; output;
      common='d'; plant='Kale'; number='5'; output;
      common='e'; plant='Lentil';  number='77'; output;
   end;
enddata; 
run;

次のプログラムは、SETステートメントと BYステートメントを使用してテーブルを
インターリーブし、結果を出力します。

/*  set with by interleaves */
data interleave (overwrite=yes);
   method run();
      set animal plant; by common; 
   end;
enddata;
run;

proc print data=interleave;
run;
quit;

アウトプット 17.3 複数の BY変数を含むインターリーブテーブル (SETステートメ
ント)

新しいテーブルの行数は、すべてのテーブルの行の合計です。行は、元のテーブル
に出現した順序で新しいテーブルに書き込まれます。

例 3: 各テーブルで異なる BY値を使用したインター
リーブ
テーブル Animalと Plantの両方に、一方のテーブルには存在するが他方のテーブル
には存在しない値が含まれています。次のプログラムでは、Animalおよび Plant入
力テーブルを作成します。
data animal(overwrite=yes);
   dcl varchar(10) common animal number;
   method init();
      common='a'; animal='Ant'; number='5'; output;
      common='c'; animal='Bird'; number=''; output;
      common='d'; animal='Cat'; number='17'; output;
      common='e'; animal='Dog'; number='9';  output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant number;
   method init();
      common='a'; plant='Grape'; number='69'; output;
      common='b'; plant='Hazelnut'; number='55'; output;

      common='c'; plant='Indigo'; number=''; output;
      common='d'; plant='Jicama';  number='14'; output;
      common='e'; plant='Kale'; number='5'; output;
      common='f'; plant='Lentil';  number='77'; output;
   end;
enddata; 
run;

このプログラムは、SETステートメントと BYステートメントを使用してこれらのテ
ーブルをインターリーブし、結果を出力します。
data interleave (overwrite=yes);
   method run();
      set animal plant; by common; 
   end;
enddata;
run;

proc print data=interleave;
run;
quit;

アウトプット 17.4 異なる BY変数を含むインターリーブテーブル (SETステートメ
ント)

結果のテーブルには、BY変数の値で並べられた 10行が含まれます。

コメントと比較
n 他の言語では、マージという用語はしばしばインターリーブを意味するために使
用されます。DS2では、マージという用語は、2つ以上のテーブルの行を 1つの
行に結合する操作に予約されています。インターリーブされたテーブルの行は
結合されません。これらは、BY変数の値の順序で元のテーブルからコピーされ
ます。

n 1つのテーブルに同じ BY値を持つ複数の行がある場合、DATAステップはそれら
の行の順序を結果に保持します。

n データプログラムを使用するには、入力テーブルを適切に並べ替える必要があり
ます。SQLでは、入力テーブルが順番に並んでいる必要はありません。

1対 1の読み取り
定義
1対 1の読み取りでは、2つ以上の SETステートメントを使用して各テーブルから
行を個別に読み取ることにより、2つ以上のテーブルの行を 1つの行に結合します。
このプロセスは、1対 1のマッチングとも呼ばれます。新しいテーブルには、すべ
ての入力テーブルのすべての列が含まれます。新しいテーブルの行数は、最小の元
のテーブルの行数です。テーブルに共通の列が含まれている場合、最後のテーブル
から読み込まれた値は、以前のテーブルから読み込まれた値を置き換えます。

構文
1対 1の読み取りには、次の形式の SETステートメントを使用します。
SET table-1;

SET table-2;

引数
table-1
テーブル名を指定します。table-1は、データプログラムが読み取る最初のテー
ブルです。

table-2
テーブル名を指定します。table-2は、データプログラムが読み取る 2番目のテ
ーブルです。

注意
複数の SETステートメントでテーブルを結合する場合は注意してください。 複数の
SETステートメントを使用して行を結合すると、望ましくない結果が生じる可能性があ
ります。この方法を使用してテーブルを結合する前に、テーブルの代表的なサンプルで
プログラムをテストしてください。

詳細については、“SET ステートメント” (SAS DS2言語リファレンス)を参照してくだ
さい。

1対 1の読み取り中の DS2処理
コンパイルフェーズ

DS2は、SETステートメントで指定された各テーブルの記述子情報を読み取り、
すべてのテーブルのすべての列とデータプログラムによって作成された列を含
むプログラムデータベクトルを作成します。

実行 - ステップ 1
DS2が最初の SETステートメントを実行すると、DS2は最初のテーブルの最初
の行をプログラムデータベクトルに読み込みます。2番目の SETステートメン
トは、2番目のテーブルの最初の行をプログラムデータベクトルに読み込みま
す。両方のテーブルに同じ列が含まれている場合、値が欠損していても、最初の
テーブルの値が 2番目のテーブルの値に置き換えられます。最後のテーブルか
ら最初の行を読み取り、データプログラム内の他のステートメントを実行した
後、DS2はプログラムデータベクトルの内容を新しいテーブルに書き込みます。
SETステートメントは、データプログラム中に作成された列または値が割り当て
られた列を除いて、プログラムデータベクトルの値を欠損値にリセットしませ
ん。

実行 - ステップ 2
DS2は、いずれかのテーブルでファイルの終わりインジケーターを検出するま
で、一方のテーブルからの読み取りを続け、次に他方のテーブルからの読み取り
を続けます。DS2は、最も短いテーブルの最後の行で処理を停止し、長いテーブ
ルから残りの行を読み取りません。

例 1: 1対 1の読み取り: 等しい行数の処理
テーブル Animalとテーブル Plantの両方に列 Commonが含まれており、その列の
値によって並べられています。次のプログラムでは、Animalおよび Plant入力テー
ブルを作成します。
data animal(overwrite=yes);
   dcl varchar(10) common animal number;
   method init();
      common='a'; animal='Ant';  output;
      common='b'; animal='Bird';  output;
      common='c'; animal='Cat';  output;
      common='d'; animal='Dog';   output;
      common='e'; animal='Eagle';  output;
      common='f'; animal='Frog';   output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant number;
   method init();
      common='a'; plant='Grape'; output;

      common='b'; plant='Hazelnut'; output;
      common='c'; plant='Indigo';  output;
      common='d'; plant='Jicama';  output;
      common='e'; plant='Kale';  output;
      common='g'; plant='Lentil';  output;
   end;
enddata; 
run;

次のプログラムは、2つの SETステートメントを使用して Animalと Plantの行を結
合し、結果を出力します。
data one2one (overwrite=yes);
   method run();
      set animal;
      set plant; 
   end;
enddata;
run;

proc print data=one2one;
run;
quit;

アウトプット 17.5 等しい行数での 1対 1の読み取り(SETステートメント)

新しいテーブルの各行には、すべてのテーブルのすべての列が含まれています。た
だし、6行目の Common列の値には"g"が含まれていることに注意してください。
Animalテーブルの 6行目の Commonの値は、Plantの値によって上書きされまし
た。Plantは、DS2が最後に読み取ったテーブルです。

行数が不均等な 1対 1の読み取り
テーブル Animalとテーブル Plantの両方に列 Commonが含まれており、その列の
値によって並べられています。テーブルの行数が異なります。次のプログラムで
は、Animalおよび Plant入力テーブルを作成します。
data animal(overwrite=yes);

   dcl varchar(10) common animal;
   method init();
      common='a'; animal='Ant'; output;
      common='a'; animal='Bird';  output;
      common='b'; animal='Cat';  output;
      common='c'; animal='Dog';  output;
      common='d'; animal='Eagle';  output;
      common='e'; animal='Frog';  output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant;
   method init();
      common='a'; plant='Grape';output;
      common='b'; plant='Hazelnut';  output;
      common='c'; plant='Indigo'; output;
      common='d'; plant='Jicama';  output;
   end;
enddata; 
run;

次のプログラムは、2つの SETステートメントを使用して Animalと Plantの行を結
合し、結果を出力します。
data one2onerowsnotequal (overwrite=yes);
   method run();
      set animal;
      set plant;
   end;
enddata;
run;
quit;

proc print data=one2onerowsnotequal;
run;

アウトプット 17.6 行数が不均等な 1対 1の読み取り(SETステートメント)

DS2は最も短いテーブルの最後の行で処理を停止するため、結果テーブルには 4つ
の行しか含まれません。

コメントと比較
n 複数の SETステートメントを他の DS2ステートメントとともに使用すると、次
の応用が可能になります。
o 1つの行を多数の行とマージする
o 行の条件付きマージ
o 同じテーブルから 2回読み取る

マッチマージ
定義
マッチマージは、共通の列の値に従って、2つ以上のテーブルの行を新しいテーブ
ルの 1つの行に結合します。新しいテーブルの行数は、すべてのテーブルの各 BYグ
ループの最大行数の合計です。マッチマージを実行するには、必要な BYステートメ
ントとともにMERGEステートメントを使用します。マッチマージを実行すると、
すべてのテーブルが BYステートメントで指定した列によって並べ替えられます。

構文
テーブルをマッチマージするには、次の形式のMERGEステートメントを使用しま
す。
MERGE table(s);
BY column(s);

引数
table
行が読み取られる既存のテーブルを少なくとも 2つ指定します。

column
テーブルを並べ替えるための各列の名前を指定します。これらの列は BY変数と
呼ばれます。

詳細については、“MERGE ステートメント” (SAS DS2言語リファレンス)および“BY 
ステートメント” (SAS DS2言語リファレンス)を参照してください。

マッチマージ中の DS2処理

コンパイルフェーズ
DS2は、MERGEステートメントで指定された各テーブルの記述子情報を読み取
り、すべてのテーブルのすべての行とデータプログラムによって作成された行を
含むプログラムデータベクトルを作成します。

実行 - ステップ 1
DS2は、MERGEステートメントで指定された各テーブルの最初の BYグループ
を調べて、どの BYグループを新しいテーブルに最初に出現させるかを決定しま
す。データプログラムは、各テーブルからその BYグループの最初の行をプログ
ラムデータベクトルに読み込み、MERGEステートメントに記述された順序でテ
ーブルを読み取ります。テーブルの BYグループに行がない場合、プログラムデ
ータベクトルには、そのテーブルに固有の行の欠損値が含まれます。

実行 - ステップ 2
入力テーブルの各行は、出力テーブルで 1回だけ使用されます。テーブルに固有
の列は、BYグループの作成中にそのテーブルが使い果たされると、欠損値また
は null値で埋められます。

実行 - ステップ 3
DS2は、すべてのテーブルのすべての BYグループからすべての行を読み取るま
で、これらのステップを繰り返します。

注意
DECIMALまたは NUMERICデータ型を持つ DS2マージの BY変数は、DOUBLEデ
ータ型に変換されます。 一致する DECIMAL列が BY変数でない場合、DECIMAL列は
DECIMALデータ型のままになります。

注意
テーブル間で DECIMALまたは NUMERICデータ型の列間に型、スケール、または
精度の不一致がある場合、その列は DOUBLEデータ型に変換されます。

RETAINを使用して DATAステップマージを作成す
る
MERGEステートメントで RETAIN引数を使用して、DATAステップマージと同様の
多対多のマッチマージを生成できます。ただし、BYグループ内の行の順序はほとん
どのデータソースで定義されていないため、RETAINを使用したMERGEには多くの
正解がある可能性があります。したがって、同じプログラムから異なる結果が得ら
れる可能性があります。どちらも正しいものです。
RETAIN引数を指定すると、特定の BYグループのデータセットの最後の行が、寄与
しているデータセットのいずれにも行がなくなるまで繰り返し使用されます。
詳細については、“MERGE ステートメント” (SAS DS2言語リファレンス)を参照して
ください。

例 1: 行の結合
テーブル Animalと Plantにはそれぞれ BY変数 Commonが含まれており、行は BY
変数の値の順に並べられています。次のプログラムでは、Animalおよび Plant入力
テーブルを作成します。
data animal(overwrite=yes);
   dcl varchar(10) common animal;
   method init();
      common='a'; animal='Ant';  output;
      common='b'; animal='Bird';  output;
      common='c'; animal='Cat';  output;
      common='d'; animal='Dog';   output;
      common='e'; animal='Eagle';  output;
      common='f'; animal='Frog';   output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant;
   method init();
      common='a'; plant='Grape'; output;
      common='b'; plant='Hazelnut'; output;
      common='c'; plant='Indigo';  output;
      common='d'; plant='Jicama';  output;
      common='e'; plant='Kale';  output;
      common='f'; plant='Lentil';  output;
   end;
enddata; 
run;

次のプログラムは、BY変数 Commonの値に従ってテーブルをマージし、結果を出
力します。
data mmerge (overwrite=yes);
   method run();
      merge animal plant;
      by common;
   end;
enddata;
run;
quit;

proc print data=mmerge;
run;
quit;

アウトプット 17.7 単純マッチマージ(MERGEステートメント)

新しいテーブルの各行には、すべてのテーブルのすべての列が含まれています。

例 2: BY変数の値が重複するマッチマージ
次の例では、テーブル Animalと Plantに BY変数 Commonの重複値が含まれてい
ます。次のプログラムでは、Animalおよび Plant入力テーブルを作成します。
data animal(overwrite=yes);
   dcl varchar(10) common animal;
   method init();
      common='a'; animal='Ant'; output;
      common='a'; animal='Ape';  output;
      common='b'; animal='Bird';  output;
      common='c'; animal='Cat';  output;
      common='d'; animal='Dog';  output;
      common='e'; animal='Eagle';  output;
   end;
enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant;
   method init();
      common='a'; plant='Apple';output;
      common='b'; plant='Banana';  output;
      common='c'; plant='Coconut'; output;
      common='c'; plant='Celery'; output;
      common='d'; plant='Dewberry';  output;
      common='e'; plant='Eggplant'; output;
   end;
enddata; 
run;

次のプログラムは、マージされたテーブル MATCH1を生成し、結果を出力します。
data mmdiffby (overwrite=yes);
   method run();
      merge animal plant;
      by common;

   end;
enddata;
run;
quit;

proc print data=mmdiffby;
run;
quit;

アウトプット 17.8 重複する BY変数を使用したマッチマージ(MERGEステートメ
ント)

出力の 2行目では、列 Plantの値は保持されません。マッチマージでは、5行目の
Animalの値も重複しませんでした。

注: MERGEステートメントは、多対多のマッチマージでデカルト積を生成しませ
ん。かわりに、少なくとも 1つのテーブルに BYグループの行があるときに、1対 1
マージを実行します。BYグループのすべての行が 1つのテーブルから読み取られ、
別のテーブルにまだ行がある場合、DS2は列を欠損値または null値で埋めます。

例 3: 不一致行でのマッチマージ
DS2が入力テーブル内の不一致行でマッチマージを実行する場合、値が欠損してい
ても、DS2はプログラムデータベクトル内のすべての列の値を保持します。テーブ
ル Animalおよび Plantには、BY変数 Commonのすべての値が含まれているわけ
ではありません。次のプログラムでは、Animalおよび Plant入力テーブルを作成し
ます。
data animal(overwrite=yes);
   dcl varchar(10) common animal;
   method init();
      common='a'; animal='Ant'; output;
      common='c'; animal='Cat';  output;
      common='d'; animal='Dog';  output;
      common='e'; animal='Eagle';  output;
   end;

enddata;
run;

data plant(overwrite=yes);
   dcl varchar(10) common plant;
   method init();
      common='a'; plant='Apple';output;
      common='b'; plant='Banana';  output;
      common='c'; plant='Coconut'; output;
      common='e'; plant='Eggplant'; output;
      common='f'; plant='Fig';  output;end;
enddata; 
run;

次のプログラムは、マージされたテーブル Mmnomrowを生成し、結果を出力しま
す。
data mmnomrow (overwrite=yes);
   method run();
      merge animal plant;
      by common;
   end;
enddata;
run;
quit;

proc print data=mmnomrow;
run;
quit;

アウトプット 17.9 不一致行でのマッチマージ(MERGEステートメント)

出力が示すように、列 Commonのすべての値が新しいテーブルに表示されます。
これには、一方のテーブルにあって他方のテーブルにない列の欠損値が含まれます。

DS2言語の予約語
次の単語は DS2言語キーワードとして予約されており、変数名として使用したり、
本来の用途と異なる方法で使用したりすることはできません。

注: 予約語は、二重引用符で囲むと変数名として使用できます。詳細と例について
は、“区切り識別子” (58ページ)を参照してください。

表 18.1 DS2予約語

特殊文字
___KPLIST

_ALL_

_HOSTNAME_

_NEW_

_LOCALNTHREAD
S_

_LOCALTHREADI
D_

_NTHREADS_

_NULL_

_RC_

_ROWSET_

_TEMPORARY_

_THREADID_

A

ABORT

AND

AS

ASM

B

BIGINT

BINARY

BY

C

CALL

CATALOG

CHAR

CHARACTER

COMMIT

CONTINUE

D

DATA

DATE

DCL

DECIMAL

DECLARE

DELETE

DESCENDING

DIM

DO

DOUBLE

DROP

DS2_OPTIONS

E

ELIF

ELSE

END

ENDDATA

ENDMODULE

ENDPACKAGE

ENDSTAGE

ENDTABLE

ENDTHREAD

EQ

ERROR

ESCAPE

F

FLOAT

FORMAT

FORTRAN

FORWARD

FROM

FUNCTION

G

GE

GLOBAL

GOTO

GROUP

GT

H

HAVING

I

IDENTITY

IF

IN

INDSNAME

INDSNUM

INFORMAT

INLINE

INPUT

INT

INTEGER

IN_OUT

K

KEEP

L

LABEL

LE

LEAVE

LIKE

LIST

LT

M

MERGE

METHOD

MISSING

MODIFY

MODULE

N

NATIONAL

NCHAR

NE

NG

NL

NOT

NULL

NUMERIC

NVARCHAR

O

ODS

OF

OR

ORDER

OTHER

OTHERWISE

OUTPUT

OVERWRITE

P

PACKAGE

PARTITION

PRECISION

PRIVATE

PROGRAM

PUT

R

REAL

REMOVE

RENAME

REPLACE

REQUIRE

RETAIN

RETURN

RETURNS

ROLLBACK

S

SELECT

SET

SMALLINT

SQLSUB

STAGE

STOP

STORED

SUBSTR

SYSTEM

T

TABLE

THEN

THIS

THREAD

THREADS

TIME

TIMESTAMP

TINYINT

TO

TRANSACTION

T_UDF

TSPL_OPTIONS

U

UNTIL

UPDATE

V

VARARRAY

VARBINARY

VARCHAR

VARLIST

VARYING

W

WHEN

WHERE

WHILE

CASでの DS2の実行方法
CASの DS2

DS2プログラムは、CASでも SASクライアントでも実行できます。CASは、分散並
列実行を可能にするクラウドベースのランタイム環境です。CASで実行される DS2
プログラムは、通常、複数のマシンに分散された複数のスレッドを使用して、イン
メモリ CASテーブルで動作します。DS2プログラムを CASで実行する方法は、SAS
クライアントで実行する方法とほとんど同じです。CASの DS2は、同じ言語要素の
ほとんどをサポートし、SASクライアントの DS2とほとんど同じ機能を提供します

CASで DS2プログラムを実行する理由
単一のマシンでのビッグデータ処理は低速です。DS2プログラムを CASで実行す
ると、複数のマシンでプログラムを実行し、そのマシンのスレッド間で処理ワーク
ロードを分割することにより、並列実行によってビッグデータの処理が高速化され
ます。

処理モード
n シングルスレッド
プログラムは、単一のマシン上の単一のスレッドで実行されます。単一のスレッ
ドがすべてのデータを処理します。非常に大きなデータセットの場合、シングル
スレッドでの実行はお勧めしません。

n マルチスレッド
プログラムは、複数のマシンに分散された複数のスレッドで並列して実行されま
す。各スレッドは、データの一部を処理します。

DS2スレッドプログラムで SET FROMステートメントを使用する DS2プログラム
のみがマルチスレッドで実行されます。他のすべての DS2プログラムは、単一のマ
シン上の単一のスレッドで実行されます。単一のスレッドがすべてのデータを順次
処理します。
DS2スレッドプログラムで SET FROMステートメントを使用する DS2プログラム
は、部分的または完全にマルチスレッドで実行されます。DS2スレッドプログラム
は常にマルチスレッドで実行されますが、DS2データプログラムは、それに含まれ
るステートメントが SET FROMステートメントと、オプションとして OUTPUTステ
ートメントだけである場合にのみマルチスレッドで実行されます。DS2データプロ
グラムに他のステートメントが含まれている場合、DS2データプログラムはシング
ルスレッドで実行されます。DS2スレッドプログラムがマルチスレッドで実行さ
れ、DS2データプログラムがシングルスレッドで実行される DS2プログラムは、部
分的にマルチスレッドです。

注: SET FROMステートメントで明示的なスレッド数が指定されている場合、DS2
プログラムは、各ワーカーで指定された数のスレッドで実行されます。たとえば、
THREADS=4で CASセッションに 3つのワーカーが含まれている場合、3つの CAS
ワーカーはそれぞれ 4つのスレッドを使用し、合計 12のスレッドが DS2プログラ
ムを実行します。SET FROMステートメントでスレッド数が明示的に指定されてい
ない場合、各 CASワーカーのスレッド数はデフォルトでライセンスされたスレッド
制限になります。

マルチスレッド処理
データプログラムが DS2スレッドから設定され、データプログラムに SET FROMス
テートメントと、オプションとして OUTPUTステートメントのみが含まれている場
合、DS2プログラムはマルチスレッドで実行されます。DS2プログラムがマルチス
レッドで実行される場合、オブザベーションは複数のスレッドで処理され、すべて
の CASワーカーに並列に分散されます。マルチスレッド実行中、各 CASワーカーは
入力テーブルのサブセットを処理し、出力テーブルのサブセットを生成します。
CASデータ管理は、各 CASワーカーをサポートし、入力テーブルのサブセットを並
列に読み取り、出力テーブルのサブセットを並列に書き込みます。
DS2プログラムをマルチスレッドで実行する方法を次に示します。
n 各 CASワーカーの各スレッドで、DS2スレッドプログラムは次のことを行いま
す。
o 入力テーブルからデータのサブセットを読み取ります。
o データのサブセットを処理します。
o データのサブセットを DS2データプログラムに送信します。

n 各 CASワーカーの各スレッドで、DS2データプログラムは次のことを行います。
o DS2スレッドプログラムからデータのサブセットを受け取ります。
o データのサブセットを出力テーブルに書き込みます。

注: DS2データプログラムは、データを出力テーブルに書き込む以外に、データの
追加処理を実行しません。

マルチスレッドの例
次の例では、DS2プログラムは carsをフィルタリングして、msrpが$100,000より
大きい車だけを表示します。プログラムは複数のスレッドで実行され、各スレッド
が carsテーブルからレコードのサブセットを読み取ります。また、各スレッドは、
レコードのサブセットを cars_luxuryテーブルに書き込みます。

cas casauto; 1
proc casutil sessref=casauto; 2
   load data=sashelp.cars;
run;

proc ds2 sessref=casauto; 3
thread cars_thd / overwrite=yes; 4
   method run();
      set cars;
      if (msrp > 100000) then do;
         output;

   end;
end;
endthread;

data cars_luxury / overwrite=yes; 5
   dcl thread cars_thd t;
   method run();
      set from t;
   end;
enddata;
run;
quit;

1 CASサーバーで casautoという名前のセッションを開始します。
2 casauto CASセッションで、sashelp.carsを SASクライアントから carsインメ
モリテーブルにロードします。

3 casauto CASセッションで DS2プログラムを実行します。CASで DS2プログラ
ムを実行するには、SESSREF=オプションが必要です。

4 各 CASワーカーの各スレッドで、スレッドプログラムは carsインメモリテーブ
ルから行のサブセットを読み取ります。次に、スレッドプログラムは、msrpが
$100,000より大きい行のサブセットをデータプログラムに送信します。

5 各 CASワーカーの各スレッドで、データプログラムはスレッドプログラムから、
msrpが$100,000より大きい行のサブセットを受け取ります。次に、データプロ
グラムは、msrpが$100,000より大きい行のサブセットを、cars_luxuryインメモ
リテーブルに書き込みます。

DS2プログラムの実行が終了すると、cars_luxuryテーブルには、 msrpが$100,000
より大きい完全な車のセットが含まれます。

シングルスレッド処理
DS2スレッドプログラムで SET FROMステートメントを使用しない DS2プログラ
ムは、シングルスレッドで実行されます。DS2プログラムがシングルスレッドで実
行される場合、プログラムは単一の CASワーカーの 1つのスレッドで実行されま
す。単一のスレッドは入力テーブルを処理し、出力テーブルを生成します。すべて
のデータオブザベーションは順次処理されます。
単一の CASワーカーの単一のスレッドで、DS2プログラムは次のことを行います。
n 入力テーブルからすべてのデータを読み取ります。
n すべてのデータを処理します。
n すべてのデータを出力テーブルに書き込みます。

注: CASでプログラムをシングルスレッドで実行すると、CASセッションで使用可
能なワーカーの数に関係なく、処理は単一のワーカーにローカライズされます。

シングルスレッドの例
次の例では、DS2プログラムは、sashelp.carsテーブル内のすべての車の平均msrp
を計算します。プログラムは、データプログラムで SET FROMステートメントを使
用しないため、シングルスレッドで実行されます。CASセッション内のワーカーの
数に関係なく、すべてのデータ処理は単一の CASワーカー内の単一のスレッドにロ
ーカライズされます。

cas casauto; 1
proc casutil sessref=casauto; 2
   load data=sashelp.cars;
run;

proc ds2 sessref=casauto; 3
data _null_; 4
   dcl double sum_msrp n;
   retain sum_msrp n;
   method init();
      sum_msrp = 0;
      n = 0; 
   end; 
   method run();
      set cars;
      sum_msrp = sum_msrp + msrp;
      n = n + 1;
   end;
   method term();
      dcl double mean_msrp having format dollar8.;
      mean_msrp = sum_msrp / n;
      put 'Mean MSRP: ' mean_msrp;   
end;
enddata;
run;
quit;

1 CASサーバーで casautoという名前のセッションを開始します。
2 casauto CASセッションで、sashelp.carsを SASクライアントから carsインメモ
リテーブルにロードします。

3 casauto CASセッションで DS2プログラムを実行します。CASで DS2プログラ
ムを実行するには、SESSREF=オプションが必要です。

4 単一の CASワーカーの単一のスレッドで、データプログラムは carsインメモリ
テーブルからすべての行を読み取り、すべての行のmsrp値を合計し、msrp値の
平均を計算して、平均を出力します。

SASログには、次のような出力が含まれます。

Mean MSRP:   $32,775
NOTE: Running DATA program on one node

部分的なマルチスレッド処理
DS2スレッドプログラムで SET FROMステートメントを使用する DS2プログラム
は、DS2スレッドプログラムがマルチスレッドで実行され、DS2データプログラム
がシングルスレッドで実行される場合、部分的にマルチスレッドで実行されます。
SET FROMまたは OUTPUT以外のステートメントを含むデータプログラムは、シン
グルスレッドで実行されます。部分的なマルチスレッド実行には、次の 2つのステ
ージがあります。
パラレルステージ
すべての CASワーカーに分散された複数のスレッドで DS2スレッドプログラム
を実行します。

シリアルステージ
単一の CASワーカー内の単一のスレッドで DS2データプログラムを実行しま
す。

パラレルステージでは、各 CASワーカーが入力テーブルのサブセットを処理し、中
間テーブルのサブセットを生成します。シリアルステージでは、単一の CASワーカ
ーが完全な中間テーブルを処理し、完全な出力テーブルを生成します。
DS2プログラムを部分的にマルチスレッドで実行する方法を次に示します。
n パラレルステージ
各 CASワーカーの各スレッドで、DS2スレッドプログラムは次のことを行いま
す。
o 入力テーブルからデータのサブセットを読み取ります。
o データのサブセットを処理します。
o データのサブセットを一時テーブルに書き込みます。

n シリアルステージ
単一の CASワーカーの単一のスレッドで、DS2データプログラムは次のことを
行います。
o 一時テーブルからすべてのデータを読み取ります。
o すべてのデータを処理します。
o すべてのデータを出力テーブルに書き込みます。

部分的にマルチスレッドで実行する理由
並列分散実行は、複数のマシンでプログラムを実行し、そのマシンのスレッド間で
処理ワークロードを分割することにより、ビッグデータの処理を高速化します。マ
ルチスレッド実行では、データ処理の各分割が独立している必要があります。デー
タを集計または結合するデータ処理アルゴリズムは、完全なマルチスレッド実行の
ためのデータ処理の独立性要件に違反しています。データ処理アルゴリズムをデー

タ処理の分割適用結合モデルにリファクタリングできる場合、処理を部分的に並列
化できます。分割適用結合戦略では、アルゴリズムはデータを分割し、他の分割と
は関係なく各データ分割にデータ処理を適用してから、分割を結合します。
MapReduceは、分割適用結合戦略の一般的な特殊化です。
部分的にマルチスレッドで実行される DS2プログラムは、スレッドプログラムがデ
ータ処理をデータ分割に並列に適用することにより、分割適用結合戦略を実装しま
す。データプログラムは、データ処理の結果をシリアルに結合します。最高のパフ
ォーマンスを得るには、できるだけ多くのデータ処理をスレッドプログラムで実行
する必要があります。同期またはシリアル化を必要とするデータ処理のみをデータ
プログラムで実行する必要があります。

注: スレッドプログラムでのデータ処理のみが、CASセッションで使用可能なワー
カー全体に分散されます。データプログラムでのデータ処理は、CASセッションで
使用可能なワーカーの数に関係なく、常に単一のワーカーにローカライズされます。

部分的なマルチスレッドの例
次の例では、DS2プログラムは、sashelp.carsテーブル内のすべての車の平均msrp
を計算します。DS2スレッドとデータプログラムからのセットは SET FROMステ
ートメントと OUTPUTステートメントに限定されないため、プログラムは部分的に
マルチスレッドで実行されます。

cas casauto; 1
 
proc casutil sessref=casauto; 2
   load data=sashelp.cars;
run;

proc ds2 sessref=casauto; 3
thread cars_thd / overwrite=yes; 4
   dcl double partial_sum_msrp partial_n;
   retain partial_sum_msrp partial_n;
   keep partial_sum_msrp partial_n;
   method init();
      partial_sum_msrp = 0;
      partial_n = 0;
   end;
   method run();
      set cars; 
      partial_sum_msrp = partial_sum_msrp + msrp; 
      partial_n = partial_n + 1;
   end;
   method term();
      output;
   end;
endthread;

data _null_; 5
   dcl thread cars_thd t;
   dcl double sum_msrp n;

   retain sum_msrp n;
   method init();
      sum_msrp = 0;
      n = 0;
   end;
   method run();
      set from t;
    sum_msrp = sum_msrp + partial_sum_msrp;
    n = n + partial_n;
  end;
  method term();
    dcl double mean_msrp having format dollar8.;
    mean_msrp = sum_msrp / n;
    put 'Mean MSRP: ' mean_msrp;
   end;
enddata;
run;
quit;

1 CASサーバーで casautoという名前のセッションを開始します。
2 casauto CASセッションで、sashelp.carsを SASクライアントから carsインメモ
リテーブルにロードします。

3 casauto CASセッションで DS2プログラムを実行します。CASで DS2プログラ
ムを実行するには、SESSREF=オプションが必要です。

4 各 CASワーカーの各スレッドで、スレッドプログラムは carsインメモリテーブ
ルから行のサブセットを読み取ります。スレッドプログラムは、行のサブセット
のmsrpを合計し、msrpの部分合計を計算します。次に、スレッドプログラム
は、 msrpの部分合計を一時インメモリテーブルに書き込みます。

5 単一の CASワーカーの単一のスレッドで、データプログラムは一時インメモリ
テーブルから msrpのすべての部分合計を読み取り、部分合計を合計してすべて
のmsrp値の合計を求め、msrp値の平均を計算して、平均を出力します。

SASログには、次のような出力が含まれます。

NOTE: Running THREAD program on all nodes
Mean MSRP:   $32,775
NOTE: Running DATA program on one node

この部分的なマルチスレッドの例は、シングルスレッドの例のリファクターである
ことに注意してください。シングルスレッドの実行を必要とするシリアルアルゴリ
ズムを分割適用結合戦略に変換できる場合、アルゴリズムをリファクタリングする
と、多くの場合、CASなどの分散並列環境で処理されるビッグデータのランタイム
パフォーマンスが大幅に向上します。

CASの DS2スレッドプログラム
CASの分析の基本的な計算モデルは、分散コンピューティングとマルチスレッドコ
ンピューティングを組み合わせたものです。データを渡す CASアクションは通常、
複数のスレッドによって行われます。各スレッドはデータの一部を受け取り、入力
テーブルの各レコードは 1つのスレッドによってのみ消費されます。これらのスレ

ッドは、実行の同時性を強調するためにワーカースレッドと呼ばれます。使用でき
る同時ワーカースレッドの最大数は、ソフトウェアライセンスによって決まります。
許可されている最大数よりも多いスレッドを DS2プログラムで指定している場合、
DS2はスレッド数をその最大数に減らします。DS2プログラムで 0スレッドが指定
されているか、スレッド数が指定されていない場合、DS2は、ソフトウェアライセ
ンスで決定されている、許可されている最大スレッド数を使用します。

Lirefと Caslib
DS2プログラムが CASで実行される場合、DS2プログラムは、SASクライアントで
作成された librefにアクセスできません。DS2プログラムで使用できるのは、CAS
セッションで作成された caslibだけです。
CASで実行される DS2プログラムでは、次の形式で caslibの CASテーブルを指定
します。
caslib.table-name

caslibにかっこまたはその他の特殊文字が含まれている場合は、caslibを二重引用
符で囲む必要があります。次に例を示します。
set "caslib(janedoe)".mytable;

コメント
/* message */コンストラクトを使用して DS2プログラムにコメントを追加し、プロ
グラムの目的を文書化したり、プログラムの異常な部分を説明したり、複雑なプロ
グラムや計算のステップを説明したりできます。

重要 ステートメント内または DS2プログラムで 1つの空白が表示される
場所ならどこにでもコメントを書くことができますが、/* message */コンスト
ラクトを使用するコメントはネストできません。

CASでサポートされていない DS2言語要素
CASでサポートされていない DS2言語要素がいくつかあります。これらの言語要
素は、"この language-elementは CASサーバーではサポートされていません。"とい
う制限付きで SAS DS2言語リファレンスで文書化されています。

CASでの BYグループ処理
SASクライアントでは、DS2 BYグループ処理によって入力テーブルの行がグループ
化され、BYステートメントの 1つ以上の列の値によって行が順序付けされます。
SASクライアントでは、事前に並べ替えられたデータが順序付けられて DS2に到着
し、BYグループが BYグループブロック全体としてスレッドに分散されます。CAS
では、データはパーティション上で本質的に順序付けられていません。BYステート
メントがない場合、各スレッドはノードに対してローカルなデータを受け取ります。
スレッドプログラムまたはデータプログラムのいずれかで BYステートメントが使
用されている場合、各スレッドが 1つ以上の BYグループ全体を参照できるように、
テーブル行がワーカー間で送信されます。グリッドを介したデータ転送はリソース
を大量に消費する可能性があるため、データプログラムで BY処理を使用して結果を
取得する前に、スレッドプログラムでハッシュテーブルを使用して各パーティショ
ンに存在する生データを集計するのが最善かどうかを検討する必要があります。入
力行に比べて個別のデータグループが比較的少ない場合は、データプログラムで BY
ステートメントを使用してデータを再分散する前に、スレッドプログラムでデータ
をそれらの少数の個別のグループにまとめる方が望ましい場合があります。

CASサーバーでの整数型変換
DS2は、SASプラットフォームで実行する場合、BIGINT、INTEGER、SMALLINT、
TINYINT整数データ型をサポートします。CASサーバーテーブルは、BIGINT 
(INT64)および INTEGER (INT32)の整数データ型のみをサポートします。
CAS runDS2アクションは、SMALLINTまたは TINYINTデータ型の変数を宣言する
DS2プログラムを CASで実行できます。プログラムが変数を CASテーブルに書き
込む場合、書き込まれた SMALLINTまたは TINYINT値は、CASテーブルの INTEGER 
(INT32)データ型に変換されます。
-2147483647の値は、CASテーブルの INTEGER (INT32)列の SAS欠損値を表しま
す。-9223372036854775808の値は、CASテーブルの BIGINT (INT64)列の SAS欠
損値を表します。-2147483647は CASテーブルの BIGINT (INT64)列の有効な値で
あるため、-2147483647はどこでも SAS欠損値を表すわけではないことに注意して
ください。
DS2では、-2147483647は BIGINT変数に対して有効ですが、INTEGER、
SMALLINT、および TINYINT変数の範囲外です。DS2で
は、-9223372036854775808は、BIGINT、INTEGER、SMALLINT、TINYINT変数
の範囲外です。範囲外の値が整数変数に割り当てられているが、変数の型でサポー
トされている場合、SAS欠損値または ANSI nullが変数に割り当てられます。したが
って、-2147483647または-9876543210が INTEGER、SMALLINT、または TINYINT

変数に割り当てられている場合、その変数には SAS欠損値または ANSI null値が割
り当てられます。
CASテーブルの INTEGER (INT32)列が DS2プログラムに読み込まれ、オブザベーシ
ョンの INTEGER (INT32)列の値が-2147483647である場合、その値は、入力時に
SAS欠損値の DS2の内部表現に変換されます。CASテーブルの BIGINT (INT64)列
が DS2プログラムに読み込まれ、オブザベーションの BIGINT (INT64)列の値
が-2147483647である場合、値は変換されずにそのまま読み取られます。

CASサーバーでの DS2ログ
DS2と CASは、SASログ機能をサポートしています。SAS Viyaプラットフォームで
のログの詳細については、SAS Viyaプラットフォーム: 管理の"ログ"を参照してくだ
さい。

CASサーバーでの DS2セッションエン
コーディング

CASサーバーでコンパイルおよび実行される DS2プログラムの DS2セッションエ
ンコーディングは、常に UTF-8です。

CASでの DS2プログラムの実行

CASで DS2プログラムを実行する方法
CASで DS2プログラムを実行するには、次の 2つの方法があります。
n PROC DS2

cas casauto;
caslib _all_ assign;

proc cas;
  session casauto;
/* This  sets the active caslib to casdata.*/
table.addCaslib / 
     caslib="casdata" 
     dataSource={srcType="path"}
     path="path-to-your-data";
 
 /* Load source data (cars) into a table.*/
   table.loadTable / 
      caslib="casdata"
      path="cars.sashdat" 
      casOut={name="cars", replace=true};  
run;
quit;

proc ds2 sessref=casauto;
thread cars_thd / overwrite=yes;
   method run();
      set casdata.cars;
      if (msrp > 100000) then do;
         put make= model= msrp=;
         output;
      end;

   end;
endthread;

data cars_luxury / overwrite=yes;
   dcl thread cars_thd t;
   method run();
      set from t threads=4;
   end;
enddata;
run;
quit;

このプログラムの説明については、“DS2プログラムの手順説明” (311ページ)を
参照してください。
ROC DS2の詳細については、“DS2 プロシジャ” (Base SASプロシジャガイド)を参
照してください。

n DS2 runDS2アクション
cas casauto;
caslib _all_ assign;

proc cas;
  session casauto;
/* This  sets the active caslib to casdata.*/
   table.addCaslib / 
      caslib="casdata" 
      dataSource={srcType="path"}
      path="path-to-your-data";

/* Load source data (cars) into a table.*/
   table.loadTable / 
      caslib="casdata"
      path="cars.sashdat" 
      casOut={name="cars", replace=true};  
run;

/*DS2 program to search for cars over $100K */
   ds2.runDS2 program="thread cars_thd / overwrite=yes;
   method run();
      set casdata.cars;
      if (msrp > 100000) then do;
         put make= model= msrp=;
         output;
      end;
   end;
endthread;

data cars_luxury / overwrite=yes;
   dcl thread cars_thd t;
   method run();
      set from t threads=4;
   end;
enddata;";
run; 
quit;

詳細については、“DS2 Action Set” (SAS Viya Platform: System Programming 
Guide)を参照してください。

ヒント Luaまたは Pythonを使用していない場合は、PROC DS2を使用して
DS2プログラムを実行することをお勧めします。PROC DS2を使用する利点
があります。詳細については、“DS2 Action Set Details” (SAS Viya Platform: 
System Programming Guide)を参照してください。

DS2プログラムの手順説明
CASで DS2プログラムを実行するには、次の操作を行います。
1 CASセッションを開始します。

CASにアクセスするには、CASセッションを開始してアクティブにします。CAS
ステートメントで CASセッションの名前を指定します。
cas casauto;

SAS Studio コードタブのビューは次のとおりです。

SASログには、次のような注が含まれます。

DS2プログラムの手順説明 311

http://documentation.sas.com/?docsetId=caspg&docsetVersion=v_044&docsetTarget=cas-ds2-TblOfActions.htm&locale=ja
http://documentation.sas.com/?docsetId=caspg&docsetVersion=v_044&docsetTarget=cas-ds2-TblOfActions.htm&locale=ja
http://documentation.sas.com/?docsetId=caspg&docsetVersion=v_044&docsetTarget=n0kk5ezhtskpncn1vgfxrv6eobgt.htm&docsetTargetAnchor=n0kk5ezhtskpncn1vgfxrv6eobgt&locale=ja
http://documentation.sas.com/?docsetId=caspg&docsetVersion=v_044&docsetTarget=n0kk5ezhtskpncn1vgfxrv6eobgt.htm&docsetTargetAnchor=n0kk5ezhtskpncn1vgfxrv6eobgt&locale=ja


  NOTE: The session CASAUTO connected successfully to Cloud Analytic Services cloud.example.com using 
port 5570. The UUID is 
       session-UUID. The user is casdemo and the active caslib is CASUSER(casdemo).
 NOTE: The SAS option SESSREF was updated with the value CASAUTO.
 NOTE: The SAS macro _SESSREF_ was updated with the value CASAUTO.
 NOTE: The session is using nnn workers.

2 caslibを関連付けます。
CASLIBステートメントを使用して、デフォルトの CASUSER caslibを、作成する
すべての SAS librefに関連付けることができます。
caslib _all_ assign;

SASログに表示される注の部分的なリストを次に示します。
 NOTE: CASLIB CASUSER(casdemo) for session CASAUTO will be mapped to SAS Library CASUSER.

3 caslibを作成し、データをロードします。
ファイルシステム上のファイルへのアクセスを提供する caslibを作成するには、
table.addCaslibアクションを、pathに設定された dataSourceオプションとと
もに使用します。この例では、CASDATA caslibは、CASAUTOセッションと/
mystore/data/casdataにあるソーステーブルとの間のインターフェイスです。
proc cas ;
   session casauto;
   /* This will set the active caslib to casdata*/
   table.addCaslib / 
      caslib="casdata" 
      dataSource={srcType="path"}
      path="/mystore/data/casdata";

詳細については、“caslib” (SAS Cloud Analytic Servcies: 基本)を参照してください。
ソースデータ(cars)を CASのインメモリテーブルにロードするには、
table.loadTableアクションを使用できます。
table.loadTable / 
   caslib="casdata" 
   path="cars.sashdat" 
   casOut={name="cars",replace=true};
run;

CASDATA caslibを説明する結果は、SAS Studioの結果タブに表示されます。

4 CASテーブルを表示します。
table.fetchアクションを使用して、インメモリテーブル(cars)からデータの行を
表示できます。
table.fetch / table="cars" to=10; 
run;

インメモリーブルから最初の 10行をフェッチした結果を次に示します。

5 DS2プログラムを実行します。
PROC DS2を sessref=オプションとともに使用すると、インメモリ CASテーブ
ルを使用してプログラムを CASで実行できます。
proc ds2 sessref=casauto;           1

thread cars_thd / overwrite=yes;     2
   method run();
      set casdata.cars;             3
      if (msrp > 100000) then do;
         put make= model= msrp=;
         output;
      end;
   end;
endthread;

data cars_luxury / overwrite=yes;    4
   dcl thread cars_thd t;
    method run();
      set from t threads=4;
    end;
enddata;

run;
quit;

1 PROC DS2ステートメントに sessref=オプションを追加して、プログラムを
CASで実行します。

2 プログラムのパラレルステージでの操作は、複数のデータ行に並列に適用さ
れます。

3 インメモリテーブルから行を読み取ります。
4 スレッドインスタンスが作成され、実行されます。
SASログに表示される結果は次のとおりです。

  NOTE: Running THREAD program on all nodes
 Make=Porsche       Model= 911 GT2 2dr                             MSRP=$192,465
 Make=Mercedes-Benz Model= CL600 2dr                               MSRP=$128,420
 Make=Mercedes-Benz Model= SL55 AMG 2dr                            MSRP=$121,770
 Make=Mercedes-Benz Model= SL600 convertible 2dr                   MSRP=$126,670
 NOTE: Created thread cars_thd in data set "casdata".cars_thd.
 NOTE: Running THREAD program on all nodes
 NOTE: Running DATA program on all nodes
 NOTE: Execution succeeded. 4 rows affected.
 104        quit;
 
 NOTE: PROCEDURE DS2 used (Total process time):
       real time           4.89 seconds
       cpu time            0.02 seconds

モデルのパブリッシュとスコアリ
ング

モデルのパブリッシュとスコアリングの概要 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 315

および CASアクション . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 315

モデルのパブリッシュとスコアリングの
概要

SCOREACCELプロシジャと CASアクションは、モデルのパブリッシュとスコアリン
グのために、CASサーバーと外部データソースへのインターフェイスを提供します。
詳細については、“In-Database Model Scoring” (SAS In-Database Products: User’s 
Guide)を参照してください。

モデルのパブリッシュとスコアリングの
ための PROC SCOREACCELおよび CAS
アクション

構文と例については、次のドキュメントを参照してください。
n “SCOREACCEL Procedure” (SAS In-Database Products: User’s Guide)

n “Run model program” (SAS Viya Platform: System Programming Guide)

n “モデルのパブリッシュとスコアリングアクションセット” (SAS Visual Analytics: 
プログラミングガイド)

n “SAS Embedded Process for Sparkアクションセット” (SAS Visual Analytics: プ
ログラミングガイド)

付録 1
式オペランドの DS2型変換

式オペランドの DS2自動型変換 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 319

式オペランドの DS2自動型変換
次の表に、式オペランドの自動型変換を示します。最初の列は"変換元"の型です。
最初の行は"変換先"の型です。

付録 2
DS2ロガー

構成ロガー . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 324

ランタイムロガー . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 324

HTTPパッケージロガー . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 326

JSONパッケージロガー . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 326

例: すべての SQL操作のログ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 328

例: トレース変数のログ . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 329
TraceVariablesロガーの呼び出し . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 329
ロガーメッセージについて  . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 330

SAS Viyaの SASログ機能への DS2ロガーの追加 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 333

DS2ロガーの概要
SASログ機能は、SASサーバーおよび SASプログラミング環境でログメッセージを
分類およびフィルタリングし、ログメッセージをさまざまな出力デバイスに書き込
むフレームワークです。サーバー環境では、管理メッセージの Admin、アプリケー
ションメッセージの App、パフォーマンスメッセージの Perfなど、定義済みのメッ
セージカテゴリに基づいてログ機能がメッセージをログに記録します。カテゴリの
メッセージは、ファイル、コンソール、およびその他のシステム宛先に同時に書き
込むことができます。ログ機能により、TRACE、DEBUG、INFO、WARN、ERROR、
FATALのしきい値に基づいてメッセージをフィルタリングすることもできます。
DS2には、構成情報とランタイム情報の両方を報告するためのロガーがいくつか用
意されています。さらに、DS2は HTTPパッケージと JSONパッケージ用のロガー
を提供します。一般に、INFOは最小限の情報を提供し、DEBUGはフィールドの問
題をデバッグするために必要なほとんど(おそらくすべて )の情報を提供します。
TRACEレベルは、関心のあるあらゆるものを提供します。

ロガーと SASログ機能の詳細については、SAS Viyaログ機能: 構成とリファレンスを
参照してください。SAS Viyaプラットフォームで使用する DS2ロガーを定義する
方法については、“SAS Viyaの SASログ機能への DS2ロガーの追加” (333ページ)を
参照してください。

注: SAS Viyaプラットフォームの SASログ機能は、32000バイトを超えるログメッ
セージを 32000バイトに切り捨てます。

構成ロガー
構成ロガーは、DS2コンパイラの起動時に情報を追跡し、ユーザーのコードを実際
に実行するためのコンテキストを提供します。
次の構成ロガーを使用できます。
App.TableServices.DS2.Config.Options

DS2コンパイラに提供されるオプションを示します。
App.TableServices.DS2.Config.Source

DS2コンパイラによって処理された DS2ソースコードを示します。
App.TableServices.DS2.Config.Version

DS2コンパイラによってロードされたすべてのスレッドカーネル拡張のバージ
ョン情報を示します。

DS2の問題を診断する場合、App.TableServices.DS2.Config.*ロガーは祖先から構成
を継承しないことを理解することが重要です。App.TableServices.DS2.Config.*ロ
ガーのデフォルトレベルは OFFです。App.TableServices.DS2.Config.*ロガーイベ
ントをサーバーのログにキャプチャする場合は、該当する各
App.TableServices.DS2.Config.*ロガーを明示的に設定する必要があります。追加
のログトラフィックがパフォーマンスに影響するため、DS2の問題を診断する場合
にのみ構成することをお勧めします。

ランタイムロガー
ランタイムロガーは実際の実行を追跡します。追跡される情報の一部は、処理され
る行ごとに生成されます。したがって、大きな入力テーブルは非常に大量のデータ
を生成する可能性があります。
次のランタイムロガーを使用できます。
App.TableServices.DS2.Runtime.Calls
実行中のすべてのメソッド呼び出しのトレースを示します。

App.TableServices.DS2.Runtime.SQL
DS2コンパイラによって準備された SQLステートメント、DS2コンパイラによ
って実行された SQLステートメント、またはその両方のすべての SQLステート
メントを示します。

App.TableServices.DS2.Runtime.Timing
コードのコンパイルと実行に費やされた時間を示します。要求された情報のレ
ベル(INFO、DEBUG、TRACE)に応じて、DS2実行中のさまざまな時点でタイミ
ング情報が提供されます。例としては、解析時間、コンパイル時間、さまざまな
監査および変換パス時間、INIT、RUN、および TERMメソッドの実行時間があり
ます。

App.TableServices.DS2.Runtime.Put
すべての PUTステートメント出力を記録します。

App.TableServices.DS2.Runtime.Log
SASセッションで SASログに送信されるすべてのメッセージを記録します。
SAS Federation Serverでは、これらのメッセージもデフォルトで診断レコード
として ODBCステートメントハンドルに追加されます。

App.TableServices.DS2.Runtime.TraceVariables
DEBUGまたは TRACEに設定すると、DS2プログラムの実行中に変数値の変更が
記録されます。トレース情報は、次の DS2ステートメントおよび関数によって
変更される変数についてのみ報告されます。
n 割り当てステートメント( = )

n Array割り当てステートメント( := )

n DOステートメントによるインデックス変数割り当て
n SETまたはMERGEステートメント
n 擬似 SUBSTR関数(割り当ての左辺に用いた場合)

n SUMステートメント( variable + expression; variable – expression; )

次の方法で変更された変数については、変数のトレース情報はレポートされませ
ん。
n ビルトイン DS2パッケージによって変更される変数データ。たとえば、

HASH.FINDメソッドがデータ変数を変更する場合、その変数の変更はトレー
スで報告されません。

n DS2自動変数の暗黙の更新。たとえば、N自動変数の暗黙的な増分はトレー
スで報告されません。

n 変数の作成中に、変数を欠損または nullに暗黙的に初期化。
n RUNメソッドの各反復前に、保持されていない変数を欠損または nullに暗黙
的に再初期化。

注: App.TableServices.DS2.Runtime.TraceVariablesロガーは、
TRACEVARIABLESオプションとともに使用されます。TRACEVARIABLESオプシ
ョンは、“DS2_OPTIONS ステートメント” (SAS DS2言語リファレンス)で設定さ
れます。TRACEVARIABLESオプションが有効でない場合、変数のトレース情報
は報告されません。TRACEVARIABLESオプションの設定方法の例については、
“例: トレース変数のログ” (329ページ)を参照してください。

DS2の問題を診断する場合、App.TableServices.DS2.Runtime.*ロガーは祖先から構
成を継承しないことを理解することが重要です。
App.TableServices.DS2.Runtime.*ロガーのデフォルトレベルは OFFです。
App.TableServices.DS2.Runtime.*ロガーイベントをサーバーのログにキャプチャ
する場合は、該当する各 App.TableServices.DS2.Runtime.*ロガーを明示的に設定
する必要があります。追加のログトラフィックがパフォーマンスに影響するため、
DS2の問題を診断する場合にのみロガーを構成することをお勧めします。

HTTPパッケージロガー
HTTPクライアントは、SASログ機能による HTTPトラフィックのログをサポートし
ています。
ロガーの名前は App.TableServices.d2pkg.HTTPです。ロガーは、次のログ記録レ
ベルをサポートしています。
INFO

Webサーバーへの接続と切断、要求情報、ステータス情報などの一般的なトラ
フィック情報を表示します。

DEBUG
すべての要求と応答のヘッダーと、本文データの最初の 64バイトを示します。
これにより、サーバーに送信されるすべてのデータを表示しなくても、クライア
ントとサーバーが何を行っているかを確認できます。

TRACE
Webサーバーとの間で送受信されるすべてのデータを表示します。

JSONパッケージロガー
JSONパッケージは、SASログ機能によるログをサポートしています。
ロガーの名前は App.tk.DS2PKG.JSONです。ロガーは、DEBUGおよび TRACEログ
記録レベルをサポートします。たとえば、システムの起動時にロガーレベルを
TRACEに設定する場合は、次の行をログ構成ファイルに追加します。
<logger name="App.tk.DS2PKG.JSON> <level value="trace"/> </logger>

そのログ構成ファイルを使用する他のユーザーのログ記録レベル設定に影響を与え
ないようにするには、SASログ機能を使用して自分のセッションを変更します。
%log4sas();
%log4sas_logger(App.tk.D2PKG.JSON, 'level=trace');

JSONパッケージは、DEBUGまたは TRACE情報をほとんど提供しません。
GETNEXTTOKENメソッドは、いくつかの TRACE情報を提供します。次のログの抜
粋は一例です。

INFO App.Program - 1   %log4sas();
INFO App.Program - 2   %log4sas_logger(App.tk.D2PKG.JSON, 'level=trace');
INFO App.Program - 3   proc ds2;
INFO App.Program - 4   ds2_options sas;
INFO App.Program - 5   data _null_;
INFO App.Program - 6      dcl package json j();
INFO App.Program - 7      dcl int i rc ttype parseFlags;
INFO App.Program - 8      dcl varchar(256) x;
INFO App.Program - 9      dcl varchar(256) token;
INFO App.Program - 10     method init();
INFO App.Program - 11        x = '[{"labl 1":[{"labl 2":" a string"}]},
                             {"labl 3":[["", 3.14, null, true, false]]}]';
INFO App.Program - 12        put x=;
INFO App.Program - 13        rc = j.createParser( x );
INFO App.Program - 14        i = 0;
INFO App.Program - 15        do while (rc < 101 );
INFO App.Program - 16           i = i + 1;
INFO App.Program - 17           token = '';
INFO App.Program - 2    The SAS System    16:33 Friday, October 21, 2016
INFO App.Program - 
INFO App.Program - 18           j.getNextToken( rc, token, ttype, parseFlags );
INFO App.Program - 19        end;
INFO App.Program - 20      end;
INFO App.Program - 21      method term();
INFO App.Program - 22         rc = j.destroyParser();
INFO App.Program - 23      end;
INFO App.Program - 24   enddata;
INFO App.Program - 25   run;
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 1, type: 16, 
                          flags: 0x0, token: [
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 2, type: 64, 
                          flags: 0x0, token: {
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 10, type: 256, 
                          flags: 0x1, token: labl 1
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 12, type: 16, 
                          flags: 0x0, token: [
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 13, type: 64, 
                          flags: 0x0, token: {
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 21, type: 256, 
                          flags: 0x1, token: labl 2
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 33, type: 256, 
                          flags: 0x0, token:  a string
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 34, type: 128, 
                          flags: 0x0, token: }
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 35, type: 32, 
                          flags: 0x0, token: ]
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 36, type: 128, 
                          flags: 0x0, token: }
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 38, type: 64, 
                          flags: 0x0, token: {
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 46, type: 256, 
                          flags: 0x1, token: labl 3
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 48, type: 16, 
                          flags: 0x0, token: [
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 49, type: 16, 
                          flags: 0x0, token: [

TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 51, type: 256, 
                          flags: 0x0, token: 
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 57, type: 512, 
                          flags: 0x4, token: 3.14
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 63, type: 1024, 
                          flags: 0x0, token: null
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 69, type: 4, 
                          flags: 0x0, token: true
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 76, type: 8, 
                          flags: 0x0, token: false
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 77, type: 32, 
                          flags: 0x0, token: ]
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 78, type: 32, 
                          flags: 0x0, token: ]
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 79, type: 128, 
                          flags: 0x0, token: }
TRACE App.tk.D2PKG.JSON - d2jsnGetNextToken: line: 1, column: 80, type: 32, 
                          flags: 0x0, token: ]
INFO App.Program - x=[{"labl 1":[{"labl 2":" a string"}]},{"labl 3":[["", 3.14,
                          null, true, false]]}]
INFO App.Program - NOTE: Execution succeeded. No rows affected.
INFO App.Program - 26         quit;

例: すべての SQL操作のログ
次の例では、すべての SQLコマンドをログに記録するログ構成ファイルを作成しま
す。構成ファイルの作成後、LOGCONFIGLOC=システムオプションを設定して、SAS
ログ機能の初期化に使用される構成ファイルの名前を指定します。
次のコードは、ログ構成ファイルを作成します。
<?xml version="1.0"?>
   <logging:configuration xmlns:logging="http://www.sas.com/xml/logging/1.0/">
   <logger name="App.TableServices.DS2.Runtime.SQL">
       <level value="trace"/>
   </logger>
   <logger name="App.TableServices.DS2" additivity="false">
       <appender-ref ref="DetailedOutput"/>
   </logger>
   <root>
       <appender-ref ref="RootLogger"/>
       <level value="info"/>
   </root>
   <appender name="DetailedOutput" class="FileAppender">
       <param name="append" value="false"/>
       <param name="FileNamePattern" value="sql.%S{App.Log}"/>
       <layout>
           <param name="ConversionPattern" value="%-5p:%sn:[%c{3}]:%m"/>
       </layout>
   </appender>
   <appender name="RootLogger" class="FileAppender">

       <param name="Append" value="false"/>
       <param name="ImmediateFlush" value="true"/>
       <param name="FileNamePattern" value="%S{App.Log}"/>
       <layout>
           <param name="ConversionPattern" value="%m"/>
       </layout>
   </appender>
   </logging:configuration>

LOGCONFIGLOC=システムオプションは、構成ファイルを config.14sとして参照す
るように設定されています。
sas -log test.log -logconfigloc config.14s

このシステムオプションの詳細については、SAS Viyaログ機能: 構成とリファレンス
を参照してください。
これにより、次のような一連のメッセージを含むファイル sql.test.logが生成されま
す。
DEBUG:00000084:[DS2.Runtime.SQL]:Found 0 NOCHANGE columns
   INFO :00000085:[DS2.Runtime.SQL]:0x0afce4b0:exec-direct:CREATE TABLE WORK.outp 
      ("i" DOUBLE, "j" DOUBLE )
   DEBUG:00000086:[DS2.Runtime.SQL]:0x0afce4b0:exec-direct:passed:rc=0x00000000
   DEBUG:00000087:[DS2.Runtime.SQL]:Found 0 NOCHANGE columns
   INFO :00000088:[DS2.Runtime.SQL]:0x0afce4b0:exec-direct:SELECT * FROM WORK.outp
   DEBUG:00000089:[DS2.Runtime.SQL]:0x0afce4b0:exec-direct:passed:rc=0x00000000
   INFO :00000095:[DS2.Runtime.SQL]:stmt=0x0afae060:prepare:SELECT * FROM WORK.outp
   DEBUG:00000096:[DS2.Runtime.SQL]:stmt=0x0afae060:prepare:passed:rc=0x00000000
   INFO :00000097:[DS2.Runtime.SQL]:stmt=0x0afae060:execute
   DEBUG:00000098:[DS2.Runtime.SQL]:stmt=0x0afae060:execute:passed:rc=0x00000000

出力ファイルの正確な内容は、DetailedOutputアペンダー内のレイアウトの
ConversionPatternパラメーターによって定義されます。DetailedOutputアペンダ
ーは、App.TableServices.DS2ロガーの定義に関連付けられています。これにより、
App.TableServices.DS2をルートとする階層内のすべてのロガーが同じファイルに
ログ記録されます。の additivity="false"修飾子は、ログメッセージが上に移動するの
を防ぎます。

例: トレース変数のログ
TraceVariablesロガーの呼び出し

トレース変数のログを有効にするには、次の操作を行います。
1 DEBUGまたは TRACEログ記録レベルのいずれかを指定して、

App.TableServices.DS2.Runtime.TraceVariablesロガーを構成します。

2 DS2プログラムで TRACEVARIABLESオプションを指定します。
TRACEVARIABLESオプションは、“DS2_OPTIONS ステートメント” (SAS DS2言
語リファレンス)で設定されます。

注: 両方のステップを完了する必要があります。これらのタスクのいずれかが実行
されていない場合、変数のトレース情報は SASログ機能に報告されません。

ロガーを構成するには、ログ構成ファイルに次の行を追加します。この例では、
DEBUGのログ記録レベルを設定します。
<logger name="App.TableServices.DS2.Runtime.TraceVariables"><level value="debug"/></logger>

TRACEVARIABLESオプションの設定方法の例を次に示します。変数をトレースす
るプログラムブロックの直前に DS2_OPTIONSステートメントを指定します。
DS2_OPTIONSステートメントは、調べるプログラムブロックごとに設定する必要
があります。

proc ds2;
ds2_options tracevariables;
package swapper / overwrite=yes;
  method swap(in_out double x, in_out double y);
    dcl double t;
    t = x;
    x = y;
    y = t;
  end;
endpackage;

ds2_options tracevariables;
data _null_;
  method init();
    dcl package swapper s();
    dcl double a[2];
    a := (10, 20);
    put 'before swap:' a[*]=;
    s.swap(a[1], a[2]);
    put 'after swap:' a[*]=;
  end;
enddata;
run; 
quit;

ロガーメッセージについて
前述のプログラムの SASログは次のとおりです。

1          proc ds2;
2          ds2_options tracevariables;
3          package swapper / overwrite=yes;
4            method swap(in_out double x, in_out double y);
5              dcl double t;
6              t = x;
7              x = y;
8              y = t;
9            end;
10         endpackage;
11
12         ds2_options tracevariables;
13         data _null_;
14           method init();
15             dcl package swapper s();
16             dcl double a[2];
17             a := (10, 20);
18             put 'before swap:' a[*]=;
19             s.swap(a[1], a[2]);
20             put 'after swap:' a[*]=;
21           end;
22         enddata;
23         run;
before swap: a[1]=10 a[2]=20
after swap: a[1]=20 a[2]=10
NOTE: Created package swapper in data set work.swapper.
NOTE: Execution succeeded.No rows affected.
24         quit;

App.TableServices.DS2.Runtime.TraceVariablesロガーによって SASログ機能に
報告されるトレースメッセージを次に示します。

2021-01-12T13:33:35,063 DEBUG [00000007] - Line 17 (data block line 5): a[1]=10
2021-01-12T13:33:35,063 DEBUG [00000007] - Line 17 (data block line 5): a[2]=20
2021-01-12T13:33:35,063 DEBUG [00000007] - Line 6 (package swapper line 4): t=10
2021-01-12T13:33:35,064 DEBUG [00000007] - Line 7 (package swapper line 5): x=20
2021-01-12T13:33:35,064 DEBUG [00000007] - Line 8 (package swapper line 6): y=10

トレースメッセージの形式は次のとおりです。
Line n1 (program_block_type program_block_name line n2): variable_name=variable_value

n1
SASログでアクティビティが発生したステートメントの行番号。

program_block_type
ステートメントを含むプログラムブロックのタイプ("package"、"thread"、また
は"data")。

program_block_name
ステートメントを含むプログラムブロックの名前。パッケージおよびスレッド
プログラムブロックの場合、パッケージまたはスレッドのユーザー定義名が名前
になります。データプログラムブロックの場合、"block"が名前です。

n2
プログラムブロックの行番号。プログラムブロックの行番号は、プログラムブロ
ックを開始する PACKAGE、THREAD、または DATAステートメントで 1から始
まり、プログラムブロックを終了する ENDPACKAGE、ENDTHREAD、または
ENDDATAステートメントに到達するまで順番に続きます。n2は、パッケージ、
スレッド、またはデータブロックの先頭を基準とした、アクティビティが発生し
たステートメントの行番号です。

SASログの行番号は、現在の SASセッションで他に何が実行されたかによって、
実行ごとに変わる可能性があるため、この番号が提供されます。プログラムブロ
ックの先頭からのステートメントの行番号は変更されません。プログラムブロ
ックの行番号がわかると、ストアド DS2パッケージまたはストアド DS2スレッ
ドが SASログのソースリストにない場合に役立ちます。

この例の詳細は次のとおりです。
n ロガー出力の最初の 2つのメッセージは、SASログの 17行目のデータプログラ
ムブロックからの Array割り当てステートメントを参照しています。
2021-01-12T13:33:35,063 DEBUG [00000007] - Line 17 (data block line 5): a[1]=10
2021-01-12T13:33:35,063 DEBUG [00000007] - Line 17 (data block line 5): a[2]=20

17行目のステートメント a := (10, 20)は、値 10と 20を含む定数リストを配列 a
に割り当てます。最初のメッセージ a[1]=10は、配列 aの最初の要素への値 10
の割り当てを記録します。2番目のメッセージ a[2]=20は、配列 aの 2番目の要
素への値 20の割り当てを記録します。各メッセージの 5行目への参照は、Array
割り当てステートメントの場所を DATAステートメントからのオフセットとし
て再度示しています。ステートメント a := (10, 20)は、DATAステートメントの
4行下にあります。

n 残りのメッセージは、swapperパッケージで発生した変換の結果として生じる変
数の割り当てについて説明しています。
2021-01-12T13:33:35,063 DEBUG [00000007] - Line 6 (package swapper line 4): t=10
2021-01-12T13:33:35,064 DEBUG [00000007] - Line 7 (package swapper line 5): x=20
2021-01-12T13:33:35,064 DEBUG [00000007] - Line 8 (package swapper line 6): y=10

6、7、および 8行目には、変数 t、x、および yに値を割り当てる変数割り当て
ステートメントが含まれています。
割り当てステートメント t=xでは、tに xの値が割り当てられます。パラメータ
ー xには、19行目の s.swapメソッド呼び出しの第 1引数 a[1]によって値 10が
渡されました。したがって、値 10が tに割り当てられます。つまり、t=10で
す。
割り当てステートメント x=yでは、xに yの値が割り当てられます。パラメータ
ー yには、第 2引数 a[2]によって値 20が渡されました。したがって、値 20が
xに割り当てられます。つまり、x=20です。
割り当てステートメント y=tでは、yに tの値が割り当てられます。変数 tには、
以前の割り当てステートメント t=xによって割り当てられた値 10がまだ含まれ
ています。したがって、y=10です。

注: 変数トレースは、特に DS2プログラムが大きい場合、または DS2プログラムに
よって入力されるデータセットが大きい場合に、大量の SASログメッセージを生成
する可能性があります。したがって、追加のログトラフィックがパフォーマンスに
悪影響を与えるため、DS2の問題を診断する場合にのみログを構成することをお勧
めします。

SAS Viyaの SASログ機能への DS2ロガ
ーの追加

SAS Viyaプラットフォームでは、プログラミングロガーは SAS Environment 
Managerのログ機能に定義されます。SAS Environment Managerを使用するに
は、管理アクセスが必要です。
コンテナーで実行されているアプリケーションは、ログをモニタリングツールから
利用できるようにするために出力を stdout (コンソール)に送信する必要がありま
す。DS2ロガーの出力をコンソールにルーティングするには、sas.compute.server 
logconfig構成スペースで DS2ロガーを計算サーバーに対して宣言する必要があり
ます。詳細については、SAS Compute Serverと計算サービスを参照してください。
sas.compute.server: logconfigインスタンスを更新する手順を参照してください。

DS2エラーの種類
DS2は、要求を完了できない場合、SASログにエラーを書き込みます。ほとんどの
エラーは、次のように分類できます。
n 使用エラー
不正または無効な要求が原因で、使用エラーが発生します。DS2使用エラーは通
常、DS2プログラムの構文、意味、または論理エラーによって発生します。コン
パイル時に構文エラーと意味エラーが検出されると、DS2プログラムのコンパイ
ルが妨げられます。正常にコンパイルされた DS2プログラムには、実行時に表
面化する論理エラーが含まれている可能性があります。

n リソースエラー
リソースサービス要求を実行できない場合、リソースエラーが発生します。さま
ざまな条件がこれらの失敗につながります。たとえば、リソースにアクセスでき
ない、リソースのしきい値にすでに達している、リソースにアクセスする権限が
ないなどの可能性があります。

DS2エラーを解決する方法
エラーを解決するには、次の操作を行います。

1 SASログでエラーメッセージを確認します。これらのメッセージは、問題の原因
を理解したり、エラーの解決に役立つアイデアを提供したりするのに役立つ場合
があります。
問題の内容を示す例を次に示します。この例では、使用エラーが発生しました。
このエラーメッセージは、DS2プログラムの 7行目またはその近くに構文エラー
があることを示しています。
ERROR: Parse encountered END when expecting ';'.
ERROR: Line 7: Parse failed:  >>> end <<<;

2 SASログでは、まず一番最初のエラーメッセージに注目します。場合によって
は、1つの問題が失敗の連鎖につながり、最初のエラーを解決すると、後続のエ
ラーも解決することがあります。

3 エラーメッセージがリソースエラーを示している場合は、システム管理者に問い
合わせて、システムリソースを監視してください。システムリソースの例には、
メモリ使用量とディスク容量が含まれます。システムリソースが少ない場合は、
リソースを増やす方法を探すか、アプリケーションがそれらをより効率的に使用
できる方法を検討してください。
リソースエラーを示す例を次に示します。このメッセージは、メモリしきい値に
達したことを示しています。
ERROR: Memory allocation failed.

4 前の 3つのステップを使用してもエラーを解決できない場合は、SASテクニカル
サポートにお問い合わせください。提供する必要がある情報のリストは次のと
おりです: SASテクニカルサポートに問い合わせる際に覚えておくべき 4つのヒ
ント。
SASログに内部エラーコードを含むエラーメッセージが含まれている場合は、
SASテクニカルサポート担当者にコードを提供してください。内部エラーコー
ドは、SASが問題の原因をより迅速に特定するのに役立つ 16進数です。
内部エラーコード(ハイライト表示)を含むエラーメッセージの例を次に示しま
す。
ERROR: DS2 internal error code 0xECCC8C8BF7C049FD. Refer to DS2 documentation 
       for resolving errors.

DS2ロガー
DS2ロガーは、エラーの解決にも役立ちます。ロガーは、実行中の DS2プログラム
から情報を取得する別のメカニズムを提供します。次に例を示します。
n App.TableServices.DS2.Config.Optionsロガーは、多くの構成設定を報告しま
す。そのうちの 1つは DS2セッションエンコーディングです。DS2プログラム
の実行に使用されているセッションエンコーディングが不明な場合は、このロガ
ーを有効にして DS2セッションエンコーディングを報告できます。

n HTTPパッケージロガーは、DS2 HTTPパッケージで生成された HTTP要求の問
題をデバッグするのに役立つ情報を提供します。

n DS2プログラムの実行中に変数値の変更をトレースする TraceVariablesロガー
は、問題の修復に役立ち、DS2コードの監査を提供できます。

これらおよびその他の DS2ロガーの詳細については、付録 2, “DS2ロガー” (323ペ
ージ)を参照してください。

